§7. Ряды Фурье. Интеграл Фурье.

7.1. Разложение функций в тригонометрические ряды Фурье.

Тригонометрическая система функций

$$1,\ \cos x,\ \sin x,\ \cos 2x,\ \sin 2x,\ \ldots,\ \cos n x,\ \sin n x,\ \ldots$$

является ортогональной на отрезке $[-\pi, \pi]$ (как, впрочем, и на всяком отрезке длины 2π), т.е. интеграл по этому отрезку от произведения любых двух различных функций этой системы равен нулю.

Если $f(x) \in L(-\pi, \pi)$ (т.е. $\int_{-\pi}^{\pi} |f(x)|\, dx < +\infty$), то существуют числа

$$a_n = \frac{1}{\pi}\int_{-\pi}^{\pi} f(x)\cos(n x)\, dx \quad n = 0,\ 1,\ 2,\ \ldots$$

$$b_n = \frac{1}{\pi}\int_{-\pi}^{\pi} f(x)\sin(n x)\, dx,\ n = 1,\ 2,\ \ldots$$

называемые коэффициентами Фурье функции $f(x)$.

Ряд

(1)

$$S(x) = \frac{a_0}{2} + \sum_{n=1}^{\infty}(a_n\cos(n x) + b_n\sin(n x)) \tag{2}$$

называется рядом Фурье функции f(x).

Члены ряда (1) можно записать в виде гармоник

$$a_n\cos(n x) + b_n\sin(n x) = A_n\cos(n x - \phi_n)$$

с амплитудой $A_n = \sqrt{a_n^2 + b_n^2}$, частотой $\omega_n = n$ и фазой $\phi_n = \operatorname{arctg}\frac{b_n}{a_n}$.

Для функции f(x) такой, что $f^2(x) \in L(-\pi, \pi)$, справедливо равенство Парсеваля

$$\frac{1}{\pi}\int_{-\pi}^{\pi} f^2(x)\, dx = \frac{a_0^2}{2} + \sum_{n=1}^{\infty}\left(a_n^2 + b_n^2\right)$$

Если же $f(x) \in L(-l, l)$, то коэффициенты Фурье записываются в виде:

$$a_0 = \frac{1}{l}\int_{-l}^{l} f(x)\, dx \tag{3}$$

$$a_n = \frac{1}{l}\int_{-l}^{l} f(x)\cos\left(\frac{\pi}{l} n x\right) dx \tag{4}$$

$$b_n = \frac{1}{l}\int_{-l}^{l} f(x)\sin\left(\frac{\pi}{l} n x\right) dx \tag{5}$$

а ряд Фурье - в виде

$$S(x) = \frac{a_0}{2} + \sum_{n=1}^{\infty}\left(a_n\cos\left(\frac{\pi}{l} n x\right) + b_n\sin\left(\frac{\pi}{l} n x\right)\right) \tag{6}$$

Суммы рядов (1), (5),(6) имеют периоды соответственно $2\pi,\ 2l,\ 2l$

Функция f(x) называется кусочно гладкой на отрезке [a;b], если сама функция f(x) и ее производная f'(x) имеют на [a;b] конечное число точек разрыва 1-ого рода.

Т е о р е м а. Если периодическая функция f(x) с периодом 2l кусочно гладка на отрезке [-l;l], то ряд Фурье (5) сходится к значению f(x) в каждой ее точке непрерывности и к значению $(f(x+0) + f(x-0))/2$ в точке разрыва, т.е.

$$\frac{1}{2}\left(f(x+0)+f(x-0)\right)=\frac{a_0}{2}+\sum_{n=1}^{\infty}\left(a_n\cos\left(\frac{\pi}{l}n\,x\right)+b_n\sin\left(\frac{\pi}{l}n\,x\right)\right) \qquad (7)$$

Если, дополнительно, f(x) непрерывна на всей оси, то ряд (6) сходится к f(x) равномерно.

Если на полуоткрытом интервали длины 2l, т.е. на интервале вида [a, $a+2\,l$) или (a, $a+2\,l$], определена какая-нибудь функция, то она может быть (единственным способом) продолжена на всю числовую прямую так, что получится функция с периодом 2l.

Отсюда следует, что если f(x) имеет на [-l,l] не более конечного числа точек разрыва и абсолютно интегрируема на этом сегменте, то она разложима в ряд фурье (5).

Пример №12.480

Разложить периодическую функцию с периодом 2l функцию в ряд Фурье, построить график.

$$f(x)=\begin{cases}1 & \text{при } 0<x<\pi\\ 0 & \text{при } -\pi<x<0\end{cases}$$

Запишем исходную кусочно-заданную функцию и построим ее график:

```
f[x_] := Piecewise[{{1, 0 < x < Pi}, {0, -Pi < x < 0}}]
Plot[f[x], {x, -Pi, Pi}, PlotRange → {-0.5, 1.5}]
```



Запишем выражения для коэффициентов Фурье по формулам (2), (3), (4)

Так как функция обращается в ноль при x<0, интегрирование ведем от 0 до π

Коэффициент a_0 вычисляется следующим образом:

```
a0 = 1/π Integrate[f[x], {x, 0, Pi}]
```

1

Коэффициент a_n

```
an = 1/π Integrate[f[x] Cos[n x], {x, 0, Pi}]
```

$$\frac{\mathrm{Sin}[n\,\pi]}{n\,\pi}$$

При любых n, $a_n = 0$, но $a_0 = 1$

Теперь коэффициент b_n

```
bn = 1/π Integrate[f[x] Sin[n x], {x, 0, Pi}]
```

$$\frac{1-\mathrm{Cos}[n\,\pi]}{n\,\pi}$$

Получили, что коэффициент b_n:

$$b_n = \begin{cases} \frac{2}{(2k+1)\pi} & \text{при } n = 2k+1 \\ 0 & \text{при } n = 2k \end{cases}$$

Запишем ряд Фурье:

$$f(x) = \frac{1}{2} + \sum_{k=0}^{\infty} \frac{2\sin((2k+1)x)}{(2k+1)\pi}$$

```
fu[x_, n_] := 1/2 + 2/π Sum[Sin[(2 k + 1) x]/(2 k + 1), {k, 0, n}]
```

```
fu[x, 4] // Expand
```

$$\frac{1}{2} + \frac{2\,\mathrm{Sin}[x]}{\pi} + \frac{2\,\mathrm{Sin}[3\,x]}{3\,\pi} + \frac{2\,\mathrm{Sin}[5\,x]}{5\,\pi} + \frac{2\,\mathrm{Sin}[7\,x]}{7\,\pi} + \frac{2\,\mathrm{Sin}[9\,x]}{9\,\pi}$$

Изобразим график исходной функции и ряда Фурье до 4ого и до 10ого члена.

```
Row[{Plot[{fu[x, 4], f[x]}, {x, -Pi, Pi}, ImageSize → 300],
  Plot[{Fu[x, 10], f[x]}, {x, -Pi, Pi}, ImageSize → 300] }]
```





Из графиков видно, что чем больше максимальный парядок разложения, тем сильнее совпадают функция и ее ряд.

В данной задаче было показано, как, на основе определений коэффициентов, разложить функцию в ряд Фурье.

Пример №12.487

Разложить периодическую функцию с периодом 2l функцию в ряд Фурье, построить график.

$$f(x) = 2x, \qquad 0 < x < 1, \; 2l = 1$$

Определим заданную функцию, и построим ее график:

```
f[x_] := 2 x
l = 1 / 2;
Plot[f[x], {x, 0, 1}]
```



Теперь запишем функции для коэффициентов Фурье по формулам (3), (4), (5):

Так как в формуле (4), при n=0, косинус обращается в единицу, то записывать выражение для коэффициента a_0 смысла нет.

```
a[n_] := Integrate[1/l f[x] Cos[Pi n x/l], {x, 0, 1}];
b[n_] := Integrate[1/l f[x] Sin[Pi n x/l], {x, 0, 1}];
```

Теперь запишем функцию тригонометрического ряда Фурье:

```
fu[x_, k_] := a[0]/2 + Sum[a[n] Cos[Pi/l n x] + b[n] Sin[Pi/l n x], {n, 1, k}]
```

Выведем выражение для ряда до 5ого члена и построим его график:

```
fu[x, 5]
Plot[{Evaluate@fu[x, 5], f[x]}, {x, 0, 1}]
```

$$1-\frac{2\,\mathrm{Sin}[2\,\pi\,x]}{\pi}-\frac{\mathrm{Sin}[4\,\pi\,x]}{\pi}-\frac{2\,\mathrm{Sin}[6\,\pi\,x]}{3\,\pi}-\frac{\mathrm{Sin}[8\,\pi\,x]}{2\,\pi}-\frac{2\,\mathrm{Sin}[10\,\pi\,x]}{5\,\pi}$$



Чтобы проверить себя найдем сумму бесконечного ряда:

```
fu[x, Infinity]
```

$$1+\frac{i\left(-\mathrm{Log}\left[1-e^{2\,i\,\pi\,x}\right]+\mathrm{Log}\left[e^{-2\,i\,\pi\,x}\left(-1+e^{2\,i\,\pi\,x}\right)\right]\right)}{\pi}$$

И построим теперь график полученной функции:

```
Plot[{f[x], Evaluate@fu[x, Infinity]}, {x, 0, 1}]
```



Графики функций совпали, следовательно, как и предполагалось, ряд сходится к функции.

7.2. Особенности разложения в ряд Фурье четных и нечетных функций.

Если функцию обладает какой-либо симметрией, то техника разложения в ряд Фурье упрощается.

В случае, если f(x) - четная функция с периодом 2l, то все коэффициенты Фурье b_n равны 0 и, следовательно, в ряде Фурье нет членов с синусами. Тогда получим ряд:

$$S(x) = \frac{a_0}{2} + \sum_{n=1}^{\infty} a_n \cos\left(\frac{\pi}{l} n x\right) \tag{8}$$

где

$$a_0 = \frac{2}{l} \int_0^l f(x)\, dx \tag{9}$$

$$a_n = \frac{2}{l} \int_0^l f(x) \cos\left(\frac{\pi}{l} n x\right) dx \tag{10}$$

Аналогично в случае, если f(x) - нечетная функция, то все коэффициенты Фурье a_n равны 0 и, следовательно, в ряде Фурье нет членов с косинусами. Тогда получим ряд:

$$S(x) = \sum_{n=1}^{\infty} b_n \sin\left(\frac{\pi}{l} n x\right) \tag{11}$$

где

$$b_n = \frac{2}{l} \int_0^l f(x) \sin\left(\frac{\pi}{l} n x\right) dx \tag{12}$$

Пример №12.482

Разложить в ряд Фурье функцию:

$$f(x) = |x|, \quad -1 < x < 1, \quad l = 1$$

Запишем и построим график заданной функции:

```
f[x_] := Abs[x]
Plot[f[x], {x, -1, 1}]
```



Так как функция четная, то в разложении будут отсутствовать члены с синусами. Тогда запишем выражение для коэффициента a_n:

```
a[n_] := Integrate[2 f[x] Cos[Pi n x], {x, 0, 1}]
```

Запишем функцию ряда:

```
fu[x_, k_] := a[0] / 2 + Sum[a[n] Cos[Pi n x], {n, 1, k}]
```

Выведем полученный ряд до 6ого члена и построим его функцию:

```
fu[x, 6]
Plot[{f[x], Evaluate@fu[x, 6]}, {x, -1, 1}]
```

$$\frac{1}{2} - \frac{4\,\mathrm{Cos}[\pi\, x]}{\pi^2} - \frac{4\,\mathrm{Cos}[3\,\pi\, x]}{9\,\pi^2} - \frac{4\,\mathrm{Cos}[5\,\pi\, x]}{25\,\pi^2}$$



Из графика заключаем, что ряд сходится к функции.

Пример №12.493

Доопределяя необходимым образом заданную в промежутке функцию до переодической, получить для нее: а) ряд Фурье по косинусам б) ряд Фурье по синусам.

$$f(x) = e^x, \;\; 0 < x < \ln 2$$

Для того чтобы получить ряд Фурье по косинусам, необходимо чтобы функция была чётной. Функция $e^{|x|}$ подходит под данное требование. Запишем и изобразим эту функцию:

```
f[x_] := Exp[Abs[x]]
Plot[f[x], {x, -Log[2], Log[2]}]
```



Теперь получим выражение для коэффициента Фурье и запишем ряд:

$$a[n\_] := \text{Integrate}\left[\frac{2}{\text{Log}[2]} f[x] \,\text{Cos}\left[\frac{\text{Pi}}{\text{Log}[2]} n\, x\right], \{x, 0, \text{Log}[2]\}\right]$$

$$fu[x\_, k\_] := a[0]/2 + \text{Sum}\left[a[n]\, \text{Cos}\left[\frac{\text{Pi}}{\text{Log}[2]} n\, x\right], \{n, 1, k\}\right]$$

Изобразим полученный ряд:

```
fu[x, 5]
Plot[{f[x], Evaluate@fu[x, 5]}, {x, -Log[2], Log[2]}]
```

$$\frac{1}{\text{Log}[2]} - \frac{6\,\text{Cos}\left[\frac{\pi x}{\text{Log}[2]}\right] \text{Log}[2]}{\pi^2 + \text{Log}[2]^2} - \frac{6\,\text{Cos}\left[\frac{3\pi x}{\text{Log}[2]}\right] \text{Log}[2]}{9\pi^2 + \text{Log}[2]^2} - \frac{6\,\text{Cos}\left[\frac{5\pi x}{\text{Log}[2]}\right] \text{Log}[2]}{25\pi^2 + \text{Log}[2]^2} + \frac{\text{Cos}\left[\frac{2\pi x}{\text{Log}[2]}\right] \text{Log}[4]}{4\pi^2 + \text{Log}[2]^2} + \frac{\text{Cos}\left[\frac{4\pi x}{\text{Log}[2]}\right] \text{Log}[4]}{16\pi^2 + \text{Log}[2]^2}$$



Теперь для того чтобы получить разложение Фурье по синусам, доопределим исходную функцию до нечетной следующим образом:

$$f(x) = \begin{cases} e^{x} & 0 < x < \ln 2 \\ -e^{-x} & -\ln 2 < x < 0 \end{cases}$$

Запишем и построим полученную функцию:

```
f[x_] := Piecewise[{{Exp[x], Log[2] > x > 0}, {-Exp[-x], -Log[2] < x < 0}}]
Plot[f[x], {x, -Log[2], Log[2]}]
```



Запишем выражения для коэффициента b_n и ряда:

```
b[n_] := Integrate[2/Log[2] f[x] Sin[Pi/Log[2] n x], {x, 0, Log[2]}]
fu[x_, k_] := Sum[b[n] Sin[Pi/Log[2] n x], {n, 1, k}]
```

Получим разложение функции до 6ого члена и построим график:

```
fu[x, 6]
Plot[{f[x], Evaluate@fu[x, 12]}, {x, -Log[2], Log[2]}]
```

$$\frac{6\pi \operatorname{Sin}\left[\frac{\pi x}{\operatorname{Log}[2]}\right]}{\pi^2+\operatorname{Log}[2]^2}-\frac{4\pi \operatorname{Sin}\left[\frac{2\pi x}{\operatorname{Log}[2]}\right]}{4\pi^2+\operatorname{Log}[2]^2}+\frac{18\pi \operatorname{Sin}\left[\frac{3\pi x}{\operatorname{Log}[2]}\right]}{9\pi^2+\operatorname{Log}[2]^2}-\frac{8\pi \operatorname{Sin}\left[\frac{4\pi x}{\operatorname{Log}[2]}\right]}{16\pi^2+\operatorname{Log}[2]^2}+\frac{30\pi \operatorname{Sin}\left[\frac{5\pi x}{\operatorname{Log}[2]}\right]}{25\pi^2+\operatorname{Log}[2]^2}-\frac{12\pi \operatorname{Sin}\left[\frac{6\pi x}{\operatorname{Log}[2]}\right]}{36\pi^2+\operatorname{Log}[2]^2}$$



В точке разрыва, значение ряда S(0) будет равно:

$$S(0) = \frac{f(+0) + f(-0)}{2}$$

Программный модуль

Анализируя последовательность действий в общем случае, напишем простую обучающую программу, которая будет раскладывать функцию в ряд Фурье на заданном интервале [h;g] с полупериодом $l = \frac{g-h}{2}$:

```
fourier[x_, func_, h_, g_, k_] := Module[{a, b, l},
  l = (g - h) / 2;
  a[n_] := Integrate[1/l func Cos[Pi n x/l], {x, h, g}];
  b[n_] := Integrate[1/l func Sin[Pi n x/l], {x, h, g}];
  a[0]/2 + Sum[a[n] Cos[Pi/l n x] + b[n] Sin[Pi/l n x], {n, 1, k}]]
```

Проведем графическую проверку полученного модуля:

Пример №12.483

Разложить в ряд Фурье функцию:

$$f(x) = |\cos(x)|, \quad -\pi < x < \pi, \quad l = \pi$$

```
fu[x_, k_] := fourier[x, Abs[Cos[x]], -Pi, Pi, k]

fu[x, 8]
Plot[{Abs[Cos[x]], Evaluate@fu[x, 8]}, {x, -Pi, Pi}]
```

$$\frac{2}{\pi} + \frac{4\,\mathrm{Cos}[2\,x]}{3\,\pi} - \frac{4\,\mathrm{Cos}[4\,x]}{15\,\pi} + \frac{4\,\mathrm{Cos}[6\,x]}{35\,\pi} - \frac{4\,\mathrm{Cos}[8\,x]}{63\,\pi}$$



При увелечении порядка разложения, ряд сходится к функции.

Пример №12.488

Разложить в ряд Фурье функцию:

$$f(x) = 10 - x, \quad 5 < x < 15, \quad l = 5$$

```
fu[x_, k_] := fourier[x, 10 - x, 5, 15, k]
```

```
fu[x, 8]
Plot[{10 - x, Evaluate@fu[x, 8]}, {x, 5, 15}]
```

$$-\frac{10\,\mathrm{Sin}\left[\frac{\pi x}{5}\right]}{\pi}+\frac{5\,\mathrm{Sin}\left[\frac{2\pi x}{5}\right]}{\pi}-\frac{10\,\mathrm{Sin}\left[\frac{3\pi x}{5}\right]}{3\pi}+\frac{5\,\mathrm{Sin}\left[\frac{4\pi x}{5}\right]}{2\pi}-\frac{2\,\mathrm{Sin}[\pi x]}{\pi}+\frac{5\,\mathrm{Sin}\left[\frac{6\pi x}{5}\right]}{3\pi}-\frac{10\,\mathrm{Sin}\left[\frac{7\pi x}{5}\right]}{7\pi}+\frac{5\,\mathrm{Sin}\left[\frac{8\pi x}{5}\right]}{4\pi}$$



На основе двух тестов будем считать, что написанный модуль проверку прошел.

7.3. Ряд Фурье в комплексной форме.

Применим тождества Эйлера для косинуса и синуса ряда (2):

$$S(t)=\frac{a_0}{2}+\sum_{n=1}^{\infty}(a_n\cos(n\,t)+b_n\sin(n\,t))=\frac{a_0}{2}+\sum_{n=1}^{\infty}\left(a_n\,\frac{e^{i\,n\,t}+e^{-i\,n\,t}}{2}+b_n\,\frac{e^{i\,n\,t}-e^{-i\,n\,t}}{2\,i}\right)=$$

$$=\frac{a_0}{2}+\sum_{n=1}^{\infty}\left(a_n\,\frac{e^{i\,n\,t}+e^{-i\,n\,t}}{2}-i\,b_n\,\frac{e^{i\,n\,t}-e^{-i\,n\,t}}{2}\right)=\frac{a_0}{2}+\sum_{n=1}^{\infty}\left(\frac{a_n-i\,b_n}{2}\,e^{i\,n\,t}+\frac{a_n+i\,b_n}{2}\,e^{-i\,n\,t}\right)$$

Введем коэффициенты a_n и b_n отрицательными номерами:

$$a_{-n}=a_n$$
$$b_{-n}=b_n$$

Для которых также справедливы формулы:

$$a_{-n}=\frac{1}{\pi}\int_{-\pi}^{\pi}f(t)\cos(-n\,t)\,dt=\frac{1}{\pi}\int_{-\pi}^{\pi}f(t)\cos(n\,t)\,dt=a_n$$

$$b_{-n}=\frac{1}{\pi}\int_{-\pi}^{\pi}f(t)\sin(-n\,t)\,dt=-\frac{1}{\pi}\int_{-\pi}^{\pi}f(t)\sin(n\,t)\,dt=-b_n$$

Продолжим преобразование:

$$f(t)=\frac{a_0}{2}+\sum_{n=1}^{\infty}\frac{a_n-i\,b_n}{2}\,e^{i\,n\,t}+\sum_{n=-\infty}^{-1}\frac{a_n-i\,b_n}{2}\,e^{i\,n\,t}=\sum_{n=-\infty}^{\infty}\frac{a_n-i\,b_n}{2}\,e^{i\,n\,t}$$

Обозначим:

$$c_n=\frac{a_n-i\,b_n}{2}=\frac{1}{2\pi}\int_{-\pi}^{\pi}f(t)\cos(n\,t)\,dt-i\,\frac{1}{2\pi}\int_{-\pi}^{\pi}f(t)\sin(n\,t)\,dt=\frac{1}{2\pi}\int_{-\pi}^{\pi}f(t)\,e^{-n\,t}\,dt$$

Итак, в комплексной форме ряд Фурье для функции с произвольным полупериодом l записывается:

$$f(t)=\sum_{n=-\infty}^{\infty}c_n\,e^{\frac{i\,n\,\pi\,x}{l}} \qquad (13)$$

$$c_n = \frac{1}{2\,l} \int_{-l}^{l} f(t)\, e^{-\frac{i n \pi t}{l}}\, dt \tag{14}$$

Пример №12.484

Представить функцию в виде ряда Фурье в комплексной форме:

$$f(t) = t^2, \ -\pi < t < \pi, \ l = \pi$$

Запишем функцию:

```
f[t_] := t²
```

По формуле (14) запишем коэффициент c_n:

```
c[n_] := Integrate[1/(2 Pi) f[t] Exp[-I n t], {t, -Pi, Pi}]
```

Запишем функцию ряда Фурье до k-ого порядка:

```
fu[t_, k_] := Sum[c[n] Exp[I n t], {n, -k, k}]
```

Получим выражение ряда до 4-ого порядка:

```
fu[t, 4]
```

$$-2\, e^{-i t} - 2\, e^{i t} + \frac{1}{2}\, e^{-2 i t} + \frac{1}{2}\, e^{2 i t} - \frac{2}{9}\, e^{-3 i t} - \frac{2}{9}\, e^{3 i t} + \frac{1}{8}\, e^{-4 i t} + \frac{1}{8}\, e^{4 i t} + \frac{\pi^2}{3}$$

Сравним, преобразовав данный результат при помощи функции ExpToTrig, с тем, что бы у нас получилось, если бы мы воспользовались модулем fourier:

```
fourier[t, f[t], -Pi, Pi, 4]
fu[t, 4] // ExpToTrig
```

$$\frac{\pi^2}{3} - 4\, \mathrm{Cos}[t] + \mathrm{Cos}[2\, t] - \frac{4}{9}\, \mathrm{Cos}[3\, t] + \frac{1}{4}\, \mathrm{Cos}[4\, t]$$

$$\frac{\pi^2}{3} - 4\, \mathrm{Cos}[t] + \mathrm{Cos}[2\, t] - \frac{4}{9}\, \mathrm{Cos}[3\, t] + \frac{1}{4}\, \mathrm{Cos}[4\, t]$$

Как и ожидалось, ответы совпадают.

Построим график, чтобы убедится, что ряд сходится к исходной функции:

```
Plot[{f[t], Evaluate@fu[t, 4]}, {t, -Pi, Pi}]
```



Программный модуль

Подобно модулю разложения в тригонометрический ряд, составим модуль разложения в ряд Фурье в комплексной форме:

```
fourierC[t_, func_, h_, g_, k_] := Module[{c, l},
  l = (g - h)/2;
  c[n_] := 1/(2 l) Integrate[func Exp[- I Pi/l n t], {t, h, g}];
  Sum[c[n] Exp[I Pi/l n t], {n, -k, k}]]
```

Пример №12.486

Представить функцию в виде ряда Фурье в комплексной форме:

$$f(t) = |\sin(t)|, \; -\pi < t < \pi, \; l = \pi$$

Выведем результат работы функции fourierC и построим графики исходной функции и полученного ряда:

```
fourierC[t, Abs[Sin[t]], -Pi, Pi, 4]
Plot[{Abs[Sin[t]], Evaluate@fourierC[t, Abs[Sin[t]], -Pi, Pi, 4]},
 {t, -Pi, Pi}]
```

$$\frac{2}{\pi} - \frac{2\,e^{-2\,i\,t}}{3\,\pi} - \frac{2\,e^{2\,i\,t}}{3\,\pi} - \frac{2\,e^{-4\,i\,t}}{15\,\pi} - \frac{2\,e^{4\,i\,t}}{15\,\pi}$$



Будем считать, что проверка прошла успешно.

7.4. Системные функции разложения в ряд Фурье.

Mathematica имеет встроенные функции разложения в ряд Фурье. Рассмотрим на примере:

Пример №12.484

Разложить функцию в ряд Фурье периода l:

$$f(x) = x^2, \; -\pi < x < \pi, \; l = \pi$$

Для решения данной задачи воспользуемся встроенной функцией FourierSeries, имеющей следующую структуру: FourierSeries[expr,t,n] , где expr - выражение или функция, которую требуется разложить в ряд Фурье, t - переменная, по которой ведется разложение, n- высший порядок разложения.

Запишем исходную функцию:

```
f[x_] := x^2
```

Теперь получим разложение в ряд до 6ого члена.

```
FourierSeries[x^2, x, 6]
```

$$-2\,e^{-i x}-2\,e^{i x}+\frac{1}{2}\,e^{-2 i x}+\frac{1}{2}\,e^{2 i x}-\frac{2}{9}\,e^{-3 i x}-\frac{2}{9}\,e^{3 i x}+\frac{1}{8}\,e^{-4 i x}+\frac{1}{8}\,e^{4 i x}-\frac{2}{25}\,e^{-5 i x}-\frac{2}{25}\,e^{5 i x}+\frac{1}{18}\,e^{-6 i x}+\frac{1}{18}\,e^{6 i x}+\frac{\pi^2}{3}$$

Как мы видим, по умолчанию, Mathematica раскладывает в ряд в комплексной форме, можем перейти в более привычной для нас, тригонометрической форме:

```
FourierSeries[x^2, x, 6] // ExpToTrig
```

$$\frac{\pi^2}{3}-4\,\mathrm{Cos}[x]+\mathrm{Cos}[2\,x]-\frac{4}{9}\,\mathrm{Cos}[3\,x]+\frac{1}{4}\,\mathrm{Cos}[4\,x]-\frac{4}{25}\,\mathrm{Cos}[5\,x]+\frac{1}{9}\,\mathrm{Cos}[6\,x]$$

Как мы видим, функция четная, функции синуса в разложении отсутствуют.

Найдем формулу n-ого коэффициента при помощи функции FourierCoefficient:

```
FourierCoefficient[x^2, x, n]
```

$$\frac{2\,(-1)^n}{n^2}$$

Напоминаем, что это коэффициент в разложении в комплексной форме c_n, он не равен коэффициентам a_n или b_n.

Также находим свободный коэффициент:

```
FourierCoefficient[x^2, x, 0]
```

$$\frac{\pi^2}{3}$$

Составим функцию ряда:

```
fu[x_, n_] := FourierSeries[t^2, t, n] /. t → x
```

```
fu[x, 6] // ExpToTrig
```

$$\frac{\pi^2}{3}-4\,\mathrm{Cos}[x]+\mathrm{Cos}[2\,x]-\frac{4}{9}\,\mathrm{Cos}[3\,x]+\frac{1}{4}\,\mathrm{Cos}[4\,x]-\frac{4}{25}\,\mathrm{Cos}[5\,x]+\frac{1}{9}\,\mathrm{Cos}[6\,x]$$

Покажем график исходной функции и ряда Фурье до 5ого порядка:

```
Plot[{f[x], Evaluate@fu[x, 5]}, {x, -Pi, Pi}]
```



Пример №12.496

Доопределяя необходимым образом заданную в промежутке функцию до переодической, получить для нее: а) ряд Фурье по косинусам б) ряд Фурье по синусам.

$$f(x) = x\sin(x),\ \ 0 < x < \pi$$

Так как данная функция четная, получим выражение ряда в общей форме.

Для этого запишем и посмотрим заданную функцию:

```
f[x_] := x Sin[x]
Plot[f[x], {x, -Pi, Pi}]
```



Так как данная функция четная, в разложении будут отсутствовать члены с синусом, тогда:

$$c_n = \frac{a_n - i\,b_n}{2} = \frac{a_n - i\,0}{2} \Rightarrow a_n = 2\,c_n$$

Получим выражение для нулевого и n-ого коэффициента:

```
FourierCoefficient[f[x], x, n]
```

$$\frac{(-1)^{1+n}}{-1+n^2}$$

Отсюда следует, что

$$a_n = \frac{2\,(-1)^{n+1}}{n^2 - 1}$$

Найдем значение коэффициента при n=1:

```
FourierCoefficient[f[x], x, 1]
```

$$-\frac{1}{4}$$

Тогда ряд запишем в виде:

$$f(x) = 1 - \frac{1}{2}\cos(x) + \sum_{n=2}^{\infty} \frac{2\,(-1)^{n+1}}{n^2 - 1}\cos(n\,x)$$

Ряд Фурье для исходной функции получим при помощи встроенной опции FourierSinSeries (аналогично существует опция FourierCosSeries), она имеет ту же структуру, что и FourierSeries:

```
FourierSinSeries[f[x], x, 8]
```

$$\frac{1}{2}\pi\,\mathrm{Sin}[x] - \frac{16\,\mathrm{Sin}[2\,x]}{9\,\pi} - \frac{32\,\mathrm{Sin}[4\,x]}{225\,\pi} - \frac{48\,\mathrm{Sin}[6\,x]}{1225\,\pi} - \frac{64\,\mathrm{Sin}[8\,x]}{3969\,\pi}$$

Изобразим график функции и ряда:

```
Plot[{f[x], Evaluate@FourierSinSeries[f[x], x, 8]}, {x, -Pi, Pi}]
```



Как видно из графика, исходная функция была доопределена до несимметричной функции, у которой отстуствует косинус в разложении в ряд.

Коэффициент b_n получим при помощи функции FourierSinCoefficient:

```
FourierSinCoefficient[f[x], x, n]
```

$$-\frac{4\,(1+(-1)^n)\,n}{(-1+n^2)^2\,\pi}$$

```
FourierSinCoefficient[f[x], x, 1]
```

$$\frac{\pi}{2}$$

Тогда ряд запишется в виде:

$$f(x)=\frac{\pi}{2}\sin(x)-\frac{4}{\pi}\sum_{n=2}^{\infty}\frac{(1+(-1)^n)\,n}{\left(-1+n^2\right)^2}\sin(n\,x)$$