Глава 1. Элементы теории функций комплексной переменной и её интерпретация в среде Mathematica

§1. Элементарные функции

1.1. Понятие функций комплексной переменной и применение средств Wolfram Mathematica для работы с ними.

Множество точек E расширенной комплексной плоскости $(z) = \mathbb{C} \bigcup \{\infty\}$ называется связным, если любые две его точки можно соединить непрерывной кривой, все точки которой принадлежат данному множеству. Связное открытое множество точек комплексной плоскости называется областью и обазначается через D, G т.п. Область D называется односвязной, если ее граница является связным множеством: в противном случае область D называется многосвязной.
Если каждому комплексному числу z, принадлежащему области D, поставлено в соответствие некоторое комплексное число w, то говорят, что в области D определена комплексная функция *w = f(z).*

Пусть $z = x + i\, y$ и $w = u + i\, v$. Тогда функция *w = f(z)* может быть представлена с помощью двух действительных функций $u = u(x,\ y)$ и $v = v(x,\ y)$ действительных переменных x и y:

$$w = f(z) = u(x,\ y) + i\, v(x,\ y)$$

Данный курс предполагает, что читатель знаком с основами работы в системе Wolfram Mathematica, однако в некоторых случаях будут приводиться пояснения синтаксиса и семантики языка Wolfram.

Рассмотрим способы задания функций в системе Wolfram Mathematica:

Функции могут быть определены правилами, действующими по шаблону:

```
in : z[x_, y_] := x + I y
```

z[x_,y_] является шаблоном, где x_ или y_ означают любые выражения. Правило гласит, что элементам перечисленным через запятую в количестве двух единиц будут присвоены имена x и y и в правой части уже будут производится преобразования с ними.

```
in : z[2, -3]
out : 2 - 3 i
```

Примечание: “функция” мнимой единицы в языке Wolfram имеет вид I (заглавная буква i) либо же i, которую можно получить по умолчанию в клетке Output, выбрать в панеле Palettes → Classrom Assistant → Basic Commands, либо же ввести вручную: (Esc+i+i+Esc).

Также после окончания использования конкретной функции, следует удалить ее определение во избежание конфликтов. Удалить определение конкретной функции можно при помощи команды Clear[название функции]. Например:

```
in : Clear[z]
```

```
in : z[2, -3]
out : z[2, -3]
```

Чтобы удалить все пользовательские определения необходимо вызвать следующую команду:

```
in : Clear["Global`*"]
```

В дальнейшем, внутри одного подпункта возможно использование функций с одинаковыми названиями, но с разным определением. Это означает, что прежде чем использовать имя функции повторно, определение необходимо сначала очистить.

Примечание: согласно принятому соглашению пользователей, имена функций и переменных, определяемых самостоятельно, начинаются со строчной (не заглавной!) буквы.

Рассмотрим также некоторые конструкции не имеющие прямого отношения к ТФКП, но часто использующиеся в книге:

Оператор ReplaceAll:

```
in : a + b + 23 + x^2 /. a → this
out : 23 + b + this + x^2
```

Оператор производит замену всего, что подходит под некоторый “шаблон”. Еще пара примеров:

```
in : {x, x^2, x^4, x^1, a^2, b^1} /. x_^p_ → p + x
out : {x, 2 + x, 4 + x, x, 2 + a, b}
```

Применение сразу нескольких правил замены:

```
in : {1, a, b, c, b^2, t, 54} /. {a → f[a], b → z[b], c → w[c]}
out : {1, f[a], z[b], w[c], z[b]^2, t, 54}
```

Данная функция будет применяться в дальнейшем, но подробно разбираться не будет.

Использование чистых функций.

Mathematica позволяет объявлять безымянные функции (так называемые “чистые” функции) для того чтобы избавиться от необходимости присваивать имена функциям для каждой небольшой операции.

Первый очевидный способ определить чистую функцию, предполагает использование команды Function, которая имеет следующее строение: Function[{блок переменных},выражение]. Например:

```
in : f = Function[{a, b, c}, a^2 + I (b + c)];

in : f[x, y, z]
out : x^2 + i (y + z)
```

Использование названия f совершенно необязательно:

```
in : Function[{a, b, c}, a^2 + I (b + c)][x, y, z]
```

```
out : x^2 + ⅈ (y + z)
```

Подобно тому, как имя функции задавать не обязательно, если вам не нужно обращаться к функции повторно, имена аргументов также можно опустить. Они заменяются указанием “номера слота” #n. В чистых функциях идет отождествление n-го заданного аргумента и знака #n. Символ # представляет собой первый аргумент. После записи выражений следует знак &. Рассмотрим некоторые примеры:

Безымянные переменные:

```
in : z = #1 + I #2 &;
```

```
in : z[2, 3]
```

```
out : 2 + 3 ⅈ
```

Безымянная функция:

```
in : #1 + I #2 &[2, 3]
```

```
out : 2 + 3 ⅈ
```

Использования знака “слот” оправдано ввиду своей краткости. В будущем на этом знаке также не будет заостряться внимание, так как предполагается, что читатель сам на примере способен понять конструкцию чистой функции.

Важно сказать также, что Wolfram Mathematica имеет огромный центр документации, где подробнейшим образом расписана информация о каждой функции существующей в Mathematica. Поэтому, если вы хотите узнать больше о строении какой-либо функции, кнопка F1 на клавиатуре - ваш верный помощник.

Вернемся непосредственно к теме параграфа - ТФКП. Рассмотрим несколько примеров, иллюстрирующих базовые возможности работы Wolfram Mathematica с функциями комплексного переменного.

Пример №11.20

Найти действительную и мнимую части функции:

$$f(z) = i\, z^* + 2\, z^2$$

В Wolfram Mathematica комплексно-сопряженное число обозначается z^*, которое вводится с клавиатуры: Esc+co+Esc.

Представим число z в алгебраическом виде: $z = x + i\, y$.

Опция, ответственная за это, имеет вид: ComplexExpand[expr]

Введем переменную z:

```
in : z = x + I y;
```

```
in : Clear[z]
```

Для того, чтобы получить число, комплексно-сопряженное данному, воспользуемся командой Conjugate[z]:

```
in : Conjugate[z]
out : Conjugate[x] - i Conjugate[y]
```

По умолчанию каждая не заданная переменная может быть и комплексной, поэтому в ответе мы видим что команда Conjugate применена также и к числам x и y. Можно избежать этого, используя команду Simplify, Refine и т.п.:

```
in : Refine[Conjugate[z], {x, y} ∈ Reals]
out : x - i y
```

Чтобы получить разложение всей функции, очевидно, необходимо ввести вместо z функцию f(z):

```
in : ComplexExpand[I Conjugate[z] + 2 z^2]
out : 2 x^2 + y - 2 y^2 + i (x + 4 x y)
```

Чтобы получить отдельно действительную или мнимую части функции, можно воспользоваться соответственно функциями Re или Im, разумеется необходимо указать что переменные x и y принадлежат к множеству действительных чисел

```
in : Refine[Im[ComplexExpand[I Conjugate[z] + 2 z^2]], {x, y} ∈ Reals]
out : x + 4 x y
```

Для получения в виде списка действительной и мнимой частей существует команда ReIm:

```
in : Refine[ReIm[ComplexExpand[I Conjugate[z] + 2 z^2]], {x, y} ∈ Reals]
out : {2 x^2 + y - 2 y^2, x + 4 x y}
```

Итак, получаем ответ задачи:

$$\mathrm{Re}[f(z)] = 2x^2 + y - 2y^2$$
$$\mathrm{Im}[f(z)] = x + 4xy$$

Пример №11.27

Определить функцию $w = f(z)$ по известным действительной и мнимой частям:

$$u = x^2 - y^2 - 2y - 1$$
$$v = 2xy + 2x$$

Выразим x и y через z:

$$z = x + i\,y\,,\ \ z^* = x - i\,y \ \rightarrow\ x = \frac{1}{2}(z + z^*),\ y = \frac{1}{2}(z - z^*)$$

Воспользуемся написанным выше, для того чтобы преобразовать выражения и получить функцию $f(z)$:

```
in : u[x_, y_] := x^2 - y^2 - 2 y - 1
     v[x_, y_] := 2 x y + 2 x
```

Найдем выражение для функции $f(z)$:

```
in : u[x, y] + I v[x, y] /. {x → 1/2 (z + z*), y → 1/2 (z - z*)} //
      ComplexExpand
```

$$out: -1 + 2 i z + z^2$$

Примечание: //ComplexExpand постфиксная форма применения команды ComplexExpand.

Итак, получен ответ: функция $f(z)$ имеет вид:

$$f(z) = z^2 + 2\,i\,z - 1$$

Пример №11.37

Найти образы указанной точки при заданном отображении:

$$z_0 = 1 - \frac{i}{2},\ w(z) = \frac{\operatorname{Im} z}{z}.$$

Запишем функцию w(z):

```
in : w[z_] := Im[z]/z
```

Очевидно, что для того чтобы найти значение функции в точке z_0, необходимо указать ее в качестве аргумента.

```
in : w[1 - I / 2]
```

$$out: -\frac{2}{5} - \frac{i}{5}$$

Графически комплексное число ($z = x + i\,y$) изображается точкой или вектором плоскости с координатами (x,y). При этом действительные числа представляются точками оси абсцисс, а чисто мнимые - точками оси ординат. Поэтому ось абсцисс называется действительной осью, а ось ординат - мнимой осью. Плоскость, на которой изображаются комплексные числа, называется комплексной плоскостью.

Модуль числа z - длина этого вектора.

Точки z и $-z$ симметричны относительно точки 0, а точки z и z^* с- относительно действительной оси.

Построим точку и ее образ на плоскости w. Для этого дважды воспользуемся функцией Graphics[{графические эл-ты},список параметров] и функцией Point[{x1,y1}]:

```
in : Row[{Graphics[{{PointSize → Large, Point[ReIm[#1]]}},
        Axes → True, AxesLabel → {Re[z], Im[z]},
        PlotRange → {{-1, 1}, {-1, 1}}, ImageSize → 180],
       Graphics[{{PointSize → Large, Point[ReIm[#2]]}},
        Axes → True, AxesLabel → {Re[w], Im[w]},
        PlotRange → {{-1, 1}, {-1, 1}}, ImageSize → 180]}] &[
    1 - I / 2, w[1 - I / 2]]
```



рис. 1.1.1

Также с помощью команды Arrow[{{x1,y1},{x2,y2}}] можно построить вектор

```
in : Graphics[{Arrow[{{0, 0}, ReIm[w[1 - I / 2]]}]}, Axes → True,
    AxesLabel → {Re, Im}, PlotRange → {{-.5, .5}, {-.5, .5}}]
```



рис. 1.1.2

Пример №11.42

Найти и построить все значения следующей функции в указанной точке:

$$w = z + z^{1/4},\ z_0 = -1$$

Для того чтобы найти все значения данной функции, воспользуемся показательной формой записи комплексного числа:

$$z = \left| z \right| e^{i(\phi + 2\pi \mathrm{k})}$$

Функция корень **4-ой** степени на комплексной плоскости имеет **4** корня, которые можно найти следующим образом:

$$z_k = \left| z \right|^{\frac{1}{4}} e^{i\left(\frac{\phi}{4} + \frac{2}{4}\pi k\right)},\ \text{где } k \text{ принимает значения } - k = 0,\ 1,\ 2,\ 3$$

Запишем 4 корня, получающиеся в результате вычисления функции $w(z_0)$:

in : `z = Table[-1 + Exp[I (Pi / 4 + Pi k / 2)], {k, 0, 3}]`

out : $\left\{-1 + e^{\frac{i\pi}{4}}, -1 + e^{\frac{3 i\pi}{4}}, -1 + e^{-\frac{3 i\pi}{4}}, -1 + e^{-\frac{i\pi}{4}}\right\}$

Применим к каждому элементу списка функцию ReIm:

in : `z = ReIm[z]`

out : $\left\{\left\{-1 + \frac{1}{\sqrt{2}}, \frac{1}{\sqrt{2}}\right\}, \left\{-1 - \frac{1}{\sqrt{2}}, \frac{1}{\sqrt{2}}\right\}, \left\{-1 - \frac{1}{\sqrt{2}}, -\frac{1}{\sqrt{2}}\right\}, \left\{-1 + \frac{1}{\sqrt{2}}, -\frac{1}{\sqrt{2}}\right\}\right\}$

Теперь изобразим 4 вектора данных точек и подпишем их с помощью команды Text[текст, {x,y}]:

in :
```
Graphics[{Text[#, #], Arrow[{{0, 0}, #}]} & /@ #, Axes → True,
    PlotRange → {{-2, 1}, {-1, 1}}, AxesLabel → {Re, Im} ] &@
  z
```



рис. 1.1.3

1.2. Основные элементарные функции комплексной переменной.

Следующие функции (как однозначные, так и многозначные) называются основными элементарными:

1. Дробно-рациональная функция

$$\frac{a_0 z^n + a_1 z^{n-1} + \ldots + a_n}{b_0 z^m + b_1 z^{m-1} + \ldots + b_m}, \; n, m \in \mathbb{N}$$

Частными случаями этой функции являются:

а) линейная функция

$$a\,z + b,\; a, b \in \mathbb{C},\; a \neq 0$$

б) степенная функция

$$z^n,\; n \in \mathbb{N}$$

в) дробно-линейная функция

$$\frac{a\,z + b}{c\,z + d},\; a, b, c, d \in \mathbb{C},\; c \neq 0,\; a\,d - b\,c \neq 0$$

г) функция Жуковского

$$\frac{1}{2}\left(z + \frac{1}{z}\right)$$

2. Показательная функция

$$e^z = e^x(\cos y + i \sin y)$$

3. Тригонометрические функции

$$\sin z = \frac{1}{2\,i}\left(e^{i\,z} - e^{-i\,z}\right),\; \cos z = \frac{1}{2}\left(e^{i\,z} + e^{-i\,z}\right),$$

$$\operatorname{tg} z = \frac{\left(e^{i\,z} - e^{-i\,z}\right)}{i\left(e^{i\,z} + e^{-i\,z}\right)},\; \operatorname{ctg} z = \frac{i\left(e^{i\,z} + e^{-i\,z}\right)}{\left(e^{i\,z} - e^{-i\,z}\right)}$$

4. Гиперболические функции

$$\operatorname{sh} z = \frac{1}{2}\,(e^z - e^{-z}),\quad \operatorname{ch} z = \frac{1}{2}\,(e^z + e^{-z}),$$

$$\operatorname{th} z = \frac{\operatorname{sh} z}{\operatorname{ch} z},\quad \operatorname{cth} z = \frac{\operatorname{ch} z}{\operatorname{sh} z}$$

5. Логарифмическая функция

$$\operatorname{Ln} z = \ln\,|z| + i\,(\arg z + 2\,k\,\pi)$$

Функция Ln z является многозначной. В каждой точке z, отличной от нуля и ∞, она принимает бесконечно много значений. Выражение $\ln|z| + i \arg z$ называется главным значением логарифмической функции и обозначается через $\ln z$. Таким образом,

$$\operatorname{Ln} z = \ln z + 2\,k\,\pi\,i$$

6. Общая степенная функция

$$z^a = e^{a\,\operatorname{Ln} z},\; a \in \mathbb{C}$$

Эта функция многозначная, ее главное значение равно $e^{a\ln z}$. Если $a = \frac{1}{n}$, где $n \in \mathbb{N}$, то получаем многозначную функцию - корень n-ой степени из комплексного числа:

$$z^{\frac{1}{n}} = e^{\frac{1}{n}(\ln|z|+i(\arg z+2k\pi))} = \sqrt[n]{|z|}\, e^{i\frac{\arg z+2k\pi}{n}}$$

7. Общая показательная функция

$$a^z = e^{z\,\mathrm{Ln}\,a}, \ a \in \mathbb{C}$$

Главное значение этой многозначной функции равно $e^{z\ln a}$. В дальнейшем при $a > 0$ полагаем $a^z = e^{z\ln a}$.

8. Обратные тригонометрические

$$\mathrm{Arccos}\, z = -i\,\mathrm{Ln}\left(z + \sqrt{z^2-1}\right), \mathrm{Arcsin}\, z = -i\,\mathrm{Ln}\left(i\,z + \sqrt{-z^2+1}\right)$$

$$\mathrm{Arctg}\, z = -\frac{i}{2}\,\mathrm{Ln}\left(\frac{1+i\,z}{1-i\,z}\right)$$

Пример №11.53

Выделить действительную и мнимую части следующей функции:

$$w = e^{1-z}$$

Запишем z, используя алгебраическую форму записи комплексного числа:

```
in : z = x + I y;
```

Теперь воспользуемся функцией ComplexExpand, чтобы получить разложение заданной функции w:

```
in : ComplexExpand[Exp[1 - z]]
out : e^(1-x) Cos[y] - i e^(1-x) Sin[y]
```

Итак получили, что:

$$u(x, y) = e^{1-x}\cos y \ - \text{ действительная часть}$$

$$v(x, y) = -e^{1-x}\sin y \ - \text{ мнимая часть}$$

Пример №11.55

Выделить действительную и мнимую части следующей функции:

$$w = \sin(z - i)$$

Воспользуемся той же командой:

```
in : ComplexExpand[Sin[z - I]]
out : Cosh[1 - y] Sin[x] - i Cos[x] Sinh[1 - y]
```

Получили, что действительная и мнимая части равны:

$$u(x,\ y) = \operatorname{ch}(1 - y)\sin x$$
$$v(x,\ y) = -\operatorname{sh}(1 - y)\cos x$$

Пример №11.62

Вычислить значения функции в указанной точке:

$$\cos(z),\ z = 1 + i$$

Если вычислить значение косинуса в точке:

```
in : Cos[1 + I]
out : Cos[1 + ⅈ]
```

то видно, что Mathematica самостоятельно не преобразует выражения. Значит, если мы хотим получить ответ в другой форме, то нам необходимо воспользоваться преобразующими функциями. В данном случае - это ComplexExpand:

```
in : Cos[1 + I] // ComplexExpand
out : Cos[1] Cosh[1] - ⅈ Sin[1] Sinh[1]
```

Итак, получили, что:

$$\cos(1 + i) = \cos(1)\operatorname{ch}(1) - i\sin(1)\operatorname{sh}(1)$$

Пример №11.65

Вычислить значения функции в указанной точке:

$$\operatorname{Ln}(z),\ \ z = -1$$

В системе Wolfram Mathematica функция натурального логарифма имеет вид Log[a]. Она принимает в том числе и комплексные значения. Попытаемся найти значение логарифма от -1:

```
in : Log[-1]
out : ⅈ π
```

Мы получили главное значение логарифма, т.е. функция логарифма в системе Mathematica определена как:

$$\operatorname{Log}[z] \leftrightarrow \ln(z)$$

Чтобы получить множество значений функции Ln(z), необходимо к функции ln(z) добавить $2\pi k$ i:

```
in : Log[-1] + 2 Pi k I // Factor
out : ⅈ (1 + 2 k) π
```

В итоге получаем ответ:

$$\operatorname{Ln}(-1) = i\pi(1 + 2k)$$

Пример №11.75

Найти значение модуля и главное значение аргумента заданной функции в указанной точке:

$$w = \sin z, \quad z = \pi + i \ln 3$$

Найдем значение функции в точке z и обозначим его за w:

```
in : w = Sin[Pi + I Log[3]]
```

out : $-\frac{4 i}{3}$

Теперь в рамках условия задачи воспользуемся функцией Abs[z], чтобы найти абсолютное значение (модуль) переменной w:

```
in : Abs[w]
```

out : $\frac{4}{3}$

Затем используем функцию Arg[z], чтобы найти аргумент числа w:

```
in : Arg[w]
```

out : $-\frac{\pi}{2}$

Итак, ответ:

$$|w| = |\sin(\pi + i \ln 3)| = \frac{4}{3}$$

$$\arg w = \arg \sin(\pi + i \ln 3) = -\frac{\pi}{2}$$

Пример №11.80

Найти все значения степени:

$$(-1)^{i}$$

По определению общая показательная функция задана, как:

$$a^{z} = e^{z \operatorname{Ln} a}$$

Используя результат задачи 11.65, приходим к выводу, что:

$$(-1)^{i} = e^{i \operatorname{Ln}(-1)} = e^{i\, i\, \pi(1+2k)} = e^{-\pi(1+2k)}, \text{ где } k \in \mathbb{Z}$$

Пример №11.87

Решить уравнения:

$$e^{iz} - 1 = 0$$

Для решения уравнений и неравенств в системе Wolfram Mathematica есть очень удобная функция - Solve[выражение, список переменных/функций],с которой каждый, кто работал ранее в Mathematica, конечно, должен быть знаком. Воспользуемся ей, чтобы получить решение исходного уравнения:

```
in : Solve[Exp[z] - I == 0, z]
```

$$out: \left\{\left\{z \to \text{ConditionalExpression}\left[\frac{i\pi}{2} + 2\,i\,\pi\,C[1],\ C[1] \in \text{Integers}\right]\right\}\right\}$$

Во-первых, обратим внимание на форму ответа - список из списков. Во-вторых, если функция-решение периодическая, то будут сгенерированы постоянные (C[1]), и при помощи функции ConditionalExpression будут указанны условия. Здесь - C[1] принадлежит множеству целых чисел (Integers). Чтобы получить само выражение-ответ, можно воспользоваться оператором “/.” :

```
in : z /. Solve[Exp[z] - I == 0, z]
```

$$out: \left\{\text{ConditionalExpression}\left[\frac{i\pi}{2} + 2\,i\,\pi\,C[1],\ C[1] \in \text{Integers}\right]\right\}$$

Однако предварительно необходимо “сгладить” глубину вложенных списков:

```
in : z /. Flatten[Solve[Exp[z] - I == 0, z]] // Factor
```

$$out: \text{ConditionalExpression}\left[\frac{1}{2}\,i\,\pi\,(1 + 4\,C[1]),\ C[1] \in \text{Integers}\right]$$

В дальнейшем всем этим операциям дополнительного внимания уделяться не будет.

Итак, ответ:

$$z = \frac{1}{2}\,i\,\pi\,(1 + 4\,k),\ \ k \in \mathbb{Z}$$

1.3. Предел и непрерывность функции комплексной переменной.

Число $A \neq \infty$ называется пределом функции $f(z)$ при $z \to z_0$ и обозначается $A = \lim_{z \to z_0} f(z)$, если для любого $\epsilon > 0$ найдется $\delta(\epsilon) > 0$ такое, что для всех $z \neq z_0$, удовлетворяющих неравенству $|z - z_0| < \delta(\epsilon)$, выполняется неравенство $|f(z) - A| \leq \epsilon$. Либо, записывая символьным языком:

$$\lim_{z \to z_0} f(z) = A \Longleftrightarrow \forall\, \epsilon > 0\ \exists\, \delta(\epsilon) > 0 : \forall\, z :\ |z - z_0| < \delta(\epsilon) :\ |f(z) - A| \leq \epsilon$$

Говорим, что $\lim_{z \to z_0} f(z) = \infty$, если для любого $R > 0$ найдется $\delta = \delta(R) > 0$ такое, что для всех $z \neq z_0$ таких, что $|z - z_0| < \delta(R)$, выполняется неравенство: $|f(z)| > R$.

Функция $f(z)$ называется непрерывной в точке z_0, если она определена в этой точке и $f(z_0) = \lim_{z \to z_0} f(z)$.

Функция $f(z)$, непрерывная в каждой точке области D, называется непрерывной в этой области. Или на языке символов:

$$f(x) \text{ непрерывна в } D \Longleftrightarrow \forall\, z_i \in D:\ f(z_i) = \lim_{z\to z_i} f(z)$$

Функция $f(z)$ называется равномерно непрерывной в области D, если для любого $\epsilon > 0$ найдется $\delta(\epsilon)$>0 такое, что для любых точек z_1 и z_2 из области D таких, что $|z_1 - z_2| < \delta(\epsilon)$, выполняется неравенство $|f(z_1) - f(z_2)| < \epsilon$.

Пример №11.92

Вычислить предел:

$$\lim_{z\to i} \frac{z^2 - 4\,i\,z - 3}{z - i}$$

Для вычисления предела в Wolfram Mathematica существует функция Limit[$f(z)$, $z \to z_0$], которая, разумеется, автоматически применяет все правила нахождения предела. Воспользуемся ею:

```
in : Limit[(z^2 - 4 I z - 3)/(z - I), z → I]
out : -2 i
```

В итоге:

$$\lim_{z\to i} \frac{z^2 - 4\,i\,z - 3}{z - i} = -2\,i$$

Примечание: в будущем дополнительное внимание функции Limit уделяться не будет.

§2. Аналитические функции. Условия Коши-Римана

2.1. Производная. Аналитичность функции

Если в точке $z \in D$ существует предел

$$\lim_{\Delta z\to 0} \frac{f(z + \Delta z) - f(z)}{\Delta z},\ z + \Delta z \in D$$

то он называется производной функции $f(z)$ в точке z и обозначается через $f'(z)$, $\frac{d}{dz} f(z)$.

Функция $f(z)$, дифференцируемая в каждой точке области D и имеющая в этой области непрерывную производную $f'(z)$, называется аналитической в области D. Будем также говорить, что $f(z)$ аналитическая в точке $z_0 \in D$, если $f(z)$ является аналитической в некоторой окрестности точки z_0.

Для того чтобы функция $f(z) = u(x, y) + i\,v(x, y)$ была аналитической в области D, необходимо и достаточно существование в этой области непрерывных частных производных функций $u(x, y)$ и $v(x, y)$, удовлетворяющих условиям Коши-Римана:

$$\frac{\partial}{\partial x}u(x,\ y) = \frac{\partial}{\partial y}v(x,\ y),\ \ \frac{\partial}{\partial y}u(x,\ y) = -\frac{\partial}{\partial x}v(x,\ y) \tag{1}$$

или, в полярных координатах:

$$\begin{aligned} &\frac{\partial}{\partial r}u(r\cos(\phi),\ r\sin(\phi)) = \frac{1}{r}\ \frac{\partial}{\partial \phi}v(r\cos(\phi),\ r\sin(\phi)),\\ &\frac{\partial}{\partial r}v(r\cos(\phi),\ r\sin(\phi)) = \frac{-1}{r}\ \frac{\partial}{\partial \phi}u(r\cos(),\ r\sin()) \end{aligned} \tag{2}$$

При выполнении условий (1) или (2) производная $f'(z)$ может быть записана соответственно:

$$f'(z) = \frac{\partial}{\partial x}u + i\ \frac{\partial}{\partial x}v = \frac{\partial}{\partial y}v - i\ \frac{\partial}{\partial y}u = \frac{\partial}{\partial x}u - i\ \frac{\partial}{\partial y}u = \frac{\partial}{\partial y}v + i\ \frac{\partial}{\partial x}v \tag{3}$$

$$f'(z) = \frac{r}{z}\left(\frac{\partial}{\partial r}u - i\ \frac{\partial}{\partial r}v\right) = \frac{1}{z}\left(\frac{\partial}{\partial \phi}v - i\ \frac{\partial}{\partial \phi}u\right) \tag{4}$$

Пример №11.115

Проверить выполнение условий Коши - Римана и в случае их выполнения найти $f'(z)$:

$$f(z) = \cos(z)$$

Запишем функцию:

```
in : f[z_] := Cos[z]
```

Для проверки выполнения условий Коши - Римана (1) выделим мнимую и действительную части функции $f(z)$ при помощи опции ComplexExpand[expr]. Затем запишем две функции $u[x,\ y]$ и $v[x,\ y]$:

```
in : ComplexExpand[f[x + I y]]

out : Cos[x] Cosh[y] - i Sin[x] Sinh[y]

in : u[x_, y_] := Cos[x] Cosh[y]
     v[x_, y_] := -Sin[x] Sinh[y]
```

Теперь составим логическую конструкцию, которая будет определять, удовлетворяет ли заданная функция условиям Коши-Римана (1).

```
in : TrueQ[ D[u[x, y], x] - D[v[x, y], y] == 0] &&
      TrueQ[D[u[x, y], y] + D[v[x, y], x] == 0]

out : True
```

Получили True следовательно, условия (1) выполняются.

Производную функции найдем, воспользовавшись формулой (3):

```
in : D[u[x, y], x] + I D[v[x, y], x]
out : -Cosh[y] Sin[x] - i Cos[x] Sinh[y]
```

Получили ответ:

$$-\sin(x)\cosh(y) - i\cos(x)\sinh(y) = -\sin(z)$$

$$f'(z) = (\cos(z))' = -\sin(z)$$

2.2. Свойства аналитических функций

Ряд свойств, характерных для дифференцируемых функций действительной переменной, сохраняется и для аналитических функций.

Если $f(z)$ и $g(z)$ – аналитические в области D функции, то:

$$\begin{gathered}(f(z) + g(z))' = f'(z) + g'(z) \\ (f(z)\,g(z))' = f'(z)\,g(z) + f(z)\,g'(z) \\ \left(\frac{f(z)}{g(z)}\right)' = \frac{f'(z)\,g(z) - f(z)\,g'(z)}{g(z)^2}, \text{ где } g(z) \neq 0 \text{ в } D\end{gathered} \tag{5}$$

Действительная и мнимая части аналитической в области D функции $f(z) = u(x, y) + i\,v(x, y)$ являются гармоническими в этой области функциями, т.е. их лапласианы равны нулю:

$$\begin{gathered}\Delta u = \frac{\partial^2}{\partial x^2}u + \frac{\partial^2}{\partial y^2}u = 0 \\ \Delta v = \frac{\partial^2}{\partial x^2}v + \frac{\partial^2}{\partial y^2}v = 0\end{gathered} \tag{6}$$

Задание действительной или мнимой части аналитической в области функции определяется с точностью до произвольной константы. Например, если $u(x, y)$- действительная часть аналитической в области D функции $f(z)$, то

$$v(x, y) = \mathrm{Im}(z) = \int_{(x_0, y_0)}^{(x,y)} -u'_y\,dx + u'_x\,dy = \int_{x_0}^{x} -u'_y(x, y_0)\,dx + \int_{y_0}^{y} u'_x(x, y)\,dy \tag{7}$$

Аналогично

$$u(x, y) = \mathrm{Re}(z) = \int_{(x_0, y_0)}^{(x,y)} v'_y\,dx - v'_x\,dy = \int_{x_0}^{x} v'_y(x, y_0)\,dx - \int_{y_0}^{y} v'_x(x, y)\,dy \tag{8}$$

Пример №11.136

Проверить гармоничность функции в указанной области и найти, если возможно, аналитическую функцию по данной ее действительной части:

$$u(x, y) = x^2 - y^2 + x\,y, \quad 0 \le |z| < +\infty$$

Для того чтобы функция была гармоничной, она должна удовлетворять уравнению Лапласа (6), т.е. ее лапласиан равен нулю:

Введем функцию $u(x, y)$ и проверим, равен ли ее лапласиан нулю. Для этого воспользуемся функцией Laplacian[f[x1 ....xn], {x1 ... .. xn}] .

```
in : u[x_, y_] := x^2 - y^2 + x y
```

```
in : Laplacian[u[x, y], {x, y}] == 0
```

```
out : True
```

Лапласиан функции равен нулю, следовательно она является гармоничной.

Теперь найдем мнимую часть аналитической функции f (z), воспользовавшись формулой (7):

```
in : Integrate[-D[u[x, y], y] /. y → y0, {x, x0, x}] +
       Integrate[D[u[x, y], x], {y, y0, y}]
```

out : $-\frac{x^2}{2} + 2\,x\,y + \frac{y^2}{2} + \frac{x_0^2}{2} - 2\,x_0\,y_0 - \frac{y_0^2}{2}$

Получили функцию вида:

$$v(x, y) = -\frac{x^2}{2} + 2\,x\,y + \frac{y^2}{2} + C,\ \text{где } C - \text{const}$$

Убедимся, что данная функция также является гармоничной:

in : $\text{Laplacian}\left[-\frac{x^2}{2} + 2\,x\,y + \frac{y^2}{2},\ \{x, y\}\right] == 0$

```
out : True
```

Теперь составим выражение $f\,(x, y)$ и приведем его к $f\,(z)$:

$$f\,(x, y) = x^2 - y^2 + x\,y + i\left(\frac{-x^2}{2} + 2\,x\,y\ + \frac{y^2}{2}\right) + C,\ \text{где } C \in \mathbb{C}$$

Из алгебраической формы записи комплексного числа z выразим x или y и произведем замену: $x\ =\ z - i\,y$. Затем раскроем скобки и упростим выражение.

in : $x^2 - y^2 + x\,y + i\left(-\frac{x^2}{2} + 2\,x\,y + \frac{y^2}{2}\right)$ /. x → z - i y // Expand //
  Simplify

out : $\left(1 - \frac{i}{2}\right) z^2$

Итак, получили, что функция $f\,(z)$ имеет вид:

$$f\,(z) = z^2 - i\,\frac{z^2}{2} + C,\ \text{где } C \in \mathbb{C}$$

§3. Конформные отображения

3.1. Геометрический смысл модуля и аргумента производной.

Пусть $w = f(z)$ - аналитическая в точке z_0 функция и $f'(z_0) \neq 0$. Тогда $k = |f'(z_0)|$ геометрически равен коэффициенту растяжения в точке z_0 при отображении $w = f(z)$. Аргумент производной $\phi = \arg f'(z_0)$ геометрически равен углу, на который нужно повернуть касательную в точке z_0 к любой гладкой кривой L, проходящей через точку z_0, чтобы получить касательную в точке $w_0 = f(z_0)$ к образу L’ этой кривой при отображении $w = f(z)$. При этом, если $\phi > 0$, то поворот происходит против часовой стрелки, а если $\phi < 0$, то по часовой.

Таким образом, геометрический смысл модуля и аргумента производной состоит в том, что при отображении, осуществляемом аналитической функцией, удовлетворяющей условию $f'(z_0) \neq 0$, $k = |f'(z_0)|$ определяет коэффициент преобразования подобия бесконечно малого линейного элемента в точке z_0, а $\phi = \arg f'(z_0)$ - угол поворота этого элемента.

К р и т е р и й к о н ф о р м н о с т и о т о б р а ж е н и я. Для того чтобы отображение области D, задаваемое функцией $w = f(z)$, было конформным необходимо и достаточно, чтобы $f(z)$ была однолистной и аналитической в области D функцией, причем $f'(z) \neq 0$ всюду в D.

Пример №11.138

Найти коэффициент растяжения k и угол поворота ϕ для заданного отображения в указанной точке:

$$w = z^2,\ z_0 = \sqrt{2}\,(1 + i)$$

Запишем функцию w:

```
in : w[z_] := z^2
```

Теперь найдем коэффициент растяжения k:

```
in : k = Abs[D[w[z], z]] /. z → √2 (1 + I)
out : 4
```

Также найдем угол поворота ϕ:

```
in : ϕ = Arg[D[w[z], z]] /. z → √2 (1 + I)
out : π/4
```

В итоге, получаем:

$$k = 4,\ \ \phi = \frac{\pi}{4}$$

Пример №11.149

Найти множество всех тех точек z_0, в которых при следующем отображении коэффициент растяжения k=1:

$$w(z) = z^2 - i\,z$$

Для того чтобы найти это множество, требуется решить уравнение:

$$|\,w'(z)\,| = 1$$

Запишем функцию $w(z)$:

```
in : w[z_] := z^2 - I z
```

Найдем ее первую производную:

```
in : w'[z]
out : -i + 2 z
```

В итоге получаем уравнение:

$$|\,2\,z - i\,| = 1$$
$$\left|\,z - \frac{i}{2}\,\right| = \frac{1}{2}$$

Это есть ничто иное, как уравнение окружности с центром в точке $\left(0,\ \frac{1}{2}\right)$ и радиусом $r = \frac{1}{2}$.

Построим график этой окружности:

```
in : With[{z = x + I y}, ContourPlot[Abs[z - I / 2] == 1 / 2, {x, -1, 1},
       {y, -1, 1}, Axes → True, AxesLabel → {Re, Im}, Frame → False]]
```



рис. 3.1.1

Итак, ответ:

$$\text{множество точек } z,\ \text{ лежащих на окружности}: \left|\,z - \frac{i}{2}\,\right| = \frac{1}{2}$$

3.2. Отображение линейной функцией

Линейная функция - однолистная, аналитичная и нигде не обращается в ноль во всей расширенной комплексной плоскости. Таким образом, она совершает конформное отображение.

Это отображение, осуществляемое линейной функцией $w = a\,z + b$, представляет собой композицию поворота ($w_1 = e^{i \arg a}\, z$), растяжения ($w_2 = |a|\, w_1$) и параллельного переноса ($w_3 = w_2 + b$). Оно имеет две неподвижные точки: $z_1 = \frac{b}{1-a}$ при $a \neq 0$ и $z_2 = \infty$.

Рассмотрим преобразование, совершаемое функцией $w = f(z)$, и способы работы с отображениями в системе Wolfram Mathematica.

$$w(z) = 2\, e^{i \frac{4\pi}{3}}\, z + \left(\frac{1}{2} + i\, \frac{1}{2}\right)$$

В отличие от большинства систем компьютерной алгебры, Wolfram Mathematica предоставляет пользователю безумно удобный и функциональный набор опций для работы с отображениями. Для начала рассмотрим способы задания областей:

Функция ImplictRegion[выражение, список переменных] задает неявно заданную область. Например, область $|x| + |y| < 1$ запишется так:

```
in : R = ImplicitRegion[Abs[x] + Abs[y] < 1, {x, y}];
```

Изобразить ее можно с помощью функции RegionPlot:

```
in : RegionPlot[R, Frame → False, Axes → True]
```



рис. 3.2.1

Еще области можно задавать функциями Triangle, Circle, Polygon и т.п., то есть функциями геометрических фигур. Рассмотрим область треугольника с вершинами (0.5,0.5), (0,0), (0.5,-0.5):

```
in : R = Triangle[{{.5, .5}, {0, 0}, {0.5, -0.5}}];
     RegionPlot[R, Frame → False, Axes → True]
```



Вот так, например, задается “область” отрезка y=x, от 0 до 1:

```
in : line = ImplicitRegion[x == y, {{x, 0, 1}, {y, 0, 1}}];
     RegionPlot[line, Frame → False, Axes → True,
      PlotRange → {{-1, 2}, {-1, 2}}]
```



рис. 3.2.3

Для чего нам это нужно? С помощью функции ParametricPlot мы будем строить результаты действия функций отображения $w_1 = f(z)$, причем действовать этим отображением будем на все x и y из выражения $z = x + i\,y$, которые находятся в некоторой допустимой области. Рассмотрим пример действия функции $w_1 = e^{i\frac{4\pi}{3}} z$, которая совершает поворот на $\frac{4\pi}{3}$ радиан.

Запишем эту функцию:

```
in : w1[z_] := Exp[I 4 Pi / 3] z
```

Теперь с помощью ParametricPlot$\left[\{f_x,\ f_y\},\ \{x,\ y\} \in \text{reg}\right]$ построим отображение треугольника с вершинами (0, 0), (0.5, 0) и $\left(0,\ \sqrt{3}\,/2\right)$. Конструкция With[$\{z = x + i\,y\}$, блок функций] просто совершает замену переменной z на $x + i\,y$ внутри блока функций. Функция ReIm нужна для того, чтобы представить функцию в виде $\{f_x,\ f_y\}$.

```
in : tr = Triangle[{{0, 0}, {0.5, 0}, {0, √3 / 2}}];
     With[{z = x + I y}, ParametricPlot[ {ReIm[z], ReIm[w1[z]]},
       {x, y} ∈ tr, Frame → False, Axes → True,
       PlotRange → {{-1, 1}, {-1, 1}}]]
```



рис. 3.2.4

Первый треугольник со сплошными сторонами - треугольник до действия на него функции $w_1(z)$, второй треугольник - результат действия функции $w_1(z)$.

Теперь рассмотрим еще два отображения функциями $w_2 = 2\,w_1$ и $w_3 = w_2 + \left(\frac{1}{2} + i\,\frac{\sqrt{3}}{2}\right)$. Второе, как уже было сказано, совершает растяжение на величину 2, третье - перенос на вектор $\left(\frac{1}{2}, \frac{\sqrt{3}}{2}\right)$. Рассмотрим, как это будет выглядеть на практике. Сначала запишем функции:

```
in : w2[z_] := 2 w1[z]
     w3[z_] := w2[z] + 1/2 + I √3/2
```

Теперь рассмотрим последовательное их действие:

```
in : With[{z = x + I y},
      ParametricPlot[ {ReIm[w1[z]], ReIm[w2[z]], ReIm[w3[z]]},
       {x, y} ∈ tr, Frame → False, Axes → True,
       PlotRange → {{-2, 2}, {-2, 2}}]]
```



Итак, получили, что функция $w(z)$ есть композиция трех функций w_1, w_2, w_3. Конечный результат - треугольник, повернутый на $\frac{4}{3}\pi$ радиан, каждая сторона которого увеличена в 2 раза и смещенный на $\left(\frac{1}{2}, \frac{\sqrt{3}}{2}\right)$.

3.3. Некоторые виды отображения

Mathematica предоставляет собой мощный инструмент, позволяющий пользователю побыть в роли исследователя. Нам более не нужно придумывать громоздкие конструкции и методы, которые позволили бы наглядно увидеть, как действует некоторая функция комплексного переменного на некоторой области.

Не будем останавливаться подробно на каждом из отображений, рассмотрим лишь несколько примеров.

Дробно-линейная функция.

Найти образ линии при отображении:

$$\text{окружность } x^2 + y^2 = 2\,y,\ \ w = \frac{1}{z}$$

Запишем функцию w:

```
in : w[z_] := 1/z
```

Преобразуем уравнение окружности:

$$x^2 + y^2 = 2\,y \iff x^2 + (y-1)^2 = 1$$

Это окружность с центром в точке $(0,\ 1)$ и радиусом 1.

На этот раз обойдемся одним параметром. Представим число z в экспоненциальной форме: $z = e^{i\,a}$. Теперь, чтобы задать окружность, нужно варьировать параметр а от 0 до 2π. Также, чтобы центр окружности был в точке (0,1), необходимо сместить аргумент на 1 вверх. Посмотрим, как это выглядит на практике:

```
in : With[{z = Exp[I a]}, ParametricPlot[{ReIm[z + I], ReIm[w[z + I]]},
    {a, 0, 2 Pi}, Frame → False, Axes → True,
    PlotRange → {{-2, 2}, {-2, 3}}]]
```



рис. 3.3.1

Итак, мы получили, что окружность $x^2 + y^2 = 2y$ трансформируется в прямую Im[z]=-0.5

Степенная функция.

Найти отображении области *D* функцией *w*:

$$D = \{z \,||\, z\,| < 1, \operatorname{Im} z > 0\},\ \ w(z) = -\left(\frac{z+1}{z-1}\right)^2$$

Запишем функцию *w*(*z*):

```
in : w[z_] := -((z + 1)/(z - 1))^2
```

Теперь с помощью алгебраического представления числа *z* посмотрим результат отображения:

```
in : With[{z = x + I y}, ParametricPlot[ReIm[{z, w[z]}],
    {x, y} ∈ ImplicitRegion[Im[z] > 0 && Abs[z] < 1, {x, y}],
    Frame → False, Axes → True, PlotRange → {{-3, 3}, {-4, 2}}]]
```



Итак, функция $w(z)$ отображает верхнюю полуокружность в нижнюю полуплоскость.

Рассмотрим еще несколько степенных функций. Пусть у нас есть область квадрата. Рассмотрим действие на него функций z^2, z^3, $\pm z^{1/2}$

Исходная область + отображение квадратичной функцией:

```
in : With[{z = x + I y}, ParametricPlot[{ReIm[z], ReIm[z^2]},
      {x, y} ∈ ImplicitRegion[-1 < x < 1 && -1 < y < 1, {x, y}],
      Frame → False, Axes → True, PlotRange → {{-3, 3}, {-3, 3}}]]
```



рис. 3.3.3

Отображение кубической функцией:

```
in : With[{z = x + I y}, ParametricPlot[ReIm[z^3],
      {x, y} ∈ ImplicitRegion[-1 < x < 1 && -1 < y < 1, {x, y}],
      Frame → False, Axes → True, PlotRange → {{-3, 3}, {-3, 3}}]]
```



Отображение функциями $\pm\sqrt{z}$

```
in : With[{z = x + I y}, ParametricPlot[ReIm[{-z^(1/2), z^(1/2)}],
       {x, y} ∈ ImplicitRegion[-1 < x < 1 && -1 < y < 1, {x, y}],
       Frame → False, Axes → True, PlotRange → {{-2, 2}, {-2, 2}}]]
```



рис. 3.3.5

Функция Жуковского.

Найти образ заданной области D при отображении w:

$$w(z) = \frac{1}{2}\left(z + \frac{1}{z}\right),\quad D = \{z \mid 0.2 \le |z| \le 2,\ 0 \le \arg z \le \pi/2\}$$

Запишем функцию:

```
in : w[z_] := 1/2 (z + 1/z)
```

Теперь построим отображение сектора кольца D:

```
in : With[{z = x + I y}, ParametricPlot[ReIm[w[z]],
      {x, y} ∈ ImplicitRegion[0 ≤ Arg[z] ≤ Pi / 2 && 0.2 ≤ Abs[z] ≤ 2,
        {x, y}], Frame → False, Axes → True,
      PlotRange → {{-1, 3}, {-3, 1}}]]
```



рис. 3.3.6

Рассмотрим другие преобразования.

Показательная функция

Найти образ области D при отображении w

$$D = \{z \mid 0 < \operatorname{Im} z < 2\pi, \operatorname{Re} z > 0\},\ w = e^z$$

Построим отображение:

```
in : With[{z = x + I y}, ParametricPlot[ReIm[Exp[z]],
      {x, y} ∈ ImplicitRegion[0 < Im[z] < Pi / 2 && Re[ z] > 0, {x, y}],
      Frame → False, Axes → True, PlotRange → {{-1, 4}, {-1, 4}}]]
```



рис. 3.3.7

Результат отображения - первая четверть без четверть окружности $|w| < 1$.

Логарифмическая функция

Найти образ области D при отображении w

$$D = \{z \mid \ |z| < 1,\ \operatorname{Im} z > 0\},\ \ w = \ln z$$

Построим отображение:

```
in : With[{z = x + I y}, ParametricPlot[ReIm[Log[z]],
    {x, y} ∈ ImplicitRegion[0 < Im[z] && Abs[z] < 1, {x, y}],
    Frame → False, Axes → True, PlotRange → {{-5, 1}, {-1, 5}}]]
```



рис. 3.3.8

Результат отображения - полоса $\operatorname{Re} w < 0,\ \ 0 < \operatorname{Im} z < \pi$.

Тригонометрические функции

Найти образ области D при отображении w

$$D = \{z \mid \ 0 < \operatorname{Re} z < \pi,\ \ \operatorname{Im} z > 0\},\ \ w = \cos z$$

Построим отображение:

```
in : With[{z = x + I y}, ParametricPlot[ReIm[Cos[z]],
    {x, y} ∈ ImplicitRegion[0 < Im[z] && 0 < Re[z] < Pi, {x, y}],
    Frame → False, Axes → True, PlotRange → {{-5, 5}, {-5, 5}}]]
```



рис. 3.3.9

Результат отображения - полосы D в нижнюю полуплоскость.

§4. Интеграл от функции комплексной переменной.

4.1. Интеграл по кривой и его вычисление.

Пусть l - дуга направленной кусочно гладкой кривой в плоскости (z), точки $z_k \in l$, $k = 1 \dots n$ разбивают дугу L на частинчные дуги, на каждой из которых выбрано по одной точке ξ_k. По определению полагаем:

$$\int_l f(z)\, dz = \lim_{\max|\Delta z_k| \to 0} \sum_{k=1}^{n} f(\xi_k)\, \Delta z_k \tag{1}$$

при условии, что предел в правой части (1) существует и не зависит ни от способа разбиения дуги L на частичные дуги, ни от выбора точек ξ_k. Если функция $f(z)$ непрерывна на L, то интеграл (1) существует.

Если $f(z) = u(x,\ y) + i\, v(x,\ y)$, то вычисление интеграла (1) сводится к вычислению двух интегралов 2-го рода:

$$\int_l f(z)\, dz = \int_l u(x,\ y)\, d\,x - v(x,\ y)\, d\,y + \ i\int_l v(x,\ y)\, d\,x + u(x,\ y)\, d\,y \tag{2}$$

Пример №11.232

Вычислить интеграл по заданному контору:

$$\int_l \left(i\, z^2 - 2\,\bar{z}\right) dz,\ \ l = \{z \mid \ |z| = 2,\ \ 0 \le \ \arg z\ \le\ \pi/2\}$$

Выделим действительную и мнимую части функции:

```
in : f[z_] := I z^2 - 2 Conjugate[z]
```

Функция Refine преобразовывает выражение при условии, что элементы x и y принадлежат множеству действительных чисел:

```
in : u[x_, y_] := Refine[Re[ComplexExpand[f[x + I y]]],
       Element[x | y, Reals]]
     v[x_, y_] := Refine[Im[ComplexExpand[f[x + I y]]],
       Element[x | y, Reals]]
```

Выведем полученные функции:

```
in : u[x, y]
     v[x, y]
```

out : $-2\,x - 2\,x\,y$

out : $x^2 + 2\,y - y^2$

Теперь изобразим контур интегрирования:

```
in : With[{z = 2 Exp[I phi]},
  ParametricPlot[ReIm[z], {phi, 0, Pi / 2}, AxesLabel → {Re, Im},
   PlotRange → {{-1, 3}, {-1, 3}} , Frame → False, Axes → True]]
```



рис. 4.1.1

По формуле (2):

$$\int_l f(z)\, dz = -\int_0^2 u(x, y) - v(x, y)\,\frac{dy}{dx}\, dx - i\int_0^2 v(x, y) + u(x, y)\,\frac{dy}{dx}\, dx$$

Перед интегралом стоит знак минус, так как обход совершается по часовой стрелке.

Вычислим криволинейный интеграл с помощью замены $y = \sqrt{(4 - x^2)}$:

```
in : -Integrate[u[x, Sqrt[4 - x^2]] - v[x, Sqrt[4 - x^2]] D[Sqrt[4 - x^2], x], {x, 0, 2}] -
  I Integrate[v[x, Sqrt[4 - x^2]] + u[x, Sqrt[4 - x^2]] D[Sqrt[4 - x^2], x], {x, 0, 2}]
```

out : $\frac{8}{3} - \frac{4}{3}\, i\, (2 + 3\pi)$

Получили ответ:

$$\int_l \left(i z^2 - 2\bar{z}\right) dz = \frac{8}{3} - i\left(\frac{8}{3} + 4\pi\right)$$

Пример №11.241

Вычислить интеграл по заданному контору:

$$\int_l \operatorname{Im} z^2 \operatorname{Re} z^3\, dz, \quad l = \left\{(x, y) \mid\ y = 3x^3,\ 0 \le x \le\ 1\right\}$$

Запишем функцию и выделим действительную и мнимую части:

```
in : f[z_] := Im[z^2] Re[z^3]
```

```
in : u[x_, y_] := Refine[Re[ComplexExpand[f[x + I y]]],
       Element[x | y, Reals]]
     v[x_, y_] := Refine[Im[ComplexExpand[f[x + I y]]],
       Element[x | y, Reals]]
```

```
in : u[x, y]
     v[x, y]
```

out : $2 x^4 y - 6 x^2 y^3$

out : 0

Изобразим контур интегрирования:

```
in : Plot[3 x^3, {x, 0, 1}, PlotRange → {{-1/2, 3/2}, {-1/2, 3}}]
```



рис. 4.1.2

Произведем замену $y = 3\,x^3$ и вычислим интеграл по формуле (2):

```
in : int = Integrate[u[x, 3 x^3] - v[x, 3 x^3] D[3 x^3, x], {x, 0, 1}] +
       I Integrate[v[x, 3 x^3] + u[x, 3 x^3] D[3 x^3, x], {x, 0, 1}]
```

out : $-\frac{51}{4} - \frac{3456\,i}{35}$

Окончательный ответ:

$$\int_l \operatorname{Im} z^2 \operatorname{Re} z^3 \, dz = -\frac{51}{4} - \frac{3456\,i}{35}$$

Изложенную методику следует применять для того, чтобы решить №№12.230-248, что рекомендуется сделать самостоятельно.

4.2. Теорема Коши. Интегральная формула Коши.

Если функция $f(z)$ аналитична в односвязной области D, ограниченной контуром Г, и γ - замкнутый контур в D, то

$$\oint_\gamma f(z)\, dz = 0 \qquad (3)$$

Если функция $f(z)$ непрерывна в замкнутой области то $\overline{D} = D \cup \Gamma$, то справедлива теорема Коши:

$$\oint_{\Gamma} f(z)\, dz = 0$$

Если функция f(z) аналитична в многосвязной области D, ограниченной контуром Γ и внутренними по отношению к нему контурами γ_1, ..., γ_n, и непрерывна в замкнутой области $\overline{D} = D \cup \Gamma^+ \cup \gamma^-{}_1 \ldots \cup \gamma^-{}_n$, то:

$$\oint_{\Gamma^+ \cup \gamma^-{}_1 \ldots \cup \gamma^-{}_n} f(z)\, dz = 0 \tag{4}$$

Отсюда следует следующая формула:

$$\oint_{\Gamma^+} f(z)\, dz = \sum_{i=1}^{n} \oint_{\gamma^+{}_i} f(z)\, dz \tag{5}$$

Если функция $f(z)$ определена и непрерывна в односвязной области D, то если $F(z)$ - одна из первообразных для f(z), то:

$$\int_{z_1}^{z_2} f(z)\, dz = F(z_2) - F(z_1) \tag{6}$$

Если функция $f(z)$ аналитична в односвязной области D, $z_0 \in D$ и $\gamma \subset D$ - замкнутый контур, охватывающий точку z_0, то справедлива *интегральная формула Коши*

$$f(z_0) = \frac{1}{2\pi i} \oint_{\gamma} \frac{f(z)}{z - z_0}\, dz \tag{7}$$

При этом функция $f(z)$ имеет всюду в D производные любого порядка, для которых справедливы формулы:

$$f^{(k)}(z_0) = \frac{k!}{2\pi i} \oint_{\gamma} \frac{f(z)}{(z - z_0)^{k+1}}\, dz \tag{8}$$

Пример №12.249

Вычислить интеграл:

$$\int_{l} e^{z}\, dz,\ l = \left\{(x, y) \mid y = x^3,\ 1 \le x \le 2\right\}$$

Чтобы вычислить интеграл по формуле (6) требуется найти пределы интегрирования. Определим z, как:

```
in : z[x_, y_] := x + I y
```

Тогда точки z_1 и z_2 будут вычисляться следующим образом:

```
in : z1 = z[x, x^3] /. x → 1
     z2 = z[x, x^3] /. x → 2
```

```
out : 1 + i

out : 2 + 8 i
```

И тогда интеграл равен:

```
in : Integrate[Exp[z], {z, z1, z2}] // ComplexExpand // TrigReduce
```

```
out : - e Cos[1] + e^2 Cos[8] - i e Sin[1] + i e^2 Sin[8]
```

В итоге, получили ответ:

$$\int_l e^z\, dz = e(e\cos(8) - \cos(1)) + i\, e(e\sin(8) - \sin(1))$$

Данный тип задач не представляет совершенно никакой сложности, поэтому №№12.249-256 рекомендуются к самостоятельному решению.

Рассмотрим следующий тип задач, для решения которых требуется использование интегральной теоремы Коши.

Пример №11.258

Вычислить интегралы (обход контуров - против часовой стрелки)

$$\text{а. } \oint_{|z|=4} \frac{e^{2z}}{z - \pi i}\, dz; \qquad \text{б. } \oint_{|z|=1} \frac{e^{2z}}{z - \pi i}\, dz;$$

Запишем функцию:

```
in : f[z_] := Exp[2 z] / (z - I Pi)
```

Из теоремы (8) выразим интеграл:

$$\oint_\gamma \frac{f(z)}{(z - z_0)^{k+1}}\, dz = 2\pi i\, \frac{f^{(k)}(z_0)}{k!} \qquad (9)$$

Изобразим контуры $|z| = 4$ и $|z| = 1$, а также точку $z = \pi i$:

```
in : Show[ContourPlot[{x^2 + y^2 == 16, x^2 + y^2 == 1}, {x, -4, 4},
      {y, -4, 4}, Frame → False, Axes → True, AxesLabel → {Re, Im},
      ImageSize → 200, PlotLegends → {"|z|=4", "|z|=1"}],
     Graphics[{PointSize[Large], Point[ReIm[π i]]}]]
```



Из графика видно, что:

В случае (б) интеграл равен нулю, так как подынтегральная функция аналитична во всей области $|z| \le 1$

В случае (а) точка $z = \pi i$ лежит внутри контура интегрирования, следовательно интеграл будет вычисляться по формуле (9):

```
in : int = 2 Pi I Exp[2 z] /. z → Pi I
```

```
out : 2 i π
```

Получаем ответ:

$$\text{а. } \oint_{|z|=4} \frac{e^{2z}}{z - \pi i}\, dz = 2 i \pi; \qquad \text{б. } \oint_{|z|=1} \frac{e^{2z}}{z - \pi i}\, dz = 0;$$

Пример №11.260

Вычислить интегралы:

$$\text{а. } \oint_{|z-1|=1} \frac{\sin\left(\frac{\pi z}{2}\right)}{z^2 - 1}\, dz; \qquad \text{б. } \oint_{|z|=4} \frac{\sin\left(\frac{\pi z}{2}\right)}{z^2 - 1}\, dz;$$

Запишем функцию:

```
in : f[z_] := Sin[Pi z/2] / (z^2 - 1)
```

Для того чтобы найти значения заданных интегралов, изобразим контуры интегрирования $|z - 1| = 1$ и $|z| = 4$, а также точки $z_1 = 1$, $z_2 = -1$:

```
in : Show[ContourPlot[{(x - 1)^2 + y^2 == 1, x^2 + y^2 == 4}, {x, -4, 4},
     {y, -4, 4}, Frame → False, Axes → True, AxesLabel → {Re, Im},
     ImageSize → 200, PlotLegends → {"|z-1|=1", "|z|=4"}],
   Graphics[{PointSize[Large], Point[{{1, 0}, {-1, 0}}]}]]
```



рис. 4.2.2

Из графика заключаем что:

В случае (а) внутри контура интегрирования находится одна точка $z_1 = 1$, поэтому по формуле (9):

```
in : int1 = 2 I Pi (Sin[Pi z/2])/(z + 1) /. z → 1

out : ⅈ π
```

В случае (б) внутри контура находятся обе точки $z_1 = 1$, $z_2 = -1$, следовательно по теореме о многосвязной области (5) имеем:

$$\oint_{|z|=4} \frac{\sin\left(\frac{\pi z}{2}\right)}{z^2-1}\, dz = \oint_{|z-1|=r} \frac{\frac{\sin\left(\frac{\pi z}{2}\right)}{z+1}}{(z-1)}\, dz + \oint_{|z+1|=r} \frac{\frac{\sin\left(\frac{\pi z}{2}\right)}{z-1}}{z+1}\, dz, \quad \text{где } r < 2$$

По Формуле (9):

```
in : int2 = 2 I Pi (((Sin[Pi z/2])/(z + 1) /. z → 1) + ((Sin[Pi z/2])/(z - 1) /. z → -1))

out : 2 ⅈ π
```

Итак ответ:

$$\text{а. } \oint_{|z-1|=1} \frac{\sin\left(\frac{\pi z}{2}\right)}{z^2-1}\, dz = i\,\pi; \qquad \text{б. } \oint_{|z|=4} \frac{\sin\left(\frac{\pi z}{2}\right)}{z^2-1}\, dz = 2\,i\,\pi;$$

Пример №11.265

Вычислить интеграл:

$$\oint_{|z-1|=1} \frac{dz}{(z-1)^3 (z+1)^3}$$

Запишем всё подынтегральное выражение:

```
in : f[z_] := 1/((z - 1)^3 (z + 1)^3)
```

На этот раз решим задачу аналитически. Имеется функция $f(z)$ и контур $|z-1| = 1$, требуется найти интеграл.

Первое, что требуется сделать, - это найти точки, в которых функция $f(z)$ теряет аналитичность. Для этого воспользуемся опцией Denominator[exrp] , которая возвращает знаменатель выражения expr, причем нас интересуют лишь те точки, которые лежат внутри контура интегрирования:

```
in : sol = z /. Solve[Abs[z - 1] < 1 && Denominator[f[z]] == 0, z]

out : {1, 1, 1}
```

Конечно, в данном случае нет ничего сложного, но будем действовать поэтапно: определим кратность каждого из корней (т.к. в формуле (9) нам требуется знать порядок производной). Для этого поступим так:

Рассмотрим уравнение: $(z-1)^3 (z+1)^3 == 0$

Решая его относительно z, получаем ответ:

```
in : solp = z /. Solve[(z - 1)^3 (z + 1)^3 == 0, z]

out : {-1, -1, -1, 1, 1, 1}
```

Здесь число повторений корня равно его кратности. Определим сколько раз встречается i-ый элемент при помощи функции Counts:

```
in : rule = Counts[solp]

out : <|-1 → 3, 1 → 3|>
```

Мы получили ассоциацию вида: <|..., i – ый корень $\rightarrow$ его кратность, ...|>

Теперь, чтобы получить кратность корня, воспользуемся следующей конструкцией:

```
in : solp[[1]] /. rule

out : 3
```

В данной конструкции происходит замена корня ($z = 1$) на его кратность (3).

Вернемся к исходной задаче и составим правило замены корня на его кратность:

```
in : rule = Counts[sol]

out : <|1 → 3|>
```

Теперь определим функцию кратности от порядкового номера корня:

```
in : k[i_] := DeleteDuplicates[sol][[i]] /. Counts[sol]

in : k[1]

out : 3
```

Введем множество уникальных корней (без повторов) при помощи функции DeleteDuplicates:

```
in : sol1 = DeleteDuplicates[sol]

out : {1}
```

Теперь, для того чтобы вычислить интеграл, нужно лишь воспользоваться формулами (9) и (5) :

```
in : int =
      2 Pi I
       Sum[
        (D[Simplify[f[z] (z - sol1[[i]])^k[i]], {z, k[i] - 1}] /
           (k[i] - 1) !) /. z → sol1[[i]], {i, 1, Length[sol1]}]
```

$$out: \frac{3\,i\,\pi}{8}$$

Получили ответ:

$$\oint_{|z-1|=1} \frac{dz}{(z-1)^3\,(z+1)^3} = \frac{3\,i\,\pi}{8}$$

Данную методику следует применять для решения подобных задач, представленных в №№11.257-11.271.

Общий алгоритм

При помощи опыта, полученного после решения задачи №11.265, напишем программу, которая будет искать значения контурных интегралов на основе интегральных теорем Коши:

Пусть имеется функция f(z), имеющая вид:

$$f(z) = \frac{\psi(z)}{(z-z_1)^{k_1}\,....(z-z_n)^{k_n}}$$

Требуется найти интеграл:

$$\oint_C f(z)\,dz;\quad C = \{z\mid\ |z-a| = b\}\text{, где } a \text{ и } b \ - \text{ параметры}$$

Для решения данной задачи составим алгоритм:

1. Нахождение точек $z_1 \dots z_n$ лежащих внутри $|z - a| = b$

2. Определение кратности каждой (k_i)

3. Комбинация формул (5) и (9).

А теперь напишем программный модуль:

```
in : integrateC[z_, expr_, a_, b_] :=
  Module[{sol, sol1, k},
   sol =
    z /.
     Solve[Abs[z - a] < b && Denominator[Simplify@TrigToExp[expr]] ==
        0, z];
   sol1 = DeleteDuplicates[sol];
   k[i_] := sol1[[i]] /. Counts[sol];
   2 I Pi
    Sum[ Limit[D[ Simplify[ expr (z - sol1[[i]])^k[i]], {z, k[i] - 1}] /
        (k[i] - 1) !, z → sol1[[i]]], {i, 1, Length[sol1]}]]
```

Разберем устройство полученного модуля:

1. Сначала решаем уравнения Abs[$z - a$] < b и Denominator[f[z]] = 0. Это позволит нам отыскать $z_1 \dots z_n$, которые лежат внутри контура интегрирования. Функция TrigToExp нужна для того, чтобы преобразовать тригонометрические функции в экспоненциальные, так как с ними удобнее работать.

2. Составляем множество уникальных корней sol1.

3. Определяем кратность i-ого уникального корня.

4. Используем формулу (9) и (5) для нахождения значения интеграла.

Продемонстрируем примеры:

Пример №12.266

```
in : integrateC[z, Sin[z]/(z + I)^3, -I, 1]

out : -π Sinh[1]
```

Пример №12.270

```
in : integrateC[z, 1/z^3 Cos[Pi/(z + 1)], 0, 1/2]

out : ⅈ π^3
```

Пример №

```
in : integrateC[z, 1/Sin[z], 0, 4]
```

```
out : -2 ⅈ π
```

Полученную функцию добавим в пользовательскую библиотеку, чтобы использовать в дальнейшем.

Глава 2. Представление рядов в системах компьютерной математики.

§1. Числовые ряды

1.1. Сходимость ряда. Абсолютная и условная сходимость. Признаки абсолютной сходимости.

Пусть имеется числовой ряд:

$$u_1 + u_2 + \ldots + u_n + \ldots = \sum_{n=1}^{\infty} u_n \tag{1}$$

К р и т е р и й К о ш и. Для того чтобы числовой ряд (1) был сходящимся, необходимо и достаточно, чтобы для любого $\epsilon > 0$, существовало $N = N(\epsilon)$ такое, что для всех $n > N$ и p=1,2.... выполнялось неравенство

$$|S_{n+p} - S_n| = |u_{n+1} + u_{n+2} + \ldots + u_{n+p}| < \epsilon$$

Н е о б х о д и м ы й п р и з н а к с х о д и м о с т и. Если ряд (1) сходится, то

$$\lim_{n\to\infty} u_n = 0 \tag{2}$$

Ряд (1) называется абсолютно сходящимся, если сходится ряд из модулей членов этого ряда, т.е. сходится ряд

$$|u_1| + |u_2| + \ldots + |u_n| + \ldots = \sum_{n=1}^{\infty} |u_n| \tag{3}$$

Если ряд (1) сходится, а ряд (3) расходится, то ряд (1) называется условно сходящимся.

П р и з н а к с р а в н е н и я р я д о в. Если члены ряда (1) для всех $n > N_0$ удовлетворяют условию $|u_n| \le b_n$, причем ряд $\sum_{n=1}^{\infty} b_n$ сходится, то ряд (1) сходится абсолютно. Если же для $n > N_1$ члены ряда (1) удовлетворяют условию $0 \le c_n \le |u_n|$, причем ряд $\sum_{n=1}^{\infty} c_n$ расходится, то ряд (1) не сходится абсолютно.

П р е д е л ь н ы й п р и з н а к с р а в н е н и я. Если ряд $\sum_{n=1}^{\infty} b_n$ сходится абсолютно и существует конечный предел $lim_{n\to\infty} \left|\frac{u_n}{b_n}\right| = q < +\infty$, то ряд (1) также сходится абсолютно.

Если же члены рядов u_n и b_n - действительные положительные числа и

$$0 < \lim_{n\to\infty} \frac{u_n}{b_n} < +\infty$$

то ряды $\sum_{n=1}^{\infty} u_n$ и $\sum_{n=1}^{\infty} b_n$ либо оба сходятся, либо оба расходятся.

П р и з н а к Д а л а м б е р а. Если члены ряда (1) таковы, что существует конечный предел

$$\lim_{n\to\infty} \left| \frac{u_{n+1}}{u_n} \right| = q \tag{4}$$

то при $0 \leq q < 1$ ряд (1) сходится абсолютно, при $q > 1$ - расходится, а при $q = 1$ требуется дополнительное исследование.

П р и з н а к К о ш и. Пусть

$$\lim_{n\to\infty} \sqrt[n]{|u_n|} = q \tag{5}$$

Тогда при $0 \leq q < 1$ ряд (1) сходится абсолютно, при $q > 1$ (1) - расходится, а при $q = 1$ требуется дополнительное исследование .

И н т е г р а л ь н ы й п р и з н а к К о ш и. Пусть функция f(x) положительна и монотонна при $x \geq 1$, и пусть для всех $n \in \mathbb{N}$ имеет равенство $f(n) = |u_n|$. Тогда числовой ряд (3) сходится абсолютно или расходится с несобственным интегралом:

$$\int_a^{\infty} f(x)\, dx, \ a \geq 1 \tag{6}$$

Пример №12.24

Используя признак сравнения или предельный признак сравнения, исследовать на сходимость следующий ряд:

$$\sum_{n=1}^{\infty} \frac{n^2+3}{4n^3+5n}$$

Так как ряд $\sum \frac{1}{n}$ расходится, и:

```
in : Limit[(n^2 + 3)/(4 n^3 + 5 n) / (1/n), n → Infinity]
out : 1/4
```

Значит, по предельному признаку сравнения исходный ряд также расходится.

Пример №12.33

Пользуясь признаком Даламбера, исследовать на сходимость следующий ряд:

$$\frac{3}{1} + \frac{3\cdot 5}{1\cdot 4} + \ldots + \frac{3\cdot 5\cdot \ldots (2n+1)}{1\cdot 4\cdot \ldots (3n-2)}$$

Запишем n-ый член ряда с помощью функции Product[f, {i, i_{max}}], которая в традиционном виде выглядит как: $\prod_{i=1}^{i_{max}} f(i)$

```
in : u[n_] := Product[2 k + 1, {k, n}] / Product[3 k - 2, {k, n}]
```

Для определения абсолютной сходимости ряда воспользуемся признаком Даламбера (4):

$$in: \texttt{Limit}\left[\texttt{Abs}\left[\frac{\texttt{u[n + 1]}}{\texttt{u[n]}}\right], \texttt{n} \to \texttt{Infinity}\right] \texttt{// N}$$

```
out : 0.6666666666666667
```

Так как $0 \le 0.67 < 1$, то данный ряд сходится.

Пример №12.45

Используя признак Коши, исследовать на сходимость следующий ряд:

$$\sum_{n=1}^{\infty} n\left(\arcsin\frac{1}{n}\right)^n$$

Запишем член ряда:

$$in: \texttt{u[n\_]} := n\left(\texttt{ArcSin}\left[\frac{1}{n}\right]\right)^n$$

Воспользуемся признаком Коши (5):

$$in: \texttt{Limit}\left[\left(\texttt{n ArcSin}\left[\frac{1}{\texttt{n}}\right]^{\texttt{n}}\right)^{1/\texttt{n}}, \texttt{n} \to \texttt{Infinity}\right]$$

```
out : 0
```

По признаку Коши ряд сходится.

Пример № 12.52

Используя интегральный признак Коши, исследовать на сходимость следующие ряды:

$$\sum_{n=3}^{\infty} \frac{1}{n \ln n\,(\ln \ln n)^2}$$

Запишем член ряда:

$$in: \texttt{u[n\_]} := \left(n\,\texttt{Log[n]}\;(\texttt{Log[Log[n]]})^2\right)^{-1}$$

Воспользуемся интегральным признаком Коши (6):

```
in : Integrate[u[x], {x, 3, Infinity}]
```

$$out: \frac{1}{\texttt{Log[Log[3]]}}$$

Получили, что данный интеграл сходится, значит и исходный ряд также сходится.

Система Wolfram Mathematica имеет встроенную функцию проверки сходимости ряда: SumConvergence[f, n]. Продемонстрируем его работу:

in : SumConvergence$\left[\frac{1}{n}, n\right]$

out : False

in : SumConvergence$\left[\frac{1}{n \operatorname{Log}[n] \operatorname{Log}[\operatorname{Log}[n]]^3}, n\right]$

in : True

in : SumConvergence$\left[\frac{2^{n^2}}{n!}, n\right]$

out : False

in : SumConvergence$\left[\frac{1}{\sqrt{n}} \operatorname{Log}\left[\frac{n+1}{n-1}\right], n\right]$

out : True

В методических целях будем применять критерии сходимости, изложенные в теоретической части, а использовать функцию SumConvergence[f,n] будем для самопроверки.

Задания под номерами №№12.19-12.86 рекомендуются к самостоятельному решению.

1.2. Признаки условной сходимости.

Признак Лейбница. Пусть члены a_n знакочередующегося ряда

$$a_1 - a_2 + a_3 + \ldots + (-1)^{n+1} a_n + \ldots \tag{7}$$

действительны и монотонно убывают, т.е.

$$a_1 > a_2 > a_3 > \ldots . > a_n \tag{8}$$

и

$$\lim_{n \to \infty} a_n = 0 \tag{9}$$

Тогда ряд (7) сходится, причем для его суммы S имеет место оценка $S < a_1$

Признак Абеля-Дирихле. Пусть члены последовательности (b_n) монотонно убывают: $b_1 > b_2 > b_3 \ldots > b_n$, и $\lim_{n\to\infty} b_n = 0$, а частичные суммы $S_n = a_1 + a_2 + \ldots + a_n$, ограничены в совокупности, т.е.

$$\left| \sum_{k=1}^{n} a_k \right| \leq M$$

тогда ряд $\sum_{n=1}^{\infty} a_n \; b_n$ сходится.

Пример №12.95

Исследовать на абсолютную и условную сходимость следующий ряд:

$$\sum_{n=2}^{\infty} (-1)^n \frac{\ln n}{n}$$

Запишем член ряда:

```
in : a[n_] := (-1)^n Log[n]/n
```

Исследуем на абсолютную сходимость ряд при помощи интегрального признака Коши:

```
in : Integrate[Abs[a[x]], {x, 2, Infinity}]

     Integrate::idiv : Integral of Log[x]/x does not converge on {2, ∞}. >>

out : ∫_2^∞ e^(-π Im[x]) Abs[Log[x]/x] dx
```

Получили сообщение о том, что интеграл расходится.

Так как ряд не сходится абсолютно, исследуем на условную сходимость по признаку Лейбница.

Покажем графически, что модули членов ряда убывают:

```
in : ListPlot[Table[Abs[a[n]], {n, 3, 10}]]
```

рис. 1.2.1

Первое условие признака Лейбница выполняется.

Найдем предел:

```
in : Limit[Abs[a[n]], n → Infinity]

out : 0
```

Условия признака Лейбница выполняются, следовательно ряд сходится условно. Проверим результат:

```
in : SumConvergence[Abs[a[n]], n]

out : False

in : SumConvergence[a[n], n]

out : True
```

Ответ: ряд сходится условно.

Данный порядок действий следует применить для решения задач под номерами №№12.90-12.109

§2. Функциональные ряды

2.1. Область сходимости функционального ряда

Пусть функции $f_n(z),\ \ n \in \mathbb{N}$, определены в области D. Выражение

$$f_1(z) + f_2(z) + \ldots + f_n(z) + \ldots = \sum_{n=1}^{\infty} f_n(z),\ \ z \in D \tag{1}$$

называется функциональным рядом. Если для $z_0 \in D$, числовой ряд $\sum_{n=1}^{\infty} f_n(z_0)$ сходится, то говорим, что функциональный ряд (1) сходится в точке z_0. Если в каждой точке $z \in D_1 \subset D$ числовые ряды $\sum_{n=1}^{\infty} f_n(z)$ сходятся, то ряд (1) называется сходящимся в области D_1.

К р и т е р и й К о ш и. Для того чтобы функциональный ряд (1) был сходящимся в области D_1, необходимо и достаточно, чтобы для любого $\epsilon > 0$ и для любого $z_0 \in D_1$, существовало $N = N\,(\epsilon, \text{z})$, такое, что

$$\left| f_{n+1}(z) + f_{n+2}(z) + \ldots + f_{n+p}(z) \right| < \epsilon,\ \text{ для всех } n > N(\epsilon, z) \text{ и } p \in \mathbb{N}. \tag{2}$$

Для определения области абсолютной сходимости функционального ряда (1) следует воспользоваться либо признаком Даламбера, либо признаком Коши. Именно, если:

$$\lim_{n \to \infty} \left| \frac{f_{n+1}(z)}{f_n(z)} \right| = q(z) \ \text{ или}$$

$$\lim_{n \to \infty} \sqrt[n]{|\, f_n(z)\, |} = q(z)\,,$$

то для определения области абсолютной сходимости ряда (1) следует решить функциональное неравенство $q(z) < 1$. При этом для изучения поведения ряда в граничных точках получаемой области, т.е. в точках, описываемых уравнением $q(z) = 1$, требуется дополнительное исследование.

Пример №12.124

Найти область сходимости ряда ($x \in \mathbb{R}$). Исследовать ряд на абсолютную сходимость.

$$\sum_{n=1}^{\infty} (-1)^n\, n^{-x}$$

Запишем общий член ряда:

```
in : f[n_, x_] := (-1)^n n^-x
```

По признаку Лейбница данный ряд сходится при любых $x \,|\, x > 0$, так как выполняются все необходимые условия.

На границе сходимости $x = 0$ получим ряд:

$$\sum_{n=1}^{\infty}(-1)^n$$

Данный ряд не является сходящимся, и следовательно область сходимости исходного ряда: $x > 0$

Ряд, составленный из модулей будет выглядеть так:

$$\sum_{n=1}^{\infty}\frac{1}{n^x}$$

Данный ряд сходится абсолютно при x>1. Докажем это, используя интегральный признак Коши:

in : `Integrate[1/n^x, {n, 1, Infinity}, Assumptions → x ∈ Reals]`

out : `ConditionalExpression[1/(-1 + x), x > 1]`

Получили, что x >1. Следовательно, область сходимости данного ряда: $x \in (1, +\infty)$.

Пример №12.125

Найти область сходимости ряда ($x \in \mathbb{R}$). Исследовать ряд на абсолютную сходимость.

$$\sum_{n=1}^{\infty}\frac{\cos(n\,x)}{n\sqrt{n}}$$

Запишем член ряда:

in : `f[n_, x_] := Cos[n x] / n^(3/2)`

Чтобы воспользоваться признаком Абеля-Дирихле необходимо доказать, что частичные суммы ряда $\sum_{k=1}^{n}\mathrm{Cos}(n\,x)$ ограничены в совокупности:

in : `Sum[Cos[k x], {k, 1, n}]`

out : `Cos[1/2 (1 + n) x] Csc[x/2] Sin[n x/2]`

Так как произведение Sin (a) Cos (b) по модулю не превосходит 1 при любых a и b, имеем:

$$\left|\sum_{n=1}^{\infty}\cos(n\,x)\right| \leq \frac{1}{\left|\sin\left(\frac{x}{2}\right)\right|} \qquad x \neq 0$$

Что и требовалось доказать.

Так как ряд $\sum_{n=1}^{\infty}\frac{1}{n^{3/2}}$ сходится, то и исходный ряд:

$$\sum_{n=1}^{\infty} \frac{\cos(n\,x)}{n\sqrt{n}}$$

по признаку Абеля-Дирихле сходится при любых $x \neq 0$. Следовательно область сходимости данного ряда: $x \in \mathbb{R} \setminus \{0\}$

Пример №12.136

Найти область абсолютной сходимости ряда:

$$\sum_{n=1}^{\infty} \frac{n\,2^n}{(z-3\,i)^{2\,n}}$$

Запишем общий член ряда:

```
in : f[n_, z_] := n 2^n / (z - 3 I)^(2 n)
```

Найдем область сходимости данного ряда по признаку Даламбера:

```
in : Limit[f[n + 1, z] / f[n, z], n → Infinity]
```

out : $\frac{2}{(-3\,i+z)^2}$

Получили, что исходный ряд сходится в кольце:

$$\frac{2}{|-3\,i+z|^2} < 1$$

$$|z-3\,i|^2 > 2$$

$$\left|z-3\,i\right| > \sqrt{2}$$

На границе области получаем следующий ряд:

$$\sum_{n=1}^{\infty} \frac{n\,2^n}{\left(\sqrt{2}\right)^{2\,n}} = \sum_{n=1}^{\infty} n$$

Так как ряд на границе не сходится, то областью сходимости исходного ряда будет: $\left|z-3\,i\right| > \sqrt{2}$

Пример №12.138

Найти область абсолютной сходимости:

$$\sum_{n=1}^{\infty} \frac{1}{n^2}\,e^{-n\,z^2}$$

Запишем общий член ряда:

in : `f[n_, z_] :=` $\frac{1}{n^2}$ `Exp[-n z^2]`

Для того чтобы определить область сходимости, воспользуемся признаком Коши:

in : `Limit[f[n, z]^(1/n), n → Infinity]`

out : e^{-z^2}

Получаем условие: $e^{-z^2} < 1 \Rightarrow \mathrm{Re}[z^2] \geq 0$.

Из алгебраической формы записи $z = x + I\, y$, $z^2 = x^2 - y^2 + 2\, I\, x\, y$ следует, что $x^2 - y^2 \geq 0$

Изобразим решение данного неравенства:

in : `RegionPlot[x^2 - y^2 ≥ 0, {x, -3, 3}, {y, -3, 3}, Axes → True, Frame → False]`



рис. 2.1.1

Из графика видно, что ряд сходится в: $D = \{z \mid -\pi/4 \leq \arg(z) \leq \pi/4 \,\&\&\, 3\pi/4 \leq \arg(z) \leq 5\pi/4\}$

С использованием описанных выше методов рекомендуется решить задания №№12.124-12.143.

2.2. Равномерная сходимость

Сходящийся в области D_1 функциональный ряд (1) называется равномерно сходящимся в этой области, если для любого $\epsilon > 0$ найдется $N = N(\epsilon)$ такое, что для остатка ряда (1)

$$R_n(z) = \sum_{k=n+1}^{\infty} f_k(z) \tag{3}$$

при всех $n > N(\epsilon)$ и $z \in D_1$ имеет место оценка:

$$|R_n(z)| < \epsilon$$

К р и т е р и й К о ш и р а в н о м е р н о й с х о д и м о с т и. Для того чтобы функциональный ряд (1) был равномерно сходящимся в области D_1, необходимо и достаточно, чтобы для любого $\epsilon > 0$ существовало $N = N(\epsilon)$ такое, что для всех $n > N(\epsilon)$ и $z \in D_1$ выполнялись неравенства

$$\left| f_{n+1}(z) + f_{n+2}(z) + \ldots + f_{n+p}(z) \right| < \epsilon,\ \ p = 1, 2 \ldots \tag{4}$$

П р и з н а к В е й е р ш т р а с с а. Пусть функциональный ряд (1) сходится в области D_1, и пусть существует сходящийся знакоположительный числовой ряд $\sum_{n=1}^{\infty} a_n$, такой, что для всех $n > N_0$ и $z \in D_1$ члены ряда (1) удовлетворяют условию:

$$| f_n(z) | \le a_n \tag{5}$$

Тогда ряд (1) сходится абсолютно и равномерно в области D_1.

Пример №1

Исследовать функциональный ряд:

$$\sum_{n=2}^{\infty} \frac{2^n (z+3)^n}{n (3 \ln(n) + 1)^2}$$

Запишем общий член ряда:

```
in : f[n_, z_] := 2^n/(n (3 Log[n] + 1)^2) (z + 3)^n
```

Найдем область сходимости с помощью критерия Коши:

```
in : Limit[f[n, z]^(1/n), n → Infinity]

out : 2 (3 + z)
```

Таким образом, область сходимости исследуемого ряда: $|z+3| < 1/2$

На границе области сходимости при $z = -2.5$ получаем ряд:

$$\sum_{n=2}^{\infty} \frac{2^n \left(\frac{1}{2}\right)^n}{n (3 \ln(n) + 1)^2} = \sum_{n=2}^{\infty} \frac{1}{n (3 \ln(n) + 1)^2} \tag{6}$$

По интегральному признаку Коши он сходится:

```
in : Integrate[f[n, -5/2], {n, 2, Infinity}]

out : 1/(3 + Log[512])
```

Так как выполняется равенство:

$$| f_n(z) | = \frac{2^n |z+3|^n}{n (3 \ln(n) + 1)^2} \le \frac{1}{n (3 \ln(n) + 1)^2}$$

ряд (6) называется мажорирующим для исследуемого. Поэтому последний по признаку Вейерштрасса сходится равномерно в области $|z+3| \le 1/2$.

Пример №12.151

Найти область сходимости и область равномерной сходимости указанного ряда:

$$\sum_{n=1}^{\infty} \frac{1}{n\,(z+2)^n}$$

Найдем область сходимости при помощи опции SumConvergence:

```
in : SumConvergence[1/(n (z + 2)^n), n]
```

$$out: \frac{1}{\text{Abs}[2+z]} \leq 1 \;\&\&\; \frac{1}{2+z} \neq 1$$

Ряд сходится в кольце $|z+2| > 1$.

Для того чтобы найти область равномерной сходимости, поступим так: найдем область сходимости следующего ряда:

$$\sum \frac{1}{n\,\alpha^n}, \quad \lim_{n\to\infty} \sqrt[n]{\frac{1}{n\,\alpha^n}} = \frac{1}{\alpha} < 1$$

Так как данный ряд является мажорирующим для исходного функционального ряда:

$$\left| \frac{1}{n\,(z+2)^n} \right| \leq \frac{1}{n\,\alpha^n}, \text{ где } \alpha > 1$$

то, областью равномерной сходимости будет: $|z+2| \geq \alpha > 1$

§3. Степенные ряды

3.1. Область сходимости и свойства степенных рядов.

Ряд

$$c_0 + c_1(z-z_0) + c_2(z-z_0)^2 + \ldots + c_n(z-z_0)^n + \ldots = \sum_{n=0}^{\infty} c_n(z-z_0)^n \tag{1}$$

называется степенным по степенях $z - z_0$. В частности, ряд

$$c_0 + c_1\, z + c_2\, z^2 + \ldots + c_n\, z^n + \ldots = \sum_{n=0}^{\infty} c_n\, z^n \tag{2}$$

является степенным по степеням z .

Т е о р е м а А б е л я. Если степенной ряд (2) сходится в точке $z = z_1 \neq 0$, то он абсолютно сходится для всех z таких, что $|z| < |z_1|$. Если же ряд (2) расходится в точке $z = z_2$, то он расходится и для всех z таких, что $|z| > |z_2|$.

Из теоремы Абеля следует, что областью сходимости степенного ряда является круг с центром в начале координат(с центром в точке z_0), радиус которого может быть определен применением либо признака Даламбера, либо признака Коши, т.е. из условий:

$$\lim_{n\to\infty}\left|\frac{c_{n+1}\,z^{n+1}}{c_n\,z}\right| = |z|\lim_{n\to\infty}\left|\frac{c_{n+1}}{c_n}\right| < 1 \tag{3}$$

$$\lim_{n\to\infty}\sqrt[n]{|c_n\,z^n|} = |z|\lim_{n\to\infty}\sqrt[n]{|c_n|} < 1 \tag{4}$$

Отсюда для вычисления радиуса R круга сходимости получаем соотношения:

$$R = \frac{1}{\lim_{n\to\infty}\sqrt[n]{|c_n|}} \quad \text{или} \quad R = \frac{1}{\lim_{n\to\infty}\left|\frac{c_{n+1}}{c_n}\right|} \tag{5}$$

Пример №12 .165

Найти область абсолютной сходимости и область равномерной сходимости ряда:

$$\sum_{n=1}^{\infty}\frac{(z-1)^n}{n^2\,2^n}$$

Радиус круга сходимости определим по формуле (5):

```
in : R = 1/Limit[(1/(n^2 2^n))^(1/n), n → Infinity]

out : 2
```

Получили, что область абсолютной сходимости исходного ряда: $|z-1| < R = 2$.

На границе имеем следующее:

$$\left|\frac{(z-1)^n}{n^2\,2^n}\right| \le \frac{1}{n^2\,2^n}$$

Следовательно по признаку Вейерштрасса ряд сходится равномерно в области: $|z-1| \le 2$

```
in :
```

Пример №12 .168

Найти область абсолютной сходимости и область равномерной сходимости ряда:

$$\sum_{n=1}^{\infty}(-1)^{n+1}\frac{2^{n+1}\,(z-4)^n}{(n+1)\sqrt{\ln(n+1)}}$$

Запишем общий член:

```
in : f[n_, z_] := (-1)^(n+1) (2^(n+1) (z - 4)^n)/((n + 1) Sqrt[Log[n + 1]])
```

Найдем при помощи критерия Даламбера область абсолютной сходимости:

```
in : Limit[Abs[f[n + 1, z]] / Abs[f[n, z]], n → Infinity]

out : 2 Abs[-4 + z]
```

Получаем, что область сходимости $|z-4| < 1/2$.

На границе сходимости при $z = 4.5$ получаем ряд:

$$\sum_{n=1}^{\infty} \frac{2^{n+1} \left(\frac{1}{2}\right)^n}{(n+1)\sqrt{\ln(n+1)}} = \sum_{n=1}^{\infty} \frac{2}{(n+1)\sqrt{\ln(n+1)}}$$

Исследуем его на сходимость по интегральному признаку Коши:

in : $\text{Integrate}\left[\dfrac{2}{(n+1)\sqrt{\text{Log}[n+1]}}, \{n, 1, \text{Infinity}\}\right]$

out : $\displaystyle\int_1^{\infty} \frac{2}{(1+n)\sqrt{\text{Log}[1+n]}}\, dn$

Получили, что ряд расходится; следовательно исходный ряд равномерно сходится в области $|z-4| \leq r < 1/2$

Исследуем ряды на границе: первый при $z = 4.5$

$$\sum_{n=1}^{\infty} \frac{(-1)^{n+1}\, 2}{(n+1)\sqrt{\ln(n+1)}}$$

Так как ряд не сходится абсолютно, исследуем его на условную сходимость:

Общий член ряда:

in : $\text{Limit}\left[\dfrac{2}{(n+1)\sqrt{\text{Log}[n+1]}}, n \to \text{Infinity}\right]$

out : 0

стремится к 0.

in : $\text{ListPlot}\left[\text{Table}\left[\dfrac{2}{(n+1)\sqrt{\text{Log}[n+1]}}, \{n, 1, 50\}\right]\right]$

Члены ряда по модулю убывают, условия критерия Лейбница выполнены, следовательно ряд при $z = 4.5$ сходится условно.

Второй при $z = 3.5$

$$\sum_{n=1}^{\infty} \frac{(-1)^{n+1}\, 2^{n+1} \left(-\frac{1}{2}\right)^n}{(n+1)\sqrt{\ln(n+1)}} = \sum_{n=1}^{\infty} \frac{-2}{(n+1)\sqrt{\ln(n+1)}} = -\sum_{n=1}^{\infty} \frac{2}{(n+1)\sqrt{\ln(n+1)}}$$

не сходится, что было получено выше.

Пример №12 .174

Найти область абсолютной сходимости и область равномерной сходимости ряда:

$$\sum_{n=1}^{\infty} \frac{(n!)^2\, z^n}{(2\,n)!}$$

Запишем общий член ряда:

$$in: \; f[n\_, z\_] := \frac{(n!)^2}{(2\,n)!} z^n$$

При помощи признака Даламбера определим область сходимости:

```
in : Limit[f[n + 1, z] / f[n, z], n → Infinity]
```

$$out: \; \frac{z}{4}$$

Область сходимости данного ряда $|z| < 4$.

На границе $z = 4$ имеем ряд:

$$\sum_{n=1}^{\infty} \frac{(n!)^2\, 4^n}{(2\,n)!}$$

По интегральному признаку Коши:

```
in : Integrate[f[n, 4], {n, 1, Infinity}]
```

out : $\int_1^\infty \frac{4^n\ (n!)^2}{(2n)!}\, dn$

Он не сходится.

На границе $z = -4$, ряд

$$\sum_{n=1}^{\infty}(-1)^n \frac{(n!)^2\, 4^n}{(2\, n)!}$$

не удовлетворяет критериям признака Лейбница:

```
in : Limit[(n!)^2 4^n / (2 n)!, n → Infinity]
```

out : ∞

Следовательно он расходится.

Таким образом исходный ряд сходится равномерно на $|z| \le r < 4$

3.2. Разложение функций в ряд Тейлора.

Т е о р е м а Т е й л о р а. Функция $f(z)$, аналитичная в круге $|z - z_0| < R$, однозначно представима в этом круге своим рядом Тейлора

$$f(z) = \sum_{n=0}^{\infty} c_n (z - z_0)^n, \tag{6}$$

коэффициенты которого определяются по формулам:

$$c_n = \frac{f^{(n)}(z_0)}{n!} = \frac{1}{2\pi i} \oint_{|z-z_0|=r} \frac{f(z)}{(z - z_0)^{n+1}}\, dz,\ n = 0,\ 1\ \ldots \tag{7}$$

С л е д с т в и е. Если функция $f(z)$ аналитична в области D и $z_0 \in D$, то в круге $|z - z_0| < R(z_0,\ D)$, где $R(z_0,\ D)$ - наименьшее расстояние от точки z_0 до границы области D или до ближайшей точки z', в которой $f(z)$ не аналитична, $f(z)$ может быть представлена в виде степенного ряда

$$f(z) = \sum_{n=0}^{\infty} \frac{f^{(n)}(z_0)}{n!}\ (z - z_0)^n \tag{8}$$

Если $z_0 = 0$, то ряд Тейлора также называют рядом Маклорена.

Разложения элементарных функций:

$$e^z = 1 + z + \frac{z^2}{2!} + \ldots + \frac{z^n}{n!} + \ldots = \sum_{n=0}^{\infty} \frac{z^n}{n!},\quad z \in \mathbb{C} \tag{9}$$

$$\cos z = 1 - \frac{z^2}{2!} + \ldots + (-1)^n \frac{z^{2n}}{(2n)!} + \ldots = \sum_{n=0}^{\infty} (-1)^n \frac{z^{2n}}{(2n)!}, \; z \in \mathbb{C} \tag{10}$$

$$\sin z = z - \frac{z^3}{3!} + \ldots + (-1)^{n+1} \frac{z^{2n+1}}{(2n+1)!} + \ldots = \sum_{n=0}^{\infty} (-1)^n \frac{z^{2n+1}}{(2n+1)!}, \; z \in \mathbb{C} \tag{11}$$

$$\ln(1+z) = z - \frac{z^2}{2} + \ldots + (-1)^{n+1} \frac{z^n}{n} + \ldots = \sum_{n=1}^{\infty} (-1)^{n+1} \frac{z^n}{n}, \quad |z| < 1 \tag{12}$$

$$\operatorname{arctg} z = z - \frac{z^3}{3} + \ldots + (-1)^{n+1} \frac{z^{2n+1}}{2n+1} + \ldots = \sum_{n=0}^{\infty} (-1)^{n+1} \frac{z^{2n+1}}{2n+1}, \quad |z| < 1 \tag{13}$$

$$\begin{gathered}(1+z)^a = 1 + a z + \frac{a(a-1)}{2!} z^2 + \ldots + \frac{a(a-1)\ldots(a-n+1)}{n!} z^n = \\ 1 + \sum_{n=1}^{\infty} \text{Binomial}[a, n]\, z^n, \quad |z| < 1\end{gathered} \tag{14}$$

$$\text{где Binomial}[a, n] = \frac{1}{n!} a(a-1)\ldots(a-n+1)$$

$$\frac{1}{1+z} = 1 - z + z^2 + \ldots + (-1)^n z^n = \sum_{n=0}^{\infty} (-1)^n z^n, \quad |z| < 1 \tag{15}$$

Пример №12.214

Разложить функцию в ряд по степеням z :

$$f(z) = e^{-z^2}$$

Рассмотрим сначала решение данной задачи “в лоб”. Так как данная нам функция аналитична во всей комплексной плоскости, то для разложения функции в ряд применима формула (8).

Запишем заданную функцию:

```
in : f[z_] := Exp[-z^2]
```

Теперь запишем выражение для n-ого коэффициента c_n:

```
in : c[n_] := D[f[z], {z, n}] / n! /. z → 0
```

Так как мы вычисляем разложение в ряд Тейлора в окрестности точки 0, то $z_0 = 0$. Такой ряд также называют рядом Маклорена.

Для того чтобы получить окончательный ответ, найдем сумму ряда, например, до 16 члена:

```
in : Sum[c[n] z^n, {n, 0, 16}]
```

$$out:\ 1 - z^2 + \frac{z^4}{2} - \frac{z^6}{6} + \frac{z^8}{24} - \frac{z^{10}}{120} + \frac{z^{12}}{720} - \frac{z^{14}}{5040} + \frac{z^{16}}{40\,320}$$

Запишем данное выражение в виде функции:

```
in : r[z_, k_] := Sum[c[n] z^n, {n, 0, k}]
```

Для того чтобы проверить полученный результат, построим графики для исходной функции и ряда до 16 члена:

```
in : Plot[{f[z], Evaluate@r[z, 16]}, {z, -2, 2}, PlotRange → All]
```

рис. 3.2.1

Примечание: функция Evaluate здесь используется для того, чтобы сначала получить разложение в ряд Тейлора, а затем выполнять подстановку в него значений z из отрезка $[-2; 2]$.

Увеличивая максимальный порядок разложения, можно добиться сходимости ряда к функции на более широком отрезке.

Программный модуль

Составим учебную программу, которая будет выполнять разложение функции в ряд Тейлора по формуле (8).

Имеем следующую тривиальную последовательность действий:

Пусть задана функция $f(z) = \cos(z)$.

```
in : f[z_] := Cos[z]
```

1. Записываем функцию коэффициента:

```
in : c[n_, z0_] := D[f[z], {z, n}] / n! /. z → z0
```

2. Находим сумму:

```
in : Sum[c[n, 0] (z - 0)^n, {n, 0, 8}]
```

$$out:\ 1 - \frac{z^2}{2} + \frac{z^4}{24} - \frac{z^6}{720} + \frac{z^8}{40\,320}$$

Соберем это всё в единый модуль:

```
in : seriesO[func_, z_, z0_, k_] := Module[{coef},
       coef[n_] := D[func, {z, n}] / n! /. z → z0;
       Sum[coef[n] (z - z0)^n, {n, 0, k}]]
```

Продемонстрируем пример:

Пример №12.215

in : `seriesO[Sin[z]^2, z, 0, 12]`

out : $z^2 - \frac{z^4}{3} + \frac{2 z^6}{45} - \frac{z^8}{315} + \frac{2 z^{10}}{14175} - \frac{2 z^{12}}{467775}$

Пример №12.242

in : `seriesO[Exp[z^2 - 4 z + 1], z, 2, 12]`

out : $\frac{1}{e^3} + \frac{(-2+z)^2}{e^3} + \frac{(-2+z)^4}{2 e^3} + \frac{(-2+z)^6}{6 e^3} + \frac{(-2+z)^8}{24 e^3} + \frac{(-2+z)^{10}}{120 e^3} + \frac{(-2+z)^{12}}{720 e^3}$

Напомним, что данный модуль использует формулу (8) для разложения функции в ряд.

Рассмотрим другие пути решения поставленной задачи.

Пример №12.231

Используя разложения основных элементарных функций, разложить функцию в ряд по степеням z и указать область сходимости полученного ряда:

$$f(z) = \int_0^z e^{-t^2/2}\, dt$$

На этот раз воспользуемся стандартными разложениями (9)-(15), для того чтобы получить разложение функции в ряд.

Подынтегральная функция - экспоненциальная, значит используя выражение (9), а также замену $h = -t^2 / 2$, получим:

in : `seriesO[Exp[h], h, 0, 8] /. h → -t^2/2`

out : $1 - \frac{t^2}{2} + \frac{t^4}{8} - \frac{t^6}{48} + \frac{t^8}{384} - \frac{t^{10}}{3840} + \frac{t^{12}}{46080} - \frac{t^{14}}{645120} + \frac{t^{16}}{10321920}$

Теперь требуется лишь проинтегрировать полученное выражение:

in : `Integrate[seriesO[Exp[h], h, 0, 8] /. h → -t^2/2, {t, 0, z}]`

out : $z - \frac{z^3}{6} + \frac{z^5}{40} - \frac{z^7}{336} + \frac{z^9}{3456} - \frac{z^{11}}{42240} + \frac{z^{13}}{599040} - \frac{z^{15}}{9676800} + \frac{z^{17}}{175472640}$

Получили разложение в ряд Тейлора до 17ого члена исходной функции, причем, так как область сходимости ряда (9) - вся комплексная плоскость, то и областью сходимости полученного ряда также является вся комплексная плоскость.

Примечание: если слегка модифицировать модуль, полученный в предыдущем примере:

```
in : seriesO1[func_, z_, z0_, k_] := Module[{coef, t},
       coef[n_] := D[func[t], {t, n}] / n! /. t → z0;
       Sum[coef[n] (z - z0)^n, {n, 0, k}]]
```

то если записать функцию при помощи безымянной переменной, например:

```
in : f = Exp[#] &;
```

то, получим следующий результат:

```
in : seriesO1[Exp[#] &, -t^2/2, 0, 8]
```

$$out: 1-\frac{t^2}{2}+\frac{t^4}{8}-\frac{t^6}{48}+\frac{t^8}{384}-\frac{t^{10}}{3840}+\frac{t^{12}}{46080}-\frac{t^{14}}{645120}+\frac{t^{16}}{10321920}$$

```
in : Integrate[seriesO1[Exp[#] &, -t^2/2, 0, 8], {t, 0, z}]
```

$$out: z-\frac{z^3}{6}+\frac{z^5}{40}-\frac{z^7}{336}+\frac{z^9}{3456}-\frac{z^{11}}{42240}+\frac{z^{13}}{599040}-\frac{z^{15}}{9676800}+\frac{z^{17}}{175472640}$$

Также вместо безымянной функции можно использовать просто заголовок подынтегральной функции:

```
in : seriesO1[Exp, -t^2/2, 0, 8]
```

$$out: 1-\frac{t^2}{2}+\frac{t^4}{8}-\frac{t^6}{48}+\frac{t^8}{384}-\frac{t^{10}}{3840}+\frac{t^{12}}{46080}-\frac{t^{14}}{645120}+\frac{t^{16}}{10321920}$$

Пример №12.218

Используя разложения основных элементарных функций, разложить функцию в ряд по степеням z и указать область сходимости полученного ряда:

$$f(z)=\sqrt[3]{27-z}$$

Преобразуем выражение:

$$f(z)=3\sqrt[3]{1-\frac{z}{27}}$$

Теперь, пользуясь стандартным разложением (14), при помощи замены $t=-z/27$ найдем разложение исходной функции в ряд:

При помощи безымянной функции:

```
in : seriesO1[3 (1 + #)^(1/3) &, -z / 27, 0, 6]
```

$$out: 3-\frac{z}{27}-\frac{z^2}{2187}-\frac{5z^3}{531441}-\frac{10z^4}{43046721}-\frac{22z^5}{3486784401}-\frac{154z^6}{847288609443}$$

При помощи замены:

```
in : seriesO[3 (1 + t)^(1/3), t, 0, 6] /. t → -z / 27
```

$$out:\ 3-\frac{z}{27}-\frac{z^2}{2187}-\frac{5z^3}{531\,441}-\frac{10z^4}{43\,046\,721}-\frac{22z^5}{3\,486\,784\,401}-\frac{154z^6}{847\,288\,609\,443}$$

Область сходимости ряда определим из:

$$|t|<1 \Longrightarrow |z|<27$$

Встроенные средства

В Wolfram Mathematica функция, которая отвечает за разложение в ряд, имеет следующую структуру: Series[$f[z]$, $\{z,\ z_0,\ n\}$] , где n- порядок разложения. Продемонстрируем пример:

```
in : Series[ Cos[z], {z, 0, 5}]
```

$$out:\ 1-\frac{z^2}{2}+\frac{z^4}{24}+O[z]^6$$

Примечание: $O[z]^6$, о-малое от z^6 - остаток ряда.

```
in : Series[Exp[-z^2], {z, 0, 16}]
```

$$out:\ 1-z^2+\frac{z^4}{2}-\frac{z^6}{6}+\frac{z^8}{24}-\frac{z^{10}}{120}+\frac{z^{12}}{720}-\frac{z^{14}}{5040}+\frac{z^{16}}{40\,320}+O[z]^{17}$$

```
in : Series[Exp[Cos[z]], {z, 0, 6}]
```

$$out:\ e-\frac{e\,z^2}{2}+\frac{e\,z^4}{6}-\frac{31\,e\,z^6}{720}+O[z]^7$$

Для того чтобы получить n-ый коэффициент разложения, существует функция SeriesCoefficient[$f[z]$, $\{z,\ z_0,\ n\}$]. Примеры:

```
in : SeriesCoefficient[Exp[-z^2], {z, 0, 8}]
```

$$out:\ \frac{1}{24}$$

```
in : SeriesCoefficient[(27 - z)^(1/3), {z, 0, 3}]
```

$$out:\ -\frac{5}{531\,441}$$

Также есть возможность получить формулу общего члена:

```
in : SeriesCoefficient[Log[1 + z], {z, 0, n}]
```

$$out:\ \begin{cases}\frac{(-1)^{1+n}}{n} & n\geq 1\\ 0 & \text{True}\end{cases}$$

Если задать $n > 1$, можно избавиться от кусочно-заданной функции:

```
in : Refine[SeriesCoefficient[Log[1 + z], {z, 0, n}], n > 1]
```

out : $\frac{(-1)^{1+n}}{n}$

Следующая конструкция позволит вывести ответ в формульном виде:

```
in : TraditionalForm@
      Inactivate[
       Sum[Refine[SeriesCoefficient[Exp[z], {z, 0, n}], n > 0] z^n,
        {n, 0, Infinity}], Sum]
```

out : $\sum_{n=0}^{\infty} \frac{z^n}{n!}$

где функция Inactivate запрещает функции, следующей за ней (Sum), вычислять результат, а TraditionalForm преобразует выражение в традиционный вид.

Составим функцию:

```
in : seriesF[f_, z_, z0_] :=
      TraditionalForm@
       Inactivate[
        Sum[Refine[SeriesCoefficient[f, {z, z0, n}], n > 5] (z - z0)^n,
         {n, 0, Infinity}], Sum]
```

Продемонстрируем пример на основе задачи №12.214:

```
in : seriesF[Exp[t], t, 0] /. t → -z^2
```

out : $\sum_{n=0}^{\infty} \frac{(-z^2)^n}{n!}$

Пример №12.232

Разложить функцию в ряд:

$$f(z) = \int_0^z \frac{\sin(t)^2}{t}\, dt$$

Получим разложение подынтегральной функции:

```
in : Sin[t]^2 / t // TrigReduce
```

out : $\frac{1 - \text{Cos}[2\,t]}{2\,t}$

Используя стандартное разложение функции $\cos(h)$ (10), где $h = 2\,t$, получаем:

$$\frac{1-\left(1-\frac{h^2}{2!}+\ldots+(-1)^n\frac{h^{2n}}{(2n)!}+\ldots\right)}{2t}=\sum_{n=1}^{\infty}\frac{(-1)^n(2t)^{2n-1}}{(2n)!}=\sum_{n=1}^{\infty}\frac{(-1)^n 4^n t^{2n-1}}{2(2n)!}$$

Теперь применяя теорему о почленном интегрировании получаем:

$$\int_0^z\sum_{n=1}^{\infty}\frac{(-1)^{n+1}4^n t^{2n-1}}{2(2n)!}\,dt=\sum_{n=1}^{\infty}\frac{(-1)^{n+1}4^n}{2(2n)!}\int_0^z t^{2n-1}\,dt=\sum_{n=1}^{\infty}\frac{(-1)^{n+1}4^n z^{2n}}{2(2n)!\,2n}$$

В итоге получаем ответ:

$$f(z)=\int_0^z\frac{\sin(t)^2}{t}\,dt=\sum_{n=1}^{\infty}\frac{(-1)^{n+1}4^{n-1}z^{2n}}{(2n)!\,n},\quad z\in\mathbb{C}.$$

Проверим полученное аналитическое решение при помощи функции Series:

```
in : Series[Integrate[Sin[t]^2/t, {t, 0, z}], {z, 0, 12}]
```

$$out:\ \frac{z^2}{2}-\frac{z^4}{12}+\frac{z^6}{135}-\frac{z^8}{2520}+\frac{z^{10}}{70\,875}-\frac{z^{12}}{2\,806\,650}+O[z]^{13}$$

```
in : Sum[((-1)^(n+1) 4^(n-1) z^(2 n))/((2 n)! n), {n, 1, 6}]
```

$$out:\ \frac{z^2}{2}-\frac{z^4}{12}+\frac{z^6}{135}-\frac{z^8}{2520}+\frac{z^{10}}{70\,875}-\frac{z^{12}}{2\,806\,650}$$

Ряды совпадают, следовательно мы получили правильный ответ.

Пример №12.237

Разложить функцию в ряд по степеням $z-z_0$ и определить область сходимости полученного ряда:

$$f(z)=\frac{1}{1-z},\ z_0=3\,i$$

Получим разложение, воспользовавшись опцией Series:

in : Series$\left[\frac{1}{1-z}, \{z, 3\,I, 6\}\right]$

out : $\left(\frac{1}{10}+\frac{3\,i}{10}\right)-\left(\frac{2}{25}-\frac{3\,i}{50}\right)(z-3\,i)-\left(\frac{13}{500}+\frac{9\,i}{500}\right)(z-3\,i)^2+\left(\frac{7}{2500}-\frac{6\,i}{625}\right)(z-3\,i)^3+\left(\frac{79}{25\,000}-\frac{3\,i}{25\,000}\right)(z-3\,i)^4+\left(\frac{11}{31\,250}+\frac{117\,i}{125\,000}\right)(z-3\,i)^5-\left(\frac{307}{1\,250\,000}-\frac{249\,i}{1\,250\,000}\right)(z-3\,i)^6+O[z-3\,i]^7$

Также можно воспользоваться созданным ранее модулем seriesO:

in : seriesO$\left[\frac{1}{1-z}, z, 3\,I, 6\right]$

out : $\left(\frac{1}{10}+\frac{3\,i}{10}\right)-\left(\frac{2}{25}-\frac{3\,i}{50}\right)(-3\,i+z)-\left(\frac{13}{500}+\frac{9\,i}{500}\right)(-3\,i+z)^2+\left(\frac{7}{2500}-\frac{6\,i}{625}\right)(-3\,i+z)^3+\left(\frac{79}{25\,000}-\frac{3\,i}{25\,000}\right)(-3\,i+z)^4+\left(\frac{11}{31\,250}+\frac{117\,i}{125\,000}\right)(-3\,i+z)^5-\left(\frac{307}{1\,250\,000}-\frac{249\,i}{1\,250\,000}\right)(-3\,i+z)^6$

Запишем ответ в стандартном виде:

in : seriesF$\left[\frac{1}{1-z}, z, 3\,I\right]$

out : $\sum_{n=0}^{\infty}(1+3\,i)^{n+1}\,10^{-n-1}\,(z-3\,i)^n$

Область сходимости определим при помощи функции SumConvergence:

in : SumConvergence$\left[(1+3\,I)^{1+n}\,10^{-1-n}\,(-3\,I+z)^n, n\right]$

out : $\sqrt{10}$ Abs$[-3\,i+z] < 10$

В итоге имеем ответ:

$$\frac{1}{1-z}=\sum_{n=0}^{\infty}\frac{(1+3\,i)^{n+1}}{10^{n+1}}(z-3\,i)^n,\quad \left|z-3\,i\right|<\sqrt{10}$$

Пример №12.250

Найти область сходимости указанного ряда и его сумму:

$$\sum_{n=0}^{\infty}(-1)^n\,a^{-2\,n-2}\,z^{2\,n},\ a\neq 0$$

Для того чтобы найти сумму, преобразуем выражение к:

$$\frac{1}{a^2}\sum_{n=0}^{\infty}(-1)^n\left(\left(\frac{z}{a}\right)^2\right)^n,\ a \neq 0$$

Суммой данной геометрической прогрессии является следующая функция:

$$\frac{1}{1+\eta}=\sum_{n=0}^{\infty}(-1)^n\,\eta^n,\ \text{где}\ \eta=\left(\frac{z}{a}\right)^2$$

Область сходимости данной суммы $|\eta| < 1$

Проверим себя:

```
in : Sum[(-1)^n a^(-2n-2) z^(2n), {n, 0, Infinity}]
```

out : $\frac{1}{a^2+z^2}$

Получили тот же самый результат.

Следовательно, получаем ответ:

$$\sum_{n=0}^{\infty}(-1)^n\,a^{-2n-2}\,z^{2n}=\frac{1}{a^2}\,\frac{1}{1+\frac{z^2}{a^2}}=\frac{1}{a^2+z^2},\quad |z| < |a|$$

§4. Применение степенных рядов

4.1. Интегрирование функций.

Разлагая подынтегральную функцию $f(t)$ в степенной ряд, можно, используя теорему об интегрировании степенных рядов, представить интеграл $\int_0^x f(t)\,dt$ в виде степенного ряда и подсчитать величину этого интеграла с заданной точностью при любом значении x из интервала сходимости полученного ряда.

Пример №12.289

Разложить указанную функцию в степенной ряд по степеням x:

$$f(x)=\int_0^x\frac{\ln\left(1+t^2\right)}{t}\,dt$$

Используем разложение функции $\ln(1+y)$:

```
in : series[y, 0, Log[1 + y]]
```

out : $\sum_{n=0}^{\infty}\frac{(-1)^{n+1}\,y^n}{n}$

Заменив $y \to t^2$ и разделив на t, получим ряд:

$$\sum_{n=0}^{\infty} (-1)^{n+1} \frac{t^{2n-1}}{n}$$

По теореме о интегрировании степенных рядов получаем:

$$\int_0^x \sum_{n=0}^{\infty} (-1)^{n+1} \frac{t^{2n-1}}{n} \, dt = \sum_{n=0}^{\infty} \frac{(-1)^{n+1}}{n} \int_0^x t^{2n-1} \, dt = \sum_{n=0}^{\infty} (-1)^{n+1} \frac{x^{2n}}{2n^2}$$

Ту же последовательность действий можно проделать, используя средства Wolfram Mathematica:

Сразу находим разложение подынтегральной функции:

```
in : Series[Log[1 + t^2] / t, {t, 0, 8}]
```

out : $t - \frac{t^3}{2} + \frac{t^5}{3} - \frac{t^7}{4} + O[t]^9$

Интегрируем по t от 0 до x и получаем искомый ряд:

```
in : Integrate[Series[Log[1 + t^2] / t, {t, 0, 10}], t] /. t → x
```

out : $\frac{x^2}{2} - \frac{x^4}{8} + \frac{x^6}{18} - \frac{x^8}{32} + \frac{x^{10}}{50} + O[x]^{12}$

Найти n-ый коэффициент можно следующим образом:

```
in : Simplify[SeriesCoefficient[Log[1 + t^2] / t, {t, 0, n}], n > 0] * 1/(n + 1)
```

out : $\frac{i\left((-i)^n - i^n\right)}{(1+n)^2}$

Множитель $\frac{1}{n+1}$ появился из-за интегрирования функции t^n.

Ответ:

$$\int_0^x \frac{\ln\left(1+t^2\right)}{t} \, dt = \sum_{n=0}^{\infty} (-1)^{n+1} \frac{x^{2n}}{2n^2}$$

Пример №12.300

Вычислить интеграл с точностью до 10^{-5}:

$$\int_0^1 \frac{\sin(t)}{t} \, dt$$

Получим разложение заданной функции в ряд :

```
in : ser = Integrate[Series[Sin[t] / t, {t, 0, 10}], t] /. t → x
```

$$out:\ x-\frac{x^3}{18}+\frac{x^5}{600}-\frac{x^7}{35\,280}+\frac{x^9}{3\,265\,920}-\frac{x^{11}}{439\,084\,800}+O[x]^{12}$$

Теперь, заменяя x на нужное нам значение x=1, получаем:

```
in : N[Normal@ser /. x → 1]
out : 0.9460830703548807
```

Функция Normal необходима, чтобы отбросить остаток ряда $O[x]^{12}$

Получили искомое значение интеграла.

4.2. Интегрирование дифференциальных уравнений с помощью рядов.

Степенные ряды широко применяются при решении дифференциальных уравнений. Для целого ряда дифференциальных уравнений показано, что решение y(t) представимо в виде степенного ряда

$$y(x)=\sum_{k=0}^{\infty}a_k(x-x_0)^k=\sum_{k=0}^{\infty}\frac{y^{(k)}(x_0)}{k!}(x-x_0)^k \qquad (1)$$

коэффициенты которого можно определить с учетом заданного уравнения различными способами.

а) Пусть требуется найти решение уравнения $y''=f(x,\ y,\ y')$, удовлетворяющее условиям $y(x_0)=y_0,\ \ y'(x_0)=y_1$, причем функция $f(x,\ y,\ y')$ в точке $(x_0,\ y_0,\ y_1)$ имеет частные производные любого порядка. Тогда коэффициенты $y^{(k)}(x_0)$ ряда (1) определяются путем последовательного дифференцирования исходного уравнения и подстановки в него x_0 и найденных уже значений $y'(x_0),\ \ y''(x_0)$ ...

Пример №12.326

Найти решение уравнения, удовлетворяющее заданным условиям:

$$y''=-x^2\,y'-2\,x\,y+1,\ \ y(0)=y'(0)=0$$

Для того чтобы записать решение в форме (1), продифференцируем данное уравнение несколько раз:

```
in : Table[D[y''[x] == -x^2 y'[x] - 2 x y[x] + 1, {x, n}], {n, 1, 6}] //
      TableForm
```

$$\begin{array}{l} y^{(3)}[x] == -2\,y[x] - 4\,x\,y'[x] - x^2\,y''[x] \\ y^{(4)}[x] == -6\,y'[x] - 6\,x\,y''[x] - x^2\,y^{(3)}[x] \\ y^{(5)}[x] == -12\,y''[x] - 8\,x\,y^{(3)}[x] - x^2\,y^{(4)}[x] \\ y^{(6)}[x] == -20\,y^{(3)}[x] - 10\,x\,y^{(4)}[x] - x^2\,y^{(5)}[x] \\ y^{(7)}[x] == -30\,y^{(4)}[x] - 12\,x\,y^{(5)}[x] - x^2\,y^{(6)}[x] \\ y^{(8)}[x] == -42\,y^{(5)}[x] - 14\,x\,y^{(6)}[x] - x^2\,y^{(7)}[x] \end{array}$$

out :

Теперь можно заметить следующую закономерность, которую мы запишем в виде:

$$y^{(n+2)}(x) = -n(n+1)\,y^{(n-1)}(x) - 2\,(n+1)\,x\,y^{(n)}(x) - x^2\,y^{(n+1)}(x), \text{ при } n > 2$$

Подставляя $x_0 = 0$ получим, следующее выражение:

$$y^{(n+2)}(0) = -n(n-1)\,y^{(n-1)}(0)$$

Или по-другому:

$$y^{(n)}(0) = -(n-1)\,(n-2)\,y^{(n-3)}(0)$$

Мы получили рекуррентное уравнение. Определим начальные условия, два из которых нам уже даны, а еще одно получается при подстановке $x_0 = 0$ в исходное уравнение:

$$y(0) = y'\,(0) = 0,\ y''\,(0) = 1$$

Составим формулу для значения n-ой производной функции y с помощью опции Piecewise:

```
in : yn[n_] :=
      Piecewise[{{-(n - 1) (n - 2) yn[n - 3], n ≥ 3}, {1, n == 2}, {0, n < 2}}]
```

Протабулируем функцию от 0 до 12:

```
in : Table[yn[n], {n, 0, 18}]

out : {0, 0, 1, 0, 0, -12, 0, 0, 504, 0, 0,
       -45 360, 0, 0, 7 076 160, 0, 0, -1 698 278 400, 0}
```

Мы получили ничто иное как коэффициенты разложения искомой функции в ряд.

Теперь запишем сам ряд по формуле (1):

in : $\mathrm{Sum}\left[\dfrac{\mathrm{yn}[k]}{k!}\,x^k,\ \{k, 0, 18\}\right]$

out : $\dfrac{x^2}{2} - \dfrac{x^5}{10} + \dfrac{x^8}{80} - \dfrac{x^{11}}{880} + \dfrac{x^{14}}{12\,320} - \dfrac{x^{17}}{209\,440}$

Введем функцию y1(x,n):

```
in : y1[x_, n_] := Sum[yn[k]/k! x^k, {k, 0, n}]
```

y1(x,n) - искомая функция, являющаяся решением исходного дифференциального уравнения.

Решим задачу другим способом.

После того как мы получили рекуррентное уравнение:

$$y^{(n)}(0) = -(n-1)(n-2)\, y^{(n-3)}(0)$$

решим его с помощью встроенной функции для решения рекуррентных уравнений RSolve[expr, f[n], n]:

```
in : sol = RSolve[{y[n] == -(n - 1) (n - 2) y[n - 3], y[0] == y[1] == 0,
       y[2] == 1}, y[n], n]
```

out : $\left\{\left\{y[n] \to \left((-1)^{1/3}\, 3^{-\frac{5}{6}-\frac{n}{3}}\, \mathrm{Gamma}\left[\frac{5}{3}\right] \mathrm{Gamma}[1+n] \left((-1)^{1+\frac{n}{3}} \sqrt{3} + (-1)^{n/3} \sqrt{3}\, \mathrm{Cos}\left[\frac{2 n \pi}{3}\right] + 3\, (-1)^{n/3}\, \mathrm{Sin}\left[\frac{2 n \pi}{3}\right]\right)\right) \Big/ \left(2\, \mathrm{Gamma}\left[1+\frac{n}{3}\right]\right)\right\}\right\}$

Мы получили громоздкую функцию, однако, если построить список ее значений от 0 до 18, то можно убедиться, что значения коэффициентов совпадают с уже ранее полученными в функции yn(n):

```
in : Table[y[n] /. Flatten@sol, {n, 0, 18}] // N
```

out : $\{0., 0., 1., 0., 0., -12.000000000000004, 0., 0., 504.00000000000006, 0., 0., -45\,359.99999999997, 0., 0., 7.076159999999998 \times 10^6, 0., 0., -1.6982783999999995 \times 10^9, 0.\}$

Значения коэффициентов не изменились, значит задавать новую функцию не имеет смысла.

Решим задачу с помощью встроенной в систему функции для решения дифференциальных уравнений DSolve[expr, y[x], x]:

```
in : sol1 = DSolve[{y''[x] == -x^2 y'[x] - 2 x y[x] + 1, y[0] == y'[0] == 0},
       y[x], x]
```

out : $\left\{\left\{y[x] \to -\frac{1}{3^{1/3}\, x}\, e^{-\frac{x^3}{3}} \left((-1)^{1/3}\, x\, \mathrm{Gamma}\left[\frac{2}{3}\right] - \left(-x^3\right)^{1/3} \mathrm{Gamma}\left[\frac{2}{3}, -\frac{x^3}{3}\right]\right)\right\}\right\}$

Введем функцию y3(x):

```
in : y3[x_] := y[x] /. sol1
```

Докажем графически, что полученная в данном случае функция совпадает с функцией y1(x,18):

```
in : Plot[{y3[x], y1[x, 18]}, {x, 0, 2}]
```

рис. 4.2.1

Как мы видим, расхождение начинается примерно с x=1.5, что, разумеется, объясняется малым порядком (n=18) разложения.

Наконец, докажем, что полученное решение верно, непосредственно подставив функцию y3(x) в исходное дифференциальное уравнение:

```
in : D[y3[x], {x, 2}] == -x^2 D[y3[x], x] - 2 x y3[x] + 1 // FullSimplify

in : True
```

Задача решена.

б) Если исходное дифференциальное уравнение линейно относительно искомой функции и ее производных, причем коэффициент при старшей производной в точке x_0 отличен от нуля, то решение следует искать в виде ряда (1) с неопределенными коэффициентами a_k, $k = 0, 1 ...$ Законность такого метода вытекает из утверждения, доказываемого в аналитической теории дифференциальных уравнений, которое мы приведем для уравнения 2-ого порядка.

Т е о р е м а 1. Если в дифференциальном уравнении

$$p_0(x)\, y'' + p_1(x)\, y' + p_2(x)\, y = f(x) \tag{2}$$

функции $p(x)$, $p_1(x)$, $p_2(x)$ и $f(x)$ аналитичны в окрестности точки x_0, и $p_0(x_0) \neq 0$, то существует решение уравнения (2), представимое в виде степенного ряда (3):

$$y(x) = \sum_{k=0}^{\infty} a_k (x - x_0)^k \tag{3}$$

Пример №12.330

Используя степенные ряды, проинтегрировать следующее дифференциальное уравнение:

$$y'' + x\, y' + y = 1, \;\; y(0) = y'(0) = 0$$

Введем функцию y(x) по формуле (3):

```
in : y[x_] := Sum[a_k x^k, {k, 0, Infinity}]
```

Подставим ее в исходное дифференциальное уравнение:

```
in : D[y[x], {x, 2}] + x D[y[x], x] + y[x] == 1 // FullSimplify //
      TraditionalForm
```

$$out: \sum_{k=0}^{\infty} (k-1)\, k\, a_k\, x^{k-2} + x \sum_{k=0}^{\infty} k\, a_k\, x^{k-1} + \sum_{k=0}^{\infty} a_k\, x^k = 1$$

Учтём начальные условия: при k=0,1 a_k=0, а при k=2 и x=0:

$$1 \cdot 2\, a_2 + 0 + 0 = 1 \Rightarrow a_2 = \frac{1}{2 \cdot 1}$$

Теперь составим следующее рекуррентное соотношение:

$$((k+2)-1)\,(k+2)\,a_{k+2} + k\,a_k + a_k = 0$$
$$(k+1)\,(k+2)\,a_{k+2} = -(k+1)\,a_k$$
$$a_{k+2} = -\frac{a_k}{k+2}$$

Запишем функцию для коэффициента a_k, заданного следующим образом:

$$a_k = \begin{cases} -\frac{a_{k-2}}{k} & k \geq 3 \\ \frac{1}{2} & k == 2 \\ 0 & k < 2 \end{cases}$$

```
in : ak[k_] := Piecewise[{{(-ak[k - 2])/k, k ≥ 3}, {0, k < 2}, {1/2, k == 2}}]
```

Протабулируем функцию на отрезке от 0 до 12:

```
in : Table[ak[k], {k, 0, 12}]
```

$$out: \left\{0,\, 0,\, \frac{1}{2},\, 0,\, -\frac{1}{8},\, 0,\, \frac{1}{48},\, 0,\, -\frac{1}{384},\, 0,\, \frac{1}{3840},\, 0,\, -\frac{1}{46\,080}\right\}$$

Теперь можно записать ряд (3):

```
in : Sum[ak[k] x^k, {k, 0, 12}]
```

$$out: \frac{x^2}{2} - \frac{x^4}{8} + \frac{x^6}{48} - \frac{x^8}{384} + \frac{x^{10}}{3840} - \frac{x^{12}}{46\,080}$$

Данный ряд является решением исходного дифференциального уравнения при заданных начальных условиях. Зададим функцию y1(x,n):

```
in : y1[x_, n_] := Sum[ak[k] x^k, {k, 0, n}]
```

Теперь решим ту же задачу, используя функцию DSolve:

```
in : sol = DSolve[{D[y[x], {x, 2}] + x D[y[x], x] + y[x] == 1, y[0] == 0,
        y'[0] == 0}, y[x], x]
```

$$out: \left\{\left\{y[x] \to e^{-\frac{x^2}{2}}\left(-1 + e^{\frac{x^2}{2}}\right)\right\}\right\}$$

$$in: y2[x\_] := e^{-\frac{x^2}{2}}\left(-1 + e^{\frac{x^2}{2}}\right)$$

Построим график функций y1(x,24) и функции полученной с помощью DSolve:

```
in : Plot[{y2[x], y1[x, 24]}, {x, -4, 4}]
```



рис. 4.2.2

Так как функции совпадают приблизительно до |x|=2.5, будем считать, что решение в окрестности точки $x_0 = 0$ получено верно.

Еще в этом можно убедится, подставив функцию в общем виде в исходное дифференциальное уравнение:

```
in : D[y2[x], {x, 2}] + x D[y2[x], x] + y2[x] == 1 // FullSimplify
out : True
```

Получили верное равенство.

Замечание: в том что ряд был получен верно, можно также убедиться, разложив функцию y2(x) в ряд по степеням x:

```
in : Series[y2[x], {x, 0, 12}]
```

$$out: \frac{x^2}{2} - \frac{x^4}{8} + \frac{x^6}{48} - \frac{x^8}{384} + \frac{x^{10}}{3840} - \frac{x^{12}}{46\,080} + O[x]^{13}$$

Ряды совпадают, значит на данном этапе будем считать, что решение получено верно.

4.3. Уравнение и функции Бесселя.

Частным случаем уравнения (2) является уравнение Бесселя

$$x^2\, y'' + x\, y' + \left(x^2 - v^2\right) y = 0 \tag{4}$$

Его решениями являются цилиндрические функции Бесселя первого рода порядка v

$$J_v(x) = a_0{}^{(v)}\, x^v \sum_{k=0}^{\infty} \frac{(-1)^k}{k!\,(v+1)\,(v+2)\ldots(v+k)} \left(\frac{x}{2}\right)^{2k} \tag{5}$$

и для нецелых v

$$J_{-\nu}(x) = a_0{}^{(\nu)}\, x^{-\nu} \sum_{k=0}^{\infty} \frac{(-1)^k}{k!\,(1-\nu)\,(2-\nu)\,...\,(k-\nu)} \left(\frac{x}{2}\right)^{2k} \tag{6}$$

Если же v - целое число, v=n, то вторым частным решением уравнения Бесселя (4) является функция Неймана (или Вебера), определяемая из соотношения

$$N_n(x) = \lim_{\nu \to n} \frac{I_\nu(x)\cos\nu\pi - I_{-\nu}(x)}{\sin\nu\pi} \tag{7}$$

являющаяся цилиндрической функцией второго рода порядка n.

Формулы (5) и (6) можно упростить с помощью гамма-функции, которая определена равенством

$$\Gamma(\nu) = \int_0^{+\infty} e^{-x}\, x^{\nu-1}\, dx$$

Этот несобственный интеграл сходится для всех $\nu > 0$. Гамма-функция - обобщение для $x > 0$ функции факториала $n!$, которая определена только если n - натуральное число. Рассмотрим предел:

$$\Gamma(1) = \int_0^{+\infty} e^{-x}\, dx = \lim_{b \to +\infty} -e^{-x}\Big|_0^b = 1$$

Теперь найдем интеграл:

$$\Gamma(\nu+1) = \int_0^{+\infty} e^{-x}\, x^\nu\, dx = \Gamma(1) = \lim_{b\to+\infty} -e^{-x}\, x^\nu\Big|_0^b + \int_0^{+\infty} \nu\, e^{-x}\, x^{\nu-1}\, dx =$$
$$= \nu \int_0^{+\infty} e^{-x}\, x^{\nu-1}\, dx$$

Таким образом мы получаем равенство:

$$\Gamma(\nu+1) = \nu\,\Gamma(\nu)$$

Это важнейшее свойство гамма-функции.

$$\Gamma(2) = 1\cdot\Gamma(1) = 1!,\ \ \Gamma(3) = 2\cdot\Gamma(2) = 2!,\ \ ...,\ \ \Gamma(n+1) = n!$$

Теперь, если положить $a_0{}^{(\nu)}$ в формулах (5) и (6)

$$a_0{}^{(\nu)} = \frac{1}{2^\nu\,\Gamma(\nu+1)} \tag{8}$$

и заметить, что

$$\Gamma(\nu+k+1) = (\nu+k)\,(\nu+k-1)\,...\,(\nu+2)\,(\nu+1)\,\Gamma(\nu+1)$$

то тогда можно записать функцию Бесселя первого рода кратко с помощью гамма-функции:

$$J_\nu(x) = \sum_{k==0}^{\infty} \frac{(-1)^k}{k!\,\Gamma(\nu+k+1)} \left(\frac{x}{2}\right)^{2k+\nu} \tag{9}$$

$$J_{-\nu}(x) = \sum_{k=0}^{\infty} \frac{(-1)^k}{k!\,\Gamma(-\nu+k+1)} \left(\frac{x}{2}\right)^{2k-\nu} \tag{10}$$

Параметрическое уравнение Бесселя порядка ν имеет вид:

$$x^2 y'' + x y' + \left(\alpha^2 x^2 - \nu^2\right) y = 0, \ \ \alpha > 0 \tag{11}$$

Заменой $t = \alpha x$ преобразовываем это уравнение в стандартное уравнение Бесселя:

$$t^2 y'' + t y' + \left(t^2 - \nu^2\right) y = 0$$

Имеющее общее решение:

$$y(x) = c_1 J_\nu(\alpha x) + c_2 J_{-\nu}(\alpha x) - \text{ если } \nu - \text{ не целое число,}$$
$$y(x) = c_1 J_\nu(\alpha x) + c_2 N_\nu(\alpha x) - \text{ если } \nu - \text{ целое число.}$$

Пример №12.342

Найти общее решение уравнения

$$x^2 y'' + x y' + \left(9 x^2 - \frac{1}{25}\right) y = 0$$

Функция Бесселя в Mathematica записывается BesselJ[v,x]. Построим два графика: $J_{1/5}(3x)$ и $J_{-1/5}(3x)$:

```
in : Plot[{BesselJ[1 / 5, x], BesselJ[-1 / 5, x]}, {x, 0, 15},
       PlotLabels → {"J1/5 (3x)", "J-1/5 (3x)"}]
```

рис. 4.3.1

Ответом, общим решением этого параметрического уравнения, является функция y(x), определяемая:

$$y(x) = c_1 J_{1/5}(3x) + c_2 J_{-1/5}(3x)$$

§5. Ряды Лорана

5.1. Ряды Лорана. Теорема Лорана.

Рядом Лорана называется ряд

$$f(z) = \sum_{n=-\infty}^{\infty} c_n (z - z_0)^n \tag{1}$$

при этом ряд

$$f_2(z) = \sum_{n=-\infty}^{-1} c_n (z - z_0)^n$$

называется *главной частью* ряда Лорана, а ряд

$$f_3(z) = \sum_{n=0}^{\infty} c_n (z - z_0)^n$$

- *правильной частью*. Если

$$\lim_{n\to\infty} \sqrt[n]{|c_{-n}|} = r < R = \frac{1}{\lim_{n\to\infty} \sqrt[n]{|c_n|}}$$

то областью сходимости ряда (1) является кольцо $K = \{ z \mid 0 \leq r < |z - z_0| < R\}$. В этом кольце К сумма ряда $f(z) = f_1(z) + f_2(z)$ является функцией аналитической, причем коэффициенты ряда c_n связаны с функцией $f(z)$ посредством формул

$$c_n = \frac{1}{2\pi i} \int_{|z-z_0|=r'} \frac{f(z)}{(z - z_0)^{n+1}}\, dz, \quad \text{где } r < r' < R,\ n + 1 > 0 \tag{2}$$

Причем если функция $f(z)$ аналитичная в кольце $|z - z_0| = r'$, то коэффициенты разложения будут вычисляться по формуле:

$$c_n = \frac{1}{n!} \left(\frac{d^n}{dz^n} f \right)(z_0) \tag{3}$$

Т е о р е м а Л о р а н а. Если функция $f(z)$ аналитична в кольце $0 \leq r < |z - z_0| < R$, то в этом кольце она единственным образом представима в виде ряда Лорана (1), коэффициенты которого вычисляются по формулам (2).

В данном разделе решение будем приводить методами функционального и алгоритмического программирования, в отличии от традиционных рутинных “бумажных” решений.

Использование подобных методов позволяет оптимальнее решать задачи, в чем и заключается преимущество *Mathematica.*

Пример №12.352

Найти все разложения указанной функции в ряд Лорана по степеням $z - z_0$ и установить области сходимости полученных разложений:

$$f(z) = \frac{1}{z(z-1)}, \quad z_0 = 1$$

Сначала рассмотрим детальное решение данной задачи. Затем усложним ее, так как на исходном примере нельзя отследить все этапы, необходимые для понимания общего алгоритма. И потом, наконец, составим программу разложения функций в ряд Лорана.

Запишем функцию:

```
in : f[z_] := 1/(z (z - 1))
```

Из структуры функции видно, что её аналитичность нарушается в точках $z_0 = 1$, $z_1 = 0$. Следовательно функция будет аналитична в кольцах: 1) $0 < |z-1| < 1$ 2) $|z-1| > 1$

Изобразим точки z_1 и z_2, а также контуры интегрирования r1 и r2, которые находятся соответственно в кольце 1) и 2):

```
in : Graphics[{Circle[{0, 0}, 0.3], Circle[{1, 0}, 0.6],
      Circle[{1, 0}, 2], PointSize[Large], Point[{0, 0}],
      Point[{1, 0}], Text[r1, {0.6, 0.6}], Text[r2, {2, 2}] ,
      Text[γ, {0.3, 0.3}]  }, PlotRange → {{-3, 3}, {-3, 3}},
     Axes → True]
```



рис. 5.1.1

Для того чтобы найти разложение в ряд Лорана в окрестности точки z_0, достаточно воспользоваться уже известной нам функцией Series,о которой говорится в разделе Ряды Тейлора и которая имеет вид: Series[$f[z]$, $\{z, z_0, n\}$] :

```
in : Series[f[z], {z, 1, 5}]
```

$$out: \frac{1}{z-1} - 1 + (z-1) - (z-1)^2 + (z-1)^3 - (z-1)^4 + (z-1)^5 + O[z-1]^6$$

Получим выражение для n-ого коэффициента:

```
in : SeriesCoefficient[f[z], {z, 1, n}]
```

$$out: \begin{cases} 1 & n == -1 \\ (-1)^{1+n} & n \geq 0 \\ 0 & \text{True} \end{cases}$$

Запишем n-ый коэффициент как функцию от n. Это понадобится нам в дальнейшем:

```
in : c1[n_] := Piecewise[{{(-1)^(n+1), n ≥ -1}, {0, n < -1}}]
```

Таким образом, мы получили разложение:

$$f(z) = \sum_{n=-1}^{\infty} (-1)^{n+1} (z-1)^n \quad , \text{ при } 0 < |z-1| < 1$$

Для того чтобы найти разложение в кольце $|z-1| > 1$, воспользуемся формулой (2):

$$c_n = \oint_{|z-1|=R} \frac{f(z)}{(z-1)^{n+1}}\, dz, \text{ где } R > 1$$

Согласно теореме Коши для многосвязной области имеем:

$$\oint_{|z-1|=R} \frac{f(z)}{(z-1)^{n+1}}\, dz = \oint_{|z-1|=r'} \frac{f(z)}{(z-1)^{n+1}}\, dz + \oint_{|z|=r'} \frac{f(z)\cdot z}{(z-1)^{n+1}\cdot z}\, dz, \quad (4)$$
$$\text{где } 0 < r' < 1$$

Здесь первый интеграл есть разложение в ряд Лорана в кольце $0 < |z-1| < 1$ и уже вычислен, а второй интеграл, исходя из интегральной формулы Коши, будет равен:

$$\oint_{|z|=r'} \frac{f(z)\cdot z^u}{(z-1)^{n+1}\cdot z^u}\, dz = \frac{1}{u!}\left(\frac{d^{u-1}}{dz^{u-1}}\frac{f(z)\cdot z^u}{(z-1)^{n+1}}\right)(0),$$

$$\text{где } u - \text{ кратность корня } z = 0 \text{ в уравнении Denominator}[f(z)] = 0$$

Теперь запишем функцию c2 от n:

```
in : c2[n_] := Simplify[f[z] z/(z - 1)^(n+1)] /. z → 0
```

Протабулируем функцию c2 (n) на отрезке от -8 до 8

```
in : Table[{n, c2[n]}, {n, -8, 8}]

out : {{-8, 1}, {-7, -1}, {-6, 1}, {-5, -1}, {-4, 1},
       {-3, -1}, {-2, 1}, {-1, -1}, {0, 1}, {1, -1}, {2, 1},
       {3, -1}, {4, 1}, {5, -1}, {6, 1}, {7, -1}, {8, 1}}
```

В нашем обозначении из (4) следует:

$$c_n = c_{1\,n} + c_{2\,n}$$

Отобразим множество вида $\{$ n, c_n $\}$, $-8 \leq n \leq 8$:

```
in : Table[{n, c1[n] + c2[n]}, {n, -8, 8}]

out : {{-8, 1}, {-7, -1}, {-6, 1}, {-5, -1},
       {-4, 1}, {-3, -1}, {-2, 1}, {-1, 0}, {0, 0}, {1, 0},
       {2, 0}, {3, 0}, {4, 0}, {5, 0}, {6, 0}, {7, 0}, {8, 0}}
```

Как мы видим, коэффициенты при степени большей -2, равны 0.

Конечно, в нашем конкретном случае функция общего члена $c(n)$ легко определяется, однако воспользуемся функцией FindSequenceFunction, которая, как следует из названия, находит функцию из последовательности. Она имеет следующую структуру:

FindSequenceFunction[{ $\{n_{-k}, a_k\}$ ....$\{n_0, a_0\}$ ....$\{n_q, a_q\}$}, n]. Применим ее к последовательности, начинающейся при n=-2 и до n=-8.

```
in : FindSequenceFunction[Table[{n, c2[n]}, {n, -8, -2}], n]
```

out : $(-1)^{10+n}$

Итак, запишем разложение в ряд Лорана в кольце $|z-1| > 1$:

```
in : Sum[(z - 1)^n (c1[n] + c2[n]), {n, -8, 8}]
```

$$out:\ \frac{1}{(-1+z)^8}-\frac{1}{(-1+z)^7}+\frac{1}{(-1+z)^6}-\frac{1}{(-1+z)^5}+\frac{1}{(-1+z)^4}-\frac{1}{(-1+z)^3}+\frac{1}{(-1+z)^2}$$

В общем виде:

$$f(z)=\sum_{n=2}^{\infty}\frac{(-1)^n}{(z-1)^n},\ \text{при}\ \ |z-1| > 1$$

Рассмотрим более сложный пример.

Пример №2

Найти все разложения указанной функции в ряд Лорана по степеням $z-z_0$ и установить области сходимости полученных разложений:

$$f(z)=\frac{1}{(z-2)^2\,(z+3)^3},\ z_0=-3$$

В данной задаче представлена более сложная функция с несколькими точками потери аналитичности разной кратности.

Запишем функцию:

$$in:\ f[z\_] := \frac{1}{(z-2)^2\,(z+3)^3}$$

Для того чтобы найти кольца аналитичности, решим уравнение $f[z]^{-1}=0$ или просто приравняем знаменатель к 0. Для того чтобы получить знаменатель функции $f(z)$, будем использовать опцию Denominator[$f[z]$]

```
in : sol = z /. Solve[Denominator[f[z]] == 0, z]
```

out : {-3, -3, -3, 2, 2}

Следующим пунктом в нашем алгоритме будет определение кратности каждого корня. Воспользуемся тем же методом, что и в главе 1 параграфе 4 “Интегралы”. Там мы определяли кратность при помощи функции Counts[list]:

```
in : rul = Counts[sol]
```

```
out : <|-3 → 3, 2 → 2|>
```

Теперь запишем функцию кратности i-ого корня:

```
in : sol = DeleteDuplicates[sol];
     u[i_] := sol[[i]] /. rul
```

Теперь нужно разделить комплексную плоскость на окружности. Изобразим графически:

```
in : Graphics[{Circle[{-3, 0}, 3], Circle[{2, 0}, 0.6],
       Circle[{-3, 0}, 6], PointSize[Large], Point[{-3, 0}],
       Point[{2, 0}], Text[r1, {0.6, 0.6}], Text[r2, {2, 2}] ,
       Text[γ, {2.7, 0.6}]  }, PlotRange → {{-8, 8}, {-8, 8}},
     Axes → True]
```



рис. 5.1.2

Для того чтобы аналитически определить кольца сходимости, поступим следующим образом: возьмем точку z_0-центр разложения и будем сравнивать расстояние от нее до корней $z_1 \ldots z_n$. Наименьшее расстояние будет соответствовать верхней границе первого кольца и нижней границе второго кольца, верхней же границе второго кольца будет соответствовать уже наименьшее расстояние до следующего корня, и так далее. Составим множество границ колец, используя функцию Sort[list], которая располагает элементы list в порядке возрастания. Сортировать же мы будем множество всех расстояний от z_0 до z_i:

```
in : a = With[{z0 = -3},
       Sort[Table[Abs[-z0 + sol[[i]]], {i, 1, Length[sol]}]]]
```

```
out : {0, 5}
```

Первый элемент равен нулю - следовательно центр разложения z_0 совпадает с одним из корней.

Получили 2 кольца: 1) $0 < |z+3| < 5$ и 2) $|z+3| > 5$

Разложение в окрестности z_0 всегда можно отыскать с помощью функции Series:

in : Series[f[z], {z, -3, 5}]

out : $\frac{1}{25\,(z+3)^3}+\frac{2}{125\,(z+3)^2}+\frac{3}{625\,(z+3)}+\frac{4}{3125}+\frac{z+3}{3125}+\frac{6\,(z+3)^2}{78125}+\frac{7\,(z+3)^3}{390625}+\frac{8\,(z+3)^4}{1953125}+\frac{9\,(z+3)^5}{9765625}+O[z+3]^6$

Получим формулу n-ого коэффициента:

in : c0 = SeriesCoefficient[f[z], {z, -3, n}]

out : $\begin{cases} \frac{3}{625} & n == -1 \\ \frac{2}{125} & n == -2 \\ \frac{1}{25} & n == -3 \\ 5^{-5-n}\,(4+n) & n \geq 0 \\ 0 & \text{True} \end{cases}$

Составим функцию $c_1(n)$ - коэффициента разложения в кольце 1):

in : c1[k_] := c0 /. n → k

Итак, получили:

$$f(z)=\sum_{n=-3}^{\infty}\frac{(4+n)\,(z+3)^n}{5^{5+n}},\ \text{при } 0<|z+3|<5$$

Разложение в кольце 2) $|z+3|>5$ будем искать аналогично формуле (4):

$$c_{2\,n}=\frac{1}{1!}\,\frac{d}{dz}\,\frac{f(z)\,(z-2)^2}{(z+3)^{n+1}},\ z\to 2$$

Запишем выражение для коэффициента $c_2(n)$:

in : c2[n_] := D$\left[\text{Simplify}\left[f[z]\,\frac{(z-\text{sol}[[2]])^{u[2]}}{(z+3)^{n+1}}\right],\ \{z,\ u[2]-1\}\right]$ /. z → 2

in : Table[{n, c2[n]}, {n, -8, 8}]

out : $\left\{\{-8, 500\}, \{-7, 75\}, \{-6, 10\}, \{-5, 1\}, \{-4, 0\}, \left\{-3, -\frac{1}{25}\right\}, \left\{-2, -\frac{2}{125}\right\}, \left\{-1, -\frac{3}{625}\right\}, \left\{0, -\frac{4}{3125}\right\}, \left\{1, -\frac{1}{3125}\right\}, \left\{2, -\frac{6}{78125}\right\}, \left\{3, -\frac{7}{390625}\right\}, \left\{4, -\frac{8}{1953125}\right\}, \left\{5, -\frac{9}{9765625}\right\}, \left\{6, -\frac{2}{9765625}\right\}, \left\{7, -\frac{11}{244140625}\right\}, \left\{8, -\frac{12}{1220703125}\right\}\right\}$

Итак, теперь, для того чтобы найти разложение в кольце, внутренний контур которого охватывает все особые точки, нужно вычислить сумму коэффициентов $c(n) = c_1(n) + c_2(n)$:

```
in : Sum[(z + 3)^n (c1[n] + c2[n]), {n, -12, 8}]
```

$$out: \frac{625000}{(3+z)^{12}} + \frac{109375}{(3+z)^{11}} + \frac{18750}{(3+z)^{10}} + \frac{3125}{(3+z)^{9}} + \frac{500}{(3+z)^{8}} + \frac{75}{(3+z)^{7}} + \frac{10}{(3+z)^{6}} + \frac{1}{(3+z)^{5}}$$

Найдем формулу коэффициента n:

```
in : c22[k_] :=
      FindSequenceFunction[Table[{n, c1[n] + c2[n]}, {n, -12, -5}], j] /.
       j → k
```

```
in : c22[n]
```

$$out: -5^{-5-n} (4+n)$$

Из получившейся последовательности действий видно, что каждый раз при увеличении радиуса кольца разложения (выход за предел текущего кольца) общий коэффициент разложения будет увеличиваться на $c_i(n)$.

В итоге, получили:

$$f(z) = -\sum_{n=-\infty}^{-5} \frac{(4+n)(z+3)^n}{5^{5+n}}, \text{ при } |z+3| > 5$$

$$f(z) = \sum_{n=-3}^{\infty} \frac{(4+n)(z+3)^n}{5^{5+n}}, \text{ при } 0 < |z+3| < 5$$

Общий алгоритм

Проанализировав последовательность действий, создадим программный модуль разложения в ряд Лорана:

```
in : loranO[z_, expr_ , z0_, r_, k_] := Module[{sol, sol1, u, c0, c1},
    c0[n_] := Evaluate[SeriesCoefficient[expr, {z, z0, n}]];
    sol1 =
     z /.
      Solve[0 < Abs[z - z0] < r &&
        Denominator[Simplify[TrigToExp[expr]]] == 0, z];
    c1[n_] := 0;
    If[sol1 ≠ {},
     sol = DeleteDuplicates[sol1];
     u[i_] := sol[[i]] /. Counts[sol1];
     c1[n_] :=
      Sum[1/(u[i] - 1)!
        Limit[D[Simplify[expr (z - sol[[i]])^u[i]/(z - z0)^(n+1)], {z, u[i] - 1}],
         z → sol[[i]]], {i, 1, Length[sol]}]];
    Sum[(z - z0)^n (c0[n] + c1[n]), {n, -k, k}]]
```

Итак, данный модуль имеет следующую построчную структуру:

1. Запись коэффициента c0[n] разложения в ряд в окрестности z_0.

2. Поиск тех корней знаменателя, которые расположены не дальше r - кольца разложения, не включая корень z0.

3. В случае если множество корней не пустое множество, определение кратности каждого корня.

4. Вычисление коэффициентов $c_i[n]$ по формуле (4).

5. Запись получившегося разложения.

Продемонстрируем пример работы:

Пример №12.354

```
in : loranO[z, 1/((z - 2) (z + 3)), 2, 6, 8]
```

$$out: \frac{15625}{(-2+z)^8}-\frac{3125}{(-2+z)^7}+\frac{625}{(-2+z)^6}-\frac{125}{(-2+z)^5}+\frac{25}{(-2+z)^4}-\frac{5}{(-2+z)^3}+\frac{1}{(-2+z)^2}$$

Пример №12.364

in : `loranO[z, z/(z^2 + 1)^2, I, 3, 8]`

out : $-\frac{112\,i}{(-i+z)^8}+\frac{48}{(-i+z)^7}+\frac{20\,i}{(-i+z)^6}-\frac{8}{(-i+z)^5}-\frac{3\,i}{(-i+z)^4}+\frac{1}{(-i+z)^3}$

Пример № x

in : `loranO[z, 1/Sin[z], 0, Pi + 1, 8]`

out : $-\frac{2\pi^6}{z^7}-\frac{2\pi^4}{z^5}-\frac{2\pi^2}{z^3}-\frac{1}{z}+\left(\frac{1}{6}-\frac{2}{\pi^2}\right)z+\left(\frac{7}{360}-\frac{2}{\pi^4}\right)z^3+\left(\frac{31}{15120}-\frac{2}{\pi^6}\right)z^5+\left(\frac{127}{604800}-\frac{2}{\pi^8}\right)z^7$

Созданную функцию будем использовать при необходимости в дальнейшем.

5.2. Характер изолированных особых точек

Точка z_0 называется правильной точкой для аналитической в области D функции $f(z)$, если существует степенной ряд $\sum_{n=0}^{\infty} c_n(z_0)\,(z-z_0)^n$ с радиусом сходимости $r(z_0)>0$ такой, что в общей части круга сходимости $|z-z_0|<r(z_0)$ и области D сумма этого ряда $\phi_{z_0}(z)$ совпадает с $f(z)$. Точки, не являющиеся правильными, называются особыми.

Точка z_0 называется изолированной особой точкой функции $f(z)$, если $f(z)$ - однозначная аналитическая функция в кольце $0<|z-z_0|<R$, а z_0 – особая точка.

Аналогично, точка $z_0=\infty$ называется изолированной особой точкой функции $f(z)$, если $f(z)$ - однозначная аналитическая функция в кольце $r<|z|<\infty$ и $z=\infty$ - особая точка.

Изолированная особая точка z_0 функции $f(z)$ называется:

Устранимой особой точкой, если существует конечный предел:

$$\lim_{z\to z_0} f(z)=a\neq\infty \qquad (5)$$

Полюсом порядка $m\geq 1$, если для функции $g(z)=f(z)^{-1}$ точка z_0 является нулем порядка m. Если z_0 – полюс, то

$$\lim_{z\to z_0} f(z)=\infty \qquad (6)$$

Существенно особой если конечного предела не существует:

$$\lim_{z\to z_0} f(z) - \text{не существует} \qquad (7)$$

Пример №12.382

Указать все конечные особые точки заданной ниже функции и определить их характер:

$$f(z) = \frac{1}{\left(z^2 + i\right)^3}$$

Запишем функцию:

```
in : f[z_] := 1/(z^2 + I)^3
```

Для того чтобы найти особые точки, решим уравнение Denominator[f[z]] = 0:

```
in : sol = Flatten[Solve[Denominator[f[z]] == 0, z]]
out : {z → -(-1)^(3/4), z → -(-1)^(3/4),
       z → -(-1)^(3/4), z → (-1)^(3/4), z → (-1)^(3/4), z → (-1)^(3/4)}
```

Мы получили 2 корня кратности 3. Для того чтобы избавиться от повторов, воспользуемся функцией DeleteDuplicates.

```
in : sol = DeleteDuplicates[sol]
out : {z → -(-1)^(3/4), z → (-1)^(3/4)}
```

Теперь исследуем характер полученных особых точек. Найдем предел в точке $z = -(-1)^{\frac{3}{4}}$:

```
in : Limit[f[z], sol[[1]]]
out : (-1)^(3/4) ∞
```

Получили, что точка $z = -(-1)^{\frac{3}{4}}$ является полюсом третьего порядка.

Проделаем то же самое для второй точки $z = (-1)^{\frac{3}{4}}$:

```
in : Limit[f[z], sol[[2]]]
out : (1 - ⅈ)/√2 ∞
```

Получается, что обе точки являются полюсами третьего порядка.

Пример 12.388

Указать все конечные особые точки заданной ниже функции и определить их характер:

$$f(z) = \frac{z(\pi - z)}{\sin(2\,z)}$$

Запишем функцию:

```
in : f[z_] := z (Pi - z) / Sin[2 z]
```

Решим уравнение вида $f^{-1}(z)$:

```
in : sol = Solve[f[z]^-1 == 0, z]
```

Solve::ifun : Inverse functions are being used by Solve, so some solutions may not be found; use Reduce for complete solution information. >>

out : { }

Так как в математике наша функция выглядит как $f(z) = (\pi - z)\, z\, \mathrm{Csc}[2\, z]$, появляются трудности с решением уравнения вида $\mathrm{Csc}[2\, z]^{-1} = 0$. Для того чтобы решить эту проблему можно задав круг, в котором будем искать корни уравнения $f(z)^{-1} = 0$:

```
in : sol = Solve[Abs[z] < 10 && f[z]^-1 == 0, z]
```

out : $\left\{\{z \to -3\pi\}, \left\{z \to -\frac{5\pi}{2}\right\}, \{z \to -2\pi\}, \left\{z \to -\frac{3\pi}{2}\right\}, \{z \to -\pi\}, \left\{z \to -\frac{\pi}{2}\right\}, \left\{z \to \frac{\pi}{2}\right\}, \left\{z \to \frac{3\pi}{2}\right\}, \{z \to 2\pi\}, \left\{z \to \frac{5\pi}{2}\right\}, \{z \to 3\pi\}\right\}$

Отсюда общее множество решений можно записать в виде:

```
in : soln[n_] := z → Pi/2 n
```

Из структуры функции видно, что особенно интересны для нас точки $z = 0$ и $z = \pi$. Исследуем функцию в данных точках:

```
in : Limit[f[z], soln[0]]
```

out : $\frac{\pi}{2}$

```
in : Limit[f[z], soln[2]]
```

out : $-\frac{\pi}{2}$

Обе точки являются устранимыми особыми точками, так как в них существует конечный предел (1).

Все остальные корни уравнения (4) будут являться полюсами первого порядка. Докажем это:

```
in : Table[{soln[i], Limit[f[z], soln[i]]}, {i, -4, 4}]
```

$$out: \left\{\left\{z\to -2\pi,\ \frac{1}{\left(i+4\pi^2\right)^3}\right\},\ \left\{z\to -\frac{3\pi}{2},\ \frac{64}{\left(4i+9\pi^2\right)^3}\right\},\ \left\{z\to -\pi,\ \frac{1}{\left(i+\pi^2\right)^3}\right\},\right.$$
$$\left\{z\to -\frac{\pi}{2},\ \frac{64}{\left(4i+\pi^2\right)^3}\right\},\ \{z\to 0,\ i\},\ \left\{z\to \frac{\pi}{2},\ \frac{64}{\left(4i+\pi^2\right)^3}\right\},$$
$$\left.\left\{z\to \pi,\ \frac{1}{\left(i+\pi^2\right)^3}\right\},\ \left\{z\to \frac{3\pi}{2},\ \frac{64}{\left(4i+9\pi^2\right)^3}\right\},\ \left\{z\to 2\pi,\ \frac{1}{\left(i+4\pi^2\right)^3}\right\}\right\}$$

Итак, имеем, что:

$$z_0 = 0,\ z_0 = \pi\ -\ \text{устранимые особые точки}$$
$$z_0 = \frac{\pi k}{2},\ k = \pm 1, -2, \pm 3 \ldots\ -\ \text{полюса первого порядка.}$$

Пример 12.391

Указать все конечные особые точки заданной ниже функции и определить их характер:

$$f(z) = e^{\frac{1}{z-3i}}$$

```
in : f[z_] := Exp[1/(z - 3 I)]
```

Изучим точку $z = 3\,i$:

```
in : f[z] // ComplexExpand
```

$$out: e^{\frac{z}{9+z^2}} \operatorname{Cos}\left[\frac{3}{9+z^2}\right] + i\, e^{\frac{z}{9+z^2}} \operatorname{Sin}\left[\frac{3}{9+z^2}\right]$$

Эта точка будет существенно особой для функции f(z), так как:

$$\lim_{z\to 3i} \sin\left(\frac{3}{z^2+9}\right) \text{не имеет конечного значения}$$

Ответ:

$$z_0 = 3\,i\ -\ \text{существенно особая точка}$$

Пример 12.399

Для заданной функции выяснить характер бесконечно удаленной особой точки:

$$f(z) = \frac{3z^5 - 5z + 2}{z^2 + z - 4}$$

```
in : f[z_] := (3 z^5 - 5 z + 2)/(z^2 + z - 4)
```

Для того чтобы исследовать бесконечно удаленную особую точку, проведем замену: $z = \frac{1}{\eta}$

```
in : f[1/η] // Simplify
```

$$out:\ \frac{3 - 5\eta^4 + 2\eta^5}{\eta^3 + \eta^4 - 4\eta^5}$$

Исследуем точку η=0:

```
in : Limit[f[1/η], η → 0]
```

```
out : ∞
```

Следовательно точка $z_0 = \infty$, бесконечно удаленная, является полюсом функции $f(z)$

§6. Вычеты и их применение.

6.1. Вычет функции и его вычисление.

Если функция $f(z)$ аналитична в некоторой окрестности точки z_0, за исключением может быть самой точки z_0, то вычетом функции $f(z)$ относительно точки z_0, обозначаемым $\mathrm{res}[f(z),\ z \to z_0]$ или $\mathrm{res}[f(z),\ z_0]$ называется число, равное значению интеграла $\frac{1}{2\pi i}\int_C f(z)\,dz$, где C - некоторый замкнутый контур, лежащий в области аналитичности $f(z)$ и содержащий внутри себя только одну особую точку z_0.

Вычет функции совпадает с коэффициентом c_{-1} разложения $f(z)$ в ряд Лорана по степеням $z - z_0$, т.е.

$$\mathrm{res}[f(z),\ z_0] = c_{-1} = \frac{1}{2\pi i}\int_{|z-z_0|=r} f(z)\,dz \tag{1}$$

Если $z_0 = \infty$ - изолированная особая точка функции $f(z)$, то

$$\mathrm{res}[f(z),\ \infty] = \frac{1}{2\pi i}\int_{C_R^-} f(z)\,dz \tag{2}$$

где $C_R^-=\{z \mid |z| = R\}$, R достаточно велико и обход контура - по часовой стрелке. Заметим, что если

$$f(z) = \sum_{n=-\infty}^{\infty} c_n z^n, \qquad r < |z| < \infty,$$

$$c_n = \frac{1}{2\pi i}\int_{|z|=R>r}\frac{f(z)}{z^{n+1}}\,dz,\ \ n = 0, \pm1, \pm2,\ \text{ то}$$

$$\operatorname{res}[f(z), \infty] = -c_{-1}. \tag{3}$$

Если z_0 – полюс первого порядка функции $f(z)$, то

$$\operatorname{res}[f(z), z_0] = \lim_{z\to z_0}(z - z_0)\,f(z) \tag{4}$$

Если z_0 – полюс порядка $m \geq 2$ функции $f(z)$, то

$$\operatorname{res}[f(z), z_0] = \frac{1}{(m-1)!}\lim_{z\to z_0}\frac{d^{m-1}}{dz^{m-1}}\left((z - z_0)^m f(z)\right) \tag{5}$$

Пример №12.416

Найти вычеты указанной функции относительно каждого из его полюсов, отличных от ∞

$$f(z) = \frac{e^z}{z^2\left(z^2 + 9\right)}$$

Найдем каждый полюс заданной функции:

```
in : f[z_] := E^z / (z^2 (z^2 + 9))
```

Для того чтобы найти особые точки функции $f(z)$, решим уравнение вида: $f(z)^{-1} = 0$.

```
in : sol = z /. Solve[f[z]^-1 == 0, z]

out : {0, 0, -3 i, 3 i}
```

Для того чтобы найти вычет в точке, воспользуемся функцией Residue[$f[z]$, $\{z, z_0\}$], которая находит вычет функции f(z) в точке z_0.

```
in : Residue[f[z], {z, 0}]

out : 1/9
```

Это можно проделать для каждого полюса, но если представить, что имеется очень много полюсов, то искать вычет в каждом отдельно - довольно трудоемкая задача.

Найдем вычеты сразу во всех точках множества полюсов.

```
in : Table[ {sol[[i]], Residue[f[z], {z, sol[[i]]}]},
       {i, 1, Length[sol]}] // DeleteDuplicates
```

out : $\left\{\left\{0, \frac{1}{9}\right\}, \left\{-3\,i, -\frac{1}{54}\,i\,e^{-3i}\right\}, \left\{3\,i, \frac{1}{54}\,i\,e^{3i}\right\}\right\}$

Элементы данного множества имеют вид: z_0 , $\operatorname{res}(f(z),\ z_0)$.

Получили три полюса, причем один из них второго порядка ($z_0 = 0$), и вычеты в каждом из них.

Пример №12.413

Найти вычеты указанной функции относительно каждого из его полюсов, отличных от ∞

$$f(z) = \frac{1}{\sin(z) - \frac{1}{2}}$$

Запишем функцию $f(z)$ и найдем полюса, решив уравнение $f(z)^{-1} = 0$:

```
in : f[z_] := 1/(Sin[z] - 1/2)
```

```
in : sol = Solve[f[z]^-1 == 0, z]
```

```
out : {{z → ConditionalExpression[π/6 + 2 π C[1], C[1] ∈ Integers]},
  {z → ConditionalExpression[5 π/6 + 2 π C[1], C[1] ∈ Integers]}}
```

Получили бесконечное число решений, следовательно и полюсов. Перепишем решение в более привычный для нас вид:

```
in : sol1 = {{z → π/6 + 2 π n}, {z → 5 π/6 + 2 π n}};
```

Или, если записать одним выражением:

```
in : sol2 = (-1)^n π/6 + π n;
```

Если функция периодична, то и функция вычета от n также будет периодичной. Докажем это составив множество $\{z_0(n),\ \mathrm{Res}(f(z),\ z \to z_0(n))\}$

```
in : Table[{sol2, Residue[f[z], {z, sol2}]}, {n, -3, 3}]
```

$$out:\ \left\{\left\{-\frac{19\pi}{6}, -\frac{2}{\sqrt{3}}\right\}, \left\{-\frac{11\pi}{6}, \frac{2}{\sqrt{3}}\right\}, \left\{-\frac{7\pi}{6}, -\frac{2}{\sqrt{3}}\right\}, \left\{\frac{\pi}{6}, \frac{2}{\sqrt{3}}\right\}, \left\{\frac{5\pi}{6}, -\frac{2}{\sqrt{3}}\right\}, \left\{\frac{13\pi}{6}, \frac{2}{\sqrt{3}}\right\}, \left\{\frac{17\pi}{6}, -\frac{2}{\sqrt{3}}\right\}\right\}$$

Ответ:

$$\mathrm{res}\left[f(z), (-1)^n \frac{\pi}{6} + \pi n\right] = (-1)^n \frac{2}{\sqrt{3}}$$

В таком же порядке следует решать задачи №№12.408-12.423.

Общий алгоритм

Составим программный модуль, который для любой функции $f(z)$ в кольце $|z - a| < b$ будет находить все полюса, а затем вычислять в них вычет.

Действовать он будет следующим образом:

1. Решение уравнения $f^{-1}(z) = 0$ при условии $|z_k - a| < b$.

2. Составление множества типа $\{z_k, \text{res}[f(z), z_k]\}$

```
in : residueK[z_, func_, a_, b_] := Module[{sol},
       sol = z /. Solve[Abs[z - a] < b && func^-1 == 0, z] // DeleteDuplicates;
       Table[{sol[[i]], Residue[func, {z, sol[[i]]}]},
        {i, 1, Length[sol]}]
      ]
```

Продемонстрируем пример его работы:

$$in:\ \text{residueK}\left[z, \frac{E^z}{z^2\,(z^2+9)}, 0, 10\right]$$

$$out:\ \left\{\left\{0, \frac{1}{9}\right\}, \left\{-3\,i, -\frac{1}{54}\,i\,e^{-3\,i}\right\}, \left\{3\,i, \frac{1}{54}\,i\,e^{3\,i}\right\}\right\}$$

$$in:\ \text{residueK}\left[z, \frac{1}{\text{Sin}[z] - \frac{1}{2}}, 0, 10\right]$$

$$out:\ \left\{\left\{-\frac{19\,\pi}{6}, -\frac{2}{\sqrt{3}}\right\}, \left\{-\frac{11\,\pi}{6}, \frac{2}{\sqrt{3}}\right\}, \left\{-\frac{7\,\pi}{6}, -\frac{2}{\sqrt{3}}\right\}, \left\{\frac{\pi}{6}, \frac{2}{\sqrt{3}}\right\}, \left\{\frac{5\,\pi}{6}, -\frac{2}{\sqrt{3}}\right\}, \left\{\frac{13\,\pi}{6}, \frac{2}{\sqrt{3}}\right\}, \left\{\frac{17\,\pi}{6}, -\frac{2}{\sqrt{3}}\right\}\right\}$$

Примечание: Данный модуль работает лишь с теми функциями, в которых полюс является нулём функции $g(z) = f^{-1}(z)$. С более сложными функциями он справиться не может.

Рассмотрим следующий номер:

Пример №12.424

Найти вычеты функции относительно точки $z_0 = 0$:

$$f(z) = e^{1/z},\ z_0 = 0$$

Введем заданную функцию:

$$in:\ f[z\_] := \text{Exp}\left[\frac{1}{z}\right]$$

Точка $z_0 = 0$ для нее является существенно особой, так как не существует предела функции при $z \to z_0$:

```
in : Limit[Cos[1 / z] + I Sin[1 / z], z → 0]
out : i Interval[{-1, 1}] + Interval[{-1, 1}]
```

Попытаемся воспользоваться встроенной функцией Residue:

```
in : Residue[f[z], {z, 0}]
```

out : $\text{Residue}\left[e^{\frac{1}{z}}, \{z, 0\}\right]$

Никакого результата мы не получили.

Для нахождения вычета определим коэффициент c_{-1} разложения функции $e^{\frac{1}{z}}$ в ряд Лорана по степеням z .

Функция разложения в ряд Series также не дала нам нужного результата.

```
in : Series[f[z], {z, 0, 4}]
```

out : $e^{\frac{1}{z}}$

Известно разложение в ряд функции e^{η}:

$$e^{\eta} = \sum_{n=0}^{\infty} \frac{\eta^n}{n!}$$

а также то, что данный ряд сходится на всей комплексной плоскости. Поэтому заменим $\eta = \frac{1}{z}$.

Тогда получается ряд:

$$e^{\frac{1}{z}} = \sum_{n=0}^{\infty} \frac{1}{n!} \left(\frac{1}{z}\right)^n$$

Запишем это разложение следующим образом:

```
in : Normal[Series[Exp[p], {p, 0, 4}]] /. p → 1/z
```

out : $1 + \frac{1}{24 z^4} + \frac{1}{6 z^3} + \frac{1}{2 z^2} + \frac{1}{z}$

Отсюда видно, что коэффициент c_{-1}=1. Следовательно ответ:

$$\text{res}\left(e^{\frac{1}{z}}, z \to 0\right) = 1$$

Примечание: номера №12.425,426 имеют точно такой же ход решения, поэтому рекомендуются к самостоятельному решению.

Пример №12.427

Найти вычеты функции относительно точки $z_0 = \infty$

$$f(z) = \sin\left(\frac{1}{z}\right), \ z_0 = \infty$$

Имеем функцию:

```
in : f[z_] := Sin[1/z]
```

Решим задачу при помощи разложения функции в ряд Лорана.

Имеем стандартное разложение:

$$\sin(\eta) = \sum_{n=0}^{\infty}(-1)^n \frac{\eta^{2n+1}}{(2n+1)!}$$

Произведя замену $\eta \to 1/z$, получаем:

$$\sin\left(\frac{1}{z}\right) = \sum_{n=0}^{\infty}(-1)^n \frac{(1/z)^{2n+1}}{(2n+1)!}$$

Получим это разложение:

```
in : Normal[Series[Sin[p], {p, 0, 7}]] /. p → 1/z
```

$$out: -\frac{1}{5040\, z^7} + \frac{1}{120\, z^5} - \frac{1}{6\, z^3} + \frac{1}{z}$$

Вычет в бесконечно удаленной точке по формуле (3) равен:

$$\mathrm{res}(f(z),\, z \to \infty) = -c_{-1}$$

поэтому, исходя разложения в ряд, получаем ответ:

$$\mathrm{res}\left(\sin\left(\frac{1}{z}\right),\, z \to \infty\right) = -1$$

Проверим себя при помощи функции Residue:

```
in : Residue[Sin[1/z], {z, Infinity}]

out : -1
```

Получили верный ответ.

Пример №12.428

Найти вычеты функции относительно точки $z_0 = \infty$

$$f(z) = \frac{1}{(z-1)^2\left(z^2+1\right)}, \quad z_0 = \infty$$

Дана функция:

```
in : f[z_] := 1/((z - 1)^2 (z^2 + 1))
```

Найдем вычет в бесконечно удаленной точке, как:

$$\operatorname{res}(f(z),\, z \to \infty) = -c_{-1}$$

Для этого воспользуемся полученным нами в разделе Ряды Лорана модулем, который мы назвали loran[z, f[z], z0, r, k] :

```
in : loranO[z, f[z], 0, 100, 7]
```

$$out: \frac{2}{z^7} + \frac{2}{z^6} + \frac{2}{z^5} + \frac{1}{z^4}$$

Примечание: для того чтобы воспользоваться функцией loranO сначала ее нужно определить.

Видим что коэффициент перед $\frac{1}{z}$ отсутствует, следовательно ответ:

$$\operatorname{res}\left(\frac{1}{(z-1)^2\left(z^2+1\right)},\, \infty\right) = 0$$

Также можно напрямую воспользоваться функцией Residue и получить тот же ответ:

$$in: \text{Residue}\left[\frac{1}{(z-1)^2\left(z^2+1\right)},\, \{z,\, \text{Infinity}\}\right]$$

```
out : 0
```

6.2. Теоремы о вычетах и их применение к вычислению контурных интегралов.

Первая теорема о вычетах. Если функция $f(z)$ аналитична в области D, за исключением изолированных особых точек $z_1, \ldots z_k$, лежащих в этой области, то для любого простого замкнутого контура $C \subset D$, охватывающего точки $z_1, \ldots z_N$:

$$\int_{C^+} f(z)\, dz = 2\pi i \sum_{k=1}^{N} \operatorname{res}(f(z);\, z_k) \tag{6}$$

Вторая теорема о вычетах. Если функция $f(z)$ аналитична во всей комплексной плоскости, за исключением изолированных особых точек $z_1, \ldots z_{N-1}$, и $z_N = \infty$, то:

$$\sum_{k=1}^{N} \operatorname{res}(f(z);\, z_k) = 0 \tag{7}$$

Пример №12.434

Используя теоремы о вычетах, вычислить следующий интеграл:

$$\int_{C^+} \frac{z}{(z-1)(z-2)}\, dz, \quad C = \{z \mid |z-2| = 2\}$$

Для того чтобы решить поставленную задачу, воспользуемся Первой теоремой о вычетах:

Найдем все полюса функции $f(z)$

```
in : f[z_] := z/((z - 1) (z - 2))
```

1) Приведем графическое решение:

Найдем полюса:

```
in : sol = z /. Solve[f[z]^-1 == 0, z]

out : {1, 2}
```

Для этого изобразим контур C, и полюса $z = 1$, $z = 2$.

```
in : Show[ ContourPlot[ (x - 2)^2 + y^2 == 4, {x, 0, 4}, {y, -2, 2},
       AxesLabel → {Re, Im}, PlotRange → {{-1, 5}, {-3, 3}} ,
       Frame → False, Axes → True],
     Graphics[{PointSize[Large], Point[ReIm[sol]]} ] ]
```



рис. 6.2.1

Из рисунка видно, что обе точки лежат внутри контура C.

Найдем интеграл по первой теореме вычетов.

```
in : int = 2 π i Sum[Residue[f[z], {z, sol[[i]]}], {i, 1, Length[sol]}]

out : 2 i π
```

2) Воспользуемся написанной нами в главе 1 параграфе 4 программой integrateC[z, f[z], a, b] :

```
in : integrateC[z, f[z], 2, 2]

out : 2 i π
```

Получили тот же самый ответ.

3) Применим новый подход с использованием вычетов. Для этого воспользуемся программой, написанной в предыдущем разделе, residueK[z, f[z], a, b]. Она находит полюса и вычеты в них:

```
in : res = residueK[z, f[z], 2, 2]
out : {{1, -1}, {2, 2}}
```

Напомним, что элемент данного множество имеет вид: $\{z_k, \mathrm{res}[f(z), z_k]\}$

Теперь, для того чтобы найти значение интеграла, воспользуемся первой теоремой о вычетах (6):

```
in : int = 2 I Pi Sum[res[[i, 2]], {i, 1, Length[res]}]
out : 2 ⅈ π
```

С использованием изложенных выше способов читателю рекомендуется решить задания №№ 12.435-449 самостоятельно.

Общий алгоритм

Составим новый программный модуль, взамен старого integrateC[z, f[z], a, b], который будет иметь те же входные параметры, но использовать первую теорему о вычетах:

```
in : integrateR[z_, func_, a_, b_] := Module[{sol},
       sol = z /. Solve[Abs[z - a] < b && func^-1 == 0, z] // DeleteDuplicates;
       2 I Pi Sum[Residue[func, {z, sol[[i]]}], {i, 1, Length[sol]}]]
```

Продемонстрируем пример работы:

Пример №12.433

$$in:\ -\mathrm{integrateR}\left[z, \frac{1}{z^4+1}, 1, 1\right] \ //\ \mathrm{Simplify}$$

$$out:\ \frac{i\pi}{\sqrt{2}}$$

Пример №12.436

$$in:\ \mathrm{integrateR}\left[z, \frac{\mathrm{Sin}[z]}{z^2+9}, 0, 4\right]$$

$$out:\ \frac{2}{3}\, i\, \pi\, \mathrm{Sinh}[3]$$

Пример №12.441

$$in:\ \mathrm{integrateR}\left[z, \frac{z+1}{\mathrm{Exp}[z]+1}, 0, 4\right] \ //\ \mathrm{Simplify}$$

$$out:\ -4\, i\, \pi$$

Пример №12.437

Используя теоремы о вычетах, вычислить следующий интеграл:

$$\oint \frac{dz}{(z-a)^n (z-b)^n}, \quad C = \{z \mid |z| = 1\}, \; n \in \mathbb{N}, \; 0 \le |a| < 1 < |b|$$

Из условий видно, что a - полюс n-ого порядка, который лежит внутри контура C.

Зададим функцию $f(z, n)$ и попробуем найти вычет в точке a с помощью уже хорошо знакомой нам функции Residue:

```
in : f[z_, n_] := 1/((z - a)^n (z - b)^n)

in : Residue[f[z, n], {z, a}]

out : Residue[(-a + z)^-n (-b + z)^-n, {z, a}]
```

В общем виде результат мы не получили.

Теперь попробуем воспользоваться определением вычета:

$$\operatorname{res}[f(z), z_0] = \frac{1}{(n-1)!} \lim_{z \to z_0} \frac{d^{n-1}}{dz^{n-1}} [(z - z_0)^n f(z)]$$

Зададим вычет, как функцию от n:

```
in : res1[n_] := 1/(n - 1)! Limit[D[(z - a)^n f[z, n], {z, n - 1}], z → a]

in : res1[n]

out : Limit[∂{z,-1+n} (-b + z)^-n, z → a] / (-1 + n)!
```

Результат в общем виде мы снова не получили. Зато можем увидеть, как зависит вычет от n:

```
in : Table[res1[i], {i, 1, 5}]
```

$$out: \left\{\frac{1}{a-b}, -\frac{2}{(a-b)^3}, \frac{6}{(a-b)^5}, -\frac{20}{(a-b)^7}, \frac{70}{(a-b)^9}\right\}$$

Из определения вычета видно:

$$\operatorname{res}[f(z), a] = \frac{1}{(n-1)!} \lim_{z \to a} \frac{d^{n-1}}{dz^{n-1}} \left[\frac{1}{(z-b)^n}\right] \tag{8}$$

Запишем:

$$\frac{d^{n-1}}{dz^{n-1}} \left[\frac{1}{(z-b)^k}\right] = (-1)^{n-1} \frac{k(k+1)\ldots(k+n-2)}{(z-b)^{k+n-1}} = (-1)^{n-1} \frac{n(n+1)\ldots(2n-2)}{(z-b)^{2n-1}} \tag{9}$$

Значит:

$$\operatorname{res}[f(z), a] = \frac{(-1)^{n-1}}{(n-1)!} \frac{n(n+1)\ldots(2n-2)}{(a-b)^{2n-1}}$$

И тогда:

$$\oint \frac{dz}{(z-a)^n (z-b)^n} = 2\pi i \frac{(-1)^{n-1}}{(n-1)!} \frac{n(n+1)\ldots(2n-2)}{(a-b)^{2n-1}},$$
$$n \in \mathbb{N},\ 0 \le |a| < 1 < |b|$$

Пример №12.438

Изменим условия предыдущей задачи на:

$$0 \le |a| < |b| < 1$$

Тогда ввиду симметрии задачи:

$$\operatorname{res}[f(z), b] = \frac{(-1)^{n-1}}{(n-1)!} \frac{n(n+1)\ldots(2n-2)}{(b-a)^{2n-1}}$$

$$\oint \frac{dz}{(z-a)^n (z-b)^n} =$$
$$2\pi i \left(\frac{(-1)^{n-1}}{(n-1)!} \frac{n(n+1)\ldots(2n-2)}{(a-b)^{2n-1}} + \frac{(-1)^{n-1}}{(n-1)!} \frac{n(n+1)\ldots(2n-2)}{(b-a)^{2n-1}} \right) =$$
$$\left(\frac{(-1)^{n-1}}{(n-1)!} \frac{n(n+1)\ldots(2n-2)}{(-1)^{2n-1}(b-a)^{2n-1}} + \frac{(-1)^{n-1}}{(n-1)!} \frac{n(n+1)\ldots(2n-2)}{(b-a)^{2n-1}} \right) =$$
$$\left(-\frac{(-1)^{n-1}}{(n-1)!} \frac{n(n+1)\ldots(2n-2)}{(b-a)^{2n-1}} + \frac{(-1)^{n-1}}{(n-1)!} \frac{n(n+1)\ldots(2n-2)}{(b-a)^{2n-1}} \right) = 0$$

Следовательно:

$$\oint \frac{dz}{(z-a)^n (z-b)^n} = 0,\ n \in \mathbb{N},\ 0 \le |a| < |b| < 1$$

6.3. Применение вычетов к вычислению определенных интегралов.

I. Интегралы вида:

$$\int_0^{2\pi} R(\sin(x), \cos(x))\, dx \tag{10}$$

где R- символ рациональной функции, с помощью замены вида $z = e^{ix}$приводятся к контурным интегралам от рациональных относительно z функций.

Пример №14.450

Используя метод описанный выше, вычислить определенный интеграл:

$$\int_0^{2\pi} \frac{1}{(a + b\cos(x))^2}\, dx,\;\; a > b > 0$$

Запишем функцию:

```
in : f[x_] := 1/(a + b Cos[x])^2
```

Мы конечно же можем вычислить данный интеграл напрямую:

```
in : Assuming[0 < b && a > b && x ∈ Reals, Integrate[f[x], {x, 0, 2 Pi}]] //
      Timing
```

$$out:\ \left\{7.956051,\ \frac{2\,a\,\pi}{\left(a^2 - b^2\right)^{3/2}}\right\}$$

но на вычисление такого простого интеграла “в лоб” *Mathematica* затрачивает целых 8 секунд, значит на более сложные интегралы времени уйдет еще больше, так что попробуем упростить задачу заменой описанной выше.

Произведем замену:

$$e^{i\,x} = z,\; dx = \frac{dz}{i\,z} \tag{11}$$

Тогда получим:

```
in : Simplify[TrigToExp[f[x]] * 1/(I z) /. <|E^(I x) → z, E^(-I x) → 1/z|>]
```

$$out:\ -\frac{4\,i\,z}{\left(b + 2\,a\,z + b\,z^2\right)^2}$$

Напомним: функция TrigToExp записывает тригонометрические функции через экспоненциальные, множитель $\frac{1}{i\,z}$ следует из (11).

Запишем функцию $f_1(z)$:

```
in : f1[z_] := - 4 I z/(b + 2 a z + b z^2)^2
```

Найдем полюса из уравнения $f_1(z)^{-1} = 0$ при условии, что $|z_k| < 1$, а также $a > b,\;\; b > 0$

```
in : sol =
    z /. DeleteDuplicates[Solve[a > b && b > 0 && Abs[z] < 1 && f1[z]^-1 == 0,
        z]] // Simplify
```

$$out:\ \left\{\text{ConditionalExpression}\left[\sqrt{-1+\frac{a^2}{b^2}}-\frac{a}{b},\ b>0\ \&\&\ a>b\right]\right\}$$

Получили один полюс:

$$in:\ z1=\sqrt{-1+\frac{a^2}{b^2}}-\frac{a}{b};$$

Теперь воспользовавшись первой теоремой о вычетах найдем значение интеграла и время, за которое он вычисляется:

```
in : int = 2 ⅈ π Residue[f1[z], {z, z1}] // Timing
```

$$out:\ \left\{0.0156001,\ -\frac{2\,a\,\pi}{\sqrt{-1+\frac{a^2}{b^2}}\ b\left(-a^2+b^2\right)}\right\}$$

Теперь осталось преобразовать ответ, чтобы убедиться, что он совпадает с ответом полученным “в лоб”:

```
in : Simplify[int[[2]], a > b && b > 0]
```

$$out:\ \frac{2\,a\,\pi}{\left(a^2-b^2\right)^{3/2}}$$

Мы получили тот же ответ, но за значительно меньший промежуток времени.

Окончательный ответ:

$$\int_0^{2\pi}\frac{1}{(a+b\cos(x))^2}\,dx=\frac{2\,a\,\pi}{\left(a^2-b^2\right)^{3/2}}$$

Данным алгоритмом можно считать любые интегралы вида:

$$\int_0^{2\pi}R(\sin(x),\ \cos(x))\,dx$$

Остальные номера, №№12.450-455, рекомендуются к самостоятельному выполнению ввиду своего однообразия.

II. Интегралы вида:

$$\int_{-\infty}^{\infty} f(x)\, dx$$

где $f\,(x)$ - функция непрерывная на $(-\infty;\,\infty)$, аналитическая в верхней полуплоскости, за исключением конечного числа особых точек z1....zn, лежащих в конечной части верхней полуплоскости, вычисляются:

$$\int_{-\infty}^{\infty} f(x)\, dx = 2\,\pi\, i \sum_{i=1}^{n} \mathrm{res}(f(z),\ z_i) \qquad (12)$$

Пример №12.461

Вычислить интеграл, используя описанный выше способ:

$$\int_{-\infty}^{\infty} \frac{x^4+1}{x^6+1}\, dx$$

Зададим функцию

```
in : f[x_] := (x^4 + 1)/(x^6 + 1)
```

Вычислим интеграл с помощью функции Integrate и измерим время.

```
in : Integrate[f[x], {x, -Infinity, Infinity}] // Timing

out : {0.2028013, 4 π/3}
```

Сравним время, затрачиваемое на вычисление данного интеграла “в лоб”, с временем, требуемым для решения данной задачи способом описанным выше.

Для этого найдем все полюса, лежащие в верхней полуплоскости (Im[z] > 0):

```
in : sol = z /. Solve[Im[z] > 0 && f[z]^-1 == 0, z]

out : {i, (-1)^(1/6), (-1)^(5/6)}
```

Изобразим их на комплексной плоскости:

```
in : Graphics[{PointSize[Large], Point[ReIm[sol]]}, Axes → True,
      PlotRange → {{-1, 1}, {0, 2}}, AxesLabel → {Re, Im}]
```



рис. 6.3.2

Теперь найдем значение интеграла согласно (12):

```
in : int = 2 ⅈ π Sum[Residue[f[z], {z, sol[[i]]}], {i, 1, Length[sol]}] //
        ComplexExpand // Timing
```

$$out:\ \left\{0.,\ \frac{4\pi}{3}\right\}$$

Получили тот же ответ, но время, затраченное на его вычисление, пренебрежимо мало.

Итак, имеем:

$$\int_{-\infty}^{\infty} f(x)\, dx = \frac{4\pi}{3}$$

Способы, описанные выше, можно применить ко всем задачам №№12.450-470, что и рекомендуется сделать самостоятельно.

6.4. Разложение в ряд Лорана по методу вычетов

Вычет функции в точке равен коэффициенту c_{-1} разложения в ряд Лорана, значит, если идти в обратном направлении, можно найти этот коэффициент. Воспользуемся этим для того, чтобы написать модуль разложения функции в ряд Лорана. Рассмотрим пример.

Пример №12.357

Найти все разложения указанной функции в ряд Лорана по степеням $z - z_0$ и установить области сходимости:

$$f(z) = \frac{z+1}{z^2 - 3z + 2},\ z_0 = 1$$

Запишем функцию:

```
in : f[z_] := (z + 1)/(z^2 - 3 z + 2)
```

Получим множество особых точек:

```
in : sol = z /. Solve[z^2 - 3 z + 2 == 0, z]

out : {1, 2}
```

Точка z_0 является полюсом первого порядка. Для того чтобы вычислить коэффициент Лорана, нам нужно вычислить интеграл:

$$c_n = \frac{1}{2\pi i}\oint_{|z-1|=r'}\frac{f(z)}{(z-1)^{n+1}}\,dz, \quad \text{где } 0 < r' < 1$$

Вычет по определению равен:

$$\operatorname{res}[f(z),\,1] = c_{-1} = \frac{1}{2\pi i}\oint_{|z-z_0|=r'} f(z)\,dz$$

Отсюда следует, что если мы будем применять функцию Residue к функции $f(z)/(z-z_0)^{n+1}$, то будем получать n-ый коэффициент Лорана:

$$\operatorname{res}\left[\frac{f(z)}{(z-1)^{n+1}},\,1\right] = c_n = \frac{1}{2\pi i}\oint_{|z-z_0|=r'}\frac{f(z)}{(z-1)^{n+1}}\,dz$$

Итак, запишем функцию n-ого коэффициента при разложении вблизи точки z_0:

```
in : c[n_] := Residue[f[z] / (z - 1)^(n+1), {z, 1}]
```

Получим разложение в ряд:

```
in : Sum[c[n] (z - 1)^n, {n, -5, 5}]
```

$$out:\ -3 - \frac{2}{-1+z} - 3\,(-1+z) - 3\,(-1+z)^2 - 3\,(-1+z)^3 - 3\,(-1+z)^4 - 3\,(-1+z)^5$$

Сравним ответ с результатом работы функции loranO:

```
in : loranO[z, f[z], 1, 1 / 2, 5]
```

$$out:\ -3 - \frac{2}{-1+z} - 3\,(-1+z) - 3\,(-1+z)^2 - 3\,(-1+z)^3 - 3\,(-1+z)^4 - 3\,(-1+z)^5$$

и с результатом Series:

```
in : Series[f[z], {z, 1, 5}]
```

$$out: -\frac{2}{z-1} - 3 - 3\,(z-1) - 3\,(z-1)^2 - 3\,(z-1)^3 - 3\,(z-1)^4 - 3\,(z-1)^5 + O[z-1]^6$$

Во всех случаях получили одинаковый ответ.

Теперь получим коэффициент разложения в кольце с радиусом $r > 1$:

$$c_n = \frac{1}{2\,\pi\,i} \oint_{|z-1|=R} \frac{f(z)}{(z-1)^{n+1}}\,dz, \quad \text{где } R > 1$$

Согласно первой теореме вычетов этот интеграл будет равен:

$$\frac{1}{2\,\pi\,i} \oint_{|z-1|=R} \frac{f(z)}{(z-1)^{n+1}}\,dz = \sum_{k=1}^{N} \mathrm{res}\left[\frac{f(z)}{(z-1)^{n+1}};\, z_k\right]$$

где z_k- полюса, охватываемые контуром интегрирования.

Отсюда, n-ый коэффициент будем искать по формуле:

$$c_n = \mathrm{res}\left[\frac{f(z)}{(z-1)^{n+1}},\, 1\right] + \sum_{k=1}^{N} \mathrm{res}\left[\frac{f(z)}{(z-1)^{n+1}};\, z_k\right], \quad \text{где } R > 1$$

где z_k- полюса, охватываемые контуром интегрирования R, за исключением полюса z_0.

Итак, первое слагаемое у нас уже вычислено, осталось найти второе:

```
in : c1[n_] := Residue[f[z]/(z - 1)^(n+1), {z, 2}]
```

Теперь чтобы получить разложение в ряд Лорана в кольце $R > 1$:

```
in : Sum[(c[n] + c1[n]) (z - 1)^n, {n, -5, 5}]
```

$$out: \frac{3}{(-1+z)^5} + \frac{3}{(-1+z)^4} + \frac{3}{(-1+z)^3} + \frac{3}{(-1+z)^2} + \frac{1}{-1+z}$$

Сравним ответы:

```
in : loran0[z, f[z], 1, 3, 5]
```

$$out: \frac{3}{(-1+z)^5} + \frac{3}{(-1+z)^4} + \frac{3}{(-1+z)^3} + \frac{3}{(-1+z)^2} + \frac{1}{-1+z}$$

Так как функции Series и SeriesCoefficient работают быстрее всего, то рациональнее по возможности использовать их. Коэффициент c_n вблизи точки z_0 оптимальнее искать именно с помощью функции SeriesCoefficient.

Соберем всё это в единый модуль:

in :
```
loranR[expr_, {z_, z0_, k_}, r_] := Module[{sol, c, c1},
  sol =
   z /.
    Solve[0 < Abs[z - z0] < r &&
      Denominator[Simplify[TrigToExp[expr]]] == 0, z];
  c[n_] := Evaluate[SeriesCoefficient[expr, {z, z0, n}]];
  If[Length[sol] ≠ 0,
   sol = DeleteDuplicates[sol];
   c1[n_] := Sum[Residue[expr/(z - z0)^(n+1), {z, sol[[i]]}],
     {i, 1, Length[sol]}], c1[n_] := 0;];
  Sum[(c1[n] + c[n]) (z - z0)^n, {n, -k, k}]]
```

Пример №12.364

in : $\text{loranR}\left[\frac{z}{\left(z^2+1\right)^2}, \{z, I, 5\}, 1/2\right]$

out : $-\frac{i}{16}-\frac{i}{4\,(-i+z)^2}+\frac{1}{16}(-i+z)+\frac{3}{64}\,i\,(-i+z)^2-\frac{1}{32}(-i+z)^3-\frac{5}{256}\,i\,(-i+z)^4+\frac{3}{256}(-i+z)^5$

in : $\text{loranR}\left[\frac{z}{\left(z^2+1\right)^2}, \{z, I, 8\}, 3\right]$

out : $-\frac{112\,i}{(-i+z)^8}+\frac{48}{(-i+z)^7}+\frac{20\,i}{(-i+z)^6}-\frac{8}{(-i+z)^5}-\frac{3\,i}{(-i+z)^4}+\frac{1}{(-i+z)^3}$

Пример №12.375

in : $\text{loranR}\left[\frac{1}{\left(z^2-4\right)\left(z^2-1\right)}, \{z, 0, 8\}, 1/2\right]$

out : $\frac{1}{4}+\frac{5\,z^2}{16}+\frac{21\,z^4}{64}+\frac{85\,z^6}{256}+\frac{341\,z^8}{1024}$

in : $\text{loranR}\left[\frac{1}{(z^2-4)(z^2-1)}, \{z, I, 5\}, 3/2\right]$

out : $-\frac{1}{15}+\frac{2}{3(-i+z)^4}+\frac{2i}{3(-i+z)^3}-\frac{1}{3(-i+z)^2}-\frac{2}{75}i(-i+z)-\frac{1}{375}(-i+z)^2-\frac{4}{625}i(-i+z)^3+\frac{19(-i+z)^4}{9375}-\frac{22i(-i+z)^5}{46875}$

in : $\text{loranR}\left[\frac{1}{(z^2-4)(z^2-1)}, \{z, I, 8\}, 5/2\right]$

out : $-\frac{49}{(-i+z)^8}-\frac{10i}{(-i+z)^7}-\frac{5}{(-i+z)^6}-\frac{4i}{(-i+z)^5}+\frac{1}{(-i+z)^4}$

Сравним скорость работы этой функции и функции loranO при больших степенях разложения:

in : $\text{Timing}\left[\text{loranR}\left[\frac{z}{(z^2+1)^2}, \{z, I, 500\}, 5/2\right];\right]$

out : {0.5616036, Null}

in : $\text{Timing}\left[\text{loranO}\left[z, \frac{z}{(z^2+1)^2}, I, 5/2, 500\right];\right]$

out : {7.176046, Null}

Так как функция loranR работает быстрее, предпочтение в использовании будем отдавать именно ей.

§7. Ряды Фурье. Интеграл Фурье.

7.1. Разложение функций в тригонометрические ряды Фурье.

Тригонометрическая система функций

$$1, \cos x, \ \sin x, \ \cos 2x, \ \sin 2x, \ \ldots, \ \cos nx, \sin nx, \ \ldots$$

является ортогональной на отрезке $[-\pi, \pi]$ (как, впрочем, и на всяком отрезке длины 2π), т.е. интеграл по этому отрезку от произведения любых двух различных функций этой системы равен нулю.

Если $f(x) \in L(-\pi, \pi)$ (т.е. $\int_{-\pi}^{\pi} |f(x)|\,dx < +\infty$), то существуют числа

$$a_n = \frac{1}{\pi}\int_{-\pi}^{\pi} f(x)\cos(nx)\,dx \quad n = 0, 1, 2, \ \ldots$$

$$b_n = \frac{1}{\pi}\int_{-\pi}^{\pi} f(x)\sin(nx)\,dx, \ \ n = 1, 2, \ \ldots$$

называемые коэффициентами Фурье функции $f(x)$.

Ряд

$$S(x)=\frac{a_0}{2}+\sum_{n=1}^{\infty}(a_n\cos(n\,x)+b_n\sin(n\,x)) \tag{1}$$

называется рядом Фурье функции $f(x)$.

Члены ряда (1) можно записать в виде гармоник

$$a_n\cos(n\,x)+b_n\sin(n\,x)=A_n\cos(n\,x-\phi_n)$$

с амплитудой $A_n=\sqrt{{a_n}^2+{b_n}^2}$, частотой $\omega_n=n$ и фазой $\phi_n=\operatorname{arctg}\frac{b_n}{a_n}$.

Для функции $f\,(x)$ такой, что $f^2(x)\in L(-\pi,\,\pi)$, справедливо равенство Парсеваля

$$\frac{1}{\pi}\int_{-\pi}^{\pi}f^2(x)\,dx=\frac{{a_0}^2}{2}+\sum_{n=1}^{\infty}\left({a_n}^2+{b_n}^2\right)$$

Если же $f(x)\ \in L(-l,\,l)$, то коэффициенты Фурье записываются в виде:

$$a_0=\frac{1}{l}\int_{-l}^{l}f(x)\,dx \tag{2}$$

$$a_n=\frac{1}{l}\int_{-l}^{l}f(x)\cos\left(\frac{\pi}{l}\,n\,x\right)dx \tag{3}$$

$$b_n=\frac{1}{l}\int_{-l}^{l}f(x)\sin\left(\frac{\pi}{l}\,n\,x\right)dx \tag{4}$$

а ряд Фурье - в виде

$$S(x)=\frac{a_0}{2}+\sum_{n=1}^{\infty}\left(a_n\cos\left(\frac{\pi}{l}\,n\,x\right)+b_n\sin\left(\frac{\pi}{l}\,n\,x\right)\right) \tag{5}$$

Суммы рядов (1) и (5) имеют периоды соответственно $2\,\pi$ и $2\,l$.

Функция $f\,(x)$ называется кусочно гладкой на отрезке $[a;\,b]$, если сама функция $f\,(x)$ и ее производная $f'\,(x)$ имеют на $[a;\,b]$ конечное число точек разрыва 1-ого рода.

Т е о р е м а. Если периодическая функция $f\,(x)$ с периодом $2\,l$ кусочно гладка на отрезке $[-l;\,l]$, то ряд Фурье (5) сходится к значению $f\,(x)$ в каждой ее точке непрерывности и к значению $(f(x+0)+f(x-0))/2$ в точке разрыва, т.е.

$$\frac{1}{2}\,(f(x+0)+f(x-0))=\frac{a_0}{2}+\sum_{n=1}^{\infty}\left(a_n\cos\left(\frac{\pi}{l}\,n\,x\right)+b_n\sin\left(\frac{\pi}{l}\,n\,x\right)\right) \tag{6}$$

Если, дополнительно, $f\,(x)$ непрерывна на всей оси, то ряд (5) сходится к $f\,(x)$ равномерно.

Если на полуоткрытом интервале длины $2\,l$, т.е. на интервале вида $[a,\ a+2\,l)$ или $(a,\ a+2\,l]$, определена какая-нибудь функция, то она может быть (единственным способом) продолжена на всю числовую прямую так, что получится функция с периодом $2\,l$.

Отсюда следует, что если $f(x)$ имеет на $[-l, l]$ не более конечного числа точек разрыва и абсолютно интегрируема на этом сегменте, то она представима рядом Фурье (5).

Пример №12.480

Разложить периодическую функцию с периодом $2\,l$ функцию в ряд Фурье, построить график.

$$f(x) = \begin{cases} 1 & \text{при } 0 < x < \pi \\ 0 & \text{при } -\pi < x < 0 \end{cases}$$

Запишем исходную кусочно-заданную функцию и построим ее график:

```
in : f[x_] := Piecewise[{{1, 0 < x < Pi}, {0, -Pi < x < 0}}]
     Plot[f[x], {x, -Pi, Pi}, PlotRange → {-0.5, 1.5}]
```



рис. 7.1.1

Запишем выражения для коэффициентов Фурье по формулам (2), (3), (4)

Так как функция обращается в ноль при $x < 0$, интегрирование ведем в пределах от 0 до π.

Коэффициент a_0 вычисляется следующим образом:

```
in : a0 = 1/π Integrate[f[x], {x, 0, Pi}]

out : 1
```

Коэффициент a_n:

```
in : an = 1/π Integrate[f[x] Cos[n x], {x, 0, Pi}]

out : Sin[n π]/(n π)
```

При любых n $a_n = 0$, но $a_0 = 1$

Коэффициент b_n:

```
in : bn = 1/π Integrate[f[x] Sin[n x], {x, 0, Pi}]
```

$$out: \frac{1 - \text{Cos}[n\,\pi]}{n\,\pi}$$

Получили, что коэффициент b_n:

$$b_n = \begin{cases} \frac{2}{(2k+1)\pi} & \text{при } n = 2k+1 \\ 0 & \text{при } n = 2k \end{cases}$$

Запишем ряд Фурье:

$$f(x) = \frac{1}{2} + \sum_{k=0}^{\infty} \frac{2\sin((2k+1)x)}{(2k+1)\pi}$$

```
in : fu[x_, n_] := 1/2 + 2/π Sum[Sin[(2 k + 1) x]/(2 k + 1), {k, 0, n}]
```

Получим выражение ряда Фурье до 4-ого члена:

```
in : fu[x, 4] // Expand
```

$$out: \frac{1}{2} + \frac{2\,\text{Sin}[x]}{\pi} + \frac{2\,\text{Sin}[3\,x]}{3\,\pi} + \frac{2\,\text{Sin}[5\,x]}{5\,\pi} + \frac{2\,\text{Sin}[7\,x]}{7\,\pi} + \frac{2\,\text{Sin}[9\,x]}{9\,\pi}$$

Изобразим графики исходной функции и ряда Фурье до 4ого и до 10ого члена.

```
in : Row[{Plot[{fu[x, 4], f[x]}, {x, -Pi, Pi}, ImageSize → 300],
       Plot[{fu[x, 10], f[x]}, {x, -Pi, Pi}, ImageSize → 300] }]
```



рис. 7.1.2

Из графиков видно, что чем больше порядок разложения, тем точнее ряд аппроксимирует функцию.

В данной задаче было показано, как на основе формул для определения коэффициентов Фурье разложить функцию в ряд Фурье.

Пример №12.487

Разложить периодическую функцию с периодом $2\,l$ функцию в ряд Фурье, построить график.

$$f(x) = 2\,x\,, \qquad 0 < x < 1,\ \ 2\,l = 1$$

Определим заданную функцию, и построим ее график:

```
in : f[x_] := 2 x
     l = 1 / 2;
     Plot[f[x], {x, 0, 1}]
```



рис. 7.1.3

Теперь запишем функции для коэффициентов Фурье по формулам (2), (3), (4):

Так как в формуле (3) при $n = 0$ косинус обращается в единицу, то записывать отдельное выражение для коэффициента a_0 смысла нет.

```
in : a[n_] := Integrate[1/l f[x] Cos[Pi n x/l], {x, 0, 1}]
     b[n_] := Integrate[1/l f[x] Sin[Pi n x/l], {x, 0, 1}]
```

Теперь запишем функцию тригонометрического ряда Фурье:

```
in : fu[x_, k_] :=
       a[0]/2 + Sum[a[n] Cos[Pi/l n x] + b[n] Sin[Pi/l n x], {n, 1, k}]
```

Выведем выражение для ряда до 5ого члена и построим его график:

```
in : fu[x, 5]
     Plot[{Evaluate@fu[x, 5], f[x]}, {x, 0, 1}]
```

out : $1 - \frac{2\,\mathrm{Sin}[2\pi x]}{\pi} - \frac{\mathrm{Sin}[4\pi x]}{\pi} - \frac{2\,\mathrm{Sin}[6\pi x]}{3\pi} - \frac{\mathrm{Sin}[8\pi x]}{2\pi} - \frac{2\,\mathrm{Sin}[10\pi x]}{5\pi}$



рис. 7.1.4

Чтобы оценить верность полученного выражения, найдем сумму ряда:

```
in : fu[x, Infinity]
```

out : $1 + \frac{1}{\pi}\, i \left(-\mathrm{Log}\left[1 - e^{2 i \pi x}\right] + \mathrm{Log}\left[e^{-2 i \pi x}\left(-1 + e^{2 i \pi x}\right)\right]\right)$

Построим график полученной функции:

```
in : Plot[{f[x], Evaluate@fu[x, Infinity]}, {x, 0, 1}]
```



рис. 7.1.5

Графики функций совпали, следовательно, как и предполагалось, ряд сходится к функции.

7.2. Особенности разложения в ряд Фурье четных и нечетных функций.

Если функцию обладает какой-либо симметрией, то техника разложения в ряд Фурье упрощается.

В случае, если $f(x)$ - четная функция с периодом $2\,l$, то все коэффициенты Фурье b_n равны 0 и, следовательно, в ряде Фурье нет членов с синусами. Тогда получим ряд:

$$S(x) = \frac{a_0}{2} + \sum_{n=1}^{\infty} a_n \cos\left(\frac{\pi}{l}\, n\, x\right) \tag{7}$$

где

$$a_0 = \frac{2}{l} \int_0^l f(x)\, dx \tag{8}$$

$$a_n = \frac{2}{l} \int_0^l f(x) \cos\left(\frac{\pi}{l}\, n\, x\right) dx \tag{9}$$

Аналогично в случае, если $f(x)$ - нечетная функция, то все коэффициенты Фурье a_n равны 0 и, следовательно, в ряде Фурье нет членов с косинусами. Тогда получим ряд:

$$S(x) = \sum_{n=1}^{\infty} b_n \sin\left(\frac{\pi}{l}\, n\, x\right) \tag{10}$$

где

$$b_n = \frac{2}{l} \int_0^l f(x) \sin\left(\frac{\pi}{l}\, n\, x\right) dx \tag{11}$$

Пример №12.482

Разложить в ряд Фурье функцию:

$$f(x) = |x|, \quad -1 < x < 1, \quad l = 1$$

Запишем заданную функцию:

```
in : f[x_] := Abs[x]
```

Функция четная, значит в разложении будут отсутствовать члены с синусами. Запишем теперь выражение для коэффициента a_n:

```
in : a[n_] := Integrate[2 f[x] Cos[Pi n x], {x, 0, 1}]
```

Зададим функцию ряда:

```
in : fu[x_, k_] := a[0] / 2 + Sum[a[n] Cos[Pi n x], {n, 1, k}]
```

Выведем полученный ряд до 6ого члена и построим его график:

```
in : fu[x, 6]
     Plot[{f[x], Evaluate@fu[x, 6]}, {x, -1, 1}]
```

$$out:\quad \frac{1}{2}-\frac{4\,\mathrm{Cos}[\pi x]}{\pi^2}-\frac{4\,\mathrm{Cos}[3\pi x]}{9\pi^2}-\frac{4\,\mathrm{Cos}[5\pi x]}{25\pi^2}$$

рис. 7.2.1

Из графика заключаем, что ряд сходится к функции.

Пример №12.493

Доопределяя необходимым образом заданную в промежутке функцию до периодической, получить для нее: а) ряд Фурье по косинусам; б) ряд Фурье по синусам.

$$f(x)=e^x,\ \ 0<x<\ln 2$$

Для того чтобы получить ряд Фурье по косинусам, необходимо чтобы функция была чётной. Функция $e^{|x|}$ удовлетворяет данному требованию. Запишем и изобразим эту функцию:

```
in : f[x_] := Exp[Abs[x]]
     Plot[f[x], {x, -Log[2], Log[2]}]
```

рис. 7.2.2

Теперь получим выражение для коэффициента Фурье и запишем ряд:

```
in : a[n_] := Integrate[2/Log[2] f[x] Cos[Pi/Log[2] n x], {x, 0, Log[2]}]
     fu[x_, k_] := a[0] / 2 + Sum[a[n] Cos[Pi/Log[2] n x], {n, 1, k}]
```

Изобразим полученный ряд:

```
in : fu[x, 5]
     Plot[{f[x], Evaluate@fu[x, 5]}, {x, -Log[2], Log[2]}]
```

$$out:\quad \frac{1}{\mathrm{Log}[2]}-\frac{6\,\mathrm{Cos}\left[\frac{\pi x}{\mathrm{Log}[2]}\right]\mathrm{Log}[2]}{\pi^2+\mathrm{Log}[2]^2}-\frac{6\,\mathrm{Cos}\left[\frac{3\pi x}{\mathrm{Log}[2]}\right]\mathrm{Log}[2]}{9\pi^2+\mathrm{Log}[2]^2}-\frac{6\,\mathrm{Cos}\left[\frac{5\pi x}{\mathrm{Log}[2]}\right]\mathrm{Log}[2]}{25\pi^2+\mathrm{Log}[2]^2}+\frac{\mathrm{Cos}\left[\frac{2\pi x}{\mathrm{Log}[2]}\right]\mathrm{Log}[4]}{4\pi^2+\mathrm{Log}[2]^2}+\frac{\mathrm{Cos}\left[\frac{4\pi x}{\mathrm{Log}[2]}\right]\mathrm{Log}[4]}{16\pi^2+\mathrm{Log}[2]^2}$$

рис. 7.2.3

Теперь, для того чтобы получить разложение Фурье по синусам, доопределим исходную функцию до нечетной следующим образом:

$$f(x) = \begin{cases} e^x & 0 < x < \ln 2 \\ -e^{-x} & -\ln 2 < x < 0 \end{cases}$$

Запишем и построим полученную функцию:

```
in : f[x_] :=
      Piecewise[{{Exp[x], Log[2] > x > 0}, {-Exp[-x], -Log[2] < x < 0}}]
     Plot[f[x], {x, -Log[2], Log[2]}]
```

рис. 7.2.4

Запишем выражения для коэффициента b_n и ряда:

```
in : b[n_] := Integrate[2/Log[2] f[x] Sin[Pi/Log[2] n x], {x, 0, Log[2]}]
     fu[x_, k_] := Sum[b[n] Sin[Pi/Log[2] n x], {n, 1, k}]
```

Получим разложение функции до 6ого члена и построим его график:

```
in : fu[x, 6]
     Plot[{f[x], Evaluate@fu[x, 12]}, {x, -Log[2], Log[2]}]
```

out :

$$\frac{6\pi \mathrm{Sin}\left[\frac{\pi x}{\mathrm{Log}[2]}\right]}{\pi^2 + \mathrm{Log}[2]^2} - \frac{4\pi \mathrm{Sin}\left[\frac{2\pi x}{\mathrm{Log}[2]}\right]}{4\pi^2 + \mathrm{Log}[2]^2} + \frac{18\pi \mathrm{Sin}\left[\frac{3\pi x}{\mathrm{Log}[2]}\right]}{9\pi^2 + \mathrm{Log}[2]^2} - \frac{8\pi \mathrm{Sin}\left[\frac{4\pi x}{\mathrm{Log}[2]}\right]}{16\pi^2 + \mathrm{Log}[2]^2} + \frac{30\pi \mathrm{Sin}\left[\frac{5\pi x}{\mathrm{Log}[2]}\right]}{25\pi^2 + \mathrm{Log}[2]^2} - \frac{12\pi \mathrm{Sin}\left[\frac{6\pi x}{\mathrm{Log}[2]}\right]}{36\pi^2 + \mathrm{Log}[2]^2}$$

рис. 7.2.5

В точке разрыва значение ряда $S(0)$ будет равно:

$$S(0) = \frac{f(+0) + f(-0)}{2}$$

Программный модуль

Проанализировав последовательность действий в общем случае, напишем простую обучающую программу, которая будет раскладывать функцию в ряд Фурье на заданном интервале $[h; g]$ с полупериодом $l = \frac{g-h}{2}$:

```
in : fourier[x_, func_, h_, g_, k_] := Module[{a, b, l},
       l = (g - h) / 2;
       a[n_] := Integrate[1/l func Cos[Pi n x/l], {x, h, g}];
       b[n_] := Integrate[1/l func Sin[Pi n x/l], {x, h, g}];
       a[0]/2 + Sum[a[n] Cos[Pi/l n x] + b[n] Sin[Pi/l n x], {n, 1, k}]]
```

Проведем графическую проверку полученного модуля:

Пример №12.483

Разложить в ряд Фурье функцию:

$$f(x) = |\cos(x)|, \quad -\pi < x < \pi, \ l = \pi$$

Зададим функцию ряда через написанный нами модуль:

```
in : fu[x_, k_] := fourier[x, Abs[Cos[x]], -Pi, Pi, k]
```

Теперь выведем разложение до 8-ого члена и построим график:

```
in : fu[x, 8]
     Plot[{Abs[Cos[x]], Evaluate@fu[x, 8]}, {x, -Pi, Pi}]
```

$$out: \ \frac{2}{\pi} + \frac{4\,\mathrm{Cos}[2\,x]}{3\,\pi} - \frac{4\,\mathrm{Cos}[4\,x]}{15\,\pi} + \frac{4\,\mathrm{Cos}[6\,x]}{35\,\pi} - \frac{4\,\mathrm{Cos}[8\,x]}{63\,\pi}$$

рис. 7.2.6

При увеличении порядка разложения ряд будет сходиться к функции.

Пример №12.488

Разложить в ряд Фурье функцию:

$$f(x) = 10 - x, \quad 5 < x < 15, \ l = 5$$

```
in : fu[x_, k_] := fourier[x, 10 - x, 5, 15, k]
```

```
in : fu[x, 8]
     Plot[{10 - x, Evaluate@fu[x, 8]}, {x, 5, 15}]
```

$$out: \ -\frac{10\,\mathrm{Sin}\left[\frac{\pi x}{5}\right]}{\pi} + \frac{5\,\mathrm{Sin}\left[\frac{2\pi x}{5}\right]}{\pi} - \frac{10\,\mathrm{Sin}\left[\frac{3\pi x}{5}\right]}{3\,\pi} + \frac{5\,\mathrm{Sin}\left[\frac{4\pi x}{5}\right]}{2\,\pi} - \frac{2\,\mathrm{Sin}[\pi\,x]}{\pi} + \frac{5\,\mathrm{Sin}\left[\frac{6\pi x}{5}\right]}{3\,\pi} - \frac{10\,\mathrm{Sin}\left[\frac{7\pi x}{5}\right]}{7\,\pi} + \frac{5\,\mathrm{Sin}\left[\frac{8\pi x}{5}\right]}{4\,\pi}$$



Разобрав два примера, будем считать, что написанный модуль проверку прошел.

7.3. Ряд Фурье в комплексной форме.

Применим тождества Эйлера для косинуса и синуса ряда (1):

$$S(t) = \frac{a_0}{2} + \sum_{n=1}^{\infty}(a_n \cos(n\,t) + b_n \sin(n\,t)) =$$

$$\frac{a_0}{2} + \sum_{n=1}^{\infty}\left(a_n\,\frac{e^{i\,n\,t} + e^{-i\,n\,t}}{2} + b_n\,\frac{e^{i\,n\,t} - e^{-i\,n\,t}}{2\,i}\right) =$$

$$= \frac{a_0}{2} + \sum_{n=1}^{\infty}\left(a_n\,\frac{e^{i\,n\,t} + e^{-i\,n\,t}}{2} - i\,b_n\,\frac{e^{i\,n\,t} - e^{-i\,n\,t}}{2}\right) =$$

$$\frac{a_0}{2} + \sum_{n=1}^{\infty}\left(\frac{a_n - i\,b_n}{2}\,e^{i\,n\,t} + \frac{a_n + i\,b_n}{2}\,e^{-i\,n\,t}\right)$$

Введем коэффициенты a_n и b_n с отрицательными номерами:

$$a_{-n} = a_n$$
$$b_{-n} = b_n$$

Для которых также справедливы формулы:

$$a_{-n} = \frac{1}{\pi}\int_{-\pi}^{\pi} f(t)\cos(-n\,t)\,dt = \frac{1}{\pi}\int_{-\pi}^{\pi} f(t)\cos(n\,t)\,dt = a_n$$

$$b_{-n} = \frac{1}{\pi}\int_{-\pi}^{\pi} f(t)\sin(-n\,t)\,dt = -\frac{1}{\pi}\int_{-\pi}^{\pi} f(t)\sin(n\,t)\,dt = -b_n$$

Продолжим преобразование:

$$f(t) = \frac{a_0}{2} + \sum_{n=1}^{\infty}\frac{a_n - i\,b_n}{2}\,e^{i\,n\,t} + \sum_{n=-\infty}^{-1}\frac{a_n - i\,b_n}{2}\,e^{i\,n\,t} = \sum_{n=-\infty}^{\infty}\frac{a_n - i\,b_n}{2}\,e^{i\,n\,t}$$

Обозначим:

$$c_n = \frac{a_n - i\,b_n}{2} =$$

$$\frac{1}{2\,\pi}\int_{-\pi}^{\pi} f(t)\cos(n\,t)\,dt - i\,\frac{1}{2\,\pi}\int_{-\pi}^{\pi} f(t)\sin(n\,t)\,dt = \frac{1}{2\,\pi}\int_{-\pi}^{\pi} f(t)\,e^{-n\,t}\,dt$$

Итак, в комплексной форме ряд Фурье для функции с произвольным полупериодом l записывается:

$$f(t) = \sum_{n=-\infty}^{\infty} c_n\,e^{\frac{i\,n\,\pi\,x}{l}} \tag{12}$$

$$c_n = \frac{1}{2\,l}\int_{-l}^{l} f(t)\,e^{-\frac{i\,n\,\pi\,t}{l}}\,dt \tag{13}$$

Пример №12.484

Представить функцию в виде ряда Фурье в комплексной форме:

$$f(t) = t^2,\ -\pi < t < \pi,\ l = \pi$$

Запишем функцию:

```
in : f[t_] := t^2
```

Используя формулу (13), запишем коэффициент c_n:

```
in : c[n_] := Integrate[1/(2 Pi) f[t] Exp[-I n t], {t, -Pi, Pi}]
```

Запишем функцию ряда Фурье до k-ого порядка:

```
in : fu[t_, k_] := Sum[c[n] Exp[I n t], {n, -k, k}]
```

Получим выражение ряда до 4-ого порядка:

```
in : fu[t, 4]
```

$$out: -2\,e^{-i\,t} - 2\,e^{i\,t} + \frac{1}{2}\,e^{-2\,i\,t} + \frac{1}{2}\,e^{2\,i\,t} - \frac{2}{9}\,e^{-3\,i\,t} - \frac{2}{9}\,e^{3\,i\,t} + \frac{1}{8}\,e^{-4\,i\,t} + \frac{1}{8}\,e^{4\,i\,t} + \frac{\pi^2}{3}$$

Сравним данный результат, преобразовав его при помощи функции ExpToTrig, с тем, что бы у нас получилось, если бы мы воспользовались модулем fourier:

```
in : fourier[t, f[t], -Pi, Pi, 4]
     fu[t, 4] // ExpToTrig
```

$$out: \frac{\pi^2}{3} - 4\,\mathrm{Cos}[t] + \mathrm{Cos}[2\,t] - \frac{4}{9}\,\mathrm{Cos}[3\,t] + \frac{1}{4}\,\mathrm{Cos}[4\,t]$$

$$out: \frac{\pi^2}{3} - 4\,\mathrm{Cos}[t] + \mathrm{Cos}[2\,t] - \frac{4}{9}\,\mathrm{Cos}[3\,t] + \frac{1}{4}\,\mathrm{Cos}[4\,t]$$

Как и ожидалось, ответы совпадают.

Построим график, чтобы убедиться, что ряд сходится к исходной функции:

```
in : Plot[{f[t], Evaluate@fu[t, 4]}, {t, -Pi, Pi}]
```



рис. 7.3.1

Программный модуль

Подобно модулю разложения в тригонометрический ряд составим модуль разложения в ряд Фурье в комплексной форме:

```
in : fourierC[t_, func_, h_, g_, k_] := Module[{c, l},
       l = (g - h)/2;
       c[n_] := 1/(2 l) Integrate[func Exp[- I Pi/l n t], {t, h, g}];
       Sum[c[n] Exp[I Pi/l n t], {n, -k, k}]]
```

Пример №12.486

Представить функцию в виде ряда Фурье в комплексной форме:

$$f(t) = |\sin(t)|, \ -\pi < t < \pi, \ l = \pi$$

Выведем результат работы функции fourierC и построим графики исходной функции и полученного ряда:

in :
```
fourierC[t, Abs[Sin[t]], -Pi, Pi, 4]
Plot[{Abs[Sin[t]], Evaluate@fourierC[t, Abs[Sin[t]], -Pi, Pi, 4]},
 {t, -Pi, Pi}]
```

out : $\frac{2}{\pi} - \frac{2\,e^{-2\,i\,t}}{3\,\pi} - \frac{2\,e^{2\,i\,t}}{3\,\pi} - \frac{2\,e^{-4\,i\,t}}{15\,\pi} - \frac{2\,e^{4\,i\,t}}{15\,\pi}$

рис. 7.3.2

Будем считать, что проверка прошла успешно.

7.4. Системные функции разложения в ряд Фурье.

Mathematica имеет встроенные функции разложения в ряд Фурье. Рассмотрим на примере:

Пример №12.484

Разложить функцию в ряд Фурье периода l:

$$f(x) = x^2,\ \ -\pi < x < \pi,\ \ l = \pi$$

Для решения данной задачи воспользуемся встроенной функцией FourierSeries, имеющей следующую структуру: FourierSeries[expr, t, n] , где expr - выражение или функция, которую требуется разложить в ряд Фурье, t - переменная, по которой ведется разложение, n - порядок разложения.

Запишем исходную функцию:

in :
```
f[x_] := x^2
```

Теперь получим разложение в ряд до 6ого члена.

in :
```
FourierSeries[x^2, x, 6]
```

out : $-2\,e^{-i\,x} - 2\,e^{i\,x} + \frac{1}{2}\,e^{-2\,i\,x} + \frac{1}{2}\,e^{2\,i\,x} - \frac{2}{9}\,e^{-3\,i\,x} - \frac{2}{9}\,e^{3\,i\,x} + \frac{1}{8}\,e^{-4\,i\,x} + \frac{1}{8}\,e^{4\,i\,x} - \frac{2}{25}\,e^{-5\,i\,x} - \frac{2}{25}\,e^{5\,i\,x} + \frac{1}{18}\,e^{-6\,i\,x} + \frac{1}{18}\,e^{6\,i\,x} + \frac{\pi^2}{3}$

Как мы видим, по умолчанию Mathematica раскладывает в ряд в комплексной форме. Также мы можем перейти к более привычной для нас, тригонометрической, форме:

in :
```
FourierSeries[x^2, x, 6] // ExpToTrig
```

out : $\frac{\pi^2}{3} - 4\,\mathrm{Cos}[x] + \mathrm{Cos}[2\,x] - \frac{4}{9}\,\mathrm{Cos}[3\,x] + \frac{1}{4}\,\mathrm{Cos}[4\,x] - \frac{4}{25}\,\mathrm{Cos}[5\,x] + \frac{1}{9}\,\mathrm{Cos}[6\,x]$

или же можем просто воспользоваться функцией FourierTrigSeries:

```
in : FourierTrigSeries[x^2, x, 6]
```

$$out: \frac{\pi^2}{3} + 4\left(-\mathrm{Cos}[x] + \frac{1}{4}\mathrm{Cos}[2\,x] - \frac{1}{9}\mathrm{Cos}[3\,x] + \frac{1}{16}\mathrm{Cos}[4\,x] - \frac{1}{25}\mathrm{Cos}[5\,x] + \frac{1}{36}\mathrm{Cos}[6\,x]\right)$$

Как мы видим, функция - четная, а функции синуса в разложении отсутствуют.

Найдем формулу n-ого коэффициента при помощи функции FourierCoefficient:

```
in : FourierCoefficient[x^2, x, n]
```

$$out: \frac{2\,(-1)^n}{n^2}$$

Напоминаем, что это коэффициент разложения в комплексной форме c_n, и он не равен коэффициентам a_n или b_n.

Так же находим и свободный коэффициент:

```
in : FourierCoefficient[x^2, x, 0]
```

$$out: \frac{\pi^2}{3}$$

Составим функцию ряда:

```
in : fu[x_, n_] := FourierSeries[t^2, t, n] /. t → x
```

```
in : fu[x, 6] // ExpToTrig
```

$$out: \frac{\pi^2}{3} - 4\,\mathrm{Cos}[x] + \mathrm{Cos}[2\,x] - \frac{4}{9}\mathrm{Cos}[3\,x] + \frac{1}{4}\mathrm{Cos}[4\,x] - \frac{4}{25}\mathrm{Cos}[5\,x] + \frac{1}{9}\mathrm{Cos}[6\,x]$$

Покажем графики исходной функции и ряда Фурье до 5ого порядка:

```
in : Plot[{f[x], Evaluate@fu[x, 5]}, {x, -Pi, Pi}]
```



Пример №12.496

Доопределяя необходимым образом заданную в промежутке функцию до переодической, получить для нее: а) ряд Фурье по косинусам б) ряд Фурье по синусам.

$$f(x) = x\sin(x),\ \ 0 < x < \pi$$

Так как данная функция - четная, получим выражение ряда в общей форме.

Для этого запишем и построим график заданной функции:

```
in : f[x_] := x Sin[x]
     Plot[f[x], {x, -Pi, Pi}]
```



рис. 7.4.2

Так как данная функция - четная, в разложении будут отсутствовать члены с синусом. Тогда:

$$c_n = \frac{a_n - i\,b_n}{2} = \frac{a_n - i\,0}{2} \Rightarrow a_n = 2\,c_n$$

Получим выражение для нулевого и n-ого коэффициента:

```
in : FourierCoefficient[f[x], x, n]
```

out : $\frac{(-1)^{1+n}}{-1+n^2}$

Отсюда следует, что

$$a_n = \frac{2\,(-1)^{n+1}}{n^2 - 1}$$

Найдем значение коэффициента при $n = 1$:

```
in : FourierCoefficient[f[x], x, 1]
```

out : $-\frac{1}{4}$

Теперь можем записать ряд в виде:

$$f(x) = 1 - \frac{1}{2}\cos(x) + \sum_{n=2}^{\infty} \frac{2(-1)^{n+1}}{n^2 - 1}\cos(n\,x)$$

Ряд Фурье по синусам для исходной функции получим при помощи встроенной опции FourierSinSeries (существует аналогичная опция FourierCosSeries), она имеет ту же структуру, что и FourierSeries:

```
in : FourierSinSeries[f[x], x, 8]
```

out : $\frac{1}{2}\pi\,\mathrm{Sin}[x] - \frac{16\,\mathrm{Sin}[2\,x]}{9\,\pi} - \frac{32\,\mathrm{Sin}[4\,x]}{225\,\pi} - \frac{48\,\mathrm{Sin}[6\,x]}{1225\,\pi} - \frac{64\,\mathrm{Sin}[8\,x]}{3969\,\pi}$

Изобразим график функции и ряда:

```
in : Plot[{f[x], Evaluate@FourierSinSeries[f[x], x, 8]},
      {x, -Pi, Pi}]
```



рис. 7.4.3

Как видно из графика, исходная функция была доопределена до несимметричной функции, у которой отсутствует косинус в разложении в ряд.

Коэффициент b_n получим при помощи функции FourierSinCoefficient:

```
in : FourierSinCoefficient[f[x], x, n]
```

out : $-\frac{4\left(1 + (-1)^{n}\right) n}{\left(-1 + n^2\right)^2 \pi}$

а также коэффициент b_1:

```
in : FourierSinCoefficient[f[x], x, 1]
```

out : $\frac{\pi}{2}$

В итоге, ряд записывается в виде:

$$f(x) = \frac{\pi}{2}\sin(x) - \frac{4}{\pi}\sum_{n=2}^{\infty}\frac{(1+(-1)^n)\,n}{\left(-1+n^2\right)^2}\sin(n\,x)$$

7.5. Интеграл Фурье. Преобразование Фурье.

Пусть функция $f\,(x)$:

1) Абсолютно интегрируема на $\mathbb{R}$, т.е. сходится интеграл: $\int_{-\infty}^{+\infty}|\,f(x)\,|\,dx$;

2) Кусочно гладкая на каждом конечном отрезке действительной оси.

Тогда справедлива интегральная формула Фурье:

$$\frac{f(x+0)-f(x-0)}{2} = \frac{1}{\pi}\int_0^{\infty}d\omega\int_{-\infty}^{\infty}f(t)\cos\omega(x-t)\,dt \tag{14}$$

Чтобы показать аналогию интеграла Фурье и ряда Фурье, проведем некоторые преобразования:

$$f(x) = \frac{1}{\pi}\int_0^{\infty}d\omega\int_{-\infty}^{\infty}f(t)\,(\cos\omega\,t\cos\omega\,x\ +\ \sin\omega\,t\sin\omega\,x)\,dt =$$

$$= \frac{1}{\pi}\int_0^{\infty}\left(\int_{-\infty}^{\infty}f(t)\cos\omega\,t\,dt\right)\cos\omega\,x\ +\left(\int_{-\infty}^{\infty}f(t)\sin\omega\,t\,dt\right)\sin\omega\,x\,d\omega$$

Введя обозначения:

$$A(\omega) = \int_{-\infty}^{\infty}f(t)\cos\omega\,t\,dt\,,\ \ B(\omega) = \int_{-\infty}^{\infty}f(t)\sin\omega\,t\,dt \tag{15}$$

Получаем:

$$f(x) = \frac{1}{\pi}\int_0^{\infty}A(\omega)\cos\omega\,x\ + B(\omega)\sin\omega\,x\,d\omega \tag{16}$$

В этом случае, в отличии от тригонометрического ряда Фурье, частота ω изменяется не дискретно, а непрерывно. И наиболее важным отличием является то, что в виде интеграла Фурье можно представить и непериодическую функцию.

Как и ряд Фурье, интеграл Фурье имеет комплексную форму записи.

Так как внутренний интеграл в выражении (14) функция четная, тогда:

$$\frac{1}{\pi}\int_0^{\infty}d\omega\int_{-\infty}^{\infty}f(t)\cos\omega(x-t)\,dt = \frac{1}{2\pi}\int_{-\infty}^{\infty}d\omega\int_{-\infty}^{\infty}f(t)\cos\omega(x-t)\,dt$$

Так как:

$$\int_{-\infty}^{\infty} f(t) \sin \omega(x-t)\, dt - \text{ нечетная функция от } \omega,$$

$$\text{то } \frac{i}{2\pi} \int_{-\infty}^{\infty} d\omega \int_{-\infty}^{\infty} f(t) \sin \omega(x-t)\, dt = 0$$

Тогда:

$$f(x) = \frac{1}{2\pi} \int_{-\infty}^{\infty} d\omega \int_{-\infty}^{\infty} f(t)\,(\cos \omega(x-t) + i \sin \omega(x-t))\, dt =$$

$$\frac{1}{2\pi} \int_{-\infty}^{\infty} d\omega \int_{-\infty}^{\infty} f(t)\, e^{i\omega(x-t)}\, dt =$$

$$= \frac{1}{\sqrt{2\pi}} \int_{-\infty}^{\infty} e^{i\omega x} \left(\frac{1}{\sqrt{2\pi}} \int_{-\infty}^{\infty} f(t)\, e^{-i\omega t}\, dt \right) d\omega$$

Функция:

$$S(\omega) = \frac{1}{\sqrt{2\pi}} \int_{-\infty}^{\infty} f(t)\, e^{-i\omega t}\, dt \tag{17}$$

Называется прямым преобразованием Фурье функции $f(x)$. Преобразованием Фурье также называют отображение $f(x) \longrightarrow S(\omega)$, которое задается формулой (17). Обратное отображение называется обратным преобразованием Фурье, и задается формулой:

$$f(x) = \frac{1}{\sqrt{2\pi}} \int_{-\infty}^{\infty} e^{i\omega x} S(\omega)\, d\omega \tag{18}$$

Комплекснозначная функция $S(\omega)$ называется также спектральной плотностью функции $f(x)$ и несет в себе значительную информацию о функции $f(x)$.

Если функция $f(x)$ четная, интегральную формулу (18) можно записать так:

$$f(x) = \int_{0}^{\infty} A(\omega) \cos \omega x\, d\omega = \frac{2}{\pi} \int_{0}^{\infty} \left(\int_{0}^{\infty} f(t) \cos \omega t\, dt \right) \cos \omega x\, d\omega$$

Обозначим:

$$f^{*}(\omega) = \sqrt{\frac{2}{\pi}} \int_{0}^{\infty} f(t) \cos \omega t\, dt \tag{19}$$

Тогда получим:

$$f(x) = \sqrt{\frac{2}{\pi}} \int_{0}^{\infty} f^{*}(\omega) \cos \omega x\, d\omega \tag{20}$$

Функция $f^{*}(\omega)$ называется косинус-преобразованием Фурье функции $f(x)$.

Аналогично, если $f(x)$ - нечетная, то можно рассмотреть синус-преобразование:

$$f_*(\omega) = \sqrt{\frac{2}{\pi}} \int_0^\infty f(t) \sin \omega\, t\, dt \tag{21}$$

$$f(x) = \sqrt{\frac{2}{\pi}} \int_0^\infty f_*(\omega) \sin \omega\, t\, dt \tag{22}$$

Пример №1

Найти преобразование Фурье функции и представить ее интегралом Фурье в комплексной форме:

$$f(x) = x\, e^{-|x|}$$

Запишем исходную функцию:

```
in : f[x_] := x Exp[- Abs[x]]
```

Найдем преобразование Фурье по формуле (17):

```
in : Integrate[1/Sqrt[2 π] t Exp[-Abs[t]] Exp[-I w t], {t, -Infinity, Infinity},
     Assumptions → 1 > Im[w] > -1]
```

$$out:\ -\frac{2\, i \sqrt{\frac{2}{\pi}}\, w}{(-i + w)^2\, (i + w)^2}$$

Будем обозначать преобразование Фурье по формуле (17) $\mathcal{F}[f]$.

Получили, что спектральная функция имеет вид:

$$\mathcal{F}[f] = S(\omega) = -\sqrt{\frac{1}{2\pi}}\, \frac{4\, i\, \omega}{(\omega^2+1)^2}$$

Значит функцию $f(x)$ можно представить в виде:

$$f(x) = x\, e^{-|x|} =$$

$$\frac{1}{\sqrt{2\pi}} \int_{-\infty}^{\infty} -\sqrt{\frac{1}{2\pi}}\, \frac{4\, i\, \omega}{(\omega^2+1)^2}\, e^{i\omega x}\, d\omega = -\frac{2\, i}{\pi} \int_{-\infty}^{\infty} \frac{\omega}{(\omega^2+1)^2}\, e^{i\omega x}\, d\omega$$

Пример №12.515

Найти преобразование Фурье функции и представить ее интегралом Фурье в комплексной форме:

$$f(t) = \begin{cases} \cos a\, t & |t| < \pi/a \\ 0 & \text{else} \end{cases}, \quad a > 0$$

Запишем функцию:

```
in : f[t_] := Piecewise[{{Cos[a t], Abs[t] < Pi / a}}, 0]
```

Построим ее график при $a = 1$:

```
in : Plot[f[t] /. a → 1, {t, -2 Pi, 2 Pi}]
```



рис. 7.5.1

В Wolfram Mathematica существует целый ряд функций для работы с преобразованием Фурье. Рассмотрим две основные функции: FourierTransform и InverseFourierTransform. Внутри системы они определены следующим образом:

$$\text{FourierTransform}[f(t), t, \omega] \ : \ \mathcal{F}[f] = S(\omega) = \sqrt{\frac{|b|}{(2\pi)^{1-a}}} \int_{-\infty}^{\infty} f(t)\, e^{i b \omega t}\, dt.$$

$$\text{InverseFourierTransform}[S(\omega), \omega, t] \ : \ \mathcal{F}^{-1}[f] =$$

$$f(t) = \sqrt{\frac{|\,b\,|}{(2\pi)^{1+a}}} \int_{-\infty}^{\infty} F(\omega)\, e^{-i b \omega t}\, d\omega.$$

где $\{a, b\}$ – параметры Фурье (FourierParameters).

В соответствии с данным нами определением отображения Фурье (17) параметры $\{a, b\} = \{0, -1\}$. Однако, по умолчанию в системе Mathematica $\{a, b\} = \{0, 1\}$. Для того чтобы изменить их стандартное значение, воспользуемся функцией SetOptions:

```
in : SetOptions[FourierTransform, FourierParameters → {0, -1}];
     SetOptions[InverseFourierTransform, FourierParameters → {0, -1}];
```

Итак, для того чтобы найти спектральную плотность заданной функции $f(t)$, воспользуемся опцией FouirierTransform:

```
in : Evaluate@FourierTransform[f[t], t, w, Assumptions → a > 0]
```

$$out: \frac{\sqrt{\frac{2}{\pi}}\ \mathrm{w\,Sin}\left[\frac{\pi\,\mathrm{w}}{\mathrm{a}}\right]}{\mathrm{a}^2-\mathrm{w}^2}$$

Такой же результат мы получим, интегрируя функцию по формуле (17):

```
in : Integrate[1/Sqrt[2 Pi] f[t] Exp[- I w t], {t, -Infinity, Infinity},
      Assumptions → a > 0]
```

$$out: -\frac{\sqrt{\frac{2}{\pi}}\ \mathrm{w\,Sin}\left[\frac{\pi\,\mathrm{w}}{\mathrm{a}}\right]}{-\mathrm{a}^2+\mathrm{w}^2}$$

Получили следующий результат:

$$\mathcal{F}[f]=S(\omega)=\frac{\sqrt{\frac{2}{\pi}}\ \omega\,\mathrm{Sin}\left[\frac{\pi\,\omega}{a}\right]}{a^2-\omega^2}$$

Значит теперь можно представить исходную функцию в виде:

$$f(t)=\frac{1}{\sqrt{2\,\pi}}\int_{-\infty}^{\infty}\frac{\sqrt{\frac{2}{\pi}}\ \omega\,\mathrm{Sin}\left[\frac{\pi\,\omega}{a}\right]}{a^2-\omega^2}\,e^{i\,\omega\,t}\,d\omega=\frac{1}{\pi}\int_{-\infty}^{\infty}\frac{\omega\,\mathrm{Sin}\left[\frac{\pi\,\omega}{a}\right]}{a^2-\omega^2}\,e^{i\,\omega\,t}\,d\omega$$

Пример №12.517

Найти пару косинус- или синус- преобразований указанной функции:

$$f(t)=\frac{1}{a^2+t^2},\ a>0.$$

Так как данная функция четная, воспользуемся формулой (19), для того чтобы найти косинус-преобразование исходной функции:

```
in : Integrate[Sqrt[2/Pi] 1/(a^2 + t^2) Cos[w t], {t, 0, Infinity},
      Assumptions → a > 0]
```

$$out: \mathrm{ConditionalExpression}\left[\frac{e^{-\mathrm{a\,Abs[w]}}\sqrt{\frac{\pi}{2}}}{\mathrm{a}},\ \mathrm{w}\in\mathrm{Reals}\right]$$

При условии, что ω принадлежит множеству действительных чисел, получили:

$$f^*(\omega) = \sqrt{\frac{\pi}{2}}\ \frac{e^{-a|\omega|}}{a}$$

Тогда заданную функцию можно записать в виде:

$$f(t) = \frac{1}{a}\int_0^\infty e^{-a\,\omega}\cos(\omega\, t)\, d\omega$$

Разумеется, Wolfram Mathematica предоставляет удобные опции FourierCosTransform и FourierSinTransform для нахождения косинус- и синус преобразований Фурье.

Пример №12.518

Найти пару косинус- или синус- преобразований указанной функции:

$$f(t) = \frac{t}{a^2+t^2},\ a > 0.$$

Так как исходная функция нечетная, воспользуемся функцией FourierSinTransofrm:

```
in : FourierSinTransform[t/(a^2 + t^2), t, w, Assumptions → a > 0]
```

```
out : e^(-a w) Sqrt[π/2]
```

Получили, что:

$$f_*(\omega) = e^{-a\,\omega}\sqrt{\frac{\pi}{2}}$$

$$f(t) = \int_0^\infty e^{-a\,\omega}\sin(\omega\, t)\, d\omega$$

7.6. Свойства преобразования Фурье

Пусть функциям $f(t)$, $g(t)$ соответствуют преобразования Фурье $F(\omega)$, $G(\omega)$. Тогда запишем основные свойства Фурье-преобразования (без доказательств):

1. Линейность

$$\mathcal{F}[a\, f(t) + b\, g(t)] = a\, F(\omega) + b\, G(\omega) \tag{23}$$

2. Запаздывание

$$\mathcal{F}[f(t-a)] = e^{-i\,\omega\, a}\, F(\omega) \tag{24}$$

3. Частотный сдвиг

$$\mathcal{F}\left[e^{-i\,a\,t} f(t)\right] = F(\omega - a) \tag{25}$$

4. Масштабирование

$$\mathcal{F}[f(a\,t)] = \frac{1}{|\,a\,|}\,F\Big(\frac{\omega}{a}\Big) \tag{26}$$

5. Преобразование Фурье n-ой производной

$$\mathcal{F}\Big[\frac{d^n}{dt^n}\,f(t)\Big] = (i\,\omega)^n\,F(\omega) \tag{27}$$

6. Дифференцирование изображения

$$\mathcal{F}[t^n\,f(t)] = i^n\,\frac{d^n}{d\omega^n}\,F(\omega) \tag{28}$$

7. Преобразование свертки

$$\mathcal{F}[(f * g)\,(t)] = \sqrt{2\,\pi}\;F(\omega)\,G(\omega) \tag{29}$$

8. Преобразование произведения

$$\mathcal{F}[f(t)\,g(t)] = \frac{1}{\sqrt{2\,\pi}}\,(F * G)\,(\omega) \tag{30}$$

9. Преобразование дельта-функции Дирака

$$\mathcal{F}[\delta(t)] = \frac{1}{\sqrt{2\,\pi}} \tag{31}$$

10. Преобразование константы

$$\mathcal{F}[a] = a\,\sqrt{2\,\pi}\;\delta(t) \tag{32}$$

11. Преобразование многочленов:

$$\begin{gathered}\mathcal{F}[t^n] = i^n\,\sqrt{2\,\pi}\;\delta^{(n)}(\omega)\\ \text{где } n - \text{ натуральное число,}\\ \delta^{(n)}(\omega) - \ n - \text{я обобщенная производная дельта} - \text{функции.}\end{gathered} \tag{33}$$

12. Изображения некоторых функций:

$$\mathcal{F}\Big[e^{-i\,a\,t}\Big] = \sqrt{2\,\pi}\;\delta(\omega - a) \tag{34}$$

$$\mathcal{F}[\cos(a\,t)] = \sqrt{2\,\pi}\;\frac{\delta(\omega - a) + \delta(\omega + a)}{2} \tag{35}$$

$$\mathcal{F}[\sin(a\,t)] = \sqrt{2\,\pi}\;\frac{\delta(\omega - a) - \delta(\omega + a)}{2\,i} \tag{36}$$

$$\mathcal{F}\Big[e^{-a\,t^2}\Big] = \frac{1}{\sqrt{2\,a}}\;e^{-\omega^2/4\,a} \tag{37}$$

$\mathcal{F}\left[e^{-t^2/2}\right] = e^{-\omega^2/2}$ – функция Гаусса совпадает со своим изображением

$$\mathcal{F}\left[\frac{1}{t}\right] = -i\sqrt{\frac{\pi}{2}}\ \mathrm{sign}(\omega) \tag{38}$$

$$\mathcal{F}\left[\frac{1}{t^n}\right] = -i\sqrt{\frac{\pi}{2}}\ \frac{(-i\,\omega)^{n-1}}{(n-1)!}\ \mathrm{sign}(\omega) \tag{39}$$

$$\mathcal{F}[\mathrm{sign}(t)] = \sqrt{\frac{2}{\pi}}\ (-i\,\omega)^{-1} \tag{40}$$

$$\mathcal{F}\left[\sqrt{2\pi}\ H(t)\right] = \frac{1}{i\,\omega} + \pi\,\delta(\omega) \tag{41}$$

$H(t)$ – функция Хевисайда.

Рассмотрим несколько тривиальных примеров применения данных свойств.

Пример №1

Найти изображение многочлена:

$$P_5(t) = 24\,t^5 - 3\,t^4 + 11\,t^2 - t + 2$$

Применив свойства 1 и 11, получаем следующий результат:

$$\mathcal{F}\left[24\,t^5 - 3\,t^4 + 11\,t^2 - t + 2\right] =$$
$$24\,\mathcal{F}\left[t^5\right] - 3\,\mathcal{F}\left[t^4\right] + 11\,\mathcal{F}\left[t^2\right] - \mathcal{F}[t] + 2\,\mathcal{F}[1] =$$
$$= \sqrt{2\pi}\ \left(24\,i^5\,\delta^{(5)}(\omega) - 3\,i^4\,\delta^{(4)}(\omega) - 11\,i^2\,\delta''(\omega) - i\delta'(\omega) + 2\,\delta(\omega)\right) =$$
$$= \sqrt{2\pi}\ \left(24\,i\,\delta^{(5)}(\omega) - 3\,\delta^{(4)}(\omega) + 11\,\delta''(\omega) - i\,\delta'(\omega) + 2\,\delta(\omega)\right)$$

Тот же результат получим при помощи FourierTransform:

```
in : FourierTransform[24 t^5 - 3 t^4 + 11 t^2 - t + 2, t, w] // TraditionalForm
```

out : $2\sqrt{2\pi}\ \delta(w) - i\sqrt{2\pi}\ \delta'(w) - 11\sqrt{2\pi}\ \delta''(w) - 3\sqrt{2\pi}\ \delta^{(4)}(w) + 24\,i\sqrt{2\pi}\ \delta^{(5)}(w)$

Пример №2

Найти изображение дифференциального выражения:

$$x'' + 2\,x' + x = \sin(t)$$

Обозначим $\mathcal{F}[x(t)] = X(\omega)$.Тогда согласно свойству 5 имеем:

$$\mathcal{F}[x'' + 2\,x' + x - e^{-t}] = \mathcal{F}[x''] + 2\,\mathcal{F}[x'] + \mathcal{F}[x] - \mathcal{F}[\sin(t)] =$$
$$= (i\,\omega)^2\,X + 2\,(i\,\omega)\,X + X - \sqrt{2\pi}\ \frac{\delta(\omega-1) - \delta(\omega+1)}{2\,i}$$

Отсюда видно, что:

$$-\,\omega^2 X + 2\,i\,\omega\,X + X = \sqrt{2\,\pi}\;\frac{\delta(\omega-1)-\delta(\omega+1)}{2\,i}$$

$$X = \sqrt{2\,\pi}\;\frac{\delta(\omega-1)-\delta(\omega+1)}{2\,i}$$

$$X(\omega) = \sqrt{2\,\pi}\;\frac{\delta(\omega-1)-\delta(\omega+1)}{2\,i\left(-\omega^2+2\,i\,\omega+1\right)}$$

Найдем функцию $x\,(t)$:

```
in : InverseFourierTransform[
      √(2 π) ((DiracDelta[(ω - 1)] - DiracDelta[(ω + 1)]) /
        (2 I (-ω^2 + 2 I ω + 1))), ω, t]
```

$$out:\; -\frac{\mathrm{Cos}[\mathrm{t}]}{2}$$

Получается, что частное решение исходного дифференциального уравнения:

$$x(t) = -\frac{1}{2}\cos(t)$$

Пример №3

Знаяизображение функции $f\,(t)$, найти изображение функции $g(t)$:

$$\mathcal{F}\left[\frac{1}{a^2+t^2}\right] = \sqrt{\frac{\pi}{2}}\;\frac{e^{-a\,\omega}}{a},\; g(t) = \frac{t^4}{a^2+t^2}$$

Согласно свойству 6 имеем:

$$\frac{t^4}{a^2+t^2} = t^4\,f\,(t),\quad \mathcal{F}\left[t^4\,f\,(t)\right] = i^4\,\frac{d^4}{d\,\omega^4}\mathcal{F}\,[f\,(t)]$$

$$\mathcal{F}\left[\frac{t^4}{a^2+t^2}\right] = \sqrt{\frac{\pi}{2}}\;\frac{d^4}{d\,\omega^4}\left(\frac{e^{-a\,\omega}}{a}\right)$$

Найдем производную 4-ого порядка:

```
in : D[Exp[-a w] / a, {w, 4}]
```

$$out:\; \mathrm{a}^3\,e^{-\mathrm{a\,w}}$$

В итоге, получаем ответ:

$$\mathcal{F}\left[\frac{t^4}{a^2+t^2}\right] = \sqrt{\frac{\pi}{2}}\;a^3\,e^{-a\,\omega}$$

7.7. Связь коэффициентов ряда Фурье с коэффициентами ряда Лорана.

Пусть функция $f(z)$ регулярна в кольце $r_1 < |z| < 1 + r_2$ $(0 \leq r_1 < 1, r_2 > 0)$ содержащем единичную окружность $|z| = 1$. Тогда она представляется в этом кольце рядом Лорана:

$$f(z) = \sum_{n=-\infty}^{\infty} c_n z^n \tag{42}$$

Откуда, полагая $z = e^{i\phi}$, получаем разложения в ряд Фурье функции:

$$F(\phi) = f\left(e^{i\phi}\right) = \sum_{n=-\infty}^{\infty} c_n e^{i\phi n} \tag{43}$$

Отсюда следует обратное: если функция $F(\phi)$ представляется в виде: $F(\phi) = f\left(e^{i\phi}\right)$, где функция $f(z)$ регулярна в кольце, то ряд (43) является рядом Фурье для $F(\phi)$.

Пример №12.352

Найти все разложения указанных функций в ряды Лорана по степеням $z - z_0$.

$$f(z) = \frac{1}{z(z-1)},\ z_0 = 1$$

Запишем функцию:

```
in : f[z_] := 1/(z (z - 1))
```

Как уже можно было заметить, Mathematica не предоставляет встроенных средств для разложения функций в ряд Лорана. В ней есть только функция Series, которая находит разложение в ряд лишь в окрестности точки z_0.

Однако, благодаря связи ряда Лорана и Фурье, а также тому, что Mathematica имеет целый ряд функций для работы с рядами Фурье, можно составить алгоритм разложения функции в ряд Лорана через ряд Фурье.

Изобразим особые точки функции $f(z)$, выберем контур С, составим множество коэффициентов ряда Фурье для этого контура, и затем сформируем из них ряд Лорана.

```
in : Graphics[{Circle[{1, 0}, 1/2], Circle[{1, 0}, 3/2],
      Text["r1", {1/2, 1/2}], Text["r2", {3/2, 3/2}],
      PointSize[Large], Point[ReIm[z /. Solve[f[z]^-1 == 0, z]]]},
     PlotRange → {{-3, 3}, {-3, 3}}, Frame → False, Axes → True,
     AxesLabel → {Re, Im}]
```



Найдем коэффициенты ряда Лорана при $0 < |z-1| < 1$. Для этого выберем контур r_1 и воспользуемся функцией FourierCoefficient.

Выберем, к примеру, $r_1 = 1/2$. Тогда получим замену: $z - z_0 = r\,e^{i\,t}$, $z = z_0 + r\,e^{i\,t}$

```
in : Table[FourierCoefficient[f[1 + 1 / 2 Exp[I t]], t, n] / (1 / 2)^n,
       {n, 1, 10}]

out : {1, -1, 1, -1, 1, -1, 1, -1, 1, -1}
```

Примечание: действие деления на r^n следует из следующего: заменив $z = r\,e^{i\,t}$ получаем, что $f\left(r\,e^{i\,t}\right) = \sum_{-\infty}^{\infty} c_n\,r^n\,e^{i\,n\,t}$, тогда, чтобы получить c_n – коэффициент Лорана, необходимо разделить получившийся коэффициент Фурье на r^n.

Составим ряд Лорана вида:

$$f(z) = \sum_{n=-k}^{k} c_n (z - z_0)^n$$

```
in : lor[z_, k_, r_] :=
      Sum[Evaluate[FourierCoefficient[f[1 + r Exp[I t]], t, n]] / (r)^n
        (z - 1)^n, {n, -k, k}]

in : lor[z, 8, 1 / 2]
```

out : $-2 + \dfrac{1}{-1+z} - (-1+z)^2 + (-1+z)^3 - (-1+z)^4 + (-1+z)^5 - (-1+z)^6 + (-1+z)^7 - (-1+z)^8 + z$

Теперь можно найти разложение функции в кольце $|z-1| > 1$:

```
in : lor[z, 8, 3]
```

$$\text{out}: \frac{1}{(-1+z)^8} - \frac{1}{(-1+z)^7} + \frac{1}{(-1+z)^6} - \frac{1}{(-1+z)^5} + \frac{1}{(-1+z)^4} - \frac{1}{(-1+z)^3} + \frac{1}{(-1+z)^2}$$

Сравним получившиеся ответы с тем, что получается при использовании функции loranO (из Главы 2 Параграфа 5):

$$\text{in}: \text{loranO}\left[z, \frac{1}{z\,(z-1)}, 1, 1/2, 8\right]$$

$$\text{loranO}\left[z, \frac{1}{z\,(z-1)}, 1, 3, 8\right]$$

$$\text{out}: -2 + \frac{1}{-1+z} - (-1+z)^2 + (-1+z)^3 - (-1+z)^4 + (-1+z)^5 - (-1+z)^6 + (-1+z)^7 - (-1+z)^8 + z$$

$$\text{out}: \frac{1}{(-1+z)^8} - \frac{1}{(-1+z)^7} + \frac{1}{(-1+z)^6} - \frac{1}{(-1+z)^5} + \frac{1}{(-1+z)^4} - \frac{1}{(-1+z)^3} + \frac{1}{(-1+z)^2}$$

Получили такие же ответы.

Программный модуль

Опыт, полученный в предыдущем примере, используем для составления программного модуля:

```
in : loranFC[expr_, {z_, z0_, k_}, r_] := Module[{c},
       c[n_] := FourierCoefficient[expr /. z → z0 + r Exp[I t], t, n] / (r)^n;
       Sum[c[n] (z - z0)^n, {n, -k, k}]]
```

Продемонстрируем пример:

$$\text{in}: \text{loranFC}\left[\frac{1}{z\,(z-1)}, \{z, 1, 8\}, 3\right]$$

$$\text{out}: \frac{1}{(-1+z)^8} - \frac{1}{(-1+z)^7} + \frac{1}{(-1+z)^6} - \frac{1}{(-1+z)^5} + \frac{1}{(-1+z)^4} - \frac{1}{(-1+z)^3} + \frac{1}{(-1+z)^2}$$

Сравним время работы получившегося алгоритма и алгоритма, работающего на основе вычетов (из Главы 2 Параграфа 6):

```
in : t1 = Timing[loranFC[1/(z (z - 1)), {z, 1, 8}, 3]]
     t2 = Timing[loranR[1/(z (z - 1)), {z, 1, 8}, 3]]
```

out : $\left\{17.40625, \frac{1}{(-1+z)^8}-\frac{1}{(-1+z)^7}+\frac{1}{(-1+z)^6}-\frac{1}{(-1+z)^5}+\frac{1}{(-1+z)^4}-\frac{1}{(-1+z)^3}+\frac{1}{(-1+z)^2}\right\}$

out : $\left\{0.03125, \frac{1}{(-1+z)^8}-\frac{1}{(-1+z)^7}+\frac{1}{(-1+z)^6}-\frac{1}{(-1+z)^5}+\frac{1}{(-1+z)^4}-\frac{1}{(-1+z)^3}+\frac{1}{(-1+z)^2}\right\}$

```
in : t1[[1]] / t2[[1]]
```

out : 557.

Как мы видим, алгоритм на основе вычетов работает почти в 600 раз быстрее. Поэтому в дальнейшем использовать метод на основе ряда Фурье для разложения функций в ряд Лорана не будем.

7.8. Ядро Дирихле. Интеграл Дирихле.

Введем обозначения: $A_n(x) = a_n \cos(n\,x) + b_n \sin(n\,x)$, где a_n, b_n – коэффициенты Фурье

$$A_n(x) = a_n \cos(n\,x) + b_n \sin(n\,x), \text{ где } a_n,\ b_n - \text{ коэффициенты Фурье}$$

$$S_n(x) = \sum_{k=0}^{n} A_k(x) \ - \text{частичные суммы ряда Фурье}$$

$$\sigma(x) = \sum_{k=0}^{\infty} A_k(x) - \text{ ряд Фурье}$$

Запишем частичную сумму ряда:

$$S_n(x) = \sum_{k=0}^{n} A_k(x) \ = \frac{1}{2\,\pi} \int_{-\pi}^{\pi} f(t)\,dt +$$

$$\sum_{k=1}^{n}\left(\frac{1}{\pi}\int_{-\pi}^{\pi} f(t)\cos(k\,t)\,dt\cos(k\,x) + \frac{1}{\pi}\int_{-\pi}^{\pi} f(t)\sin(k\,t)\,dt\sin(k\,x)\right)$$

По свойствам интеграла, внося знак суммы под знак интеграла, получаем:

$$S_n(x) = \int_{-\pi}^{\pi} f(t)\,\frac{1}{\pi}\left(\frac{1}{2} + \sum_{k=1}^{n}(\cos(k\,t)\cos(k\,x) + \sin(k\,t)\sin(k\,x))\right)dt =$$

$$\int_{-\pi}^{\pi} f(t)\,\frac{1}{\pi}\left(\frac{1}{2}+\sum_{k=1}^{n}\cos\left(k\,(t-x)\right)\right)dt$$

Тригонометрический полином вида:

$$D_n(t)=\frac{1}{\pi}\left(\frac{1}{2}+\sum_{k=1}^{n}\cos\left(k\,t\right)\right) \tag{44}$$

называется ядром Дирихле.

Подставляя эту функцию в формулу для частичной суммы, получаем следующее выражение:

$$S_n(x)=\int_{-\pi}^{\pi} f(t)\,D_n(t-x)\,dt \tag{45}$$

которое называется интегралом Дирихле.

Пример

Найдем выражение для ядра:

```
in : 1/Pi (1 / 2 + Sum[Cos[k t], {k, 1, n}]) // FullSimplify
out : Csc[t/2] Sin[(1/2 + n) t] / (2 π)
```

Получили, что:

$$D_n(t)=\frac{1}{\pi}\left(\frac{1}{2}+\sum_{k=1}^{n}\cos\left(k\,t\right)\right)=\frac{1}{2\,\pi}\,\frac{\sin\left(\left(\frac{1}{2}+n\right)t\right)}{\sin\left(\frac{t}{2}\right)} \tag{46}$$

Глава 3 Операционное исчисление в СКМ Mathematica

§1. Преобразование Лапласа

1.1. Определение и свойства преобразования Лапласа.

Преобразованием Лапласа функции $f(t)$, $t\in\mathbb{R}$ (которая, может принимать и комплексные значения), называется функция $F(p)$ комплексной переменной p, определяемая следующим равенством:

$$F(p)=\int_{0}^{+\infty} e^{-p\,t}\,f(t)\,dt \tag{1}$$

Оригиналом называется всякая функция $f(t)$, удовлетворяющая следующим условиям:

1) $f(t)=0$ при $t<0$, причем принимается, что $f(0)=f(+0)$;

2) Существуют такие постоянные σ и M, что:

$$\left| f(t) \right| < M\, e^{\sigma\, t} \text{ при } t > 0 \tag{2}$$

(величина $\sigma_0 = \inf \sigma$ называется показателем роста функции $f(t)$);

3) На любом конечном отрезке $[0; T]$ функция может иметь лишь конечное число точек разрыва, причем только 1-ого рода.

Если $f(t)$ - оригинал, то стоящий в правой части равенства (1) интеграл Лапласа сходится абсолютно и равномерно в полуплоскости $\operatorname{Re} p \geqslant \sigma > \sigma_0$. При этом функция $F(p)$ является аналитической в полуплоскости $\operatorname{Re} p > \sigma_0$ и называется изображением функции $f(t)$.

Соответствие между оригиналом $f(t)$ и его изображением $F(p)$ символически записывается в виде $f(t) \doteq F(p)$

Пример №13.10

Рассмотрим простейший пример: Используя формулу (1), найти изображение для следующего оригинала:

$$f(t) = \begin{cases} t & 0 \le t < 2 \\ \frac{1}{2}(4 - t) & 2 \le t < 4 \\ 0 & 4 \le t \end{cases}$$

Запишем исходную кусочно-заданную функцию:

```
in : f[t_] := Piecewise[{{t, 0 ≤ t < 2}, {1/2 (4 - t), 2 ≤ t < 4}}, 0]
```

Для наглядности построим ее график :

```
in : Plot[f[t], {t, -2, 6}]
```



рис. 1.1.1

Очевидно, что данная функция является оригиналом, т.к. соблюдены все условия изложенные выше.

Для того чтобы найти изображение, воспользуемся формулой (1). Обозначать изображения будем $fp(p)$, т.к. функции начинающиеся с большой буквы, в основном, являются системными, т.е. встроенными.

```
in : fp[p_] := Integrate[Exp[-p t] f[t], {t, 0, Infinity}]
```

```
in : fp[p_] := Integrate[Exp[-k t] f[t], {t, 0, Infinity}] /. k → p
```

Существует принципиальное отличие, между двумя выражениями, записанными выше. Первое интерпретируется следующим образом: каждый раз при вызове функции $\mathrm{fp}(p_1)$ в переменную p, стоящую внутри интеграла, передается значение p_1 и затем вычисляется весь интеграл. Можно догадаться, что это замедляет работу программы, поэтому используется второй способ: сначала вычисляется интеграл, зависящий от k, затем переменная k просто заменяется на значение p_1. Данный способ работает быстрее, т.к. интеграл не будет вычисляться столько раз, сколько мы вызвали функцию $\mathrm{fp}(p)$.

Выведем полученную функцию:

```
in : fp[p]
```

out : $$\frac{e^{-4\,p}\left(1-3\,e^{2\,p}+2\,e^{4\,p}-2\,e^{2\,p}\,p\right)}{2\,p^2}$$

Ответ:

$$f(t) \doteq \frac{e^{-4\,p}\left(1-3\,e^{2\,p}+2\,e^{4\,p}-2\,e^{2\,p}\,p\right)}{2\,p^2}$$

Пример №1

Найти изображение функции Хевисайда:

$$H(t)=\eta(t)=\begin{cases}1 & t\geq 0\\ 0 & t<0\end{cases}$$

Функция Хевисайда в Mathematica является встроенной. Конечно, можно определить ее самостоятельно при помощи кусочно-заданной функции PieceWise, но в этом нет необходимости. Функция PieceWise записывается следующим образом:

```
in : h[t_] := HeavisideTheta[t]
     Plot[h[t], {t, -2, 2}]
```

рис. 1.1.2

Так как функция Хевисайда является оригиналом, то ее изображение:

```
in : hp[p_] := Integrate[Exp[-k t] h[t], {t, 0, Infinity}] /. k → p
```

```
in : Simplify[hp[p], Re[p] > 0]
```

out : $$\frac{1}{p}$$

Получили $H(t) \doteq p^{-1}$.

1.2 Свойства преобразования Лапласа.

1. С в о й с т в о л и н е й н о с т и.

$$\sum_{k=1}^{n} C_k\, f_k(t) \doteq \sum_{k=1}^{n} C_k\, F_k(p), \quad \operatorname{Re} p > \max\{\sigma_1, \sigma_2, \ \dots \sigma_n\}$$

2. Т е о р е м а п о д о б и я.

$$f\left(\frac{t}{\alpha}\right) \doteq \frac{1}{\alpha}\, F\left(\frac{p}{\alpha}\right), \ \operatorname{Re} p > \alpha\, \sigma_0$$

3. Т е о р е м а с м е щ е н и я. Умножению оригинала на $e^{\alpha t}$, $\alpha \in \mathbb{R}$, соответствует смещение аргумента изображения на α, т.е.:

$$e^{\alpha t}\, f(t) \doteq F(p-\alpha), \ \operatorname{Re}(p-\alpha) > \sigma_0$$

4. Т е о р е м а з а п а з д ы в а н и я. Запаздыванию оригинала на τ соответствует умножение изображения на $e^{-p\tau}$:

$$\eta(t-\tau)\, f(t) \doteq e^{-p\tau}\, F(p), \ \operatorname{Re} p \ > \ \sigma_0$$

5. Д и ф ф е р е н ц и р о в а н и е о р и г и н а л а. Если $f(t)$ и ее производные $f^{(k)}(t)$ являются оригиналами, то для любого k=1,2...n

$$f^{(k)}(t) \doteq p^k\, F(p) - \left(p^{k-1}\, f(0) + p^{k-2}\, f'(0) + \dots + f^{(k-1)}(0)\right)$$

В частности,

$$f'(t) \doteq p\, F(p) - f(0), \ \operatorname{Re} p > \ \sigma_0$$

6. И н т е г р и р о в а н и е о р и г и н а л а:

$$\int_0^t f(\tau)\, d\tau \doteq \frac{F(p)}{p}, \ \operatorname{Re} p > \ \sigma_0$$

7. Д и ф ф е р е н ц и р о в а н и е и з о б р а ж е н и я. Умножению оригинала на множитель t соответствует умножению изображения на -1 и дифференцирование его по аргументу p:

$$\genfrac{}{}{0pt}{}{t^n}{\square}\ f(t) \doteq (-1)^n\, F^{(n)}(p), \ n = 1, 2 \dots$$

8. И н т е г р и р о в а н и е и з о б р а ж е н и я. Если $f(t)/t$ является оригиналом, то

$$\frac{1}{t}\, f(t) \doteq \int_p^{+\infty} F(q)\, dq$$

9. Д и ф ф е р е н ц и р о в а н и е и и н т е г р и р о в а н и е п о п а р а м е т р у. Если $f(t, \alpha) \doteq F(p, \alpha)$ и функции $\dfrac{\partial}{\partial\alpha} f(t, \alpha)$ и $\int_{\alpha 1}^{\alpha 2} f(t, \alpha)\, d\alpha$, рассматриваемые как функции переменной t, являются оригиналами, то

$$\frac{\partial}{\partial\alpha} f(t,\alpha) \doteq \frac{\partial}{\partial\alpha} F(p,\alpha) \quad \text{и} \quad \int_{\alpha 1}^{\alpha 2} f(t,\alpha)\, d\alpha = \int_{\alpha 1}^{\alpha 2} F(p,\alpha)\, d\alpha$$

10. Т е о р е м а Б о р е л я о б и з о б р а ж е н и и с в е р т к и. Свертке оригиналов

$$f_1 * f_2 = \int_0^t f_1(\tau)\, f_2(t-\tau)\, d\tau = \int_0^t f_1(t-\tau)\, f_2(\tau)\, d\tau$$

соответствует произведение изображений, т. е.

$$f_1 * f_2 = F_1(p) \cdot F_2(p)$$

11. И н т е г р а л Д ю а м е л я. Если $f(t) \doteq F(p)$ и $g(t) \doteq G(p)$, то

$$p\, F(p)\, G(p) \doteq f(o)\, g(t) + (f' * g)\,(t) = g(0)\, f(t) + (g' * f)\,(t)$$

Ниже приведен список изображений наиболее распространенных функций:

$$\begin{array}{ll} \eta(t) \doteq \dfrac{1}{p} & t^n \doteq \dfrac{n!}{p^{n+1}} \\ \operatorname{ch} \beta t \doteq \dfrac{p}{p^2 - \beta^2} & \operatorname{sh} \beta t \doteq \dfrac{\beta}{p^2 - \beta^2} \\ e^{\alpha t} \doteq \dfrac{1}{p - \alpha} & t^n e^{\alpha t} \doteq \dfrac{n!}{(p-\alpha)^{n+1}} \\ \cos \beta t \doteq \dfrac{p}{p^2 + \beta^2} & \sin \beta t \doteq \dfrac{\beta}{p^2 + \beta^2} \end{array} \qquad (3)$$

С помощью свойств преобразования Лапласа и списка основных изображений можно найти изображения большинства функций встречающихся на практике.

Рассмотрим пример:

Пример №13.19

Найти изображение заданной функции:

$$f(t) = e^{-t} + 3\, e^{-2t} + t^2$$

Согласно свойству линейности изображение суммы функций равно сумме изображений. Используя список изображений (3), находим :

$$e^{-t} = \frac{1}{p+1}$$

$$3\, e^{-2t} = \frac{3}{p+2}$$

$$t^2 = \frac{2}{p^3}$$

В итоге, изображение исходной функции:

$$f(t) \doteq F(p)$$

$$F(p) = \frac{1}{p+1} + \frac{3}{p+2} + \frac{2}{p^3}$$

Пример №13.22

Найти изображение заданной функции:

$$f(t) = \sin^2(t-a)$$

Применяя тригонометрические тождества, приведем функцию к виду:

$$\sin^2(t-a) = \frac{1}{2}\,(1-\cos(2\,(t-a)))$$

$$\cos(2\,(t-a)) = \cos 2\,t\cos 2\,a \;+\; \sin 2\,t\sin 2\,a$$

$$\sin^2(t-a) = \frac{1}{2}\,(1-\cos 2\,t\cos 2\,a \;-\; \sin 2\,t\sin 2\,a)$$

Воспользуемся свойством линейности и списком изображений:

$$\cos 2\,t \doteq \frac{p}{p^2+4},\quad \sin 2\,t \doteq \frac{2}{p^2+4}$$

Получаем изображение исходной функции:

$$F(p) = \frac{1}{2}\left(\frac{1}{p} - \frac{p\cos 2\,a}{p^2+4} - \frac{2\sin 2\,a}{p^2+4}\right)$$

Пример №13.32

Найти изображение заданной функции:

$$f(t) = e^{-t}\sin^2(t)$$

Чтобы найти изображение, воспользуемся выражением, полученным в предыдущем примере. Так как $a = 0$, то:

$$\sin(t)^2 \doteq \frac{1}{2}\left(\frac{1}{p} - \frac{p}{p^2+4}\right)$$

Теперь согласно теореме смещения получаем ответ:

$$e^{-t}\sin^2(t) \doteq \frac{1}{2}\left(\frac{1}{p+1} - \frac{p}{(p+1)^2+4}\right)$$

Пример №13.36

Найти изображение заданной функции:

$$f(t) = \int_0^t e^{-\frac{1}{2}(t-\tau)}\,\tau\,d\tau$$

Заданная функция представляет из себя свертку двух функций:

$$e^{-\frac{1}{2}t} * t = \int_0^t e^{-\frac{1}{2}(t-\tau)}\, \tau \, d\tau = \int_0^t e^{-\frac{1}{2}\tau}(t-\tau)\, d\tau$$

По теореме Бореля об изображении свертки получаем изображение:

$$e^{-\frac{1}{2}t} * t \doteq \frac{1}{p^2}\,\frac{1}{p+\frac{1}{2}} = \frac{2}{p^2(2\,p+1)}$$

Пример №13.44

Найти изображение дифференциального выражения при заданных начальных условиях:

$$x^{IV}(t) + 4\,x'''(t) + 2\,x''(t) - 3\,x'(t) - 5, \quad x(0) = x'(0) = x''(0) = x'''(0) = 0$$

Обозначим изображение функции $x\,(t)$ как $X\,(p)$, тогда согласно свойству дифференцирования оригинала получаем:

$$x^{IV}(t) \doteq p^4\,X(p) - \left(p^3\,f(0) + p^2\,f'(0) + p\,f''(0) + f'''(0) = p^4\,X(p)\right.$$

Так как начальные условия нулевые, получим следующее изображение исходного выражения:

$$X(p)\left(p^4 + 4\,p^3 + 2\,p^2 - 3\,p\right) - \frac{5}{p}$$

Все рассмотренные выше примеры мало относятся к теме данной книги (компьютерные технологии). Они были решены традиционными методами: в каждой задаче приходилось выбирать нужные свойства и каждый раз смотреть в таблицу изображений. Разумеется, процесс решения данного класса задач может быть существенно упрощен с использованием внутренних функций Mathematica. Рассмотрим простейший пример.

Пример №13.27

Найти изображение заданной функции:

$$f(t) = \sin t - \, t\cos t$$

В обычном случае нам было бы необходимо воспользоваться рядом свойств (линейности и дифференцирования изображения), чтобы с помощью изображений известных простейших функций получить изображение заданной.

Функция LaplaceTransform заменяет все эти действия. Рассмотрим пример работы:

```
in : LaplaceTransform[Cos[t], t, p]
```

out : $\dfrac{p}{1+p^2}$

```
in : LaplaceTransform[t Cos[t], t, p]
```

out : $\dfrac{-1+p^2}{\left(1+p^2\right)^2}$

Данная функция сама подбирает нужные свойства и находит изображение.

Изображение заданной функции:

```
in : LaplaceTransform[Sin[t] - Cos[t] t, t, p]
```

$$out: -\frac{-1+p^2}{(1+p^2)^2}+\frac{1}{1+p^2}$$

Рассмотрим ряд примеров:

Пример №13.25

$$f(t) = \text{sh}\,(3\,t)\cos\,(2\,t)$$

```
in : LaplaceTransform[Sinh[3 t] Cos[2 t], t, p]
```

$$out: -\frac{3\,(13-p^2)}{(13-6\,p+p^2)\,(13+6\,p+p^2)}$$

$$F(p) = -\frac{3\,(13-p^2)}{(13-6\,p+p^2)\,(13+6\,p+p^2)}$$

Пример №13.37

$$f(t) = \int_0^t (t-\tau)^2 \cos 2\,\tau\,d\tau$$

```
in : LaplaceTransform[Integrate[(t - τ)^2 Cos[2 τ], {τ, 0, t}], t, p] //
    Simplify
```

$$out: \frac{2}{4\,p^2+p^4}$$

$$F(p) = \frac{2}{4\,p^2+p^4}$$

Пример №13.42

$$f(t) = \int_0^t \frac{\cos\beta\,\tau - \cos\alpha\,\tau}{\tau}\,d\tau$$

```
in : LaplaceTransform[Integrate[(Cos[β τ] - Cos[α τ]) / τ, {τ, 0, t}],
    t, p]
```

$$out: \frac{\text{Log}\left[1+\frac{p^2}{\alpha^2}\right]}{2\,p}+\frac{\text{Log}[\alpha]}{p}-\frac{\text{Log}\left[1+\frac{p^2}{\beta^2}\right]}{2\,p}-\frac{\text{Log}[\beta]}{p}$$

$$F(p) = \frac{1}{2\,p} \ln\left(\frac{\alpha^2 + p^2}{\beta^2 + p^2}\right)$$

Также с использованием данной функции становится удобно находить изображения дифференциальных выражений:

Пример №13.42

Найти изображение дифференциального выражения при заданных начальных условиях:

$$x'''(t) + 6\,x''(t) + x'(t) - 2\,x(t), \qquad x(0) = x'(0) = 0, \quad x''(0) = 1$$

Применим функцию LaplaceTransofrm непосредственно ко всему выражению:

```
in : LaplaceTransform[x'''[t] + 6 x''[t] + x'[t] - 2 x[t], t, p]

out : -2 LaplaceTransform[x[t], t, p] + p LaplaceTransform[x[t], t, p] +
      p^3 LaplaceTransform[x[t], t, p] - x[0] - p^2 x[0] +
      6 (p^2 LaplaceTransform[x[t], t, p] - p x[0] - x′[0]) - p x′[0] - x″[0]
```

Обратим внимание на LaplaceTransorm[x[t],t,p]. Данная функция, зависящая от p, есть ни что иное, как изображение функции x(t). Заменим его на X(p):

```
in : LaplaceTransform[x'''[t] + 6 x''[t] + x'[t] - 2 x[t], t, p] /.
      LaplaceTransform[x[t], t, p] → X[p]

out : -x[0] - p^2 x[0] - 2 X[p] + p X[p] + p^3 X[p] +
      6 (-p x[0] + p^2 X[p] - x′[0]) - p x′[0] - x″[0]
```

Теперь тем же способом можно учесть начальные условия:

```
in : LaplaceTransform[x'''[t] + 6 x''[t] + x'[t] - 2 x[t], t, p] /.
      {LaplaceTransform[x[t], t, p] → X[p], x[0] → 0, x'[0] → 0,
       x''[0] → 1}

out : -1 - 2 X[p] + p X[p] + 6 p^2 X[p] + p^3 X[p]
```

В итоге, получили:

$$-1 + X(p)\left(-2 + p + 6\,p^2 + p^3\right)$$

Модуль

Составим программный модуль, который будет находить изображения дифференциальных выражений с учетом начальных условий (или без них):

```
in : laplaceD[t_, expr_, rule_] := Module[{},
       Collect[
        LaplaceTransform[expr, t, p] /.
          LaplaceTransform[x[t], t, p] → X[p] /. rule, X[p]]]
```

Пример работы:

Пример №13.46

Найти изображение дифференциального выражения при заданных начальных условиях:

$$x''(t) + 5\,x'(t) - 7\,x(t) + 2, \qquad x(0) = \alpha,\ \ x'(0) = 0$$

```
in : laplaceD[t, x''[t] + 5 x'[t] - 7 x[t] + 2, {x[0] → α, x'[0] → 0}]
```

$$out: \ \frac{2}{p} - 5\,\alpha - p\,\alpha + \left(-7 + 5\,p + p^2\right) X[p]$$

Пример №13.113

Найти изображение дифференциального выражения при заданных начальных условиях:

$$x''(t) - 2\,x'(t) + 2\,x = \sin t,\ \ x(0) = 0,\ \ x'(0) = 1$$

```
in : laplaceD[t, x''[t] - 2 x'[t] + 2 x[t] == Sin[t], {x[0] → 0, x'[0] → 1}]
```

$$out: \ -1 + \left(2 - 2\,p + p^2\right) X[p] == \frac{1}{1 + p^2}$$

Пример №13.53

Найти изображение следующей функции:

$$f(t) = \begin{cases} \frac{h}{\tau}\,t & 0 \le t < \tau \\ h & \tau \le t \le 2\,\tau \\ \frac{-h}{\tau}\,(t - 3\,\tau) & 2\,\tau \le t < 3\,\tau \\ 0 & t \ge 3\,\tau \end{cases}$$

Данный пример не представляет абсолютно никакой сложности, поэтому решим его при помощи функции LaplaceTransform.

Запишем функцию:

```
in : f[t_] :=
      Piecewise[{{h / τ t, 0 ≤ t < τ}, {h, τ ≤ t < 2 τ},
        {-h/τ (t - 3 τ), 2 τ ≤ t < 3 τ}}]
```

Построим ее график, при $h = 1$, $\tau = \pi$:

```
in : Plot[f[t] /. {h → 1, τ → Pi}, {t, -1 Pi, 4 Pi}]
```

рис.1.2.1

Теперь найдем изображение:

```
in : Simplify[LaplaceTransform[f[t], t, p], τ > 0]
```

$$out: \frac{e^{-3 p \tau} \left(-1 + e^{p \tau}\right)^2 \left(1 + e^{p \tau}\right) h}{p^2 \tau}$$

На данном этапе оставим тему поиска изображений.

§2. Восстановление оригинала по изображению

2.1. Формула обращения. Теоремы разложения.

Если $f(t)$ – оригинал и $F(p)$ – его изображение, то в любой точке непрерывности $f(t)$ справедлива формула обращения Меллина

$$f(t) = \frac{1}{2 \pi i} \int_{\sigma - i\infty}^{\sigma + i\infty} F(p)\, e^{p t}\, dp$$

где интегрирование производится по любой прямой $\operatorname{Re} p = \sigma,\ \ \sigma > \sigma_0$.

Непосредственное применение формулы обращения чсто затруднительно, и обычно пользуются теоремами разложения, являющимися следствиями из нее:

П е р в а я т е о р е м а р а з л о ж е н и я. Если функция $F(p)$ аналитична в некоторой окрестности бесконечно удаленной точки и ее разложение в ряд по степеням $1/p$ имеет вид

$$F(p) = \sum_{n=0}^{\infty} \frac{a_n}{p^{n+1}},$$

то функция

$$f(t) = \sum_{n=0}^{\infty} a_n \frac{t^n}{n!},\ \ t \geq\ 0\ (f(t) = 0 \text{ при } t < 0)$$

является оригиналом, имеющим изображение $F(p)$.

В т о р а я т е о р е м а р а з л о ж е н и я. Если изображение $F(p)$ является однозначной функцией и имеет лишь конечное число особых точек $p_1,\ p_2,\ \ldots\ p_n$, лежащих в конечной части полуплоскости $\operatorname{Re} p \leqslant \sigma_0$, то

$$f(t) = \sum_{k=1}^{n} \operatorname{res}[e^{p t} F(p);\ p_k] \tag{1}$$

Если в частности, $F(p) = P_m(p) / Q_n(p)$, где $P_m(p)$ и $Q_n(p)$ - многочлены степеней m и n соответственно (n>m), $p_1,\ p_2,\ \ldots\ p_r$ - корни многочлена $Q_n(p)$ с кратностями, соответственно равными $j_1, j_2,\ \ldots\ j_r$, то

$$f(t) = \sum_{k=1}^{r} \frac{1}{(j_k - 1)!} \lim_{p \to p_k} \frac{d^{j_k - 1}}{d p^{j_k - 1}} \left((p - p_k)^{j_k} F(p) e^{p t} \right) \tag{2}$$

Примечание. Если все корни p_i - простые ($j_i = 1$), то формула (2) упрощается до вида:

$$f(t) = \sum_{k=1}^{n} \frac{P_m(p_k)}{Q_n'(p_k)} e^{p_k t} \tag{3}$$

Пример 13.88

Пользуясь первой теоремой разложения, найти оригинал для заданной функции

$$F(p) = \frac{1}{p} \cos\left(\frac{1}{p}\right)$$

Применяя первую теорему, находим разложение в ряд $F(p)$ в окрестности бесконечно удаленной точки.
Вычислим n-ый коэффициент этого разложения:

```
in : Series[1/p Cos[1/p], {p, Infinity, 7}]
```

$$out: \frac{1}{p} - \frac{1}{2 p^3} + \frac{1}{24 p^5} - \frac{1}{720 p^7} + O\left[\frac{1}{p}\right]^8$$

```
in : c[n_] := SeriesCoefficient[1/p Cos[1/p], {p, Infinity, n}]
```

```
in : Refine[c[n], n > 1]
```

$$out: \frac{i\, i^n \left(-1 + (-1)^n\right)}{2\,(-1+n)!}$$

Итак, получили разложение изображения $F(p)$ в ряд:

$$\frac{1}{p} \cos\left(\frac{1}{p}\right) = \sum_{n=1}^{\infty} c_n \frac{1}{p^n} = \sum_{n=0}^{\infty} c_{n+1} \frac{1}{p^{n+1}}$$

По первой теореме находим оригинал данного изображения изображения:

$$f(t) = \sum_{n=0}^{\infty} c_{n+1} \frac{t^n}{n!}$$

Запишем выражение функции оригинала:

```
in : f[t_, k_] := Sum[c[n + 1] / n! t^n, {n, 0, k}]
```

in : `f[t, 10]`

out : $1-\frac{t^2}{4}+\frac{t^4}{576}-\frac{t^6}{518400}+\frac{t^8}{1625702400}-\frac{t^{10}}{13168189440000}$

Преобразуем выражение для коэффициента:

in : `Refine[c[n + 1] / n!, n > 0]`

out : $\frac{i\, i^{1+n}\left(-1+(-1)^{1+n}\right)}{2\,(n!)^2}$

$$\frac{i\, i^{1+n}\left(-1+(-1)^{1+n}\right)}{2\,(n!)^2}=\begin{cases}\frac{-2\,(i)^{2+n}}{2\,(n!)^2} & n=2m\\ 0 & n=2m+1\end{cases}$$

В итоге функция-оригинал имеет вид:

$$f(t)=\sum_{m=0}^{\infty}(-1)^m\frac{t^{2m}}{((2m)!)^2}$$

Разумеется, каждый раз проделывать такую утомительную последовательность действий не рационально. Составим программный обучающий модуль, который будет выдавать нам ответ в традиционной форме.

Программный модуль (обучающая программа).

Проанализируем основную последовательность необходимых действий на основе примера №13.89:

$$F(p)=\sin\left(\frac{1}{p}\right)$$

Сначала находим выражение для n+1 коэффициента:

in : `c[n_] := SeriesCoefficient[Sin[1 / p], {p, Infinity, n + 1}]`

in : `c[n]`

out : $\begin{cases}\frac{i\, i^{1+n}\left(-1+(-1)^{1+n}\right)}{2\,(1+n)!} & 1+n\geq 0\\ 0 & \text{True}\end{cases}$

Затем уходим от кусочно-заданной функции при помощи функции Refine:

in : `Refine[c[n], n > 1]`

out : $\frac{i\, i^{1+n}\left(-1+(-1)^{1+n}\right)}{2\,(1+n)!}$

Мы можем также преобразовать выражение для коэффициента с учетом того, что n - четное число:

```
in : Refine[c[n], n > 1 && Mod[n, 2] == 0] // Simplify
```

$$out: \frac{i^n}{(1+n)!}$$

Но, к сожалению, способ делать это автоматически сложен и в нем нет необходимости.

Теперь нужно домножить коэффициент на $t^n / n!$, и просуммировать от 0 до бесконечности. Но, так как на выходе мы хотим получить лишь запись в традиционной форме, воспользуемся функцией Inactivate, которая, как следует из названия, дезактивирует функцию, следующую за ней:

```
in : Inactivate[ Sum[x^n, {n, 1, 45}], Sum]
```

$$out: \sum_{n=1}^{45} x^n$$

Далее добавим опцию TraditionalForm для формульного отображения:

```
in : TraditionalForm@
      Inactivate[Sum[Refine[c[n], n > 1] t^n / n!, {n, 0, Infinity}], Sum]
```

$$out: \sum_{n=0}^{\infty} \frac{i\, i^{n+1} \left((-1)^{n+1} - 1\right) t^n}{2\, n!\, (n+1)!}$$

Итак, после того как мы изучили основные шаги, соберем это всё как и прежде в единый модуль:

Программный модуль

```
in : findOriginalS[p_, F_, t_] :=
      Module[{}, TraditionalForm@
        Inactivate[
         Sum[Refine[SeriesCoefficient[F, {p, Infinity, n + 1}] * t^n / (n) !,
           n ≥ 8], {n, 0, Infinity}], Sum]]
```

Конечно, раз мы не объявляем никаких локальных переменных, то модуль здесь просто не нужен. Его применение объясняется лишь стилистикой данной книги.

Продемонстрируем пример:

```
in : findOriginalS[p, 1/p Exp[1/p], t]
```

$$out: \sum_{n=0}^{\infty} \frac{t^n}{(n!)^2}$$

```
in : findOriginalS[p, 1/(2 p) Log[(p + 1)/(p - 1)], t]
```

out : $\sum_{n=0}^{\infty} \frac{\left((-1)^{n+1}+1\right) t^n}{2\, n\, n!}$

Также мы можем составить модуль, который будет считать сумму бесконечного сходящегося ряда:

```
in : findOriginal[p_, F_, t_] :=
      Module[{}, TraditionalForm@
        Sum[Refine[SeriesCoefficient[F, {p, Infinity, n + 1}] * t^n / (n) !,
          n ≥ 8], {n, 0, Infinity}]]
```

```
in : findOriginal[p, 1/(2 p) Log[(p + 1)/(p - 1)], t] // FullSimplify
```

out : $\mathrm{Shi}(t) - \frac{i\,\pi}{2}$

```
in : findOriginal[p, 1/p Exp[1/p], t]
```

out : $I_0\left(2\sqrt{t}\right)$

Вдаваться в подробности полученных специальных функций мы не будем. Информацию о них можно получить в учебниках по математическому анализу, или же в центре документации.

Пример №13.93

Пользуясь второй теоремой разложения, найти оригинал для заданной функции:

$$F(p) = \frac{p}{p^2 + 4\,p + 5}$$

Согласно второй теореме разложения (1), оригинал функции найдем, как сумму вычетов функции $e^{p\,t}\,F(p)$.

Для этого найдем полюса исходной функции $F(p)$:

```
in : F[p_] := p / (p^2 + 4 p + 5)
```

```
in : sol = p /. Solve[Denominator[F[p]] == 0, p]
```

out : {-2 - i, -2 + i}

Теперь, чтобы найти функцию-оригинал, просуммируем вычеты:

```
in : Sum[Residue[Exp[t p] F[p], {p, sol[[i]]}], {i, 1, Length[sol]}] //
      FullSimplify
```

$out:\ e^{-2t}(\mathrm{Cos}[t] - 2\,\mathrm{Sin}[t])$

Получили, что:

$$\frac{p}{p^2 + 4\,p + 5} \doteq e^{-2\,t}(\cos(t) - 2\sin(t))$$

Программный модуль

Анализируя несложную последовательность действий,на основе опыта предыдущей задачи составим программный модуль:

```
in : findOriginalR[p_, F_, t_] := Module[{sol},
       sol = p /. Solve[Denominator[F] == 0, p];
       FullSimplify@Sum[Residue[Exp[p t] F, {p, sol[[i]]}],
         {i, 1, Length[sol]}]]
```

Продемонстрируем пример работы:

Пример 13.94

```
in : findOriginalR[p, (p + 2) / ((p + 1) (p - 2) (p^2 + 4)), t]
```

$$out:\ \frac{1}{30}(-3\,\mathrm{Cos}[2\,t] - 2\,\mathrm{Cosh}[t] + 5\,\mathrm{Cosh}[2\,t] - 6\,\mathrm{Sin}[2\,t] + 2\,\mathrm{Sinh}[t] + 5\,\mathrm{Sinh}[2\,t])$$

Пример 13.99

```
in : findOriginalR[p, p^5 / (p^6 - 1), t]
```

$$out:\ \frac{1}{3}\left(2\,\mathrm{Cos}\left[\frac{\sqrt{3}\,t}{2}\right]\mathrm{Cosh}\left[\frac{t}{2}\right] + \mathrm{Cosh}[t]\right)$$

Пример 13.104

```
in : findOriginalR[p, p^3 / ((p^4 - 1) (p^4 + 4)), t]
```

$$out:\ \frac{1}{10}(\mathrm{Cos}[t] + \mathrm{Cosh}[t] - 2\,\mathrm{Cos}[t]\,\mathrm{Cosh}[t])$$

Встроенные средства

Mathematica имеет встроенную функцию обратного преобразования InverseLaplaceTransofrm, которая имеет схожую структуру с функцией LaplaceTransform.

Рассмотрим примеры применения:

Пример 13.80

in : InverseLaplaceTransform$\left[\frac{p}{(p^2-4)(p^2+1)}, p, t\right]$

out : $\frac{e^{-2t}}{10} + \frac{e^{2t}}{10} - \frac{\mathrm{Cos}[t]}{5}$

Пример 13.86

in : InverseLaplaceTransform$\left[\frac{1}{p-2} + \frac{\mathrm{Exp}[-p]}{p} + 3\,\frac{\mathrm{Exp}[-4\,p]}{p^2+9}, p, t\right]$

out : e^{2t} + HeavisideTheta[-1 + t] + HeavisideTheta[-4 + t] Sin[3 (-4 + t)]

Пример 13.89

in : InverseLaplaceTransform$\left[\frac{1}{p}\,\mathrm{Exp}\left[1/p^2\right], p, t\right]$

out : InverseLaplaceTransform$\left[\frac{e^{\frac{1}{p^2}}}{p}, p, t\right]$

Пример 13.100

in : InverseLaplaceTransform$\left[\frac{p^3}{(p^2+1)^3}, p, t\right]$

out : $\frac{1}{8}\left(t^2\,\mathrm{Cos}[t] + 3\,t\,\mathrm{Sin}[t]\right)$

Как мы видим, функция InverseLaplaceTransofrm работает практически со всеми видами функций-изображений и имеет широкую область применений, но в сложных случаях (13.89) не дает результата. В дальнейшем при необходимости поиска оригинала изображения мы будем пользоваться ей, если это, конечно, приведет к положительному результату.

§3. Применение операционного исчисления

3.1. Решение линейных дифференциальных уравнений и систем уравнений с постоянными коэффициентами.

Для того чтобы найти решение $x(t)$ линейного дифференциального уравнения с постоянными коэффициентами

$$x^{(n)} + a_1\,x^{(n-1)} + \ldots + a_n\,x = f(t) \quad (1)$$

(где $f(t)$ - оригинал), удовлетворяющее начальным условиям

$$x(0)=x_0,\ x'(0)=(x')_0,\ \ldots,\ x^{(n-1)}(0)=\left(x^{(n-1)}\right)_0 \qquad (2)$$

следует применить к обеим частям этого уравнения преобразование Лапласа, т.е. от уравнения (1) с условиями (2) перейти к операторному уравнению

$$\left(p^n + a_1\,p^{n-1} + \ldots + a_n\right) X(p) + Q(p) = F(p),$$

где $X\,(p)$- изображение искомого решения, $F\,(p)$- изображение функции $f\,(t)$, а $Q\,(p)$ - некоторый многочлен, коэффициенты которого зависят от начальных данных (2) и который тождественно равен нулю, если $x_0=(x')_0=.\,..=\left(x^{(n-1)}\right)_0=0$. Решим операторное уравнение относительно $X\,(p)$:

$$X(p)=\frac{F(p)-Q(p)}{L(p)}$$

$$L(p)=p^n+a_1\,p^{n-1}+\ldots+a_n - \text{ характеристический многочлен}$$

и найдя оригинал для $X\,(p)$, мы получим искомое решение $x\,(t)$. Если считать $x_0,\ (x')_0,\ \ldots,\ \left(x^{(n-1)}\right)_0$ произвольными постоянными, то найденное решение будет общим решением уравнения (1). Совершенно аналогично решаются и системы линейных дифференциальных уравнений с постоянными коэффициентами. Отличие будет лишь в том, что вместо одного операторного уравнения мы получим систему таких уравнений, которые будут линейными относительно изображений искомых функций.

Пример №13.105

Найти общее решение дифференциального уравнения:

$$x''+9\,x=\cos 3\,t$$

Найдем изображение этого уравнения при помощи модуля laplaceD из Главы 3 Параграфа 1:

```
in : dif = laplaceD[t, x''[t] + 9 x[t] == Cos[3 t], {}]
```

$$out:\ -p\,x[0]+\left(9+p^2\right)X[p]-x'[0]==\frac{p}{9+p^2}$$

Мы получили изображение исходного уравнения, выразим в нём искомую функцию X(p):

```
in : Solve[dif, X[p]]
```

$$out:\ \left\{\left\{X[p]\to\frac{p+9\,p\,x[0]+p^3\,x[0]+9\,x'[0]+p^2\,x'[0]}{\left(9+p^2\right)^2}\right\}\right\}$$

Выше мы нашли выражение для изображения искомой функции, теперь найдем саму функцию:

```
in : InverseLaplaceTransform[(p + 9 p x[0] + p^3 x[0] + 9 x′[0] + p^2 x′[0])/(9 + p^2)^2,
      p, t]

out : 1/6 (t Sin[3 t] + 6 Cos[3 t] x[0] + 2 Sin[3 t] x′[0])
```

Итак, мы получили решение дифференциального уравнения:

$$x(t) = \frac{1}{6}\,(2\sin(3\,t)\,x'(0) + 6\,x(0)\cos(3\,t) + t\sin(3\,t))$$

Обозначив начальные условия $x\,(0) = C_1$, $x'\,(0)\,/\,3 = C_2$, получим общее решение в привычной форме:

$$x(t) = \mathrm{C}_1\cos(3\,t) + \mathrm{C}_2\sin(3\,t)\ + \frac{1}{6}\,t\sin(3\,t)$$

Пример №13.111

Найти решение дифференциального уравнения при заданных начальных условиях:

$$x'' + 2\,x' + x = e^{-t},\ x(0) = 1,\ x'\,(0) = 0$$

Найдем изображение:

```
in : dif = laplaceD[t, x''[t] + 2 x'[t] + x[t] == Exp[-t],
        {x[0] → 1, x'[0] → 0}]

out : -2 - p + (1 + 2 p + p^2) X[p] == 1/(1 + p)
```

Получим выражение для изображения:

```
in : Solve[dif, X[p]]

out : {{X[p] → (3 + 3 p + p^2)/(1 + p)^3}}
```

Теперь найдем решение:

```
in : InverseLaplaceTransform[(3 + 3 p + p^2)/(1 + p)^3, p, t]

out : 1/2 e^-t (2 + 2 t + t^2)
```

Решение исходного дифференциального уравнения:

$$x(t) = \frac{1}{2}\,e^{-t}\left(2 + 2\,t + t^2\right)$$

Программный модуль

Составим простой программный модуль для решения дифференциального уравнения:

```
in : dsolve[t_, expr_, rule_] := Module[{im, xp},
       im = LaplaceTransform[expr, t, p] /.
          LaplaceTransform[x[t], t, p] → X[p] /. rule;
       xp = X[p] /. Flatten@Solve[im, X[p]];
       InverseLaplaceTransform[xp, p, t] // Simplify
      ]
```

Пример №13.117

Найти решение дифференциального уравнения:

$$x^{\mathrm{IV}} - x = \operatorname{sh} t,\ \ x(0) = x'(0) = x''(0) = 0,\quad x'''(0) = 1$$

Решим задачу при помощи созданного нами модуля:

```
in : dsolve[t, x''''[t] - x[t] == Sinh[t],
       {x[0] → 0, x'[0] → 0, x''[0] → 0, x'''[0] → 1}] // FullSimplify
```

$$out:\ \frac{1}{4}\,(t\,\mathrm{Cosh}[t] - \mathrm{Sin}[t])$$

Итак, получаем ответ:

$$x(t) = \frac{1}{4}\,(t\,\operatorname{ch}(t) - \sin(t))$$

Пример №13.121

Найти при нулевых начальных условиях решение следующего дифференциального уравнения:

$$x'' + x' = f(t),\ \text{где } f(t) = \begin{cases} \cos t & 0 \le t < \pi \\ 0 & t \ge \pi \end{cases}$$

Сначала запишем функцию $f\,(t)$:

```
in : f[t_] := Piecewise[{{Cos[t], 0 ≤ t < π}, {0, t ≥ π}}]
```

Теперь найдем решение:

```
in : dsolve[t, x''[t] + x'[t] == f[t], {x[0] → 0, x'[0] → 0}] //
      FullSimplify
```

$$out:\ \frac{1}{2}\,e^{-t}\left(1 + e^{\pi}\,\mathrm{HeavisideTheta}[-\pi + t] + e^{t}\,(-1 + \mathrm{HeavisideTheta}[-\pi + t])\,(\mathrm{Cos}[t] - \mathrm{Sin}[t])\right)$$

Таким образом, ответ:

$$x(t) = \frac{1}{2} e^{-t} (e^{\pi} H(t-\pi) + e^{t} (H(t-\pi) - 1) (\cos(t) - \sin(t)) + 1),$$

Теперь рассмотрим решение систем дифференциальных уравнений.

Пример №13.129

Найти общее решение системы дифференциальных уравнений:

$$\begin{cases} x'' + y' = t \\ y'' - x' = 0 \end{cases}$$

Принцип решения не изменился: находим изображение системы дифференциальных уравнений, затем решение для функции-изображения, а потом саму функцию-оригинал.

Чтобы найти изображение системы применим функцию LaplaceTransofrorm, которая обладает атрибутом Listable, то есть применяется для каждого элемента списка.

```
in : LaplaceTransform[{x''[t] + y'[t] == t, y''[t] - x'[t] == 0},
      t, p]
```

```
out : {p^2 LaplaceTransform[x[t], t, p] +
        p LaplaceTransform[y[t], t, p] - p x[0] - y[0] - x′[0] == 1/p^2,
      -p LaplaceTransform[x[t], t, p] + p^2 LaplaceTransform[y[t], t, p] +
        x[0] - p y[0] - y′[0] == 0}
```

Получили систему уравнений изображений. Произведем необходимую замену:

```
in : dif = LaplaceTransform[{x''[t] + y'[t] == t, y''[t] - x'[t] == 0},
       t, p] /. {LaplaceTransform[x[t], t, p] → X[p],
       LaplaceTransform[y[t], t, p] → Y[p]}
```

```
out : {-p x[0] + p^2 X[p] - y[0] + p Y[p] - x′[0] == 1/p^2,
      x[0] - p X[p] - p y[0] + p^2 Y[p] - y′[0] == 0}
```

Теперь необходимо решить уравнения относительно $X(p)$ и $Y(p)$:

```
in : Solve[dif, {X[p], Y[p]}]
```

```
out : {{X[p] → -((-1 - p x[0] - p^3 x[0] - p^2 x′[0] + p y′[0]) / (p^2 (1 + p^2))),
       Y[p] → -((-1 - p^2 y[0] - p^4 y[0] - p^2 x′[0] - p^3 y′[0]) / (p^3 (1 + p^2)))}}
```

а также найти выражения для функций оригиналов:

```
in : InverseLaplaceTransform[-(-1 - p x[0] - p^3 x[0] - p^2 x′[0] + p y′[0])/(p^2 (1 + p^2)),
      p, t]
     InverseLaplaceTransform[
      -((-1 - p^2 y[0] - p^4 y[0] - p^2 x′[0] - p^3 y′[0]) / (p^3 (1 + p^2))),
      p, t]

out : t - Sin[t] + x[0] + Sin[t] x′[0] - y′[0] + Cos[t] y′[0]

out : -1 + t^2/2 + Cos[t] + y[0] + x′[0] - Cos[t] x′[0] + Sin[t] y′[0]
```

где $x(0),\ y(0),\ x'(0),\ y'(0)$ - произвольные константы.

Решение системы:

$$x(t) = \sin(t)\,x'(0) + \cos(t)\,y'(0) + t - \sin(t) + x(0) - y'(0)$$

$$y(t) = \frac{t^2}{2} - \cos(t)\,x'(0) + \sin(t)\,y'(0) + \cos(t) + x'(0) + y(0) - 1$$

Пример №13.133

Найти решение системы дифференциальных уравнений при заданных начальных условиях:

$$\begin{cases} x'' - y' = 0 \\ x - y'' = 2\sin t \end{cases}, \quad x(0) = -1,\ x'(0) = y(0) = y'(0) = 1$$

Запишем список начальных условий:

```
in : cond = {x[0] → -1, y'[0] → 1, y[0] → 1, x'[0] → 1};
```

Найдем изображение системы:

```
in : dif =
      LaplaceTransform[{x''[t] - y'[t] == 0, x[t] - y''[t] == 2 Sin[t]},
         t, p] /. {LaplaceTransform[x[t], t, p] → X[p],
         LaplaceTransform[y[t], t, p] → Y[p]} /. cond

out : {p + p^2 X[p] - p Y[p] == 0, 1 + p + X[p] - p^2 Y[p] == 2/(1 + p^2)}
```

Найдем решение относительно функций оригиналов:

```
in : Solve[dif, {X[p], Y[p]}]

out : {{X[p] → -(-1 + p)/(1 + p^2), Y[p] → -(-1 - p)/(1 + p^2)}}
```

Теперь можно получить решение исходной системы:

```
in : InverseLaplaceTransform[-(-1 + p)/(1 + p^2), p, t]
     InverseLaplaceTransform[-(-1 - p)/(1 + p^2), p, t]

out : -Cos[t] + Sin[t]

out : Cos[t] + Sin[t]
```

Таким образом, ответ:

$$x(t) = -\cos(t) + \sin(t)$$
$$y(t) = \cos(t) + \sin(t)$$

Программный модуль

Мы можем составить программный модуль для решения систем дифференциальных уравнений с двумя неизвестными функциями:

```
in : dssolve[t_, expr_, rules_] := Module[{zam, im, sol},
       zam = {LaplaceTransform[x[t], t, p] → X[p],
         LaplaceTransform[y[t], t, p] → Y[p]};
       im = LaplaceTransform[expr, t, p] /. zam /. rules;
       sol = Flatten@Solve[im, {X[p], Y[p]}];
       Print["x[t]=",
        Simplify@InverseLaplaceTransform[X[p] /. sol, p, t]];
       Print["y[t]=",
        Simplify@InverseLaplaceTransform[Y[p] /. sol, p, t]];]
```

Рассмотрим пример работы:

Пример №13.135

Найти решение системы дифференциальных уравнений при заданных начальных условиях:

$$\begin{cases} x'' - y' = e^t \\ x' + y'' - y = 0 \end{cases}, \quad x(0) = 1,\ y(0) = -1,\ x'(0) = y'(0) = 0$$

Воспользуемся только что написанным модулем для поиска решения:

```
in : dssolve[t, {x''[t] - y'[t] == Exp[t], x'[t] + y''[t] - y[t] == 0},
      {x[0] → 1, y[0] → -1, x'[0] → 0, y'[0] → 0}]
```

$$x[t] = \left(1 + \frac{t^2}{2}\right)$$

$$y[t] = \left(-e^t + t\right)$$

Получили ответ:

$$x(t) = 1 + \frac{t^2}{2}$$

$$y(t) = -e^t + t$$

Пример №13.137

Проинтегрировать при нулевых начальных условиях систему дифференциальных уравнений:

$$\begin{cases} x'' - y' = f_1(t) \\ y' + x = f_2(t) \end{cases}, \quad f_1(t) = \begin{cases} 1 & 0 \le t < \pi \\ 0 & t \ge \pi \end{cases}, \quad f_2(t) = \begin{cases} t & 0 \le t < \pi/2 \\ \pi - t & \pi/2 \le t < \pi \\ 0 & t \ge \pi \end{cases}$$

Запишем функции f_1 и f_2:

```
in : f1[t_] := Piecewise[{{1, 0 ≤ t < Pi}}]
     f2[t_] := Piecewise[{{t, 0 ≤ t < Pi / 2}, {Pi - t, Pi / 2 ≤ t < Pi}}]
```

Получим их графики:

```
in : Plot[{f1[t], f2[t]}, {t, -1, 4}]
```



рис. 3.1.1

Теперь можно найти решение:

```
in : dssolve[t, {x''[t] - y'[t] == f1[t], x[t] + y'[t] == f2[t]},
      {x[0] → 0, y[0] → 0, x'[0] → 0, y'[0] → 0}]
```

$$x[t] = \left(1 + t - \text{Cos}[t] + (\pi - 2t - 2\,\text{Cos}[t])\,\text{HeavisideTheta}\left[-\frac{\pi}{2} + t\right] - \text{Sin}[t] + \text{HeavisideTheta}[-\pi + t]\,(-1 - \pi + t - \text{Cos}[t] + \text{Sin}[t])\right)$$

$$y[t] = \left(1 - t - \text{Cos}[t] + 2\,\text{HeavisideTheta}\left[-\frac{\pi}{2} + t\right]\,(-1 + \text{Sin}[t]) + \text{Sin}[t] + \text{HeavisideTheta}[-\pi + t]\,(1 - \pi + t + \text{Cos}[t] + \text{Sin}[t])\right)$$

Примечание: можно было бы составить модуль для решения систем с произвольным количеством неизвестных функций, но это слишком трудоемко и выходит за рамки данного курса.

3.2. Решение линейных интегральных и

интегро-дифференциальных уравнений.

Используя теорему свертывания, можно легко найти изображения решений интегральных уравнений Вольтерра 1-ого и 2-ого рода в том случае, когда ядром в соответствующем уравнении служит функция вида $K(t-\tau)$, где $K(t)$ - оригинал. Этот метод применим и к интегро-дифференциальных уравнениям с таким же ядром.

Пример №13.139

Решить следующее интегральное уравнение:

$$\int_0^t \operatorname{ch}(t-\tau)\, x(\tau)\, d\tau = \operatorname{ch} t - \cos t$$

Метод решения подобных уравнений ничем не отличается от рассмотренного ранее метода решения дифференциальных уравнений.

Напомним, что по теореме свертывания:

$$\int_0^t f(t-\tau)\, x(\tau)\, d\tau = (f * x)(t) \doteq F(p)\cdot X(p)$$

Найдем изображение исходного уравнения:

```
in : LaplaceTransform[Integrate[Cosh[t - τ] x[τ], {τ, 0, t}] ==
       Cosh[ t] - Cos [t], t, p]
```

out : $\dfrac{p\,\text{LaplaceTransform}[x[t],t,p]}{-1+p^2} == \dfrac{p}{-1+p^2} - \dfrac{p}{1+p^2}$

Как и раньше, обозначим LaplaceTransform[$x[t]$, t, p] как $X[p]$.

```
in : dif =
      LaplaceTransform[Integrate[Cosh[t - τ] x[τ], {τ, 0, t}] ==
         Cosh[ t] - Cos [t], t, p] /. LaplaceTransform[x[t], t, p] → X[p]
```

out : $\dfrac{p\,X[p]}{-1+p^2} == \dfrac{p}{-1+p^2} - \dfrac{p}{1+p^2}$

Теперь из этого уравнения необходимо выразить функцию $X(p)$:

```
in : Solve[dif, X[p]]
```

out : $\left\{\left\{X[p] \to \dfrac{2}{1+p^2}\right\}\right\}$

Зная выражение для изображения, найдем функцию-решение исходного интегрального уравнения:

```
in : InverseLaplaceTransform[2/(1 + p^2), p, t]

out : 2 Sin[t]
```

Итак, ответ:

$$x(t) = 2\sin t$$

Если посмотреть внимательно, то можно убедиться, что последовательность действий, примененная для решения интегрального уравнения, полностью совпадает с последовательностью решения дифференциальных уравнений, следовательно, мы можем воспользоваться написанным нами ранее модулем dsolve с пустыми начальными условиями. Убедимся в этом:

```
in : dsolve[t, Integrate[Cosh[t - τ] x[τ], {τ, 0, t}] == Cosh[ t] - Cos [t],
      {}]

out : 2 Sin[t]
```

Получили, как и ожидалось, тот же самый ответ. Следовательно, данный модуль можно в каком-то роде назвать универсальным.

Пример №13.141

Решить интегро-дифференциальное уравнение:

$$\int_0^t e^{t-\tau}\sin(t-\tau)\,x(\tau)\,d\tau = x'' - x' + e^t(1-\cos t),\ \ x(0) = x'(0) = 1$$

Данное уравнение содержит и интеграл и дифференциал неизвестной функции, поэтому называется интегро-дифференциальным уравнением.

Однако, способ решения и такого уравнения ничем не отличается от рассмотренного ранее.

Поэтому воспользуемся модулем dsolve:

```
in : dsolve[t, Integrate[Exp[t - τ] Sin[t - τ] x[τ], {τ, 0, t}] ==
       x''[t] - x'[t] + Exp[t] (1 - Cos[t]), {x[0] → 1, x'[0] → 1}]

out : e^t
```

Убедимся, что полученный ответ верен, подставив полученную функцию в исходное уравнение:

```
in : Integrate[Exp[t - τ] Sin[t - τ] x[τ], {τ, 0, t}] ==
       D[x[t], {x, 2}] - D[x[t], x] + Exp[t] (1 - Cos[t]) /.
      x[t_] → Exp[t]

out : -e^t (-1 + Cos[t]) == e^t (1 - Cos[t])
```

Функция удовлетворяет уравнению, следовательно получено верное решение.

Ответ:

$$x(t) = e^t$$

Пример №13.143

Проинтегрировать уравнение Абеля:

$$\int_0^t \frac{x(\tau)}{\sqrt{t-\tau}}\, d\tau = \pi$$

Разумеется, и данное уравнение входит в класс задач, решаемых с помощью функции dsolve. Воспользуемся ей:

```
in : dsolve[t, Integrate[x[τ] (t - τ)^(-1/2), {τ, 0, t}] == Pi, {}]
```

out : $\frac{1}{\sqrt{t}}$

Получили, что решением уравнения Абеля является функция:

$$x(t) = \frac{1}{\sqrt{t}}$$

Выполним проверку, подставив эту функцию в исходное уравнение:

```
in : Integrate[x[τ] (t - τ)^(-1/2), {τ, 0, t}] == Pi /. x[t_] → 1/√t
out : ConditionalExpression[True, Re[t] > 0 && Im[t] == 0]
```

Получили верное равенство при условии, что действительная часть *t* больше нуля, а мнимая часть равна нулю.

Следует сказать, что данное уравнение является частным случаем следующего:

Пример №13.144

Проинтегрировать уравнение Абеля:

$$\int_0^t \frac{x(\tau)}{(t-\tau)^\alpha}\, d\tau = t^\beta,\ 0 \le \alpha < 1,\ \ \beta > -1$$

Решим его с помощью функции dsolve:

```
in : dsolve[t, Integrate[x[τ] (t - τ)^(-α), {τ, 0, t}] == t^β, {}]
```

out : $\frac{t^{-1+\alpha+\beta}\,\mathrm{Gamma}[1+\beta]}{\mathrm{Gamma}[1-\alpha]\,\mathrm{Gamma}[\alpha+\beta]}$

Решение запишем в традиционной форме:

$$x(t) = \frac{\Gamma(1+\beta)}{\Gamma(1-\alpha)}\,\frac{t^{-1+\alpha+\beta}}{\Gamma(\alpha+\beta)}$$

где Γ(t) - гамма-функция.

Убедимся в правильности решения:

```
in : Integrate[x[τ] (t - τ)^-α, {τ, 0, t}] == t^β /.
       x[t_] → t^(-1+α+β) Gamma[1 + β] / (Gamma[1 - α] Gamma[α + β])

out : ConditionalExpression[True,
       Re[α] < 1 && Re[α + β] > 0 && Re[t] > 0 && Im[t] == 0]
```

Оно верно при $\alpha < 1,\ a + \beta > 0$ или $\beta > -1$, что полностью совпадает с условиями задачи.

3.3. Вычисление несобственных интегралов.

Один из способов вычисления несобственных интегралов вида $\int_0^\infty f(t)\, dt$ основан на применении теоремы операционного исчисления о связи “конечного” значения оригинала и “начального” значения изображения. Докажем эту теорему:

Теорема:

Если $f(t) \doteq F(p)$ и существует конечный предел $\lim\limits_{t\to+\infty} f(t) = f(+\infty)$, то

$$f(+\infty) = \lim_{p\to 0} p\, F(p)$$

Доказательство:

Воспользуемся теоремой подобия:

$$f(a\, t) \doteq \frac{1}{a}\, F\left(\frac{p}{a}\right)$$

Значению $f(+\infty)$ соответствует:

$$\lim_{a\to\infty} f(a\, t) = f(+\infty)$$

Тогда, согласно теореме подобия, получим:

$$\lim_{a\to+\infty} f(a\, t) = f(+\infty) = \lim_{a\to+\infty} \frac{1}{a}\, F\left(\frac{p}{a}\right)$$

введем обозначение :$p = p / a$, тогда:

$$\lim_{a\to+\infty} \frac{1}{a}\, F\left(\frac{p}{a}\right) = \lim_{p\to+0} p\, F(p)$$

Отсюда:

$$f(+\infty) = \lim_{p\to+0} p\, F(p)$$

Из этой теоремы и соотношения:

$$\int_0^t f(t)\, dt = \frac{1}{p} F(p)$$

при условии сходимости интеграла следует соотношение:

$$\int_0^{+\infty} f(t)\, dt = F(0) \tag{3}$$

Пример №13.150

Вычислить несобственный интеграл:

$$\int_0^{+\infty} t^{\mu}\, e^{-\alpha t} \ln t\, dt,\ \ \alpha,\, \mu > -1$$

Сначала попробуем напрямую посчитать значения данного интеграла, используя функцию Integrate:

```
in : Integrate[t^μ Exp[-α t] Log[t], {t, 0, Infinity},
       Assumptions → α > 0 && μ > -1]

out : -α^(-1-μ) μ Gamma[μ] (EulerGamma - HarmonicNumber[μ] + Log[α])
```

Время затраченное на вычисление данного интеграла очень велико (около 10 минут). Из этого можно сделать вывод, что использование вспомогательного метода описанного выше вполне рационально, и даже необходимо.

Найдем изображение подынтегральной функции с помощью LaplaceTransform:

```
in : i = LaplaceTransform[t^μ Exp[-α t] Log[t], t, p]

out : -(p + α)^(-1-μ) μ Gamma[μ] (EulerGamma - HarmonicNumber[μ] + Log[p + α])
```

EulerGamma- постоянная Эйлера, численно равная $\gamma \simeq 0.577216$; HarmonicNumber[μ] функция гармонического числа, которая задается:

$$\text{HarmonicNumber}[\mu] = H_{\mu} = \sum_{i=1}^{\mu} \frac{1}{i}$$

Итак, мы получили изображение подынтегральной функции. Теперь, чтобы получить значение несобственного интеграла, просто подставим $p = 0$:

```
in : i /. p → 0

out : -α^(-1-μ) μ Gamma[μ] (EulerGamma - HarmonicNumber[μ] + Log[α])
```

Данный ответ сходится с тем, что мы получили, воспользовавшись функцией Integrate.

Пусть функции

$$f(t,u) \text{ и } \psi(t,u) = \int_0^{+\infty} \phi(u)\, f(t,u)\, du$$

являются оригиналами. Тогда, применяя теорему об интегрировании по параметру, будем иметь:

$$\psi(t) \doteq \Psi(p) = \int_0^{+\infty} \phi(u)\, F(p,u)\, du$$

Поэтому, если интеграл определяющий $\Psi(p)$, можно вычислить, то для отыскания интеграла ψ(t) достаточно найти оригинал для $\Psi(p)$,т.е.

$$\int_0^{+\infty} \phi(u)\, f(t,u)\, du \doteq \int_0^{+\infty} \phi(u)\, F(p,u)\, du \tag{4}$$

Пример №13.152

Вычислить несобственный интеграл:

$$\int_0^{+\infty} e^{-t u^2}\, du$$

В данной задаче имеем:

$$\phi(u) = 1,\ \ f(t,u) = e^{-t u^2}$$

Чтобы вычислить значение интеграла, найдем изображение функции f:

```
in : LaplaceTransform[Exp[-t u^2], t, p]
```

$$out: \frac{1}{p + u^2}$$

Теперь найдем значение следующего интеграла:

$$\Psi(p) = \int_0^{+\infty} \frac{1}{p + u^2}\, du$$

Для этого воспользуемся функцией Integrate:

```
in : Integrate[1/(p + u^2), {u, 0, Infinity}]
```

$$out: \text{ConditionalExpression}\left[\frac{\pi}{2\sqrt{p}},\ \text{Im}[p] \neq 0 \ || \ \text{Re}[p] \geq 0\right]$$

Теперь просто найдем оригинал получившейся функции Ψ:

```
in : InverseLaplaceTransform[π/(2 Sqrt[p]), p, t]
```

```
out : Sqrt[π]/(2 Sqrt[t])
```

Получили, что:

$$\int_0^{+\infty} e^{-t u^2} \, du = \frac{\sqrt{\pi}}{2 \sqrt{t}}$$

Проверим ответ, вычислив значение первоначального интеграла с помощью функции Integrate:

```
in : Integrate[Exp[-t u^2], {u, 0, Infinity}]
```

```
out : ConditionalExpression[Sqrt[π]/(2 Sqrt[t]), Re[t] > 0]
```

Ответ получили тот же, следовательно полученное решение верно.

Т е о р е м а П а р с е в а л я. Если

$f_1(t) \doteq F_1(p),\ f_2(t) \doteq F_2(p)$ и $F_1(p),\ F_2(p)$ аналитичны при $\operatorname{Re} p \geq 0$, то

$$\int_0^{\infty} f_1(u) F_2(u) \, du = \int_0^{\infty} F_1(v) f_2(v) \, dv \tag{5}$$

Пример №13.153

Вычислить несобственный интеграл, используя теорему Парсеваля:

$$\int_0^{+\infty} \frac{e^{-\alpha u} - e^{-\beta u}}{\sqrt{u}} \, du$$

Введем следующие обозначения:

$$f_1(u) = e^{-\alpha u} - e^{-\beta u},\ F_2(u) = \frac{1}{\sqrt{u}}$$

Тогда, согласно теореме Парсеваля (5), для того чтобы вычислить значение исходного интеграла, необходимо найти изображение функции f_1 и оригинал функции F_2. Сделаем это при помощи функций LaplaceTransform, и InverseLaplaceTransform:

```
in : LaplaceTransform[Exp[-α u] - Exp[-β u], u, v]
```

```
out : 1/(v + α) - 1/(v + β)
```

$$F_1(v) = \frac{1}{v+\alpha} - \frac{1}{v+\beta}$$

```
in : InverseLaplaceTransform[1/Sqrt[u], u, v]
```

```
out : 1/(Sqrt[π] Sqrt[v])
```

$$f_2(v) = \frac{1}{\sqrt{\pi}\ \sqrt{v}}$$

Теперь найдем значение исходного интеграла по формуле (5):

```
in : Integrate[1/(Sqrt[π] Sqrt[v]) (1/(v + α) - 1/(v + β)), {v, 0, Infinity},
        Assumptions → α > 0 && β > 0]
```

```
out : Sqrt[π] (1/Sqrt[α] - 1/Sqrt[β])
```

Получили ответ:

$$\int_0^{+\infty} \frac{e^{-\alpha u} - e^{-\beta u}}{\sqrt{u}}\, du =$$

$$\frac{1}{\sqrt{\pi}} \int_0^{+\infty} \frac{1}{\sqrt{v}} \left(\frac{1}{v+\alpha} - \frac{1}{v+\beta} \right) du = \sqrt{\pi} \left(\frac{1}{\sqrt{\alpha}} - \frac{1}{\sqrt{\beta}} \right)$$

3.4. Суммирование рядов.

Методы операционного исчисления могут быт использованы при суммировании числовых и функциональных рядов.

Пусть $F(p)$ является изображением для $f(t)$ (область аналитичности $F(p)$: $\operatorname{Re} p \geqslant k$), тогда сумма ряда S

$$S = \sum_{n=k}^{\infty} (\pm 1)^n F(n)$$

может быть найдена по формуле:

$$S = (\pm 1)^k \int_0^{\infty} \frac{e^{-k t} f(t)}{1 \mp e^{-t}}\, dt \tag{6}$$

Докажем это.

По условию:

$$F(p)=\int_0^\infty e^{-p\,t}\,f(t)\,dt \text{ (по определению преобразования Лапласа)}$$

Имеем:

$$\frac{(\pm 1)^k\,e^{-k\,t}}{1\mp e^{-t}}=\sum_{n=k}^{\infty}(\pm 1)^n\,e^{-n\,t}$$

```
in : Sum[(-1)^n Exp[- n t], {n, k, Infinity}]
```

$$out:\ \frac{(-1)^k\,e^{t-k\,t}}{1+e^t}$$

Поэтому:

$$(\pm 1)^k\int_0^{+\infty}\frac{e^{-k\,t}\,f(t)}{1\mp e^{-t}}\,dt=\int_0^{+\infty}f(t)\sum_{n=k}^{\infty}(\pm 1)^n\,e^{-n\,t}\,dt=$$

$$=\sum_{n=k}^{\infty}(\pm 1)^n\int_0^{+\infty}f(t)\,e^{-n\,t}\,dt=\sum_{n=k}^{\infty}(\pm 1)^n\,F(n)$$

Пример №13.156

Используя формулу (6), найти сумму числового ряда:

$$\sum_{n=1}^{\infty}\frac{n}{\left(n^2-\frac{1}{4}\right)^2}$$

Данный ряд имеет следующую структуру:

$$\sum_{n=1}^{\infty}(+1)^n\,F(n),\ \text{ где } F(n)=\frac{n}{\left(n^2-\frac{1}{4}\right)^2}$$

Тогда, для того чтобы вычислить его сумму, необходимо найти оригинал функции $F(n)$. Сделаем это, обратившись к функции InverseLaplaceTransofrm:

```
in : InverseLaplaceTransform[n/(n^2 - 1/4)^2, n, t]
```

$$out:\ \frac{1}{2}\,e^{-t/2}\left(-1+e^t\right)t$$

Итак, получили, что:

$$f(t)=\frac{1}{2}\,e^{-t/2}\,(-1+e^t)\,t$$

Теперь вычислим сумму ряда по формуле (6):

$$\sum_{n=1}^{\infty} \frac{n}{\left(n^2 - \frac{1}{4}\right)^2} = (+1)^1 \int_0^{+\infty} f(t)\, \frac{e^{-1\,t}}{1 - e^{-t}}\, dt$$

```
in : Integrate[1/2 e^(-t/2) (-1 + e^t) t Exp[-t]/(1 - Exp[-t]), {t, 0, Infinity}]

out : 2
```

Сумма исходного ряда равна:

$$\sum_{n=1}^{\infty} \frac{n}{\left(n^2 - \frac{1}{4}\right)^2} = 2$$

Убедимся в этом, вычислив сумму ряда напрямую:

```
in : Sum[n/(n^2 - 1/4)^2, {n, 1, Infinity}]

out : 2
```

Получили тот же ответ.

Пусть $f\,(t)$ является оригиналом функции $F(p)$ (область аналитичности $F\,(p)$: $\operatorname{Re} p \geqslant 0$). Пусть, кроме того, $\Phi\,(t,\, x)$ - производящая функция бесконечной последовательности функции $\phi_n(x)$:

$$\Phi(t,\, x) = \sum_{n=0}^{\infty} \phi_n(x)\, t^n$$

Сумма $S\,(x)$ сходящегося на $[a,\, b]$ функционального ряда:

$$\sum_{n=0}^{\infty} F(n)\, \phi_n(x)$$

может быть найдена по формуле:

$$S(x) = \int_0^{+\infty} f(t)\, \Phi(e^{-t},\, x)\, dt \qquad (7)$$

Докажем это:

$$\int_0^{+\infty} f(t)\, \Phi(e^{-t},\, x)\, dt = \int_0^{+\infty} f(t) \sum_{n=0}^{\infty} \phi_n(x)\, e^{-n\,t}\, dt =$$

$$= \sum_{n=0}^{\infty} \phi_n(x) \int_0^{+\infty} f(t)\, e^{-n\,t}\, dt = \sum_{n=0}^{\infty} \phi_n(x)\, F(n) = S(x)$$

Пример №13.160

Используя формулу (7), с помощью подходящей производящей функции просуммировать следующий ряд:

$$\sum_{n=0}^{\infty}(-1)^n \frac{x^{2n+1}}{2n+1}$$

Главная задача при использовании формулы (7) - выбрать наиболее удобную производящую функцию. Пусть Φ(t,x):

$$\Phi(t, x) = \sum_{n=0}^{\infty}(-1)^n x^{2n+1} t^n = x\sum_{n=0}^{\infty}(-1)^n x^{2n} t^n = x\sum_{n=0}^{\infty}(-1)^n \left(x^2 t\right)^n$$

$$\sum_{n=0}^{\infty}(-1)^n \left(x^2 t\right)^n = \frac{1}{1+x^2 t} \; - \text{ сумма геометрической прогрессии.}$$

В итоге:

$$\Phi(t, x) = \frac{x}{1+x^2 t}$$

Запишем функцию Φ от t и x:

```
in : Φ[t_, x_] := x/(1 + x^2 t)
```

Теперь найдем оригинал функции $F(n)$, которая в нашем случае равна:

$$F(n) = \frac{1}{2n+1}$$

Воспользуемся InverseLaplaceTransform:

```
in : InverseLaplaceTransform[1/(2 n + 1), n, t]

out : e^(-t/2)/2
```

Получили, что оригинал функции $F(p)$:

$$f(t) = \frac{e^{-t/2}}{2}$$

Осталось лишь по формуле (7) вычислить интеграл:

```
in : Integrate[e^(-t/2)/2 Φ[Exp[-t], x], {t, 0, Infinity},
       Assumptions → -1 < x < 1]

out : ArcTan[x]
```

Итак, получили, что сумма исходного ряда равна:

$$\sum_{n=0}^{\infty}(-1)^n \frac{x^{2n+1}}{2n+1} = \arctan x$$

Воспользуемся функцией Sum, чтобы подтвердить это:

```
in : Sum[(-1)^n x^(2 n+1)/(2 n + 1), {n, 0, Infinity}]

out : ArcTan[x]
```

Ответы совпали, поэтому будем считать, что получено верное решение.

§4. Дискретное преобразование Лапласа и его применение

4.1. Z-преобразование и дискретное преобразование Лапласа.

Z-преобразованием числовой (действительной или комплексной) бесконечной последовательности (a_n) называется функция комплексной переменной $F(z)$, определяемая при $|z| > R = \lim_{n\to\infty} \sqrt[n]{|a_n|}$ рядом Лорана

$$F(z) = \sum_{n=0}^{\infty} \frac{a_n}{z^n} \tag{1}$$

и аналитически продолженная в круг $|z| < R$. Если последовательность удовлетворяет условию $|a_n| < M\, e^{\alpha n}$ (M>0, α - постоянная), то функция $F(z)$ будет аналитической в области $|z| > e^{\alpha}$, т.е. вне круга с центром в нулевой точке и радиусом $R = e^{\alpha}$.

Формула (1) дает разложение $F(z)$ в ряд Лорана в окрестности бесконечно удаленной точки (являющейся правильной точкой $F(z)$), поэтому для восстановления последовательности (a_n) по ее Z-преобразованию надо $F(z)$ любым способом разложить в ряд Лорана в окрестности бесконечно удаленной точки; в частности, можно воспользоваться формулой для определения коэффициентов этого разложения:

$$a_n = \frac{1}{2\pi i} \int_C F(z)\, z^{n-1}\, dz \tag{2}$$

(C - контур, внутри которого лежат все особые точки функции $F(z)$)

Введем решетчатую функцию $f(n)$, полагая что $a_n = f(n)$. По-прежнему $f(n)$ удовлетворяет условию $|f(n)| < M\, e^{\alpha n}$, и примем дополнительно, что $f(n) = 0$, при $n < 0$; такие решетчатые функции будем называть дискретным оригиналом.
Дискретное преобразование Лапласа функции $f(n)$ мы получим, если в Z-преобразовании положим $z = e^q$:

$$F(q)=\sum_{n=0}^{\infty} f(n)\, e^{-n q} \tag{3}$$

Связь между дискретным оригиналом и его изображением будем обозначать:

$$f(n) \doteq F(q), \text{ либо } \mathcal{D}[f]=F(q)$$

Изображение $F(q)$ - функция комплексной переменной с периодом 2π, при этом в основной полосе $-\pi < \operatorname{Im} q < \pi$ она аналитична только при Re q > α. Таким образом, все ее особые точки лежат в этой полосе слева от прямой Re q = α.

Из формулы (3) вытекает следующая формула обращения дискретного преобразования Лапласа:

$$f(n)=\frac{1}{2\pi i}\int_{\gamma-\pi i}^{\gamma+\pi i} F(q)\, e^{n q}\, dq \tag{4}$$

Свойства дискретного преобразования Лапласа

1. Л и н е й н о с т ь.

$$\sum_{j=0}^{\infty} C_j\, f_j(n) \doteq \sum_{j=0}^{\infty} C_j\, F_j(q) \tag{5}$$

2. Ф о р м у л а с м е щ е н и я.

$$e^{\alpha n} f(n) \doteq F(q-a) \tag{6}$$

3. Ф о р м у л ы з а п а з д ы в а н и я и о п е р е ж е н и я.

$$\begin{gathered} f(n-k) \doteq e^{-k q} F(q) \\ f(n+k) \doteq e^{k q}\left(F(q)-\sum_{r=0}^{k-1} f(r)\, e^{-r q}\right) \end{gathered} \tag{7}$$

4. Д и ф ф е р е н ц и р о в а н и е п о п а р а м е т р у.

$$\text{Если } f(n, x) \doteq F(q, x), \text{ то } \frac{\partial}{\partial x} f(n, x) \doteq \frac{\partial}{\partial x} F(q, x) \tag{8}$$

5. Д и ф ф е р е н ц и р о в а н и е и и н т е г р и р о в а н и е и з о б р а ж е н и я.

$$\begin{gathered} n^k f(n) \doteq (-1)^k \frac{d^k}{dq^k} F(q) \\ \frac{f(n)}{n} \doteq \int_q^{\infty} (F(s)-f(0))\, ds \end{gathered} \tag{9}$$

6. И з о б р а ж е н и е к о н е ч н ы х р а з н о с т е й о р и г и н а л а.

$$\Delta^k f(n) \doteq (e^q - 1)^k F(q) - e^q \sum_{r=0}^{k-1} (e^q - 1)^{k-r-1} \Delta^r f(0) \tag{10}$$

7. Изображение конечных сумм оригинала.

$$\text{Если, } g(n) = \sum_{k=0}^{n-1} f(k), \text{ то } g(n) \doteq \frac{F(q)}{e^q - 1} \tag{11}$$

8. Умножение изображений

$$\text{Если } f_1(n) * f_2(n) = \sum_{r=0}^{n} f_1(r)\, f_2(n-r), \text{ то } f_1(n) * f_2(n) \doteq F_1(q) \cdot F_2(q) \tag{12}$$

Приведем список изображений основных решетчатых функций:

$$f(n) = \begin{cases} C & n = 0 \\ 0 & n \neq 0 \end{cases}, \qquad F(q) = C$$

$$\eta(n) = \begin{cases} 1 & n \geq 0 \\ 0 & n < 0 \end{cases}, \qquad F(q) = \frac{e^q}{e^q - 1}$$

$$f(n) = a^n, \qquad F(q) = \frac{e^q}{e^q - a}$$

$$f(n) = e^{\alpha n}, \qquad F(q) = \frac{e^q}{e^q - e^\alpha}$$

$$f(n) = n, \qquad F(q) = \frac{e^q}{(e^q - 1)^2}$$

$$f(n) = n^2, \qquad F(q) = \frac{e^q (e^q + 1)}{(e^q - 1)^3}$$

$$f(n) = \frac{n^{[2]}}{2!} = \frac{n(n-1)}{2} = C_n^2, \qquad F(q) = \frac{e^q}{(e^q - 1)^3}$$

$$f(n) = \frac{n^{[k]}}{k!} = \frac{n(n-1)\ldots(n-k+1)}{k!} = C_n^k, \qquad F(q) = \frac{e^q}{(e^q - 1)^{k+1}}$$

$$f(n) = \sin \beta n, \qquad F(q) = \frac{e^q \sin \beta}{e^{2q} + 1 - 2 e^q \cos \beta}$$

$$f(n) = \cos \beta n, \qquad F(q) = \frac{e^q (e^q - \cos \beta)}{e^{2q} + 1 - 2 e^q \cos \beta}$$

$$f(n) = \operatorname{sh} \beta n, \qquad F(q) = \frac{e^q \operatorname{sh} \beta}{e^{2q} + 1 - 2 e^q \operatorname{ch} \beta}$$

$$f(n) = \operatorname{ch} \beta n, \qquad F(q) = \frac{e^q (e^q - \operatorname{ch} \beta)}{e^{2q} + 1 - 2 e^q \operatorname{ch} \beta}$$

$$f(n) = \frac{n^{[k]}}{k!} e^{\alpha n} = C_n^k e^{\alpha n}, \qquad F(q) = \frac{e^{q+k\alpha}}{(e^q - e^{\alpha})^{k+1}}$$

$$f(n) = \frac{n^{[k]}}{k!} a = C_n^k a^n, \qquad F(q) = \frac{a^k e^q}{(e^q - a)^{k+1}}$$

Пример №13.174

Найти изображение следующей функции:

$$f(n) = e^{a n} \cos \beta n$$

Воспользовавшись формулой смещения,

$$e^{a n} f(n) \doteq F(q - a)$$

получим, что изображение заданной функции равно:

$$F(q) = \frac{e^{q-a}(e^{q-a} - \cos \beta)}{e^{2q-2a} + 1 - 2 e^{q-a} \cos \beta}$$

Пример №13.177

Найти изображение следующей функции:

$$f(n) = n^2 a^n$$

Согласно свойству 5 (дифференцирование изображения), имеем формулу:

$$n^2 f_1(n) \doteq \frac{d^2}{dq^2} F_1(q)$$

$$f_1(n) \doteq F_1(q), \;\; f_1(n) = a^n$$

Найдем вторую производную от функции F_1(q):

in : $D\left[\frac{\mathtt{Exp[q]}}{\mathtt{Exp[q]-a}}, \{q, 2\}\right]$ **// Simplify**

out : $\frac{a e^q (a + e^q)}{(-a + e^q)^3}$

Получаем, что:

$$f(n) = n^2 a^n \doteq \frac{a e^q (a + e^q)}{(-a + e^q)^3}$$

Разумеется каждый раз искать и комбинировать необходимые свойства слишком утомительно, да и не нужно. Воспользуемся встроенной в систему функцией ZTransform, которая, как очевидно из названия, совершает Z-преобразование заданной дискретной функции.

in : ZTransform$\left[n^2\, a^n,\, n,\, z\right]$

out : $\dfrac{a\, z\, (a+z)}{(-a+z)^3}$

Теперь произведем замену:

in : ZTransform$\left[n^2\, a^n,\, n,\, z\right]$ /. z → Exp[q]

out : $\dfrac{a\, e^q\, \left(a+e^q\right)}{\left(-a+e^q\right)^3}$

Получили тот же самый ответ. Функция ZTransform, как видно из примера, имеет ту же структуру, что и функции LaplaceTransform и InverseLaplaceTransform.

Рассмотрим еще несколько примеров:

Пример №13.180

Найти изображение следующей функции:

$$f(n)=\frac{\sin\beta\, n}{n}$$

in : ZTransform[Sin[β n] / n, n, z] /. z → Exp[q]

out : $\beta+\text{ArcTan}\left[\dfrac{\text{Sin}[\beta]}{e^q-\text{Cos}[\beta]}\right]$

Вместо замены z на e^q, можно сразу в аргументах функции ZTransform указывать не z, а e^q:

in : ZTransform[Sin[β n] / n, n, Exp[q]]

out : $\beta+\text{ArcTan}\left[\dfrac{\text{Sin}[\beta]}{e^q-\text{Cos}[\beta]}\right]$

Получили:

$$\frac{\sin\beta\, n}{n}\doteq\beta+\arctan\frac{\sin(\beta)}{e^q-\cos(\beta)}$$

Пример №13.181

Найти решетчатую функцию по её изображению:

$$F(q)=\frac{e^q}{(e^q-1)\left(e^{2\,q}-4\right)}$$

Для того чтобы найти решетчатую функцию, разложим изображение на простые дроби:

```
in : Apart[1/((Exp[q] - 1) (Exp[2 q] - 4))]
```

$$out: \frac{1}{4\left(-2+e^{q}\right)}-\frac{1}{3\left(-1+e^{q}\right)}+\frac{1}{12\left(2+e^{q}\right)}$$

Домножим на экспоненту:

```
in : Expand[Apart[1/((Exp[q] - 1) (Exp[2 q] - 4))] Exp[q]]
```

$$out: \frac{e^{q}}{4\left(-2+e^{q}\right)}-\frac{e^{q}}{3\left(-1+e^{q}\right)}+\frac{e^{q}}{12\left(2+e^{q}\right)}$$

Получили 3 простые дроби. По свойству линейности находим оригинал каждой дроби отдельно:

$$\frac{e^{q}}{4\left(-2+e^{q}\right)} \doteq \frac{1}{4} \quad 2^{n}=2^{n-2}$$

$$-\frac{e^{q}}{3\left(-1+e^{q}\right)}=-\frac{1}{3} \quad 1^{n}=-\frac{1}{3}$$

$$\frac{e^{q}}{12\left(-(-2)+e^{q}\right)}=\frac{1}{12}(-2)^{n}=\frac{1}{3}(-1)^{n} 2^{-2+n}$$

Таким образом, получили:

$$\frac{e^{q}}{4\left(-2+e^{q}\right)}-\frac{e^{q}}{3\left(-1+e^{q}\right)}+\frac{e^{q}}{12\left(2+e^{q}\right)} \doteq 2^{-2+n}-\frac{1}{3}+\frac{1}{3}(-1)^{n} 2^{-2+n}$$

Решим задачу вторым способом.

С помощью замены e^q на z и используя формулу обращения (2) сведем задачу поиска оригинала к вычислению контурного интеграла:

$$\frac{e^{q}}{\left(e^{q}-1\right)\left(e^{2 q}-4\right)} \rightarrow \frac{z}{(z-1)\left(z^{2}-4\right)}$$

$$a_{n}=\frac{1}{2 \pi i} \int_{C} \frac{z z^{n-1}}{(z-1)\left(z^{2}-4\right)} d z$$

Для этого обратимся integrateR из главы 2 параграфа 6 “Вычеты и их применение”:

```
in : integrateR[z, 1/(2 Pi I) z^n/((z - 1) (z^2 - 4)), 0, 4] // Simplify // Expand
```

$$out: -\frac{1}{3}+2^{-2+n}+\frac{1}{3}(-1)^{n} 2^{-2+n}$$

Получили тот же самый ответ.

Пример №13.182

Найти решетчатую функцию по её изображению:

$$F(q) = \frac{e^{q}}{\left(e^{4q} + 1\right)}$$

Воспользуемся встроенной функцией InverseZTransform, предварительно заменив e^{q} на z.

```
in : InverseZTransform[z/(z^4 + 1), z, n] // FullSimplify // TrigFactor //
       Simplify

out : Cos[1/4 (1 + n) π] Sin[n π/2]
```

С помощью небольших манипуляций приходим к удовлетворяющей нас форме ответа.

Итак, получили, что:

$$\frac{e^{q}}{\left(e^{4q} + 1\right)} \doteq \sin\left(\frac{\pi n}{2}\right) \cos\left(\frac{1}{4} \pi (n+1)\right)$$

Программный модуль

Так как в системе Mathematica отсутствует функция обратного дискретного преобразования, напишем ее сами:

```
in : inverseDTransform[expr_, q_, n_] := Module[{},
       InverseZTransform[expr /. Exp[p_] → z^(p/q), z, n]]
```

Пример №13.183

Найти решетчатую функцию по её изображению:

$$F(q) = \frac{e^{2q}}{\left(e^{2q} + 2 e^{q} + 2\right)}$$

Решим задачу с помощью функции inverseDTransform:

```
in : inverseDTransform[Exp[2 q]/(Exp[2 q] + 2 Exp[q] + 2), q, n] // ExpToTrig //
       TrigFactor // FullSimplify

out : 2^(n/2) (Cos[3 n π/4] - Sin[3 n π/4])
```

Получили ответ:

$$f(n) = 2^{n/2} \left(\cos\left(\frac{3 \pi n}{4}\right) - \sin\left(\frac{3 \pi n}{4}\right)\right) \doteq F(q) = \frac{e^{2q}}{\left(e^{2q} + 2 e^{q} + 2\right)}$$

4.2. Решение разностных уравнений.

Пусть дано уравнение

$$a_0\, x(n+k) + a_1(n+k-1) + \ldots + a_k\, x(n) = \phi(n) \tag{13}$$

($a_0,\ a_1,\ \ldots,\ a_k$ – постоянные) с заданными (или произвольными) начальными условиями. Правая часть уравнения (13) - решетчатая функция $\phi(n)$ - предполагается оригиналом.

Полагая $x(n) \doteq X(q)$ и применяя формулу опережения (свойство 3, б), составляем операторное уравнение и определяем из него X(q). Затем одним из способов по изображению найдем искомое решение x(n).

Если исходное уравнение было задано не через последовательные значения неизвестной функции, а через ее конечные разности, т.е. имеет вид

$$b_0\, \Delta^k\, x(n) + b_1\, \Delta^{k-1}\, x(n) + \ldots + b_k\, x(n) = \phi(n) \tag{14}$$

то вследствие громоздкости формул для отыскания изображений конечных разностей решетчатых функций (свойство 6) его следует предварительно преобразовать к виду (13) при помощи известных формул, связывающих конечные разности функции с ее последовательными значениями:

$$\Delta^r\, x(n) = x(n+r) - C_r^1\, x(n+r-1) + C_r^2\, x(n+r-2) + \ldots + (-1)^r\, x(n) \tag{15}$$

Аналогично решаются и системы разностных уравнений

Пример №13.189

Решить линейное разностное уравнение:

$$x_{n+2} - 3\, x_{n+1} - 10\, x_n = 0,\ \ x_0 = 3,\ x_1 = -1$$

Обозначим $x_n \doteq X(q)$, и с помощью формулы опережения найдем следующие изображения:

$$x_{n+1} \doteq e^q (X(q) - x_0) = e^q\, X(q) - 3\, e^q$$
$$x_{n+2} \doteq e^{2q} (X(q) - x_0 - x_1\, e^{-q}) = e^{2q}\, X(q) - 3\, e^{2q} + e^q$$

Запишем эти выражения в исходное уравнение и найдем выражение для изображения исходной функции:

$$e^{2q}\, X(q) - 3\, e^{2q} + e^q - 3\, e^q\, X(q) + 9\, e^{2q} - 10\, X(q) = 0$$
$$X(q) = \frac{e^q\,(-10 + 3\, e^q)}{-10 - 3\, e^q + e^{2q}}$$

Теперь найдем оригинал с помощью функции inverseDTransform:

in : $\text{inverseDTransform}\left[\dfrac{e^q\,(-10 + 3\, e^q)}{-10 - 3\, e^q + e^{2q}}, q, n\right]$ // FullSimplify

out : $\dfrac{1}{7}\left((-1)^n\, 2^{4+n} + 5^{1+n}\right)$

Итак, получили, что решение исходного уравнения - есть функция x_n:

$$x_n = \frac{1}{7}\left((-1)^n 2^{4+n} + 5^{1+n}\right)$$

Пример №13.190

Решить линейное разностное уравнение:

$$x_{n+2} + x_{n+1} + x_n = 0, \;\; x_0 = 1, \, x_1 = -1$$

В этот раз найдем изображение всего уравнения:

```
in : ZTransform[x[n + 2] + x[n + 1] + x[n] == 0, n, Exp[q]]
```

```
out : -e^q x[0] - e^(2 q) x[0] - e^q x[1] + ZTransform[x[n], n, e^q] +
      e^q ZTransform[x[n], n, e^q] + e^(2 q) ZTransform[x[n], n, e^q] == 0
```

Заменим ZTransform[$x[n]$, n, Exp[q]] на функцию $X[q]$:

```
in : ZTransform[x[n + 2] + x[n + 1] + x[n] == 0, n, Exp[q]] /.
      ZTransform[x[n], n, Exp[q]] → X[q]
```

```
out : -e^q x[0] - e^(2 q) x[0] - e^q x[1] + X[q] + e^q X[q] + e^(2 q) X[q] == 0
```

Теперь учтем начальные условия:

```
in : ZTransform[x[n + 2] + x[n + 1] + x[n] == 0, n, Exp[q]] /.
      {ZTransform[x[n], n, Exp[q]] → X[q], x[0] → 1, x[1] → -1}
```

```
out : -e^(2 q) + X[q] + e^q X[q] + e^(2 q) X[q] == 0
```

Выразим функцию $X[q]$:

```
in : Solve[-e^(2 q) + X[q] + e^q X[q] + e^(2 q) X[q] == 0, X[q]]
```

$$out: \left\{\left\{X[q] \to \frac{e^{2q}}{1 + e^q + e^{2q}}\right\}\right\}$$

Осталось лишь применить функцию inverseDTransform, для того чтобы найти функцию $x[n]$:

$$in: \text{inverseDTransform}\left[\frac{e^{2q}}{1 + e^q + e^{2q}}, q, n\right]$$

$$out: \text{ChebyshevU}\left[n, -\frac{1}{2}\right]$$

В ответе получаем функцию многочлена Чебышева 2-ого рода, который в привычном для нас виде записывается как:

$$x(n) = \frac{2}{\sqrt{3}} \sin \frac{2\pi(n+1)}{3}$$

Пример №13.191

Решить линейное разностное уравнение:

$$x_{n+2} - \sqrt{3}\, x_{n+1} + x_n = 0,\ \ x_0 = \frac{1}{2},\ x_1 = \frac{\sqrt{3}}{2}$$

Внимательно изучив последовательность действий, примененных в предыдущем примере, для решения разностных уравнений составим программный модуль , который имеет схожее строение с функцией dsolve из главы 3 параграфа 3 “Применения операционного исчисления”:

```
in : rSolve[n_, expr_, rules_] := Module[{im, sol, X, z},
       im = ZTransform[expr, n, z] /. rules /.
         ZTransform[x[n], n, z] → X[z];
       sol = Flatten@Solve[im, X[z]];
       InverseZTransform[X[z] /. sol, z, n]]
```

Теперь применим его:

```
in : rSolve[n, x[n + 2] - √3 x[n + 1] + x[n] == 0,
      {x[0] → 1 / 2, x[1] → √3 / 2}]
```

$$out:\ \frac{1}{2}\,\text{ChebyshevU}\left[n, \frac{\sqrt{3}}{2}\right]$$

Мы вновь получили многочлен Чебышева 2-ого рода, который в привычном виде он записывается, как:

$$x(n) = \sin\frac{\pi(n+1)}{6}$$

Пример №13.195

Решить систему линейных разностных уравнений:

$$\begin{cases} x_{n+1} - x_n + y_n = 3^n \\ y_{n+1} + 2\,x_n = -3^n \end{cases}\qquad x_0 = 3,\ y_0 = 0$$

Так как принципиальная последовательность действий не изменилась, составим программный модуль для решения систем линейных рекуррентных уравнений:

```
in : rsSolve[n_, expr_, rules_] := Module[{zam, im, sol},
       zam = {ZTransform[x[n], n, z] → X[z],
         ZTransform[y[n], n, z] → Y[z]};
       im = ZTransform[expr, n, z] /. zam /. rules;
       sol = Flatten@Solve[im, {X[z], Y[z]}];
       Print["x[n]=", Simplify@InverseZTransform[X[z] /. sol, z, n]];
       Print["y[n]=", Simplify@InverseZTransform[Y[z] /. sol, z, n]];]
```

Теперь решим с помощью него исходную систему линейных разностных уравнений:

```
in : rsSolve[n, {x[n + 1] - x[n] + y[n] == 3^n, y[n + 1] + 2 x[n] == -3^n},
  {x[0] → 3, y[0] → 0}]
```

$x[n] = (-1)^n + 2^n + 3^n$

$y[n] = 2\ (-1)^n - 2^n - 3^n$

Мы получили ответ:

$$x(n) = (-1)^n + 2^n + 3^n$$
$$y(n) = 2\,(-1)^n - 2^n - 3^n$$

Полученный модуль можно добавить в пользовательскую библиотеку для применения в будущем.