§3. Степенные ряды

3.1. Область сходимости и свойства степенных рядов.

Ряд

$$c_0 + c_1(z - z_0) + c_2(z - z_0)^2 + \ldots + c_n(z - z_0)^n + \ldots = \sum_{n=0}^{\infty} c_n(z - z_0)^n \qquad (1)$$

называется степенным по степенях $z - z_0$. В частности, ряд

$$c_0 + c_1 z + c_2 z^2 + \ldots + c_n z^n + \ldots = \sum_{n=0}^{\infty} c_n z^n \qquad (2)$$

является степенным по степеням z .

Т е о р е м а А б е л я. Если степенной ряд (2) сходится в точке $z = z_1 \neq 0$, то он абсолютно сходится для всех z таких, что $|z| < |z_1|$. Если же ряд (2) расходится в точке $z = z_2$, то он расходится и для всех z таких, что $|z| > |z_2|$.

Из теоремы Абеля следует, что областью сходимости степенного ряда является круг с центром в начале координат(с центром в точке z_0), радиус которого может быть определен применением либо признака Даламбера, либо признака Коши, т.е. из условий:

$$\lim_{n\to\infty} \left| \frac{c_{n+1} z^{n+1}}{c_n z} \right| = |z| \lim_{n\to\infty} \left| \frac{c_{n+1}}{c_n} \right| < 1 \qquad (3)$$

$$\lim_{n\to\infty} \sqrt[n]{|c_n z^n|} = |z| \lim_{n\to\infty} \sqrt[n]{|c_n|} < 1 \qquad (4)$$

Отсюда для вычисления радиуса R круга сходимости получаем соотношения:

$$R = \frac{1}{\lim_{n\to\infty} \sqrt[n]{|c_n|}} \quad \text{или} \quad R = \frac{1}{\lim_{n\to\infty} \left| \frac{c_{n+1}}{c_n} \right|} \qquad (5)$$

Пример №12 .165

Найти область абсолютной сходимости и область равномерной сходимости ряда:

$$\sum_{n=1}^{\infty} \frac{(z-1)^n}{n^2\, 2^n}$$

Радиус круга сходимости определим по формуле (5):

$$\mathbf{R = 1 \Big/ Limit\left[\left(\frac{1}{n^2\, 2^n}\right)^{1/n},\ n \to Infinity\right]}$$

2

Получили, что область абсолютной сходимости исходного ряда: $|z - 1| < R = 2$.

На границе имеем:

$$\left| \frac{(z-1)^n}{n^2\, 2^n} \right| \le \frac{1}{n^2\, 2^n}$$

Следовательно, по принзнаку Вейерштрасса, ряд сходится равномерно в области: $|z - 1| \le 2$

Пример №12 .168

Найти область абсолютной сходимости и область равномерной сходимости ряда:

$$\sum_{n=1}^{\infty} (-1)^{n+1} \frac{2^{n+1} (z-4)^n}{(n+1)\sqrt{\ln(n+1)}}$$

Запишем общий член:

```
f[n_, z_] := (-1)^(n+1) 2^(n+1) (z - 4)^n / ((n + 1) Sqrt[Log[n + 1]])
```

Найдем, при помощи критерия Даламбера, область абсолютной сходимости:

```
Limit[Abs[f[n + 1, z]] / Abs[f[n, z]], n → Infinity]
```

2 Abs[-4 + z]

Получаем, что область сходимости $|z-4| < 1/2$.

На границе сходимости, при $z = 4.5$, имеем:

$$\sum_{n=1}^{\infty} \frac{2^{n+1} \left(\frac{1}{2}\right)^n}{(n+1)\sqrt{\ln(n+1)}} = \sum_{n=1}^{\infty} \frac{2}{(n+1)\sqrt{\ln(n+1)}}$$

Исследуем на сходимость, по интегральному признаку Коши:

```
Integrate[2 / ((n + 1) Sqrt[Log[n + 1]]), {n, 1, Infinity}]
```

$$\int_1^{\infty} \frac{2}{(1+n)\sqrt{\text{Log}[1+n]}} \, dn$$

Получили, что ряд расходится абсолютно, следовательно исходный ряд, равномерно сходится в области $|z-4| \le r < 1/2$

Исследуем ряды на границе: 1. $z = 4.5$

$$\sum_{n=1}^{\infty} \frac{(-1)^{n+1}\, 2}{(n+1)\sqrt{\ln(n+1)}}$$

Так как ряд не сходится абсолютно, исследуем на условную сходимость:

Общий член ряда:

```
Limit[2 / ((n + 1) Sqrt[Log[n + 1]]), n → Infinity]
```

0

Стремится к 0.

```
ListPlot[Table[2 / ((n + 1) Sqrt[Log[n + 1]]), {n, 1, 50}]]
```



Члены ряда по модулю убывают, условия критерия Лейбница выполнены, следовательно ряд при $z = 4.5$ сходится условно.

2. $z = 3.5$

$$\sum_{n=1}^{\infty} \frac{(-1)^{n+1} 2^{n+1} \left(-\frac{1}{2}\right)^n}{(n+1)\sqrt{\ln(n+1)}} = \sum_{n=1}^{\infty} \frac{-2}{(n+1)\sqrt{\ln(n+1)}} = -\sum_{n=1}^{\infty} \frac{2}{(n+1)\sqrt{\ln(n+1)}}$$

Ранее было получено, что данный ряд не сходится (см. выше).

Пример №12 .174

Найти область абсолютной сходимости и область равномерной сходимости ряда:

$$\sum_{n=1}^{\infty} \frac{(n!)^2 z^n}{(2n)!}$$

Запишем общий член ряда:

$$f[n\_, z\_] := \frac{(n!)^2}{(2n)!} z^n$$

При помощи признака Даламбера, определим область сходимости:

```
Limit[f[n + 1, z] / f[n, z], n → Infinity]
```

$$\frac{z}{4}$$

Область сходимости данного ряда $|z| < 4$.

На границе $z = 4$ имеем ряд:

$$\sum_{n=1}^{\infty} \frac{(n!)^2 4^n}{(2n)!}$$

Который по интегральному признаку Коши:

```
Integrate[f[n, 4], {n, 1, Infinity}]
```

$$\int_1^{\infty} \frac{4^n (n!)^2}{(2n)!} \, dn$$

Не сходится.

На границе $z = -4$, ряд,

$$\sum_{n=1}^{\infty} (-1)^n \frac{(n!)^2 4^n}{(2n)!}$$

Не удовлетворяет критериям признака Лейбница:

$$\text{Limit}\left[\frac{(n!)^2 4^n}{(2n)!}, n \to \text{Infinity}\right]$$

$$\infty$$

Следовательно, не сходится абсолютно и условно.

Таким образом исходный ряд сходится равномерно на $|z| \le r < 4$

2. Разложение функций в ряд Тейлора.

Т е о р е м а Т е й л о р а. Функция $f(z)$, аналитичная в круге $|z - z_0| < R$, однозначно представима в этом круге своим рядом Тейлора

$$f(z) = \sum_{n=0}^{\infty} c_n (z - z_0)^n, \tag{6}$$

коэффициенты которого определяются по формулам:

$$c_n = \frac{f^{(n)}(z_0)}{n!} = \frac{1}{2\pi i}\oint_{|z-z_0|=r}\frac{f(z)}{(z-z_0)^{n+1}}\,dz,\ n = 0,\ 1\ ... \tag{7}$$

С л е д с т в и е. Если функция $f(z)$ аналитична в области D и $z_0 \in D$, то в круге $|z - z_0| < R(z_0, D)$, где $R(z_0, D)$ - наименьшее растояние от точки z_0 до границы области D или до ближайшей точки z', в которой $f(z)$ не аналитична, $f(z)$ может быть представлена в виде степенного ряда

$$f(z) = \sum_{n=0}^{\infty}\frac{f^{(n)}(z_0)}{n!}(z-z_0)^n \tag{8}$$

Если $z_0 = 0$, то ряд Тейлора также называют рядом Маклорена.

Разложения элементарных функций:

$$e^z = 1 + z + \frac{z^2}{2!} + ... + \frac{z^n}{n!} + ... = \sum_{n=0}^{\infty}\frac{z^n}{n!},\quad z \in \mathbb{C} \tag{9}$$

$$\cos z = 1 - \frac{z^2}{2!} + ... + (-1)^n\,\frac{z^{2n}}{(2n)!} + ... = \sum_{n=0}^{\infty}(-1)^n\,\frac{z^{2n}}{(2n)!},\ z \in \mathbb{C} \tag{10}$$

$$\sin z = z - \frac{z^3}{3!} + ... + (-1)^{n+1}\,\frac{z^{2n+1}}{(2n+1)!} + ... = \sum_{n=0}^{\infty}(-1)^n\,\frac{z^{2n+1}}{(2n+1)!},\ z \in \mathbb{C} \tag{11}$$

$$\ln(1+z) = z - \frac{z^2}{2} + ... + (-1)^{n+1}\,\frac{z^n}{n} + ... = \sum_{n=1}^{\infty}(-1)^{n+1}\,\frac{z^n}{n},\quad |z| < 1 \tag{12}$$

$$\operatorname{arctg} z = z - \frac{z^3}{3} + ... + (-1)^{n+1}\,\frac{z^{2n+1}}{2n+1} + ... = \sum_{n=0}^{\infty}(-1)^{n+1}\,\frac{z^{2n+1}}{2n+1},\quad |z| < 1 \tag{13}$$

$$(1+z)^a = 1 + a\,z + \frac{a(a-1)}{2!}z^2 + ... + \frac{a(a-1)\,...\,(a-n+1)}{n!}z^n = 1 + \sum_{n=1}^{\infty}\frac{\text{Binomial}[a,\,n]}{n!}z^n,\quad |z| < 1 \tag{14}$$

$$\text{где Binomial}[a,\,n] = a(a-1)\,...\,(a-n+1)$$

$$\frac{1}{1+z} = 1 - z + z^2 + ... + (-1)^n z^n = \sum_{n=0}^{\infty}(-1)^n z^n,\quad |z| < 1 \tag{15}$$

Пример №12.214

Разложить функцию в ряд по степеням z :

$$f(z) = e^{-z^2}$$

Рассмотрим сначала решение данной задачи “в лоб”. Так как данная нам функция аналитична во всей комплексной плоскости, то для разложения функции в ряд применима формула (8).

Запишем заданную функцию:

```
f[z_] := Exp[-z^2]
```

Теперь запишем выражение для n-ого коэффициента c_n:

```
c[n_] := D[f[z], {z, n}] / n! /. z → 0
```

Так как мы вычисляем разложение в ряд Тейлора в окрестности точки 0, то $z_0 = 0$. Такой ряд также называют рядом Маклорена.

Для того чтобы получить окончательный ответ, найдем сумму ряда, например до 16 члена:

```
Sum[c[n] z^n, {n, 0, 16}]
```

$$1 - z^2 + \frac{z^4}{2} - \frac{z^6}{6} + \frac{z^8}{24} - \frac{z^{10}}{120} + \frac{z^{12}}{720} - \frac{z^{14}}{5040} + \frac{z^{16}}{40320}$$

Запишем данное выражение в виде функции:

```
r[z_, k_] := Sum[c[n] z^n, {n, 0, k}]
```

Для того чтобы проверить полученный результат, построим графики функции для исходной функции и ряда до 16 члена:

```
Plot[{f[z], Evaluate@r[z, 16]}, {z, -2, 2}, PlotRange → All]
```



Примечание: функция Evaluate здесь используется для того, чтобы сначала получить разложение в ряд Тейлора, а затем в него происходила подстановка значений z отрезка [−2; 2].

Увеличивая максимальный порядок разложения, можно добиться сходимости ряда к функции на более широком отрезке.

Програмный модуль

Составим учебную программу, которая будет выполнять разложение функции в ряд Тейлора по формуле (8).

Имеем следующую тривиальную последовательность действий:

Пусть задана функция $f(z) = \cos(z)$.

```
f[z_] := Cos[z]
```

1. Записываем функцию коэффициента:

```
c[n_, z0_] := D[f[z], {z, n}] / n! /. z → z0
```

2. Находим сумму:

```
Sum[c[n, 0] (z - 0)^n, {n, 0, 8}]
```

$$1-\frac{z^2}{2}+\frac{z^4}{24}-\frac{z^6}{720}+\frac{z^8}{40320}$$

Соберем это всё в единый модуль:

```
seriesO[func_, z_, z0_, k_] := Module[{coef},
  coef[n_] := D[func, {z, n}] / n! /. z → z0;
  Sum[coef[n] (z - z0)^n, {n, 0, k}]]
```

Продемонстрируем пример:

```
seriesO[Sin[z]^2, z, 0, 12]
```

$$z^2-\frac{z^4}{3}+\frac{2z^6}{45}-\frac{z^8}{315}+\frac{2z^{10}}{14175}-\frac{2z^{12}}{467775}$$

Пример №12.242

```
seriesO[Exp[z^2 - 4 z + 1], z, 2, 12]
```

$$\frac{1}{e^3}+\frac{(-2+z)^2}{e^3}+\frac{(-2+z)^4}{2\,e^3}+\frac{(-2+z)^6}{6\,e^3}+\frac{(-2+z)^8}{24\,e^3}+\frac{(-2+z)^{10}}{120\,e^3}+\frac{(-2+z)^{12}}{720\,e^3}$$

Напомним, что данный модуль использует формулу (8) для разложения функции в ряд.

Рассмотрим другие пути решения поставленной задачи.

Пример №12.231

Используя разложения основных элементарных функций, разложить функцию в ряд по степеням z и указать область сходимости полученного ряда:

$$f(z)=\int_0^z e^{-t^2/2}\,dt$$

На этот раз воспользуемся стандартными разложениями (9)-(15), для того чтобы получить разложение функции в ряд.

Подынтегральная функция - экспоненциальная, значит используя выражение (9), а также замену $h=-t^2/2$, получим:

```
seriesO[Exp[h], h, 0, 8] /. h → -t^2 / 2
```

$$1-\frac{t^2}{2}+\frac{t^4}{8}-\frac{t^6}{48}+\frac{t^8}{384}-\frac{t^{10}}{3840}+\frac{t^{12}}{46\,080}-\frac{t^{14}}{645\,120}+\frac{t^{16}}{10\,321\,920}$$

Теперь требуется лишь проинтегрировать полученное выражение:

```
Integrate[seriesO[Exp[h], h, 0, 8] /. h → -t^2 / 2, {t, 0, z}]
```

$$z-\frac{z^3}{6}+\frac{z^5}{40}-\frac{z^7}{336}+\frac{z^9}{3456}-\frac{z^{11}}{42\,240}+\frac{z^{13}}{599\,040}-\frac{z^{15}}{9\,676\,800}+\frac{z^{17}}{175\,472\,640}$$

Получили разложение в ряд Тейлора до 17ого члена исходной функции, причем, так как область сходимости ряда (9) - вся комплексная плоскость, то и областью сходимости полученного ряда также является вся комплексная плоскость.

Примечание: если слегка модифицировать модуль, полученный в предыдущем примере:

```
seriesO1[func_, z_, z0_, k_] := Module[{coef, t},
  coef[n_] := D[func[t], {t, n}] / n! /. t → z0;
  Sum[coef[n] (z - z0)^n, {n, 0, k}]]
```

Если записать функцию при помощи безымянной переменной (см. Help -> Slot), например:

```
f = Exp[#] &;

f[z]
```

e^z

То, получим следующий результат:

```
seriesO1[Exp[#] &, -t^2 / 2, 0, 8]
```

$$1-\frac{t^2}{2}+\frac{t^4}{8}-\frac{t^6}{48}+\frac{t^8}{384}-\frac{t^{10}}{3840}+\frac{t^{12}}{46\,080}-\frac{t^{14}}{645\,120}+\frac{t^{16}}{10\,321\,920}$$

```
Integrate[seriesO1[Exp[#] &, -t^2 / 2, 0, 8], {t, 0, z}]
```

$$z-\frac{z^3}{6}+\frac{z^5}{40}-\frac{z^7}{336}+\frac{z^9}{3456}-\frac{z^{11}}{42\,240}+\frac{z^{13}}{599\,040}-\frac{z^{15}}{9\,676\,800}+\frac{z^{17}}{175\,472\,640}$$

Также вместо безымянной функции, можно использовать просто заголовок подынтегральной функции:

```
seriesO1[Exp, -t^2/2, 0, 8]
```

$$1-\frac{t^2}{2}+\frac{t^4}{8}-\frac{t^6}{48}+\frac{t^8}{384}-\frac{t^{10}}{3840}+\frac{t^{12}}{46\,080}-\frac{t^{14}}{645\,120}+\frac{t^{16}}{10\,321\,920}$$

Пример №12.218

Используя разложения основных элементарных функций, разложить функцию в ряд по степеням z и указать область сходимости полученного ряда:

$$f(z)=\sqrt[3]{27-z}$$

Преобразуем выражение:

$$f(z)=3\sqrt[3]{1-\frac{z}{27}}$$

Теперь, пользуясь стандартным разложением (14), при помощи замены $t=-z/27$, найдем разложение исходной функции в ряд:

При помощи безымянной функции:

```
seriesO1[3 (1 + #)^(1/3) &, -z / 27, 0, 6]
```

$$3-\frac{z}{27}-\frac{z^2}{2187}-\frac{5\,z^3}{531\,441}-\frac{10\,z^4}{43\,046\,721}-\frac{22\,z^5}{3\,486\,784\,401}-\frac{154\,z^6}{847\,288\,609\,443}$$

При помощи замены:

```
seriesO[3 (1 + t)^(1/3), t, 0, 6] /. t → -z / 27
```

$$3-\frac{z}{27}-\frac{z^2}{2187}-\frac{5\,z^3}{531\,441}-\frac{10\,z^4}{43\,046\,721}-\frac{22\,z^5}{3\,486\,784\,401}-\frac{154\,z^6}{847\,288\,609\,443}$$

Область сходимости ряда определим из:

$$|t|<1\Longrightarrow|z|<27$$

Встроенные средства

В Wolfram Mathematica функция, которая отвечает за разложение в ряд, она имеет следующую структуру: Series[$f[z]$, $\{z,\ z_0,\ n\}$] , где n- порядок разложения. Продемонстрируем пример:

```
Series[ Cos[z], {z, 0, 5}]
```

$$1-\frac{z^2}{2}+\frac{z^4}{24}+O[z]^6$$

Примечание: $O[z]^6$, о-малое от z^6 - остаток ряда.

```
Series[Exp[-z^2], {z, 0, 16}]
```

$$1-z^2+\frac{z^4}{2}-\frac{z^6}{6}+\frac{z^8}{24}-\frac{z^{10}}{120}+\frac{z^{12}}{720}-\frac{z^{14}}{5040}+\frac{z^{16}}{40\,320}+O[z]^{17}$$

```
Series[Exp[Cos[z]], {z, 0, 6}]
```

$$e-\frac{e\,z^2}{2}+\frac{e\,z^4}{6}-\frac{31\,e\,z^6}{720}+O[z]^7$$

Для того, чтобы получить n-ый коэффициент разложения, существует функция SeriesCoefficient[$f[z]$, $\{z, z_0, n\}$]. Примеры:

```
SeriesCoefficient[Exp[-z^2], {z, 0, 8}]
```

$$\frac{1}{24}$$

```
SeriesCoefficient[(27 - z)^(1/3), {z, 0, 3}]
```

$$-\frac{5}{531441}$$

Также есть возьможность получить общий член:

```
SeriesCoefficient[Log[1 + z], {z, 0, n}]
```

$$\begin{cases} \frac{(-1)^{1+n}}{n} & n \geq 1 \\ 0 & \text{True} \end{cases}$$

Или если задать, что $n > 1$, можно избавиться от кусочно-заданной функции:

```
Refine[SeriesCoefficient[Log[1 + z], {z, 0, n}], n > 1]
```

$$\frac{(-1)^{1+n}}{n}$$

Следующая конструкция, позволит вывести ответ в формульном виде:

```
TraditionalForm@
 Inactivate[Sum[Refine[SeriesCoefficient[Exp[z], {z, 0, n}], n > 0] z^n,
   {n, 0, Infinity}], Sum]
```

$$\sum_{n=0}^{\infty} \frac{z^n}{n!}$$

Где функция Inactivate запрещает функции, следующей за ней (Sum) вычеслять результат, а TraditionalForm преобразует выражение в традиционный вид.

Составим функцию:

```
seriesF[f_, z_, z0_] :=
 TraditionalForm@
  Inactivate[Sum[Refine[SeriesCoefficient[f, {z, z0, n}], n > 5 ] (z - z0)^n,
    {n, 0, Infinity}], Sum]
```

```
seriesF[1/(-z + 1), z, 0]
```

$$\sum_{n=0}^{\infty} z^n$$

Продемонстрируем пример на основе задачи №12.214:

```
seriesF[Exp[t], t, 0] /. t → -z^2
```

$$\sum_{n=0}^{\infty} \frac{(-z^2)^n}{n!}$$

Пример №12.232

Разложить функцию в ряд:

$$f(z) = \int_0^z \frac{\sin(t)^2}{t} \, dt$$

Получим разложение подинтегральной функции:

```
Sin[t]^2 / t // TrigReduce
```

$$\frac{1 - \mathrm{Cos}[2\,t]}{2\,t}$$

Зная стандартное разложение функции cos(h) (10), где $h = 2\,t$, получаем:

$$\frac{1-\left(1-\frac{h^2}{2!}+...+(-1)^n\,\frac{h^{2\,n}}{(2\,n)!}+...\right)}{2\,t}=\sum_{n=1}^{\infty}\frac{(-1)^n\,(2\,t)^{2\,n-1}}{(2\,n)\,!}=\sum_{n=1}^{\infty}\frac{(-1)^n\,4^n\,t^{2\,n-1}}{2\,(2\,n)\,!}$$

Теперь, применяя теорему о почленном интегрировании получаем:

$$\int_0^z\sum_{n=1}^{\infty}\frac{(-1)^{n+1}\,4^n\,t^{2\,n-1}}{2\,(2\,n)\,!}\,dt=\sum_{n=1}^{\infty}\frac{(-1)^{n+1}\,4^n}{2\,(2\,n)\,!}\int_0^z t^{2\,n-1}\,dt=\sum_{n=1}^{\infty}\frac{(-1)^{n+1}\,4^n\,z^{2\,n}}{2\,(2\,n)\,!\,2\,n}$$

В итоге получаем ответ:

$$f(z)=\int_0^z\frac{\sin(t)^2}{t}\,dt=\sum_{n=1}^{\infty}\frac{(-1)^{n+1}\,4^n\,z^{2\,n}}{2\,(2\,n)\,!\,2\,n},\quad z\in\mathbb{C}.$$

Проверим полученное аналитическое решение при помощи функции Series:

```
Series[Integrate[Sin[t]^2 / t, {t, 0, z}], {z, 0, 12}]
```

$$\frac{z^2}{2}-\frac{z^4}{12}+\frac{z^6}{135}-\frac{z^8}{2520}+\frac{z^{10}}{70\,875}-\frac{z^{12}}{2\,806\,650}+O[z]^{13}$$

```
Sum[(-1)^(n+1) 4^n z^(2 n) / (2 (2 n)! 2 n), {n, 1, 6}]
```

$$\frac{z^2}{2}-\frac{z^4}{12}+\frac{z^6}{135}-\frac{z^8}{2520}+\frac{z^{10}}{70\,875}-\frac{z^{12}}{2\,806\,650}$$

Ряды совпадают, следовательно получили правильный ответ.

Пример №12.237

Разложить функцию в ряд по степеням $z - z_0$ и опредить область сходимости полученного ряда:

$$f(z)=\frac{1}{1-z},\ z_0=3\,i$$

Получим разложение, воспользовавшись опцией Series:

```
Series[1/(1 - z), {z, 3 I, 6}]
```

$$\left(\frac{1}{10}+\frac{3\,i}{10}\right)-\left(\frac{2}{25}-\frac{3\,i}{50}\right)(z-3\,i)-\left(\frac{13}{500}+\frac{9\,i}{500}\right)(z-3\,i)^2+\left(\frac{7}{2500}-\frac{6\,i}{625}\right)(z-3\,i)^3+\left(\frac{79}{25\,000}-\frac{3\,i}{25\,000}\right)(z-3\,i)^4+\left(\frac{11}{31\,250}+\frac{117\,i}{125\,000}\right)(z-3\,i)^5-\left(\frac{307}{1\,250\,000}-\frac{249\,i}{1\,250\,000}\right)(z-3\,i)^6+O[z-3\,i]^7$$

Также можно воспользоваться созданным ранее модулем seriesO:

```
seriesO[1/(1 - z), z, 3 I, 6]
```

$$\left(\frac{1}{10}+\frac{3\,i}{10}\right)-\left(\frac{2}{25}-\frac{3\,i}{50}\right)(-3\,i+z)-\left(\frac{13}{500}+\frac{9\,i}{500}\right)(-3\,i+z)^2+\left(\frac{7}{2500}-\frac{6\,i}{625}\right)(-3\,i+z)^3+\left(\frac{79}{25\,000}-\frac{3\,i}{25\,000}\right)(-3\,i+z)^4+\left(\frac{11}{31\,250}+\frac{117\,i}{125\,000}\right)(-3\,i+z)^5-\left(\frac{307}{1\,250\,000}-\frac{249\,i}{1\,250\,000}\right)(-3\,i+z)^6$$

Запишем ответ в стандартном виде:

```
seriesF[1/(1 - z), z, 3 I]
```

$$\sum_{n=0}^{\infty}(1+3\,i)^{n+1}\,10^{-n-1}\,(z-3\,i)^n$$

Область сходимости определим при помощи функции SumConvergence:

```
SumConvergence[(1 + 3 I)^(1+n) 10^(-1-n) (-3 I + z)^n, n]
```

$$\sqrt{10}\ \mathrm{Abs}[-3\,i+z] < 10$$

В итоге имеет ответ:

$$\frac{1}{1-z}=\sum_{n=0}^{\infty}\frac{(1+3\,i)^{n+1}}{10^{n+1}}(z-3\,i)^n,\quad \left|z-3\,i\right|<\sqrt{10}$$

Пример №12.250

Найти область сходимости указанного ряда и его сумму:

$$\sum_{n=0}^{\infty}(-1)^n\,a^{-2\,n-2}\,z^{2\,n},\ \ a\neq 0$$

Для того чтобы найти сумму, преобразуем выражение к:

$$\frac{1}{a^2}\sum_{n=0}^{\infty}(-1)^n\left(\left(\frac{z}{a}\right)^2\right)^n,\ \ a\neq 0$$

Суммой данной геометрической прогрессии является следующая функция:

$$\frac{1}{1+\eta}=\sum_{n=0}^{\infty}(-1)^n\,\eta^n,\ \ \text{где}\ \eta=\left(\frac{z}{a}\right)^2$$

Область сходимости данной суммы $|\eta| < 1$

Проверим себя:

```
Sum[(-1)^n a^(-2 n-2) z^(2 n), {n, 0, Infinity}]
```

$$\frac{1}{a^2+z^2}$$

Получили тот же самый результат.

Следовательно, получаем ответ:

$$\sum_{n=0}^{\infty}(-1)^n\,a^{-2\,n-2}\,z^{2\,n}=\frac{1}{a^2}\,\frac{1}{1+\frac{z^2}{a^2}}=\frac{1}{a^2+z^2},\quad |z| < |a|$$